

Initiation à la Simulation Numérique en
Mécanique des Fluides à l'Aide de
CASTEM2000

Recueil d'Exemples Commentés

Cours ENSTA MF307

H. Paillère & F. Dabbene
CEA Saclay

Département de Mécanique et de Technologie
Service d'Etudes Mécaniques et Thermiques

Table des matières

1	Transport d'un champ scalaire par convection et diffusion	7
1.1	Convection-diffusion : cas test de Smith et Hutton	8
1.1.1	Formulation Eléments Finis Explicite	8
1.1.2	Formulation Eléments Finis Mixtes avec Hybridation . . .	16
1.2	Convection instationnaire d'un cône	27
1.2.1	Formulation Eléments Finis Explicite	27
1.2.2	Formulation Eléments Finis Implicite	34
1.2.3	Formulation Eléments Finis Mixtes avec Hybridation . . .	45
1.3	Equation de conservation linéaire	59
1.3.1	Formulation Eléments Finis Explicite Conservative	59
1.4	Equation de conservation non-linéaire : cas d'une solution avec discontinuité	68
1.4.1	Formulation Eléments Finis Explicite Conservative	68
1.4.2	Formulation Eléments Finis Explicite Non-Conservative . .	77
1.5	Equation de conservation non-linéaire : cas d'une solution sans discontinuité	83
1.5.1	Formulation Eléments Finis Explicite Non-Conservative . .	83
1.6	Transport diffusif : équation de la chaleur	91
1.6.1	Formulation Eléments Finis Implicite	91
2	Ecoulements incompressibles laminaires	97
2.1	Couche limite sur plaque plane	98
2.1.1	Formulation Eléments Finis Semi-Implicite	98
2.2	Ecoulement autour d'une marche descendante	105
2.2.1	Formulation Eléments Finis Semi-Implicite	105
2.3	Convection naturelle dans une cavité carrée	112

2.3.1	Formulation Eléments Finis Semi-Implicite	112
2.4	Tourbillon dans une cavité carrée	121
2.4.1	Formulation Eléments Finis Semi-Implicite	121
2.5	Convection naturelle dans un fluide à nombre de Prandtl nul	132
2.5.1	Formulation Eléments Finis Semi-Implicite	132
2.6	Convection Rayleigh-Marangoni	140
2.6.1	Formulation Eléments Finis Semi-Implicite	140
3	Ecoulements incompressibles turbulents	151
3.1	Décroissance de la turbulence homogène	151
3.1.1	Formulation Eléments Finis Semi-Implicite	151
3.2	Écoulement dans une conduite cylindrique	159
3.2.1	Formulation Eléments Finis Semi-Implicite	159
3.3	Convection naturelle turbulente dans une cavité carrée	174
3.3.1	Formulation Eléments Finis Semi-Implicite	174
4	Ecoulements bi-fluides à phase dispersée	183
4.1	Motion of a particle in a fluid at rest	184
4.1.1	Formulation Eléments Finis Semi-Implicite	184
4.2	Gas flow induced by a dispersed phase	189
4.2.1	Formulation Eléments Finis Semi-Implicite	189
5	Ecoulements compressibles	199
5.1	Calcul d'un Tube à Choc	200
5.1.1	Formulation Volumes Finis Explicite Conservative	200
5.2	Ecoulement supersonique dans un canal avec rampe à 15°	232
5.2.1	Formulation Volumes Finis Explicite Conservative	232
5.3	Injection d'un mélange de gaz dans une enceinte fermée	247
5.3.1	Formulation Eléments Finis Semi-Implicite Non-Conservative	247
6	Hydrogéologie	262
6.1	Convection naturelle dans un milieu géologique	263
6.1.1	Formulation Eléments Finis Mixtes avec Hybridation	263
6.2	Convection naturelle en milieu poreux dans une cavité carrée	284

6.2.1	Formulation Eléments Finis Mixtes avec Hybridation	284
6.3	Hydraulique en milieu poreux : influence de la Salinité	295
6.3.1	Formulation Eléments Finis Mixtes avec Hybridation	295
6.4	Cas d'une inclusion circulaire	306
6.4.1	Formulation Eléments Finis Mixtes avec Hybridation	306
7	Trajectoires et lignes de courant	315
7.1	Calcul de trajectoires	315
7.1.1	Formulations EF/EFMH	315
7.2	Lignes de courant	321
7.2.1	Formulation Eléments Finis Implicite	321

Nomenclature	
<i>I</i>	implicite
<i>E</i>	explicite
<i>EF</i>	éléments finis standard
<i>EFM1</i>	éléments finis sous-intégrés (sur quadrangles)
<i>EFMH</i>	éléments finis mixtes hybrides
<i>VF</i>	volumes finis (schéma décentré 1er ordre)
<i>SUPG</i>	schéma "Streamline Upwind Petrov Galerkin"
	option de décentrement : stabilisation
<i>SUPG-CC</i>	schéma "Streamline Upwind Petrov Galerkin"
	option de décentrement : stabilisation+capture discontinuités
<i>SUPG-DC</i>	idem que SUPG-CC
<i>CENTREE</i>	aucun décentrement
<i>NOCONS</i>	forme non-conservative
<i>CONS</i>	forme conservative

Attention :

Pour imposer une option (par exemple une des 3 options de décentrement, SUPGDC, SUPG ou CENTREE) à un opérateur, quelques précautions sont nécessaires. Les options se précisent grâce à la directive 'OPTI' rangée dans la table 'EQEX'. Toutes les options relatives à un opérateur doivent être listées sur une seule ligne.

Exemple :

OUI	NON
'OPTI' 'CONS' 'SUPG'	'OPTI' 'CONS'
'ZONE' \$MT 'OPER' 'TSCAL' ...	'OPTI' 'SUPG'
	'ZONE' \$MT 'OPER' 'TSCAL' ...

Il faut préciser les options pour chaque opérateur.

Introduction

Ce document rassemble près d'une trentaine d'exemples commentés d'utilisation du logiciel CASTEM2000 pour traiter des problèmes de Mécanique des Fluides, tels que le transport par convection et diffusion, les écoulements incompressibles laminaires ou turbulents, les écoulements compressibles, ainsi que les écoulements liés à l'hydrogéologie. Ces exemples représentent des cas-tests simples mais significatifs où l'on compare les résultats numériques obtenus avec les opérateurs de mécanique des fluides de CASTEM2000 à des résultats théoriques, expérimentaux ou numériques, référencés dans la littérature.

Pour chaque cas-test, le problème physique à traiter, ainsi que les méthodes numériques (type de schéma, maillage, etc) sont indiqués. Les résultats numériques sont représentés graphiquement, avec dans la mesure du possible, une superposition de la solution de référence. Le listing du fichier GIBIANE est aussi donné, permettant au lecteur de suivre la démarche utilisée pour traiter le problème dans CASTEM2000.

Le rapport est organisé de la façon suivante :

- le chapitre 1 traite des phénomènes de transport (convection et diffusion), caractéristiques des problèmes de mécanique des fluides. Plusieurs aspects sont abordés : convection pure, convection-diffusion, problèmes stationnaires ou instationnaires, problème non-linéaire, discontinuités, diffusion pure, etc.
- le chapitre 2 est consacré aux écoulements de fluide incompressible en régime laminaire, gouvernés par les équations de Navier-Stokes. L'approximation de Boussinesq permet de traiter les fluides faiblement dilatables.
- le chapitre 3 est consacré aux écoulements de fluide incompressible en régime turbulent. La turbulence est modélisée par le modèle $k - \varepsilon$.
- le chapitre 4 est consacré aux écoulements de type bi-fluide dont l'une des phases est dispersée. Ce type de modélisation s'applique par exemple à des problèmes d'aspersion (entraînement d'air par des gouttelettes).
- le chapitre 5 est consacré aux écoulements compressibles, soit de fluides parfaits (équations d'Euler de l'aérodynamique, écoulements avec chocs), soit de fluides visqueux dans le cas d'écoulements faiblement compressibles pour lesquels l'approximation de Boussinesq n'est plus valable.
- le chapitre 6 est consacré aux problèmes d'hydrogéologie, dont les écoulements sont gouvernés par les équations de Darcy.
- le chapitre 7 est consacré à la résolution des problèmes de trajectoire et à la détermination des lignes de courant pour les écoulements incompressibles.

Ce rapport a été rédigé en partie grâce aux contributions des personnes suivantes : par ordre alphabétique, F. Auriol, A. Beccantini, L.-V. Benet, D. Broc, C. Caroli, A. Chene, N. Coulon, P. Galon, S. Gounand, J.-P. Magnaud et A. Pavie.

H. Paillère & F. Dabbene
Service d'Etudes Mécaniques et Thermiques
CEA Saclay.

Chapitre 1

Transport d'un champ scalaire par convection et diffusion

1.1 Convection-diffusion : cas test de Smith et Hutton

1.1.1 Formulation Eléments Finis Explicite

<i>Opérateurs utilisés</i>	<i>TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation P1 ou Q1, schéma SUPG, E</i>
<i>Nature du problème</i>	<i>Convection-diffusion dans un champ de vitesse donné. Comparaison avec solutions d'un workshop.</i>
<i>Classification</i>	<i>2D plan, convection-diffusion, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/smithhutton.dgibi</i>

Description du problème :

Ce cas-test, proposé dans le cadre d'un workshop [1], concerne la résolution de l'équation de convection-diffusion,

$$\frac{\partial c}{\partial t} + \vec{\lambda} \cdot \nabla c = \nu \Delta c$$

sur le domaine $\Omega = \{(x, y), (x, y) \in [-1, 1] \times [0, 1]\}$, avec le champ de vitesse $\vec{\lambda}(x, y)$ donné par :

$$\vec{\lambda}(x, y) = 2y(1 - x^2)\vec{1}_x - 2x(1 - y^2)\vec{1}_y$$

On impose sur les bords du domaine (à l'exception de la frontière de sortie que l'on laisse libre) les conditions aux limites suivantes :

$$c(x, 0) = 1 + \tanh [(2x + 1)\alpha] \quad \text{sur } y = 0 \text{ et } -1 \leq x \leq 0$$

$$c(x, y) = 1 - \tanh \alpha \begin{cases} \text{sur } x = -1 \text{ et } 0 \leq y \leq 1 \\ \text{sur } y = 1 \text{ et } -1 \leq x \leq 1 \\ \text{sur } x = 1 \text{ et } 0 \leq y \leq 1 \end{cases}$$

On traite les problèmes suivants :

- $\nu = 10^{-1}$ (correspondant à $Pe = 10$ dans [1])
 - $\nu = 10^{-2}$ (correspondant à $Pe = 100$ dans [1])
 - $\nu = 10^{-6}$ (correspondant à $Pe = 10^6$ dans [1])
- avec dans tous les cas, $\alpha = 10$.




FIG. 1.1.1 – Maillage 40×20 constitué de triangles et champ de vitesse pour le cas-test de Smith et Hutton

Modélisation :

Un maillage 40×20 constitués de triangles a été choisi. L'opérateur **TSCAL** (schéma de convection-diffusion EF Streamline Upwind Petrov-Galerkin) a été utilisé pour traiter l'équation de convection-diffusion. Pour le cas où la convection est dominante ($\nu = 10^{-6}$), l'option SUPG-CC a aussi été utilisée. La solution stationnaire est obtenue après 2000 itérations.

Analyse des résultats :

Les solutions sont comparées aux solutions de référence de [1]. L'accord est excellent pour les cas où la diffusion physique est importante. Pour le cas à haut Peclet, des oscillations sont présentes avec l'option SUPG. Cependant, le profil très raide est bien capturé. L'option SUPG-CC permet de se débarrasser des oscillations, au prix d'un lissage du profil.

Référence :

- 1– R.M. Smith and A.G. Hutton, *The numerical treatment of advection : a performance comparison of current methods*, Numerical Heat Transfer, vol. 5, pp.439-461, 1982.




FIG. 1.1.2 – Cas $\nu = 10^{-1}$, TSCAL option SUPG




FIG. 1.1.3 – Cas $\nu = 10^{-2}$, TSCAL option SUPG





FIG. 1.1.4 – Cas $\nu = 10^{-6}$, TSCAL option SUPG



Jeu de données :

```

COMPLET = FAUX ;
'OPTI' 'DIME' 2 'ISOV' 'SULI' ;
'SI' ( COMPLET ) ;
  'OPTI' 'ELEM' 'TRI3' ;
  NX =40 ; NY=20 ;
  nbiter = 2000 ;
  tolera = 0.006 ;
'SINON' ;
  'OPTI' 'ELEM' 'QUA4' ;
  NX = 20 ; NY = 10 ;
  nbiter = 250 ;
  tolera = 0.012 ;
finsi ;
graph = 'N' ;
*****
* ADVECTION-DIFFUSION: CAS TEST DE SMITH ET HUTTON      *
* REFERENCE : NUMERICAL HEAT TRANSFER, VOL.5, p.439, 1982 *
*****

*****
* PROCEDURE POUR ESTIMER LA CONVERGENCE *
*****

DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;

RV = RVX.'EQEX' ;

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;

ERR = KOPS (RV.INCO.'CN') - (RV.INCO.'CNM1') ;

ELI = MAXI ERR 'ABS' ;
ELI = (LOG (ELI + 1.0E-20))/(LOG 10.) ;
MESSAGE 'ITER ' RV.PASDETPS.'NUPASDT' ' ERREUR LINF ' ELI ;
IT = PROG RV.PASDETPS.'NUPASDT' ;
ER = PROG ELI ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER ;
FINSI ;

RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' (RV.INCO.'CN') ;

as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

*****
* DIFFUSION *

```

```

* Pe = 10, 100, 500, 1E3, 1E6 *
*****

Pe = 100 ;
DIF = 1.0/Pe ;

*****
* MAILLAGE *
*****

A1 = -1.0 0.0 ;
A2 = 1.0 0.0 ;
A3 = 1.0 1.0 ;
A4 = -1.0 1.0 ;
A0 = 0.0 0.0 ;

FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;
FENTREE = A1 'DROI' (NX/2) A0 ;

PA3 = FHAU POIN 'PROC' A3 ;
PA4 = FHAU POIN 'PROC' A4 ;
P34 = CHAN POI1 FHAU ;
MA3 = P34 ELEM 'CONTENANT' PA3 ;
MA4 = P34 ELEM 'CONTENANT' PA4 ;

DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;

*****
* CREATION MODELES NAVIER_STOKES *
*****

MDOMTOT = CHAN DOMTOT QUAF ;
MFENTREE= CHAN FENTREE QUAF ;

ELIM MDOMTOT MFENTREE 1.E-3 ;
$DOMTOT = MODE MDOMTOT 'NAVIER_STOKES' LINE ;
  DOMA $DOMTOT 'IMPR' ;
$FENTREE = MODE MFENTREE 'NAVIER_STOKES' LINE ;

*****
* INITIALISATION DU CHAMP DE VITESSE *
*****

XX YY = 'COOR' (DOMA $DOMTOT SOMMET) ;
VXSH = (2.*YY)*(1.0-(XX*XX)) ;
VYSH = (-2.*XX)*(1.0-(YY*YY)) ;
VXO = NOMC 'UX' VXSH ;
VYO = NOMC 'UY' VYSH ;
VX = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' VXO ;
VY = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' VYO ;

```

```
CHVIT=KCHT $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY') (VX ET VY);
```

```
*****
* PROFIL DE LA SOLUTION A L'ENTREE *
*****
```

```
XBAS = COOR 1 FENTREE ;
XBAS = NOMC 'UX' XBAS ;
TOTO = 2.0*XBAS ;
TOTO = TOTO + 1.0 ;
TOTO = 10.0*TOTO ;
SOLUTION = TANH TOTO ;
SOLUTION = 1.0 + SOLUTION ;
CHP1 = KCHT $FENTREE 'SCAL' 'SOMMET' 'COMP' 'UX' SOLUTION ;
CHP1 = NOMC 'CN' CHP1 ;
```

```
*****
* SOLUTION SUR FRONTIERES GAUCHE, HAUT, DROIT *
*****
```

```
C1 = (1.0 - (TANH 10.0)) ;
C2 = C1*(-1.0) ;
```

```
RV1 = EQEX $DOMTOT 'ITMA' nbiter 'ALFA' 0.7
      'ZONE' $DOMTOT 'OPER' CALCUL
      'OPTI' 'SUPG'
      'ZONE' $DOMTOT 'OPER' TSCAL DIF 'VITESSE' 0.0 'INCO' 'CN'
      'OPTI' 'CENTREE'
      'ZONE' $DOMTOT 'OPER' DFDT 1. 'CN' 'DELTAT' 'INCO' 'CN'
      'CLIM' 'CN' 'TIMP' FENTREE CHP1
      'CLIM' 'CN' 'TIMP' FHAU C1
      'CLIM' 'CN' 'TIMP' FGAU C1
      'CLIM' 'CN' 'TIMP' FDRO C1
      'CLIM' 'CN' 'TIMP' MA3 C2
      'CLIM' 'CN' 'TIMP' MA4 C2 ;
```

```
RV1.INCO = TABLE 'INCO' ;
RV1.INCO.'CN' = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
RV1.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
RV1.INCO.'VITESSE' = CHVIT ;
RV1.INCO.'IT' = PROG 1 ;
RV1.INCO.'ER' = PROG 0. ;
```

```
EXEC RV1 ;
```

```
*****
* ANALYSE DES RESULTATS *
*****
```

```
EVOL1 = EVOL 'CHPO' (RV1.INCO.'CN') SCAL FBAS ;
EVOL2 = EVOL 'CHPO' XX SCAL FBAS ;
LIX = 'EXTR' EVOL2 ORDO ;
LIU = 'EXTR' EVOL1 ORDO ;
```

```

EVOL3 = EVOL 'MANU' 'X' LIX 'U(X,0)' LIU ;

LIXT = PROG 0. 0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8 0.9 1.0 ;
SI ( Pe 'EGA' 10 ) ;
LIUT = PROG 1.989 1.402 1.146 0.946 0.775 0.621 0.480 0.349
          0.227 0.111 0.000 ;
FINSI ;
SI ( Pe 'EGA' 100 ) ;
LIUT = PROG 2.000 1.940 1.836 1.627 1.288 0.869 0.480 0.209
          0.070 0.017 0.000 ;
FINSI ;
SI ( Pe 'EGA' 500 ) ;
LIUT = PROG 2.000 2.000 1.998 1.965 1.702 0.947 0.242 0.023
          0.001 0.000 0.000 ;
FINSI ;
SI ( Pe 'EGA' 1E3 ) ;
LIUT = PROG 2.000 2.000 2.000 1.985 1.841 0.951 0.154 0.001
          0.000 0.000 0.000 ;
FINSI ;
SI ( Pe 'EGA' 1E6 ) ;
LIUT = PROG 2.000 2.000 2.000 1.999 1.964 1.000 0.036 0.001
          0.000 0.000 0.000 ;
FINSI ;

EVOL4 = EVOL 'MANU' 'X' LIXT 'Uref(X,0)' LIUT ;

LIUC = IPOL LIXT LIX LIU ;
NP = DIME LIXT ;
ERRO = 0. ;
REPETER BLOC1 NP ;
    UCAL = EXTRAIRE LIUC &BLOC1 ;
    UREF = EXTRAIRE LIUT &BLOC1 ;
    ERRO = ERRO + ((UCAL-UREF)*(UCAL-UREF)) ;
FIN BLOC1 ;
ERRO = ERRO/NP ;
ERRO = ERRO '**' 0.5 ;
MESSAGE 'ERREUR ' ERRO ;
SI ( ERRO > tolera ) ;
    ERREUR 5 ;
FINSI ;

SI ('EGA' graph '0') ;
*****
* SORTIES GRAPHIQUES *
*****

TRACE DOMTOT ;

UNCH = VECT CHVIT 0.1 UX UY ROUGE ;
TRACE UNCH DOMTOT (CONT DOMTOT) ;

TRAC DOMTOT (RV1.INCO.'CN') (CONT DOMTOT) ;

```

```
EVOL5 = EVOL 'MANU' 'ITERATIONS' (RV1.INCO.'IT') 'LOG|E|inf'  
      (RV1.INCO.'ER') ;  
DESS EVOL5 'XBOR' 0. ( 'FLOT' NBITER) 'YBOR' -20.0 0.0 ;  
  
TAB1 = TABLE ;  
TITRE = 'SOLUTION CASTEM2000' ;  
TAB1.2 = 'MARQ LOSA NOLI ' ;  
TAB1.'TITRE' = TABLE ;  
TAB1.'TITRE'. 1 = 'MOT' 'SOLUTION_CASTEM2000' ;  
TAB1.'TITRE'. 2 = 'MOT' 'SOLUTION_REFERENCE' ;  
DESS (EVOL3 ET EVOL4) 'LEGE' TAB1  
      'MIMA' 'XBOR' -1.0 1.0 'YBOR' -1.0 3.0      ;  
FINSI ;  
  
FIN ;
```


1.1.2 Formulation Eléments Finis Mixtes avec Hybridation

<i>Procédure utilisée</i>	<i>DARCYTRA</i>
<i>Méthodes numériques</i>	<i>EFMH, éléments HYT3 et HYQ4, schéma implicite</i>
<i>Nature du problème</i>	<i>Recherche de la solution stationnaire du transport par diffusion-convection d'un champ scalaire, le champ de vitesse étant connu. Comparaison avec les solutions d'un workshop.</i>
<i>Classification</i>	<i>2D plan, convection-diffusion, état stationnaire</i>
<i>Jeu de données</i>	

Description du problème :

Ce cas-test, proposé dans le cadre d'un workshop [1], concerne la résolution de l'équation de convection-diffusion,

$$\frac{\partial c}{\partial t} + \vec{\lambda} \cdot \nabla c = \nu \Delta c$$

sur le domaine $\Omega = \{(x, y), (x, y) \in [-1, 1] \times [0, 1]\}$, avec le champ de vitesse $\vec{\lambda}(x, y)$ donné par :

$$\vec{\lambda}(x, y) = 2y(1 - x^2)\vec{1}_x - 2x(1 - y^2)\vec{1}_y$$

On impose sur les bords du domaine (à l'exception de la frontière de sortie laissée libre) les conditions aux limites suivantes :

$$c(x, 0) = 1 + \tanh [(2x + 1)\alpha] \quad \text{sur } y = 0 \text{ et } -1 \leq x \leq 0$$

$$c(x, y) = 1 - \tanh \alpha \begin{cases} \text{sur } x = -1 \text{ et } 0 \leq y \leq 1 \\ \text{sur } y = 1 \text{ et } -1 \leq x \leq 1 \\ \text{sur } x = 1 \text{ et } 0 \leq y \leq 1 \end{cases}$$

On traite les problèmes suivants :

- ▷ $\nu = 10^{-1}$ (correspondant à $Pe = 10$ dans [1])
 - ▷ $\nu = 10^{-2}$ (correspondant à $Pe = 100$ dans [1])
 - ▷ $\nu = 10^{-6}$ (correspondant à $Pe = 10^6$ dans [1])
- avec dans tous les cas, $\alpha = 10$.

Modélisation :

Un maillage 40×20 constitué de triangles ou de quadrangles est choisi. La procédure DARCYTRA est utilisée afin de résoudre ce problème de transport par diffusion-convection. On utilise donc une méthode de discrétisation spatiale par éléments finis mixtes avec hybridation. La méthode de discrétisation temporelle choisie est implicite (θ -méthode avec $\theta = 1$).




La solution stationnaire est obtenue après 5 pas de temps et un pas de temps de 2000. La différence entre deux itérés successifs est alors inférieure à 10^{-10} .

Analyse des résultats :

Les solutions sont comparées aux solutions de référence de [1]. L'accord est bon dans tous les cas en éléments quadrangulaires, pour les deux premiers cas en éléments triangulaires. Pour le cas où la convection est dominante ($\nu = 10^{-6}$) des problèmes d'instabilités numériques sont apparus en utilisant les éléments triangulaires. Notons toutefois que la solution obtenue est convergée. Les problèmes d'instabilité à haut Peclet pour ce type d'éléments devront être analysés.

Références :

- [1] R.M. Smith and A.G. Hutton, *The numerical treatment of advection : a performance comparison of current methods*, Numerical Heat Transfer, vol. 5, pp.439-461, 1982.
- [2] F. Dabbene, *Schémas de diffusion-convection en éléments finis mixtes hybrides*, Rapport interne CEA, DMT95/613, Saclay, 1995.

FIG. 1.1.6 – Cas $\nu = 10^{-1}$, EFMH implicite, maillage \square FIG. 1.1.7 – Cas $\nu = 10^{-2}$, EFMH implicite, maillage \square FIG. 1.1.8 – Cas $\nu = 10^{-6}$, EFMH implicite, maillage \square




FIG. 1.1.9 – Cas $\nu = 10^{-1}$, EFMH implicite, maillage Δ




FIG. 1.1.10 – Cas $\nu = 10^{-2}$, EFMH implicite, maillage Δ




FIG. 1.1.11 – Cas $\nu = 10^{-6}$, EFMH implicite, maillage Δ

Jeu de données :

```

*
*-- Utilisation du modèle DARCY pour la résolution du
*-- transport par diffusion-convection d'un champ scalaire.
*
*-----
*= Cas test Smith et Hutton, The numerical treatment of advection : a
*= performance comparison of current methods, Numerical Heat Transfer,
*= vol. 5, pp. 439-461, 1982.
*-----
*
*-----
* Options générales
*-----
*
GRAPH = '0' ;
'OPTI' 'DIME' 2 'ELEM' 'QUA4' ;
'OPTI' 'ISOV' 'SULI' ;
'OPTI' 'TRAC' 'PS' ;
'OPTI' 'ECHO' 1 ;
*
*=====
* Choix du cas traité et du schéma de transport
*=====
*
* IK = 1 à 5 : diffusion de 0.1, 0.01, 0.005, 0.001 ou 1.d-6
* TETD : theta-méthode (schéma en temps pour flux diffusif).
* TETC : theta-méthode (schéma en temps pour flux convectif).
*
IK = 5 ;
SCH = 0 ;
TETD = 1.0D0 ;
TETC = 1.0D0 ;
*
*=====
* MAILLAGE
*=====
*
*- Création des points supports des DROITES
*
L = 2.D0 ;
LS2 = L / 2.D0 ;
H = LS2 ;
X0 = -1.D0 * LS2 ;
X1 = X0 + L ;
Y0 = 0.D0 ;
Y1 = Y0 + H ;

```

```

INUMX = 40 ;
INUX2 = INUMX / 2 ;
INUMY = 20 ;
XNUMY = 'FLOT' INUMX ;
INUX1 = INUMX - 1 ;
DX = X1 - X0 / INUMX ;
DX1 = DX / 2.D0 ;
DY = Y1 - Y0 / INUMY ;
*
B1 = X0 Y0 ;
B2 = 0. Y0 ;
B3 = X1 Y0 ;
H1 = X0 Y1 ;
H3 = X1 Y1 ;
*
DIR1 = 1.D0 0.D0 ;
*
*- Création des DROITES frontieres
*
DRBA1 = B1 'DROI' INUX2 B2 ;
DRBA2 = B2 'DROI' INUX2 B3 ;
DRBAS = DRBA1 'ET' DRBA2 ;
DRDRO = B3 'DROI' INUMY H3 ;
DRHAU = H3 'DROI' INUMX H1 ;
DRGAU = H1 'DROI' INUMY B1 ;
DRCOT = DRBAS 'ET' DRDRO 'ET' DRHAU 'ET' DRGAU ;
PELIM = DX1 / (50.D0 * INUMX) ;
*
*- Création et orientation du maillage GEOMETRIQUE
*
PTOT1 = 'DALL' DRBAS DRDRO DRHAU DRGAU ;
PTOT2 = 'ORIE' PTOT1 ;
*
*- Creation maillage HYBRIDE y compris sous-objets (cond. limites)
*
HYTOT = 'DOMA' PTOT2 PELIM ;
GEOBL1 = DRDRO 'ET' DRHAU 'ET' DRGAU ;
DOMBL1 = 'DOMA' GEOBL1 'INCL' HYTOT PELIM ;
GEOBL2 = DRBA1 ;
DOMBL2 = 'DOMA' GEOBL2 'INCL' HYTOT PELIM ;
VOLU1 = 'DOMA' HYTOT 'VOLUME' ;
CHYB1 = 'DOMA' HYTOT 'SURFACE' ;
CHYB2 = 'DOMA' HYTOT 'NORMALE' ;
*
*- Construction de la ligne joignant les faces pour post-traitement
*
XG = X0 + DX1 ;
XD = X1 - DX1 ;
MX1 = XG 0. ;
MX3 = XD 0. ;
DRMIX = MX1 'DROI' INUX1 MX3 ;
'ELIM' HYTOT.FACE DRMIX PELIM ;
*

```

```

*
*=====
* INITIALISATIONS
*=====
*
*
*-----
* = Donnees physiques =
*-----
* VK    : Tenseur de diffusion isotrope
*
'SI' ( IK 'EGA' 1) ;
      VK = 1.D-1 ;
'FINSI' ;
'SI' ( IK 'EGA' 2) ;
      VK = 1.D-2 ;
'FINSI' ;
'SI' ( IK 'EGA' 3) ;
      VK = 5.D-3 ;
'FINSI' ;
'SI' ( IK 'EGA' 4) ;
      VK = 1.D-3 ;
'FINSI' ;
'SI' ( IK 'EGA' 5) ;
      VK = 1.D-6 ;
'FINSI' ;
*
*-----
* = Donnees numeriques =
*-----
*
* TMIN   : Temps initial
* TMAX   : Temps final
* TSUP   : Temps bidon superieur à TMAX pour conditions aux limites
* DELTAT : Pas de temps de calcul
* DELTAS : Pas de temps de sauvegarde
*
TMIN   = 0.DO      ;
TMAX   = 10000.DO ;
TSUP   = 2.4 * TMAX ;
DELTAT = TMAX / 5.DO ;
DELTAS = TMAX / 5.DO ;
*
*- Vitesses aux faces
*
XF YF = 'COOR' HYTOT . 'FACE' ;
SPEEX = 'NOMC' 'VX' (2.*YF*(1.-(XF*XF))) 'NATU' 'DISCRET' ;
SPEEY = 'NOMC' 'VY' (2.*XF*(YF*YF - 1.)) 'NATU' 'DISCRET' ;
SPEEF = SPEEX 'ET' SPEEY ;
*
*- Flux aux faces
*
MOT1 = 'MOTS' 'VX' 'VY' ;
MOT2 = 'MOTS' 'UX' 'UY' ;
SOUR1 = 'PSCA' SPEEF CHYB2 MOT1 MOT2 ;
SOUR1 = CHYB1 * SOUR1 ;

```



```

*- Solution de référence
*
LIXT = 'PROG' 0. 0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8 0.9 1.0 ;
'SI' ( IK 'EGA' 1 ) ;
    LIUT = 'PROG' 1.989 1.402 1.146 0.946 0.775 0.621 0.480 0.349
           0.227 0.111 0.000 ;
'FINSI' ;
'SI' ( IK 'EGA' 2 ) ;
    LIUT = 'PROG' 2.000 1.940 1.836 1.627 1.288 0.869 0.480 0.209
           0.070 0.017 0.000 ;
'FINSI' ;
'SI' ( IK 'EGA' 3 ) ;
    LIUT = 'PROG' 2.000 2.000 1.998 1.965 1.702 0.947 0.242 0.023
           0.001 0.000 0.000 ;
'FINSI' ;
'SI' ( IK 'EGA' 4 ) ;
    LIUT = 'PROG' 2.000 2.000 2.000 1.985 1.841 0.951 0.154 0.001
           0.000 0.000 0.000 ;
'FINSI' ;
'SI' ( IK 'EGA' 5 ) ;
    LIUT = 'PROG' 2.000 2.000 2.000 1.999 1.964 1.000 0.036 0.001
           0.000 0.000 0.000 ;
'FINSI' ;
EVOL4 = 'EVOL' 'MANU' 'X' LIXT 'Uref(X,0)' LIUT ;
*
*- Table de légende
*
TAB1 = 'TABLE' ;
TAB1.1 = 'TITR SOLUTION_CASTEM2000' ;
TAB1.2 = 'MARQ LOSA NOLI TITR SOLUTION_REFERANCE' ;
*
*- Récupération de l'évolution contenant l'abscisse des points
*- de la droite le long de laquelle on compare la solution
*- calculée et la solution analytique.
*
XX1 YY1 = 'COOR' DRMIX ;
EVOL2 = 'EVOL' 'ROUG' 'CHPO' XX1 'SCAL' DRMIX ;
LIX = 'EXTR' EVOL2 'ORDO' ;
*
*-----
REPETER VISURESU NTSOR ;
*-----
*
IRESU = IRESU + 1 ;
INDI1 = ISOR1.IRESU ;
TTRA = TRANS2 . 'TEMPS' . INDI1 ;
CON1 = 'CONT' HYTOT.MAILLAGE ;
*
'TITR' LTI3 ;
ETP1 = TRANS2 . 'TRACE_CHARGE' . INDI1 ;
EP1 = TRANS2 . 'CHARGE' . INDI1 ;
EPR1 = 'KCHA' HYTOT 'CHAM' EP1 ;
EVOL1 = 'EVOL' 'ROUG' 'CHPO' ETP1 'TH' DRMIX ;

```

```
LIU   = 'EXTR' EVOL1 'ORDO'           ;
EVOL3 = 'EVOL' 'ROUG' 'MANU' 'X' LIX 'U(X,0)' LIU ;
CP1   = 'ELNO' HYTOT EP1             ;
'SI' ('EGA' INDI1 NTSOR) ;
*   'TRAC' MODHYB EPR1 CON1 ;
    'TRAC' HYTOT.MAILLAGE CP1 DRCOT ;
    'DESS' (EVOL3 'ET' EVOL4) 'LEGE' 'MIMA' 'YBOR' -1.0 3.0 TAB1 ;
'FINSI' ;
INDI2 = INDI1 - 1 ;
ERR   = 'ABS' ((TRANS2.CHARGE.INDI1) - (TRANS2.CHARGE.INDI2)) ;
'MAXI' ERR 'LIST' ;
*
*-----
FIN VISURESU ;
*-----
*
*- Tracé du champ de vitesse et de la norme de la vitesse
*
*VVVV = 'VECT' SPEEF 0.25 'VX' 'VY' 'VERT' ;
*'TRAC' VVVV (HYTOT.'MAILLAGE') ;
*TRAC (HYTOT.MAILLAGE) NORMS ;
*
'FIN' ;
```

1.2 Convection instationnaire d'un cône

1.2.1 Formulation Eléments Finis Explicite

<i>Opérateurs utilisés</i>	<i>TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation P1 ou Q1, schéma SUPG, E</i>
<i>Nature du problème</i>	<i>Convection instationnaire d'un cône de concentration, dans un champ de vitesse circulaire</i>
<i>Classification</i>	<i>2D plan, convection, instationnaire</i>
<i>Jeu de données</i>	

Description du problème :

Ce cas-test traite la convection pure d'un cône de concentration dans un champ de vitesse circulaire,

$$\frac{\partial c}{\partial t} + \vec{\lambda} \cdot \nabla c = 0$$

où $\vec{\lambda} = y\vec{1}_x - x\vec{1}_y$. On résoud cette équation sur le domaine Ω , $(x, y) \in [-4, 4]^2$, avec comme solution initiale un cône unitaire centré en $(0, -1)$ et de rayon 1,

$$c(x, y, t = 0) = 1 - \min \left[1, (x^2 + (y + 1)^2)^{1/2} \right]$$

On cherche à calculer la solution au bout du temps $t = 2\pi$, correspondant à une rotation complète. La solution exacte de ce problème de convection pure est évidemment identique à la solution initiale.

Discretisation du problème :

On traite le problème avec l'opérateur de convection-diffusion explicite **TSCAL**, et l'on compare les solutions numériques pour :

- des maillages constitués d'éléments linéaires et bilinéaires ;
- des options différentes de l'opérateur (SUPG-CC, SUPG ou CENTREE)

On rappelle que cette formulation explicite correspond à une matrice masse diagonale non-consistante. Il est connu que la diagonalisation des termes temporels dans une formulation variationnelle entraîne une diffusion numérique supplémentaire.

La diffusion physique dans ce problème étant nulle, seule la diffusion numérique est responsable de l'aplatissement du cône. On s'intéresse aux valeurs suivantes :

$$\max_{\Omega} c, \quad \min_{\Omega} c, \quad \iint_{\partial\Omega} c \, d\Omega$$




FIG. 1.2.1 – Convection d'un cône dans un champ de vitesse circulaire.

sachant que pour la solution exacte,

$$\max_{\Omega} c_{ex} = 1., \quad \min_{\Omega} c_{ex} = 0., \quad \iint_{\partial\Omega} c_{ex} d\Omega = \frac{\pi}{3}$$

Résultats :

Les résultats des différents calculs sont résumés dans le tableau suivant :

maillage	option	max c	min c	$\iint_{\partial\Omega} c d\Omega$
quadrangle	SUPG-CC	0.13139	-1.76×10^{-6}	1.0419
quadrangle	SUPG	0.24009	-9.77×10^{-3}	1.0421
quadrangle	CENTREE	0.78738	-0.24066	1.0401
triangle	SUPG-CC	0.17597	-5.20×10^{-6}	1.0420
triangle	SUPG	0.28336	-1.40×10^{-2}	1.0421
triangle	CENTREE	0.85884	-0.31343	1.0432
solution exacte		1.0	0.0	1.0472
solution exacte, projetée		1.0	0.0	1.0421

On note que dans tous les cas, la masse est conservée avec une assez bonne précision. L'option centrée est celle qui restitue le pic de concentration avec le plus de précision, les options de décentrement ayant tendance à aplanir le cône. En contrepartie, les oscillations non-physiques sont plus importantes avec l'option centrée. Les solutions sur triangles semblent être plus précises que sur quadrangles. Peut-être est-ce dû à la sous-intégration sur ces éléments.



FIG. 1.2.2 – Rotation d'un cône : maillage 40×40 quadrangles. solution exacte, opérateur TSCAL (explicite) avec les options SUPG-CC, SUPG et CENTREE. Isoconcentrations ($\Delta c = 0.1$).




FIG. 1.2.3 – Rotation d'un cône : maillage 40×40 triangles. solution exacte, opérateur TSCAL (explicite) avec les options SUPG-CC, SUPG et CENTREE. Isoconcentrations ($\Delta c = 0.1$).

Jeu de données :

```
DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;
```

```
RV = RVX.'EQEX' ;
```

```
RV.PASDETPS.'DELTAT' = 2.*3.14159/800. ;
```

```
FINPROC ;
```

```
*****
* CONVECTION CIRCULAIRE D'UN CONE DE CONCENTRATION *
*****
```

```
opti dime 2 ;
opti elem qua4 ;
opti elem tri3 ;
opti isov sul1 ;
*opti trac ps ;
```

```
*****
* MAILLAGE *
*****
```

```
NX = 40 ; NY = 40 ;
A1 = -4.0 -4.0 ;
A2 = 4.0 -4.0 ;
A3 = 4.0 4.0 ;
A4 = -4.0 4.0 ;
FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;
```

```
CNT = FBAS ET FDRO ET FHAU ET FGAU ;
DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;
```

```
*****
* CREATION TABLES DOMAINE *
*****
```

```
$DOMTOT = 'DOMA' DOMTOT ;
```

```
*****
* INITIALISATION DU CHAMP DE VITESSE *
*****
```

```
XX YY = COOR $DOMTOT.MAILLAGE ;
UU = NOMC 'UX' YY 'NATU' 'DISCRET' ;
VV = KOPS XX '*' -1.0 ;
```



```

VV = NOMC 'UY' VV 'NATU' 'DISCRET' ;

CHVIT = KCHT $DOMTOT 'VECT' 'SOMMET' (UU ET VV) ;

*****
* SOLUTION INITIALE *
*****

RR = KOPS (KOPS (KOPS YY '+' 1.) '*' (KOPS YY '+' 1.))
      '+' (KOPS XX '*' XX) ;
RR = KOPS RR '**' 0.5 ;
RR = KOPS RR '>|' 1.0 ;
RR = KOPS RR '-' 1.0 ;
RR = ABS RR ;

*****
* TABLE EQEX *
*****

TF = 2.*3.14159 ;
DT = TF/800. ;

MESSAGE 'PAS DE TEMPS IMPOSE = ' DT ;

RV = EQEX $DOMTOT 'ITMA' 800 'ALFA' 1.0
      'ZONE' $DOMTOT 'OPER' 'CALCUL'
* 'OPTI' 'SUPG'
      'ZONE' $DOMTOT
'OPER' TSCAL 1E-10 'VITESSE' 0.0 'INCO' 'CN'
      'ZONE' $DOMTOT 'OPER' 'CALCUL' ;

RV = EQEX RV
'CLIM' 'CN' 'TIMP' CNT 0. ;

RV.INCO = TABLE 'INCO' ;
RV.INCO.'CN' = KCHT $DOMTOT 'SCAL' 'SOMMET' RR ;
RV.INCO.'VITESSE' = CHVIT ;

VOL = DOMA (RV.'DOMAINE') 'VOLUME' ;
VOLT = SOMT VOL ;

MASTI = SOMT (KOPS (NOEL $DOMTOT RV.INCO.'CN') '*' VOL) ;
MESSAGE 'MASSE INITIALE ' MASTI ;

*UNCH = VECT RV.INCO.'VITESSE' 1. UX UY JAUNE ;
*TRACE UNCH DOMTOT CNT ;

EXEC RV ;

*****
* ON TRACE LES RESULTATS *
*****

```

```
LI = PROG 0. 0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8 0.9 1.0 ;

MESSAGE 'TEMPS = ' RV.PASDETPS.'TPS' ;
MESSAGE 'MAX = ' (MAXI RV.INCO.'CN') ;
MESSAGE 'MIN = ' (MINI RV.INCO.'CN') ;
TRACE RR DOMTOT CNT LI ;
TRACE RV.INCO.'CN' DOMTOT CNT LI ;

opti dime 3 ;
P1 = -5. -5. 2. ;
MONTAGNE RR DOMTOT 2. 'SOLUTION INITIALE' P1 2 ;
MONTAGNE RV.INCO.'CN' DOMTOT
      2. 'SOLUTION APRES UNE ROTATION' P1 2 ;

MASTF = SOMT (KOPS (NOEL $DOMTOT RV.INCO.'CN') '* ' VOL) ;
MESSAGE 'MASSE FINALE ' MASTF ;

ERRMA = ABS (( MASTF-MASTI))/MASTI ;
ERRMA = ERRMA*100. ;
MESSAGE 'PERTE DE MASSE = ' ERRMA ' %' ;
MAXC = MAXI RV.INCO.'CN' ;
MESSAGE 'MAXIMUM SOL = ' MAXC ;
fin;
```

1.2.2 Formulation Eléments Finis Implicite

<i>Opérateurs utilisés</i>	<i>KONV DFDT</i>
<i>Méthodes numériques</i>	<i>EF, approximation Q1, schéma en temps : ordre 1 IMPL-PG (Euler Petrov-Galerkin), ordre 2 CN (Crank-Nicolson Galerkin), ordre 2 CN-PG (Crank-Nicolson Petrov-Galerkin), ordre 4 CNG (Crank-Nicolson généralisé Galerkin)</i>
<i>Nature du problème</i>	<i>Convection instationnaire d'un cône de concentration, dans un champ de vitesse circulaire</i>
<i>Classification</i>	<i>2D plan, convection, instationnaire</i>
<i>Jeu de données</i>	

Description du problème :

Ce cas-test traite la convection pure d'un cône de concentration dans un champ de vitesse circulaire,

$$\frac{\partial c}{\partial t} + \vec{\lambda} \cdot \nabla c = 0$$

où $\vec{\lambda} = y\vec{1}_x - x\vec{1}_y$. On résoud cette équation sur le domaine Ω , $(x, y) \in [-1, 1]^2$, avec comme solution initiale un cône unitaire centré en $(0, -0.5)$ et de rayon 0.3.

On cherche à calculer la solution au bout du temps $t = 2\pi$, correspondant à une rotation complète. La solution exacte de ce problème de convection pure est évidemment identique à la solution initiale.

Résultats :

maillage	option	max c	min c	$\iint_{\partial\Omega} c \, d\Omega$
quadrangle	IMPL-PG	0.23283	-1.86×10^{-2}	9.5848×10^{-2}
quadrangle	CN	0.78233	-0.209	9.3315×10^{-2}
quadrangle	CN-PG	0.51570	-3.16×10^{-3}	9.5821×10^{-2}
quadrangle	CNG	0.85939	-8.23×10^{-2}	9.4548×10^{-2}
solution exacte		1.0	0.0	9.4248×10^{-2}
solution exacte, projetée		1.0	0.0	9.3803×10^{-2}




FIG. 1.2.4 – Rotation d'un cône : maillage 20×20 quadrangles. solution exacte, opérateurs DFDT KONV (Coupe $x'ox$) et (Coupe $y'oy$) avec les schéma CNG (ordre 4) CN (ordre 2) CN-PG (ordre 2) IMPL-PG (ordre 1).




FIG. 1.2.5 – Rotation d'un cône : maillage 20×20 quadrangles. Schéma implicites, CNG (ordre 4), CN (ordre 2), CN-PG (ordre 2), IMPL-PG (ordre 1).




FIG. 1.2.6 – Rotation d'un cône : maillage 20×20 quadrangles. Schéma implicites, CN-SUPG (ordre 2), IMPL-SUPG (ordre 1).




FIG. 1.2.7 – Rotation d'un cône : maillage 20×20 quadrangles. solution exacte, opérateur DFDT KONV avec les schéma CN-SUPG (ordre 2) IMPL-SUPG (ordre 1).

Commentaires :

Les tests ont été réalisés sur un maillage 20X20 et non 40X40 comme précédemment. Dans chacun des cas le nombre de CFL est de 1. Les deux premières figures (1.2.4) et (1.2.5) permettent de comparer la diffusion numérique, l'erreur de phase et la stabilité de quatre schémas en temps implicites basés sur une formulation variationnelle consistante : Galerkin ou Petrov-Galerkin.

schéma d'Euler implicite Petrov-Galerkin (ordre 1 en temps) (IMPL-PG)

schéma de Crank Nicolson Galerkin (ordre 2 en temps) (CN)

schéma de Crank Nicolson Petrov-Galerkin (ordre 2 en temps) (CN-PG)

schéma de Crank Nicolson généralisé Galerkin (ordre 4 en temps) (CNG)

En ce qui concerne la diffusion numérique, les quatre schémas se classent hiérarchiquement dans l'ordre de précision en temps des schémas. On note cependant un bien meilleur comportement du schéma (CN) par rapport au schéma (CN-PG) tout deux schémas du second ordre. Par contre le schéma (CN) exhibe un déphasage (à peu près une maille) par rapport à (CN-PG). En proportion le passage à l'ordre 4 (CNG) n'apporte pas grand chose de plus concernant la diffusion numérique. Par contre l'erreur de déphasage est quasiment nulle. Une explication peut être la trop grande différence entre l'ordre de l'approximation en temps (ordre 4) et l'ordre d'approximation en espace (ordre 2). Du point de vue de la stabilité ce sont les schémas (CN-PG) et (IMPL-PG) qui présentent le moins d'oscillations.

Un autre point non montré ici est le maintien des performances du schéma (CN-PG) à CFL=2 alors que les autres schémas se détériorent rapidement.

On a reporté pour mémoire sur les figures (1.2.6) et (1.2.7) les résultats des deux schémas implicites Euler et Crank-Nicolson (CFL=1.) où la matrice masse n'est pas consistante, (IMPL-SUPG) et (CN-SUPG). Ces deux schémas sont respectivement d'ordre 1 et 2 en temps. On constate une grande stabilité mais une très grande diffusion l'ordre 2 en temps n'étant d'aucune utilité.

Jeu de données : Fichier cone.dgibi

```

*-----
*= Transport d'un cone par un champ de vitesse à rotationnel constant.
*= Comparaison de schéma en temps implicites
*-----
*
*-----
* Options générales
*-----
*
GRAPH = '0' ;
ERR1=5.e-6 ;
COMPLET = FAUX ;
SI ( COMPLET ) ;
DIAM=2. ;
SINON ;
DIAM=0.5 ;
FINSI ;

'OPTI' 'DIME' 2 'ELEM' 'QUA4' ;
'OPTI' 'ISOV' 'SULI' ;
'OPTI' 'ECHO' 1 ;
*
*=====
* MAILLAGE
*=====
*
*- Création des points supports des DROITES
*
L = 2.DO ; LS2 = L / 2.DO ;
H = L ; HS2 = H / 2.DO ;
X0 = -1.DO * LS2 ; X1 = X0 + L ;
Y0 = -1.DO * HS2 ; Y1 = Y0 + H ;
INUMX = 20 ;
INUMY = INUMX ;
XNUMY = 'FLOT' INUMX ;
INUX1 = INUMX - 1 ;
INUY1 = INUMY - 1 ;
XO1 = X0 + X1 * 0.5D0 ;
YO1 = Y0 + Y1 * 0.5D0 ;
DX = X1 - X0 / INUMX ;
DY = Y1 - Y0 / INUMY ;
DX1 = DX / 2.DO ;
DY1 = DY / 2.DO ;
*
B1 = X0 Y0 ;
B3 = X1 Y0 ;
H1 = X0 Y1 ;
H3 = X1 Y1 ;
*

```



```

*
*- Création des DROITES frontieres
*
DRBAS = B1 'DROI' INUMX B3 ;
DRDRO = B3 'DROI' INUMY H3 ;
DRHAU = H3 'DROI' INUMX H1 ;
DRGAU = H1 'DROI' INUMY B1 ;
DRCOT = DRBAS 'ET' DRDRO 'ET' DRGAU 'ET' DRHAU ;
PELIM = DX1 / (5.DO * INUMX) ;
*
*- Creation maillage GEOMETRIQUE
*
PTOT1 = 'DALL' DRBAS DRDRO DRHAU DRGAU ;
*
*- Creation maillage HYBRIDE y compris sous-objets (cond. limites)
*
HYTOT = 'DOMA' PTOT1 PELIM ;
XVOL= DOMA HYTOT 'VOLUME' ;
*
*- Coordonnées des points par classe de points
*
XS YS = 'COOR' HYTOT . 'MAILLAGE' ;
*
* XCEN : Abscisse du sommet du cone
* YCEN : Ordonnée du sommet du cone
* RCEN : Rayon du cone
*
XCEN = 0.DO ;
YCEN = -0.5DO ;
RCEN = 3.0DO * DX ;

*=====
* INITIALISATIONS
*=====
*
* -----
* = Conditions initiales =
* -----
* h(x,y) = f(r)
* avec f(r) = max( h(1.-r/rcen) , 0.)
* On se sert des égalités suivantes :
* m = min(f,g) = (f+g - abs(f-g)) / 2
* M = max(f,g) = (f+g + abs(f-g)) / 2
*
HAUT = 1.DO ;
*
*- Charge aux sommets
*
CRS1 = (XS-XCEN)*(XS-XCEN) + ((YS-YCEN)*(YS-YCEN)) ** 0.5DO ;
CRS2 = 1.DO - (CRS1 / RCEN) * HAUT ;
CRS3 = 'ABS' CRS2 ;
CRS4 = CRS2 + CRS3 / 2.DO ;

```

```

SI ( COMPLET ) ;
daxh=drbas plus (0. 0.5) ;
daxv=drgau plus (1. 0.) ;
elim (daxh et daxv et ptot1) 1.e-5 ;
SINON ;
daxh=drbas plus (0. 1.) ;
daxv=drgau plus (0.5 0.) ;
elim (daxh et daxv et ptot1) 1.e-5 ;
FINSI ;
*
*- Vitesses aux sommets
*
SPEEX = 'NOMC' 'UX' 'YS' 'NATU' 'DISCRET' ;
SPEEY = 'NOMC' 'UY' 'XS' 'NATU' 'DISCRET' ;
SPEEY = (-1.DO) * SPEEY ;

TN      = KCHT HYTOT SCAL SOMMET CRS4      ;
evolth=evol 'CHPO' TN daxh ;
evoltv=evol 'CHPO' TN daxv ;
s2c= SOMT ((NOEL HYTOT TN)*XVOL) ;
mess ' Solution exacte projetée ' s2c ;

DEBPROC TEST ;
ARGU OPDFDT*MOT OPKONV*MOT BETA*FLOTTANT CFL*FLOTTANT
      DIAM*FLOTTANT GRAPH*MOT ;
DT=CFL*0.1 ;
NBIT=enti (3.146 * DIAM / DT) + 1 ;
      mess 'NBIT=' nbit ' DT=' dt;
coef = 1.e-20 ;

RV = 'EQEX' HYTOT ITMA NBIT
      'OPTI' 'EF' OPDFDT
      'ZONE' HYTOT 'OPER' 'DFDT' 1. 'TN' DT 'UN' coef 'INCO' 'TN'
      'OPTI' 'EF' 'SEMI' BETA OPKONV
      'ZONE' HYTOT 'OPER' 'KONV' 1. 'UN' COEF DT 'INCO' 'TN'
      'CLIM' 'TN' 'TIMP' DRCOT 0. ;

RV.'INCO'          = TABLE 'INCO'          ;
RV.'INCO'.'TN'     = KCHT HYTOT SCAL SOMMET CRS4      ;
RV.'INCO'.'TO'     = KCHT HYTOT SCAL SOMMET 0.        ;
RV.'INCO'.'UN'     = KCHT HYTOT VECT SOMMET SPEEX SPEEY ;

u = vect RV.'INCO'.'UN' 1. ux uy ;
*trac u ptot2;
*trac RV.'INCO'.'TN' ptot1;

rv.'OMEGA'= 1. ;
rv.'NITER'=1 ;

exec rv ;

maxc=maxi (RV.'INCO'.'TN') ; minc=mini (RV.'INCO'.'TN');

```

```

mess 'MAXC=' maxc 'MINC=' minc ;
TN=(RV.'INCO'.'TN') ;
s2c= SOMT ((NOEL HYTOT TN)*XVOL) ;
mess ' S2C= ' s2c ;

SI ('EGA' graph '0' );
opti dime 3 ;
P1=-5. -5. 2. ;
montagne (RV.'INCO'.'TN') ptot1 2.
'SOLUTION APRES UNE ROTATION' P1 2 ;
opti dime 2 ;
trac RV.'INCO'.'TN' ptot1 (prog -0.1 pas 0.1 0.9) drcot ;
FINSI ;
FINPROC RV ;

* SCHEMA CRANK-NICOLSON GENERALISE GALERKIN (CNG)

CFL=1. ;
BETA=0.5 ;
OPDFDT='CENTREE' ;
OPKONV='CNG' ;
rv = TEST OPDFDT OPKONV BETA CFL DIAM GRAPH ;
evolh=evol 'CHPO' (RV.'INCO'.'TN') daxh ;
evolv=evol 'CHPO' (RV.'INCO'.'TN') daxv ;
evolth=evolth et evolh ;
evoltv=evoltv et evolv ;

SI ( COMPLET ) ;
lrr=prog 1.74749E-35 -1.99218E-03 7.05737E-03 -4.28096E-03 -4.02347E-03
1.67639E-02 -1.03904E-03 -3.81883E-03 -1.18897E-03 2.74398E-02
-2.86245E-05 -2.63234E-02 5.74495E-02 3.16722E-01 6.70855E-01
8.59386E-01 6.86044E-01 3.10690E-01 5.87646E-02 -2.73409E-02
3.52053E-34 ;

SINON ;
lrr=prog 5.46903E-36 2.09259E-03 2.88562E-03 1.12206E-03 -4.76120E-03
-9.60266E-04 -4.48346E-03 6.03275E-02 3.35740E-01 7.30999E-01
9.08812E-01 6.66444E-01 2.81078E-01 1.99606E-02 -2.04501E-02
1.80881E-02 -3.02145E-03 -8.24603E-03 -4.76773E-03 1.34593E-02
-3.43162E-34 ;

FINSI ;
lr=extr evolv 'ORDO';
ER=resu( abs (lr - lrr) ) ;
mess ' Erreur sur CNG ' er ;
si ( er > err1 ) ; erreur 5 ; finsi ;
FINSI ;

* SCHEMA CRANK-NICOLSON GALERKIN (CN)

CFL=1. ;
BETA=0.5 ;
OPDFDT='CENTREE' ;
OPKONV='CENTREE' ;
rv = TEST OPDFDT OPKONV BETA CFL DIAM GRAPH ;

```

```

evolh=evol 'CHPO' (RV.'INCO'.'TN') daxh ;
evolv=evol 'CHPO' (RV.'INCO'.'TN') daxv ;
evolth=evolth et evolh ;
evoltv=evoltv et evolv ;

SI ( COMPLET ) ;
lrr=prog
-1.04066E-34  2.91445E-03 -1.00833E-02  1.50326E-02 -9.15096E-03
-2.67805E-03  1.53555E-02  3.27736E-03  3.51457E-03  2.15993E-02
-4.75657E-17 -1.02280E-02  2.47179E-02  2.96349E-01  5.91085E-01
 7.31227E-01  5.41867E-01  2.21427E-01  4.23593E-02 -1.08779E-02
 1.80257E-34;
SINON ;
lrr=prog 9.58956E-37  1.73659E-03 -2.57124E-03  1.73350E-03 -3.51837E-04
-1.31922E-03  1.28652E-02  8.03178E-02  2.99399E-01  6.56892E-01
 9.00858E-01  7.45170E-01  3.25613E-01  1.36989E-02 -7.24436E-02
-2.25911E-02  6.70052E-03 -5.58714E-03  2.41537E-03  6.56414E-03
-1.09803E-34 ;
FINSI ;
  lr=extr evolv 'ORDO';
  ER=resu( abs (lr - lrr) ) ;
  mess ' Erreur sur CN ' er ;
  si ( er > err1 ) ; erreur 5 ; finsi ;

* SCHEMA CRANK-NICOLSON PETROV-GALERKIN (CN-PG)

CFL=1. ;
BETA=0.5 ;
OPDFDT='SUPG' ;
OPKONV='SUPG' ;
rv = TEST OPDFDT OPKONV BETA CFL DIAM GRAPH ;
evolh=evol 'CHPO' (RV.'INCO'.'TN') daxh ;
evolv=evol 'CHPO' (RV.'INCO'.'TN') daxv ;
evolth=evolth et evolh ;
evoltv=evoltv et evolv ;

SI ( COMPLET ) ;
lrr=prog
 5.77432E-38  1.31358E-06 -5.21667E-05 -1.15643E-05  2.05947E-04
 1.80299E-04 -6.60164E-04 -1.11064E-03  8.69688E-04 -1.45458E-03
 1.33018E-02 -5.65454E-03  1.01072E-01  2.90655E-01  4.62448E-01
 5.15698E-01  4.29900E-01  2.69022E-01  1.18361E-01  2.83177E-02
-5.78273E-34 ;
SINON ;
lrr=prog 1.96845E-36 -1.73228E-04 -2.73666E-04  1.88624E-03 -3.02943E-05
-1.47515E-02 -4.38510E-03  1.23668E-01  3.97215E-01  6.70257E-01
 7.33005E-01  5.42977E-01  2.64419E-01  7.09012E-02 -2.74620E-03
-1.14302E-02 -4.69231E-03 -5.55237E-04  2.33049E-04  6.94725E-05
 5.43458E-38 ;
FINSI ;
  lr=extr evolv 'ORDO';
  ER=resu( abs (lr - lrr) ) ;
  mess ' Erreur sur CN-PG ' er ;

```

```

    si ( er > err1 ) ; erreur 5 ; finsi ;

* SCHEMA EULER IMPLICITE PETOV-GALERKIN (IMPL-PG)

CFL=1. ;
BETA=1. ;
OPDFDT='SUPG' ;
OPKONV='SUPG' ;
rv = TEST OPDFDT OPKONV BETA CFL DIAM GRAPH ;
evolh=evol 'CHPO' (RV.'INCO'.'TN') daxh ;
evolv=evol 'CHPO' (RV.'INCO'.'TN') daxv ;
evolth=evolth et evolh ;
evoltv=evoltv et evolv ;

SI ( COMPLET ) ;
lrr=prog
  4.18862E-37 -3.96745E-05 -7.28934E-06  1.41207E-04  4.21582E-04
  6.60423E-04  5.20803E-04 -2.65280E-04 -2.09047E-03 -6.75981E-03
  1.28871E-02 -1.29009E-02  4.49156E-02  1.44254E-01  2.23218E-01
  2.32834E-01  1.76272E-01  9.55129E-02  3.27831E-02  1.25697E-03
  -3.34441E-35;
SINON ;
lrr=prog 7.40684E-36 -2.75671E-04 -1.46179E-03 -1.94593E-03 5.00567E-03
  3.22896E-02  9.36961E-02  1.92874E-01  3.12966E-01  4.16705E-01
  4.62149E-01  4.26676E-01  3.22806E-01  1.93018E-01  8.45311E-02
  2.20555E-02  2.65230E-05 -2.16917E-03 -4.85344E-04 6.16728E-05
  2.36981E-36 ;
FINSI ;
  lr=extr evolv 'ORDO';
  ER=resu( abs (lr - lrr) ) ;
  mess ' Erreur sur IMPL-PG ' er ;
  si ( er > err1 ) ; erreur 5 ; finsi ;

si ('EGA' graph '0' );
titre 'Coupe ox à y=1/2 ' ;
TAB1=TABLE;
TAB1.1='TIRR MARQ CROI REGU TITR Tini ' ;
TAB1.2='MARQ ETOI REGU TITR CNG CFL 1. ' ;
TAB1.3='MARQ CROI REGU TITR CN CFL 1. ' ;
TAB1.4='MARQ LOSA REGU TITR CNPG CFL 1. ' ;
TAB1.5='MARQ CARR REGU TITR IMPLPG CFL 1. ' ;
DESS EVOLTH 'TITX' 'Ox' LEGE TAB1 ;
titre 'Coupe oy à x=0 ' ;
DESS EVOLTV 'TITX' 'Oy' LEGE TAB1 ;
FINSI ;

```

1.2.3 Formulation Eléments Finis Mixtes avec Hybridation

<i>Procédure utilisée</i>	<i>DARCYTRA</i>
<i>Méthodes numériques</i>	<i>EFMH, élément HYQ4, schéma explicite décentré, schéma explicite de type tenseurs visqueux, schéma de Crank-Nicolson et schéma implicite</i>
<i>Nature du problème</i>	<i>Convection instationnaire d'un cône de concentration, dans un champ de vitesse circulaire</i>
<i>Classification</i>	<i>2D plan, convection, instationnaire</i>
<i>Jeu de données</i>	

Description du problème :

Ce cas-test traite la convection pure d'un cône de concentration dans un champ de vitesse circulaire,

$$\frac{\partial c}{\partial t} + \vec{\lambda} \cdot \nabla c = 0$$

où $\vec{\lambda} = y\vec{1}_x - x\vec{1}_y$. On résoud cette équation sur le domaine Ω , $(x, y) \in [-1, 1]^2$, avec comme solution initiale un cône unitaire centré en $(0, -0.5)$ et de rayon 0.3.

On cherche à calculer la solution au bout du temps $t = 2\pi$, correspondant à une rotation complète. La solution exacte de ce problème de convection pure est évidemment identique à la solution initiale.

Modélisation :

On traite le problème avec la procédure DARCYTRA qui calcule le transport par diffusion-convection d'un champ scalaire par un schéma basé sur une discrétisation spatiale par EFMH [1].

Le coefficient de diffusion est choisi afin que le Peclet de maille soit de l'ordre de 10^5 . La diffusion est alors négligeable devant la convection. On peut donc utiliser cette procédure pour un problème de convection pure.

Nous comparons dans ce cas test différents schémas numériques sur un maillage régulier constitué de quadrangles. Les conditions aux limites sont mixtes : pour chaque face du maillage appartenant à la frontière, on regarde si la vitesse est entrante ou sortante. Si elle est entrante on impose à zero la valeur moyenne de la concentration sur la face ; sinon, on impose le flux diffusif à zero.

Les schémas considérés ici sont :

- ▷ un schéma implicite ;
- ▷ un schéma de Crank-Nicolson ;
- ▷ un schéma explicite de type tenseurs visqueux ;
- ▷ un schéma explicite décentré ;

Les deux premiers schémas correspondent à une intégration temporelle par une θ -méthode avec $\theta = 1$ et $\theta = 0.5$, la discrétisation spatiale étant centrée. Le schéma de Crank-Nicolson est d'ordre 2 en temps et en espace alors que le schéma implicite centré n'est que d'ordre 1 en temps.

Les schémas explicites sont stabilisés par ajout de viscosité artificielle. L'ordre du schéma des tenseurs-visqueux est théoriquement 2 en temps et en espace alors que le schéma décentré n'est que d'ordre 1 en temps et en espace.

La diffusion physique étant nulle, seule la diffusion numérique est responsable de l'aplatissement du cône. C'est pourquoi on s'intéresse aux extrema des grandeurs scalaires calculées par l'approche EFMH : la valeur moyenne du champ de concentration par élément (noté c) et la valeur moyenne du champ de concentration par face (encore appelée trace de concentration et noté T_c). On aura ainsi une mesure locale de l'effet de la diffusion numérique.

Une mesure globale de cet effet consiste à évaluer au bout d'un tour la quantité de matière Q présente dans le domaine de calcul et de comparer cette quantité à la quantité initiale de matière présente dans le domaine. La matière ayant quitté le domaine de calcul au cours de la rotation du cône permet de mesurer la perte de masse par diffusion numérique. Le maillage utilisé est pénalisant pour ce dernier test puisque la frontière du domaine est proche du cône (seulement deux mailles entre le cône et la frontière).

Ces différentes valeurs permettent de comparer les différents schémas utilisés. De plus, la trace de concentration au bout d'un tour est comparée au champ initial sur une coupe verticale et une coupe horizontale passant par le centre du cône. Les lignes de coupe sont visualisées figure 1.2.9.

Afin de comparer les différents schémas, le même pas de temps est utilisé, en explicite comme en implicite. On effectue un tour en 800 pas de temps. Cela correspond à une CFL de 0.1.

Analyse des résultats :

Comme le schéma explicite décentré n'est que d'ordre 1 en temps et en espace, il doit être le plus diffusif et le principe du maximum doit être vérifié. Les schémas de Crank-Nicolson et des tenseurs visqueux étant d'ordre 2 en temps et en espace, la solution calculée après un tour doit présenter des oscillations numériques. De ce fait, le principe du maximum ne peut pas être respecté. Le schéma implicite n'étant que d'ordre 1 en temps, la viscosité numérique ainsi introduite doit atténuer les oscillations numériques, l'ordre 2 en espace permettant d'avoir moins de diffusion que dans le cas explicite décentré.

L'étude des différents indicateurs regroupés dans le tableau 1.2.1 et les coupes visualisées figures 1.2.10 et 1.2.11 sont comparables avec les résultats obtenus par éléments finis (cf fiches de validation en EF sur ce cas test dans une configuration légèrement différente et [2] pour cette configuration calculée en éléments finis avec TRIO-EF) et conformes à la théorie.

Mis à part le schéma explicite décentré qui ajoute de la diffusion dans toutes les directions, on voit sur les coupes réalisées que les autres schémas sont peu diffusifs dans la direction transverse à l'écoulement. Par contre, ils présentent un léger retard de phase.

Références :

- [1] DABBENE F., *Schémas de diffusion-convection en éléments finis mixtes hybrides*, Rapport interne CEA, DMT95/613, Saclay, 1995.

[2] LE GARREC S., *Modélisation et simulation numérique de l'écoulement et de la dispersion d'un contaminant dans une colonne pulsée*, pp.212-213, Thèse de Doctorat, CNAM, 1992.

	$\min Tc$	$\max Tc$	$\min c$	$\max c$	$Q \times 10^{-2}$
Solution initiale	0.	0.833	0.	0.764	9.439
Implicite	-0.02	0.684	-0.02	0.648	9.428
Crank-Nicolson	-0.08	0.851	-0.07	0.795	9.430
Explicite décentré	0.	0.053	4d-5	0.052	5.922
Explicite tenseurs visqueux	-0.08	0.813	-0.08	0.782	9.439

TAB. 1.2.1 – Convection d'un cône dans un champ de vitesse circulaire. Comparaison de la diffusion numérique des différents schémas.




FIG. 1.2.8 – Convection d'un cône dans un champ de vitesse circulaire. Visualisation de la trace du cône et du maillage




FIG. 1.2.9 – Convection d'un cône dans un champ de vitesse circulaire. Visualisation de la trace du cône et des coupes verticale et horizontale sur lesquelles seront effectuées des post-traitements.




FIG. 1.2.10 – Convection d'un cône dans un champ de vitesse circulaire. Valeurs moyennes de la concentration par face, coupe horizontale.




FIG. 1.2.11 – Convection d'un cône dans un champ de vitesse circulaire. Valeurs moyennes de la concentration par face, coupe verticale.

Jeu de données :

```

*
*-- Utilisation du modèle DARCY pour la résolution du
*-- transport par diffusion-convection d'un champ scalaire.
*
*-----
*= Transport d'un cône par un champ de vitesse à rotationnel constant.
*= Selection du schéma de transport par la variable SCH.
*-----
*= Données issues de la thèse de S.LE GARREC, Modélisation et
*= simulation numérique de l'écoulement et de la dispersion d'un
*= contaminant dans une colonne pulsée, Thèse CNAM, 1992, pp212-213.
*-----
*
*-----
* Options générales
*-----
*
GRAPH = '0' ;
'OPTI' 'DIME' 2 'ELEM' 'QUA4' ;
'OPTI' 'ISOV' 'SURF' ;

```

```

'OPTI' 'ECHO' 1          ;
*
*=====
* Choix du schéma de transport
*=====
*
* SCH=0 : schéma centré,
* SCH=1 : schéma des Tenseurs visqueux,
* SCH=2 : schéma décentré amont (Upwind).
* TETD  : theta-méthode pour le schéma en temps (flux diffusif).
* TETC  : theta-méthode pour le schéma en temps (flux convectif).
* Implicite
*SCH = 0 ; TETD = 1.0D0 ; TETC = 1.0D0 ;
* Crank-Nicolson
*SCH = 0 ; TETD = 0.5D0 ; TETC = 0.5D0 ;
* Tenseurs Visqueux
*SCH = 1 ; TETD = 0.0D0 ; TETC = 0.0D0 ;
* Upwind (explicite décentré
SCH = 2 ; TETD = 0.0D0 ; TETC = 0.0D0 ;
*=====
* MAILLAGE
*=====
*
*
*- Création des points supports des DROITES
*
L      = 2.D0          ;
LS2    = L / 2.D0      ;
H      = L              ;
HS2    = H / 2.D0      ;
X0     = -1.D0 * LS2   ;
X1     = X0 + L        ;
Y0     = -1.D0 * HS2   ;
Y1     = Y0 + H        ;
INUMX  = 20            ;
INUMY  = INUMX         ;
XNUMY  = 'FLOT' INUMX ;
INUX1  = INUMX - 1     ;
INUY1  = INUMY - 1     ;
XO1    = X0 + X1 * 0.5D0 ;
YO1    = Y0 + Y1 * 0.5D0 ;
DX     = X1 - X0 / INUMX ;
DY     = Y1 - Y0 / INUMY ;
DX1    = DX / 2.D0     ;
DY1    = DY / 2.D0     ;
*
B1 = X0 Y0 ;
B3 = X1 Y0 ;
H1 = X0 Y1 ;
H3 = X1 Y1 ;
*
DIR1 = 1.D0 0.D0 ;
*

```

```

*- Création des DROITES frontieres
*
DRBAS = B1 'DROI' INUMX B3 ;
DRDRO = B3 'DROI' INUMY H3 ;
DRHAU = H3 'DROI' INUMX H1 ;
DRGAU = H1 'DROI' INUMY B1 ;
DRCOT = DRBAS 'ET' DRDRO 'ET' DRGAU 'ET' DRHAU ;
PELIM = DX1 / (5.DO * INUMX) ;
*
*- Creation maillage GEOMETRIQUE
*
PTOT1 = 'DALL' DRBAS DRDRO DRHAU DRGAU ;
PTOT2 = 'ORIE' PTOT1 ;
*
*- Creation maillage HYBRIDE y compris sous-objets (cond. limites)
*
HYTOT = 'DOMA' PTOT2 PELIM ;
HYCOT = 'DOMA' DRCOT 'INCL' HYTOT PELIM ;
VOLU1 = 'DOMA' HYTOT 'VOLUME' ;
CHYB1 = 'DOMA' HYTOT 'SURFACE' ;
CHYB2 = 'DOMA' HYTOT 'NORMALE' ;
MCHYB = 'DOMA' HYTOT 'ORIENTAT' ;
*
*- Construction de la ligne joignant les faces pour post traitement
*
* XCEN : Abscisse du sommet du cone
* YCEN : Ordonnée du sommet du cone
* RCEN : Rayon du cone
*
XCEN = 0.DO ;
YCEN = -0.5DO ;
RCEN = 3.0DO * DX ;
*
YTEST = XNUMY / 2.DO ;
ITEST = INUMY / 2 ;
YTES2 = 'FLOT' ITEST ;
'SI' ( YTEST 'EGA' YTES2 PELIM ) ;
  XG = X0 + DX1 ;
  YB = Y0 + DY1 ;
  XD = X1 - DX1 ;
  YH = Y1 - DY1 ;
  MX1 = XG YCEN ;
  MY1 = XCEN YB ;
  MX3 = XD YCEN ;
  MY3 = XCEN YH ;
  DRMIX = MX1 'DROI' INUX1 MX3 ;
  DRMIY = MY1 'DROI' INUY1 MY3 ;
'SINO' ;
  XG = X0 ;
  YB = Y0 ;
  XD = X1 ;
  YH = Y1 ;
  MX1 = XG YCEN ;

```



```

*
*- Vitesses aux faces
*
SPEEX = 'NOMC' 'VX' YF 'NATU' 'DISCRET' ;
SPEEY = 'NOMC' 'VY' XF 'NATU' 'DISCRET' ;
SPEEY = (-1.DO) * SPEEY ;
SPEEF = SPEEX 'ET' SPEEY ;
VVVV = 'VECT' SPEEF 0.25 'VX' 'VY' 'ROUG' ;
*
*- Flux aux faces
*
MOT1 = 'MOTS' 'VX' 'VY' ;
MOT2 = 'MOTS' 'UX' 'UY' ;
SOUR1 = 'PSCA' SPEEF CHYB2 MOT1 MOT2 ;
SOUR1 = CHYB1 * SOUR1 ;
SOUR2 = 'NOMC' 'FLUX' SOUR1 ;
SOUR3 = (-1.DO) * SOUR2 ;
*
*- Vitesses aux centres
*
SPEEX = 'NOMC' 'VX' YC 'NATU' 'DISCRET' ;
SPEEY = 'NOMC' 'VY' XC 'NATU' 'DISCRET' ;
SPEEY = (-1.DO) * SPEEY ;
SPEEC = SPEEX 'ET' SPEEY ;
*
*- Coefficient d'emmagasinement
*
COEMM = 'MANU' 'CHML' HYTOT.MAILLAGE 'CK' CEPS ;
*
*- Tenseur de diffusion TV ou UPWIND
*
'SI' (SCH 'NEG' 0) ;
  MODHYB = 'MODL' HYTOT 'DARCY' 'ANISOTROPE' ;
  CHVK = 'MANU' 'CHPO' HYTOT.CENTRE 1 'SCAL' VK ;
'SI' (SCH 'EGA' 1) ;
  NORX = DELTAT / 2.DO * YC * YC + CHVK ;
  NORV = DELTAT / 2.DO * XC * XC + CHVK ;
  NOXY = (-0.0DO) * DELTAT / 2.DO * XC * YC ;
  LTIO = 'CHAINE' 'Lega3 - TV - Th t/tetadc' ;
  LTI2 = 'CHAINE' 'Lega3 - TV - Th tetadc' ;
  LTI3 = 'CHAINE' LTI2 TETD TETC ;
'SINO' ;
  NORV = XC*XC + (YC*YC) ** 0.5DO ;
  NORX = DX * YC * YC * (NORV ** -1.DO) ;
  NORV = DY * XC * XC * (NORV ** -1.DO) ;
  NOXY = 0.DO * YC ;
  LTIO = 'CHAINE' 'Lega3 - Upwind - Th t/tetadc' ;
  LTI2 = 'CHAINE' 'Lega3 - Upwind - Th tetadc' ;
  LTI3 = 'CHAINE' LTI2 TETD TETC ;
'FINS' ;
DIFNX = 'KCHA' HYTOT 'CHAM' NORX ;
DIFNY = 'KCHA' HYTOT 'CHAM' NORV ;
DIFXY = 'KCHA' HYTOT 'CHAM' NOXY ;

```

```

      MATI2 = 'MATE' MODHYB 'DIRECTION' DIR1
              'K11' DIFNX 'K22' DIFNY 'K21' DIFXY ;
'SINO'
      MODHYB = 'MODL' HYTOT 'DARCY' 'ISOTROPE' ;
      NORV   = XC*XC + (YC*YC) ** 0.5D0 ;
      CHVK   = (1.D0 / PECM) * (VOLU1 ** 0.5D0) * NORV ;
      DIFXX  = 'MAXI' CHVK ;
      LTI0   = 'CHAINE' 'Lega3 - Implicite - Th t/tetadc' ;
      LTI2   = 'CHAINE' 'Lega3 - Implicite - Th tetadc' ;
      LTI3   = 'CHAINE' LTI2 TETD TETC ;
      MATI2 = 'MATE' MODHYB 'K' DIFXX ;
'FINS'
*
*
*
*
*
* h(x,y) = f(r)
* avec f(r) = max( h(1.-r/rcen) , 0.)
* On se sert des égalités suivantes :
*   m = min(f,g) = (f+g - abs(f-g)) / 2
*   M = max(f,g) = (f+g + abs(f-g)) / 2
*
HAUT = 1.D0 ;
*
*- Charge aux centres de gravité
*
CRC1 = (XC-XCEN)*(XC-XCEN) + ((YC-YCEN)*(YC-YCEN)) ** 0.5D0 ;
CRC2 = 1.D0 - (CRC1 / RCEN) * HAUT ;
CRC3 = 'ABS' CRC2 ;
CRC4 = CRC2 + CRC3 / 2.D0 ;
TCHYB = 'NOMC' 'H' CRC4 'NATU' 'DISCRET' ;
MCHO = 'KCHA' HYTOT 'CHAM' TCHYB ;
MO = 'INTG' MODHYB MCHO ;
*
FLHYB = 'MANU' 'CHPO' HYTOT.FACE 1 'FLUX' 0.D0 ;
*
*
*- Trace de Charge aux faces
*
CRF1 = (XF-XCEN)*(XF-XCEN) + ((YF-YCEN)*(YF-YCEN)) ** 0.5D0 ;
CRF2 = 1.D0 - (CRF1 / RCEN) * HAUT ;
CRF3 = 'ABS' CRF2 ;
CRF4 = CRF2 + CRF3 / 2.D0 ;
TFHYB = 'NOMC' 'TH' CRF4 'NATU' 'DISCRET' ;
*
*- Charge aux sommets
*
CRS1 = (XS-XCEN)*(XS-XCEN) + ((YS-YCEN)*(YS-YCEN)) ** 0.5D0 ;
CRS2 = 1.D0 - (CRS1 / RCEN) * HAUT ;
CRS3 = 'ABS' CRS2 ;
CRS4 = CRS2 + CRS3 / 2.D0 ;
TSHYB = 'NOMC' 'T' CRS4 'NATU' 'DISCRET' ;

```



```
DARCYTRA TRANS2 ;
*
*- Sauvegarde
*
'OPTI' 'SAUV' UP ;
*'OPTI' 'SAUV' IMPL ;
*'OPTI' 'SAUV' TV ;
*'OPTI' 'SAUV' CN ;
'SAUV' ;
*
FIN ;
```

Post Traitement :

```

*
*=====
* Post traitement du cone traité en EFMH
*=====
*
*   Chaque calcul ayant été sauvé, on relit chaque fichier
*   qu'on post-traite, puis on assemble et on trace.
*
*- Coupes dans le cas implicite
*
'OPTI' 'REST' 'IMPL' ;
'REST' ;
ETP1 = TRANS2 . 'TRACE_CHARGE' . 1          ;
XEVO2 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIX ;
YEVO2 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIY ;
*
*- Coupes dans le cas Crank-Nicolson
*
'OPTI' 'REST' CN ;
'REST' ;
ETP1 = TRANS2 . 'TRACE_CHARGE' . 1          ;
XEVO3 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIX ;
YEVO3 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIY ;
*
*- Coupes dans le cas Tenseurs-Visqueux
*
'OPTI' 'REST' TV ;
'REST' ;
ETP1 = TRANS2 . 'TRACE_CHARGE' . 1          ;
XEVO4 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIX ;
YEVO4 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIY ;
*
*- Coupes dans le cas explicite décentré
*
'OPTI' 'REST' UP ;
'REST' ;
ETP1 = TRANS2 . 'TRACE_CHARGE' . 1          ;
XEVO5 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIX ;
YEVO5 = 'EVOL' 'BLAN' 'CHPO' ETP1 'TH' DRMIY ;
*
*- Tracé du maillage, de la trace du cone et des coupes
*
'OPTI' 'DIME' 2 'ELEM' 'QUA4' 'TRAC' 'PS' ;
PTR1 = XCEN - RCEN ;
PTR2 = XCEN + RCEN ;
PTR3 = YCEN + RCEN ;

```

```

PTR4 = YCEN - RCEN ;
PPC1 = XCEN PTR3 ;
PPC2 = PTR1 YCEN ;
PPC3 = XCEN PTR4 ;
PPC4 = PTR2 YCEN ;
PPCEN = XCEN YCEN ;
CC1 = 'CERC' 25 PPC1 PPCEN PPC2 ;
CC2 = 'CERC' 25 PPC2 PPCEN PPC3 ;
CC3 = 'CERC' 25 PPC3 PPCEN PPC4 ;
CC4 = 'CERC' 25 PPC4 PPCEN PPC1 ;
CCT = CC1 'ET' CC2 'ET' CC3 'ET' CC4 ;
'TITR' ' ' ;
'TRAC' (CCT 'ET' (HYTOT.'MAILLAGE')) ;
'TRAC' (DRCOT 'ET' CCT 'ET' DRMIX 'ET' DRMIY) ;
*
*- Table des Légendes pour tracé des coupes
*
TAB1 = 'TABLE' ;
TAB1.1 = ' TITR Solution_initiale' ;
TAB1.2 = 'MARQ CARR TITR Implicite' ;
TAB1.3 = 'MARQ CROI TITR Crank-Nicolson' ;
TAB1.4 = 'MARQ PLUS TITR Tenseurs_Visqueux' ;
TAB1.5 = 'MARQ LOSA TITR Upwind' ;
*
*- Solution initiale
*
ETPO = TRANS2 . 'TRACE_CHARGE' . 0 ;
XEVO1 = 'EVOL' 'BLAN' 'CHPO' ETPO 'TH' DRMIX ;
YEVO1 = 'EVOL' 'BLAN' 'CHPO' ETPO 'TH' DRMIY ;
*
*- Tracé des coupes
*
'TITR' ' ' ;
'DESS' (XEVO1 'ET' XEVO2 'ET' XEVO3 'ET' XEVO4 'ET' XEVO5) 'LEGE' TAB1;
'DESS' (YEVO1 'ET' YEVO2 'ET' YEVO3 'ET' YEVO4 'ET' YEVO5) 'LEGE' TAB1;
*
'FIN' ;

```

1.3 Equation de conservation linéaire

1.3.1 Formulation Eléments Finis Explicite Conservative

<i>Opérateurs utilisés</i>	<i>TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation P1 ou Q1, schéma SUPG</i>
<i>Nature du problème</i>	<i>Transport d'un scalaire dans un champ de vitesse à divergence non-nulle.</i>
<i>Classification</i>	<i>2D plan, convection, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/consmasse.dgibi</i>

Description du problème :

On s'intéresse à la solution stationnaire du problème de conservation linéaire sur le domaine $\Omega = \{(x, y) \in [0, 2] \times [0, 1]\}$:

$$\frac{\partial \phi}{\partial t} + \nabla \cdot (\underline{u}\phi) = 0 \quad \forall \underline{x} \in \Omega,$$

avec \underline{u} le champ de vitesse à divergence non-nulle défini par :

$$\underline{u} = \begin{pmatrix} \alpha x + \beta y + \gamma \\ 0 \end{pmatrix}, \quad \nabla \cdot \underline{u} = \alpha \neq 0$$

et la condition aux limites en entrée, $\phi = 1$ sur $x = 0$.

Ce problème admet pour solution stationnaire la fonction suivante :


$$\phi_{ex}(x, y) = \frac{\beta y + \gamma}{\alpha x + \beta y + \gamma}$$

qui vérifie évidemment la condition de conservation sur le flux :

$$I(x) = I(0) = \frac{\beta}{2} + \gamma, \quad \forall x, 0 \leq x \leq 2$$

où $I(x)$ est le flux à travers une section du domaine d'abscisse x et de normale unitaire $\underline{1}_x$,

$$I(x) = \int_0^1 \underline{u}(x, y) \phi(x, y) \cdot \underline{1}_x \, dy$$

FIG. 1.3.1 – Solution exacte $\phi_{ex}(x, y)$ **Modélisation :**

On s'intéresse au problème défini par les paramètres $\alpha = 1$, $\beta = 2$ et $\gamma = 1$. On résout ce problème à l'aide de l'opérateur de convection TSCAL, avec l'option CONS (Conservatif) et l'option de décentrement SUPG. On effectue 2 calculs, l'un sur un maillage de quadrangles contenant 40×20 éléments, l'un sur triangles contenant le double d'éléments (subdivision de chaque quadrangle en deux).

Synthèse des résultats :

On s'intéresse à deux indicateurs d'erreur de conservation : L'erreur relative entre le flux entrant et le flux sortant, $\Delta I/I$, et l'erreur relative sur tout le domaine et évaluée en norme L_∞ discrète entre la solution numérique ϕ et la solution exacte ϕ_{ex} .

maillage	$\Delta I/I$	$\ \Delta\phi/\phi\ _{L_\infty}$
□	$1.4 \times 10^{-5} \%$	0.7 %
△	$1.0 \times 10^{-4} \%$	0.2 %

Ces résultats montrent que les erreurs de conservation de l'opérateur Eléments Finis TSCAL (avec les options CONS et SUPG) sont très faibles, que ce soit avec des maillages de quadrangles ou de triangles.

Jeu de données :

```

COMPLET = FAUX ;
SI ( COMPLET ) ;
    NX = 40 ; NY = 20 ;
    NITER = 800 ;
    CFL = 0.9 ;
    TOLCONV = -5.0 ;
    TOLDEBI = 8.0E-4 ;
    TOLSOLU = 0.8 ;
SINON ;
    NX = 8 ; NY = 4 ;
    NITER = 200 ;
    CFL = 0.9 ;
    TOLCONV = -4.0 ;
    TOLDEBI = 3.0E-3 ;
    TOLSOLU = 1.7 ;
FINSI ;
GRAPH = 'N' ;

*****
* TESTE LA CONSERVATION DE LA MASSE POUR L'EQUATION SOUS FORME *
* CONSERVATIVE : dC/dt + div ( U C ) = 0 AVEC U CHAMP DE VITESSE *
* A DIVERGENCE NON NULLE *
* COMPARAISON AVEC SOLUTION ANALYTIQUE EXACTE *
* ON TESTE SUCCESSIVEMENT MAILLAGE QUA4 ET MAILLAGE TRI3 *
*****

DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;

RV = RVX.'EQEX' ;
CN = RV.INCO.'CN' ;

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;

ERR = KOPS (RV.INCO.'CN') - (RV.INCO.'CNM1') ;
ELI = MAXI ERR 'ABS' ;
ELI = (LOG (ELI + 1.0E-20))/(LOG 10.) ;
MESSAGE 'ITER ' RV.PASDETPS.'NUPASDT' ' ERREUR LINF ' ELI ;
IT = PROG RV.PASDETPS.'NUPASDT' ;
ER = PROG ELI ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER ;
FINSI ;

RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' (RV.INCO.'CN') ;

as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

```

```

opti dime 2 ;
opti elem qua4 ;
opti isov sul1 ;
titre 'Conservation de la masse' ;

*****
* DIFFUSION *
*****
DIF = 1.0E-20 ;

*****
* MAILLAGE *
*****

A1 = 0.0 0.0 ;
A2 = 2.0 0.0 ;
A3 = 2.0 1.0 ;
A4 = 0.0 1.0 ;

FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;

PA1 = FBAS POIN 'PROC' A1 ;
PA2 = FBAS POIN 'PROC' A2 ;
P12 = CHAN POI1 FBAS ;
MA1 = P12 ELEM 'CONTENANT' PA1 ;
MA2 = P12 ELEM 'CONTENANT' PA2 ;

DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;

*****
* CREATION DU MODELE NAVIER_STOKES *
*****

MDOMTOT = CHAN DOMTOT QUAF ;
$DOMTOT = MODE MDOMTOT 'NAVIER_STOKES' LINE ; DOMA $DOMTOT 'IMPR' ;
DOMTOT = DOMA $DOMTOT MAILLAGE ;
CNT = 'CONT' DOMTOT ;

FDRO = CHAN FDRO QUAF ;
FGAU = CHAN FGAU QUAF ;
ELIM (MDOMTOT ET FDRO ET FGAU ) 1.E-3 ;

$FDRO = MODE FDRO 'NAVIER_STOKES' LINE ;
$FGAU = MODE FGAU 'NAVIER_STOKES' LINE ;
FDRO = DOMA $FDRO 'MAILLAGE' ;
FGAU = DOMA $FGAU 'MAILLAGE' ;

*****
* INITIALISATION DU CHAMP DE VITESSE *

```

```

*****

alpha = 1. ;
beta = 2. ;
gamma = 1. ;

XX = COOR 1 (DOMA $DOMTOT MAILLAGE) ;
YY = COOR 2 (DOMA $DOMTOT MAILLAGE) ;
XX = KCHT $DOMTOT 'SCAL' 'SOMMET' XX ;
YY = KCHT $DOMTOT 'SCAL' 'SOMMET' YY ;
U1 = KOPS XX '*' alpha ;
U2 = KOPS YY '*' beta ;
U3 = gamma ;
U = KOPS ( KOPS U1 '+' U2) '+' U3 ;
CEXACT = KOPS ( KOPS U2 '+' U3 ) '/' U ;
U = NOMC 'UX' U 'NATU' 'DISCRET' ;
VX = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' U ;
VY = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' 0.0 ;
CHVIT=KCHT $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY') (VX ET VY);

*****
* TABLE EQEX CONTENANT L'EQUATION A RESOUDRE *
*****

RV = EQEX $DOMTOT 'ITMA' NITER 'ALFA' CFL
      'ZONE' $DOMTOT
      'OPER' CALCUL
      'OPTI' 'CONS' 'SUPG'
      'OPER' TSCAL DIF 'VITESSE' 0.0 'CN' 'INCO' 'CN'
      'OPTI' 'CENTREE'
      'OPER' DFDT 1. 'CN' 'DELTAT' 'INCO' 'CN'
      'CLIM' 'CN' 'TIMP' FGAU 1.0 ;

RV.INCO = TABLE 'INCO' ;
RV.INCO.'CN' = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
RV.INCO.'VITESSE'= CHVIT ;
RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' 1. ;
RV.INCO.'IT' = PROG 1 ;
RV.INCO.'ER' = PROG 0. ;
RV.INCO.'NBIT' = 0 ;

EXEC RV ;

*****
* ANALYSE DES RESULTATS : *
* ON TESTE SUR LA CONVERGENCE VERS LA SOLUTION STATIONNAIRE *
* SUR LA PRECISION ENTRE DEBITS ENTRANT/SORTANT *
* SUR LA PRECISION SOLUTIONS NUMERIQUE/EXACTE *
*****

GX = KOPS VX '*' RV.INCO.'CN' ;
GY = KOPS VY '*' RV.INCO.'CN' ;
GX = NOMC 'UX' GX 'NATU' 'DISCRET' ;

```



```

GY = NOMC 'UY' GY 'NATU' 'DISCRET' ;
G = KCHT $DOMTOT VECT SOMMET (GX ET GY) ;
QENTREE = DBIT G $FGAU ;
QSORTIE = DBIT G $FDRO ;
ERQ = (QENTREE+QSORTIE)/QENTREE ;
ERQ = ABS(ERQ)*100. ;
MESSAGE 'ERREUR RELATIVE ENTRE DEBIT ENTRANT ET SORTANT '
        ERQ ;

ERR = KOPS CEXACT '-' RV.INCO.'CN' ;
ERR = KOPS ERR '/' CEXACT ;
ERR = KOPS ERR '*' 100. ;
ERR = ABS ERR ;
MESSAGE 'ERREUR RELATIVE ENTRE SOL. NUMERIQUE ET EXACTE'
        (MAXI ERR) ;

SI ( (MINI RV.INCO.'ER') > TOLCONV ) ;
        ERREUR 5 ;
FINSI ;
SI ( ERQ > TOLDEBI ) ;
        ERREUR 5 ;
FINSI ;
SI ( (MAXI ERR) > TOLSOLU ) ;
        ERREUR 5 ;
FINSI ;

*****
* TRACE DES RESULTATS *
*****

SI ( 'EGA' graph '0' ) ;

VV = VECT RV.INCO.'VITESSE' 0.1 UX UY JAUNE ;
TRACE DOMTOT 'TITR' 'MAILLAGE' ;
TRACE VV DOMTOT CNT 'TITR' 'CHAMP DE VITESSE' ;
TRAC DOMTOT CEXACT CNT 'TITR' 'SOLUTION EXACTE' ;
TRAC DOMTOT (RV.INCO.'CN') CNT 'TITR' 'SOLUTION NUMERIQUE' ;

LISO = PROG 0. 2. 4. 6. 8. 10. 12. 14. 16. 20. ;
TRACE ERR DOMTOT (CONT DOMTOT) LISO 'TITR' 'ERREUR RELATIVE' ;

EVOL4 = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
        (RV.INCO.'ER') ;
DESS EVOL4 'XBOR' 0. 2000. 'YBOR' -12.0 0.0
        'TITR' 'HISTOIRE DE CONVERGENCE' ;
FINSI ;

*****
* ON TESTE MAINTENANT SUR DES TRI3 *
*****

opti elem tri3 ;
DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;

```

```

*****
* CREATION DU MODELE NAVIER_STOKES *
*****

MDOMTOT = CHAN DOMTOT QUAF ;
$DOMTOT = MODE MDOMTOT 'NAVIER_STOKES' LINE ; DOMA $DOMTOT 'IMPR' ;
DOMTOT = DOMA $DOMTOT MAILLAGE ;
CNT = 'CONT' DOMTOT ;

FDRO = CHAN FDRO QUAF ;
FGAU = CHAN FGAU QUAF ;
ELIM (MDOMTOT ET FDRO ET FGAU ) 1.E-3 ;

$FDRO = MODE FDRO 'NAVIER_STOKES' LINE ;
$FGAU = MODE FGAU 'NAVIER_STOKES' LINE ;

FDRO = DOMA $FDRO 'MAILLAGE' ;
FGAU = DOMA $FGAU 'MAILLAGE' ;

*****
* INITIALISATION DU CHAMP DE VITESSE *
*****

alpha = 1. ;
beta = 2. ;
gamma = 1. ;

XX = COOR 1 (DOMA $DOMTOT MAILLAGE) ;
YY = COOR 2 (DOMA $DOMTOT MAILLAGE) ;
XX = KCHT $DOMTOT 'SCAL' 'SOMMET' XX ;
YY = KCHT $DOMTOT 'SCAL' 'SOMMET' YY ;
U1 = KOPS XX '*' alpha ;
U2 = KOPS YY '*' beta ;
U3 = gamma ;
U = KOPS ( KOPS U1 '+' U2 ) '+' U3 ;
CEXACT = KOPS ( KOPS U2 '+' U3 ) '/' U ;
U = NOMC 'UX' U 'NATU' 'DISCRET' ;
VX = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' U ;
VY = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' 0.0 ;
CHVIT=KCHT $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY') (VX ET VY);

*****
* TABLE EQEX CONTENANT L'EQUATION A RESOUDRE *
*****

RV = EQEX $DOMTOT 'ITMA' NITER 'ALFA' CFL
      'ZONE' $DOMTOT
      'OPER' CALCUL
      'OPTI' 'CONS' 'SUPG'
      'OPER' TSCAL DIF 'VITESSE' 0.0 'CN' 'INCO' 'CN'
      'OPTI' 'CENTREE'
      'OPER' DFDT 1. 'CN' 'DELTAT' 'INCO' 'CN'

```

```

      'CLIM' 'CN' 'TIMP' FGAU 1.0 ;

RV.INCO = TABLE 'INCO' ;
RV.INCO.'CN' = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
RV.INCO.'VITESSE' = CHVIT ;
RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' 1. ;
RV.INCO.'IT' = PROG 1 ;
RV.INCO.'ER' = PROG 0. ;
RV.INCO.'NBIT' = 0 ;

EXEC RV ;

*****
* ANALYSE DES RESULTATS : *
* ON TESTE SUR LA CONVERGENCE VERS LA SOLUTION STATIONNAIRE *
* SUR LA PRECISION ENTRE DEBITS ENTRANT/SORTANT *
* SUR LA PRECISION SOLUTIONS NUMERIQUE/EXACTE *
*****

GX = KOPS VX '*' RV.INCO.'CN' ;
GY = KOPS VY '*' RV.INCO.'CN' ;
GX = NOMC 'UX' GX 'NATU' 'DISCRET' ;
GY = NOMC 'UY' GY 'NATU' 'DISCRET' ;
G = KCHT $DOMTOT VECT SOMMET (GX ET GY) ;
QENTREE = DBIT G $FGAU ;
QSORTIE = DBIT G $FDRO ;
ERQ = (QENTREE+QSORTIE)/QENTREE ;
ERQ = ABS(ERQ)*100. ;
MESSAGE 'ERREUR RELATIVE ENTRE DEBIT ENTRANT ET SORTANT '
        ERQ ;

ERR = KOPS CEXACT '-' RV.INCO.'CN' ;
ERR = KOPS ERR '/' CEXACT ;
ERR = KOPS ERR '*' 100. ;
ERR = ABS ERR ;
MESSAGE 'ERREUR RELATIVE ENTRE SOL. NUMERIQUE ET EXACTE'
        (MAXI ERR) ;

SI ( (MINI RV.INCO.'ER') > TOLCONV ) ;
        ERREUR 5 ;
FINSI ;
SI ( ERQ > TOLDEBI ) ;
        ERREUR 5 ;
FINSI ;
SI ( (MAXI ERR) > TOLSOLU ) ;
        ERREUR 5 ;
FINSI ;

*****
* TRACE DES RESULTATS *
*****

SI ( 'EGA' graph '0' ) ;

```

```
VV = VECT RV.INCO.'VITESSE' 0.1 UX UY JAUNE ;
TRACE DOMTOT 'TITR' 'MAILLAGE' ;
TRACE VV DOMTOT CNT 'TITR' 'CHAMP DE VITESSE' ;
TRAC DOMTOT CEXACT CNT 'TITR' 'SOLUTION EXACTE' ;
TRAC DOMTOT (RV.INCO.'CN') CNT 'TITR' 'SOLUTION NUMERIQUE' ;

LISO = PROG 0. 2. 4. 6. 8. 10. 12. 14. 16. 20. ;
TRACE ERR DOMTOT (CONT DOMTOT) LISO 'TITR' 'ERREUR RELATIVE' ;

EVOL4 = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
        (RV.INCO.'ER') ;
DESS EVOL4 'XBOR' 0. 2000. 'YBOR' -12.0 0.0
        'TITR' 'HISTOIRE DE CONVERGENCE' ;
FINSI ;

FIN ;
```

1.4 Equation de conservation non-linéaire : cas d'une solution avec discontinuité

1.4.1 Formulation Eléments Finis Explicite Conservative

<i>Opérateurs utilisés</i>	<i>TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation P1 ou Q1, schéma SUPG</i>
<i>Nature du problème</i>	<i>Résolution d'une équation de convection non-linéaire, avec formation d'une solution discontinue (choc).</i>
<i>Classification</i>	<i>2D plan, convection, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/burgerC.dgibi</i>

Description du problème :

Ce cas-test, proposé dans [1], concerne la résolution d'une équation de conservation non-linéaire, écrite sous forme conservative,

$$\frac{\partial u}{\partial t} + \nabla \cdot \underline{F}(u) = 0, \quad \underline{F}(u) = \begin{pmatrix} \frac{u^2}{2} \\ u \end{pmatrix}$$

sur le domaine $\Omega = [0, 1]^2$, et avec les conditions aux limites suivantes :

$$\begin{aligned} u(x, 0) &= 1.5 - 2.0x, & 0 \leq x \leq 1, \\ u(0, y) &= 1.5, & 0 \leq y \leq 1, \\ u(1, y) &= -0.5, & 0 \leq y \leq 1. \end{aligned}$$

La solution analytique de ce problème contient une discontinuité ou choc, qui se forme par convergence des caractéristiques issues de la frontière $y = 0$.

On peut également résoudre le problème en résolvant la forme non-conservative de l'équation de conservation, qui s'écrit :

$$\frac{\partial u}{\partial t} + \underline{\lambda} \cdot \nabla u = 0, \quad \underline{\lambda} = \frac{\partial \underline{F}(u)}{\partial u} = \begin{pmatrix} u \\ 1 \end{pmatrix}$$

Modélisation :

L'opérateur de convection **TSCAL**, correspondant au schéma élément fini SUPG (Streamline Upwind Petrov Galerkin), permet de traiter de tels problèmes par ajout d'un terme dit de 'capture de choc'. C'est l'option par défaut de **TSCAL** ('SUPG-CC'). Pour traiter une équation de conservation sous forme conservative, il faut également sélectionner l'option 'CONS' de **TSCAL**. Deux maillages de 31×31 nœuds chacun ont été considérés, l'un constitué de triangles, l'autre de quadrangles.

Analyse des résultats :

L'opérateur **TSCAL** (SUPG ou SUPG-CC) a été testé sur les deux maillages. Les isolignes des solutions ainsi qu'une coupe à l'ordonnée $y = 0.7$, superposée au profil correspondant à la solution exacte montrent que l'option SUPG-CC permet de réduire les oscillations par rapport à l'option SUPG. Des améliorations seront apportées dans une version ultérieure de l'opérateur afin de supprimer les oscillations restantes. Les histoires de convergence montrent que dans tous les cas, la solution numérique peut être considérée comme convergée.

Référence :

- 1- S. Spekreijse. Multigrid Solution of Monotone Second-Order Discretizations of Hyperbolic Conservation Laws. *Mathematics of Computation*, 49(179) : 135-155, 1987.




FIG. 1.4.1 – Maillages de 31×31 nœuds constitués de triangles (1800 éléments) ou de quadrangles (900 éléments)




FIG. 1.4.2 – Histoires de convergence du schéma SUPG appliqué à la forme conservative de l'équation de conservation




FIG. 1.4.3 – schéma SUPG sur triangles




FIG. 1.4.4 – schéma SUPG-CC sur triangles




FIG. 1.4.5 – schéma SUPG sur quadrangles




FIG. 1.4.6 – schéma SUPG-CC sur quadrangles

Jeu de données :

```

COMPLET = FAUX ;
SI ( COMPLET ) ;
    NX = 30 ; NY = 30 ;
    NITER = 500 ;
    CFL = 0.8 ;
SINON ;
    NX = 10 ; NY = 10 ;
    NITER = 180 ;
    CFL = 0.8 ;
FINSI ;
GRAPH = 'N' ;

*****
* EQUATION DE CONVECTION NON-LINEAIRE Ut + div (F(U)) = 0 *
* TSCAL OPTION CONS (conservatif) SUPGDC (par défaut)      *
*****

*****
* PROCEDURE POUR TRACER COUPES *
*****

DEBPROC TRCE ;
ARGU M*'MAILLAGE' ;
nb = nbel M ;
R = (M poin 1) droi 1 (M poin 2) ;
i=2 ; itma = nb-2 ;
repete bloc1 itma ;
R = R et ((M poin i) droi 1 (M poin (i+1))) ;
i=i+1 ;
fin bloc1 ;
FINPROC R ;

*****
* PROCEDURE POUR CALCULER CHAMP DE VITESSE ET *
* TESTER LA CONVERGENCE *
*****

DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;

RV = RVX.'EQEX' ;
CN = RV.INCO.'CN' ;

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;

ERR = KOPS (RV.INCO.'CN') - (RV.INCO.'CNM1') ;

```

```

ELI = MAXI ERR 'ABS' ;
ELI = (LOG (ELI + 1.0E-20))/(LOG 10.) ;
MESSAGE 'ITER ' RV.PASDETPS.'NUPASDT' ' ERREUR LINF ' ELI ;
IT = PROG RV.PASDETPS.'NUPASDT' ;
ER = PROG ELI ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER ;
FINSI ;

VY = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' 1.0 ;
VX1 = KOPS CN '/' 2. ;
VX = NOMC 'UX' VX1 ;
VV = KCHT $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY')
      (VX ET VY) ;
RV.INCO.'VITESSE' = VV ;

RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' (RV.INCO.'CN') ;
as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

*****

opti dime 2 ;
opti elem tri3 ;
opti isov lign ;
titre 'Equation de Burger' ;

*****
* DIFFUSION *
*****
DIF = 1.0E-10 ;

*****
* MAILLAGE *
*****

A1 = 0.0 0.0 ;
A2 = 1.0 0.0 ;
A3 = 1.0 1.0 ;
A4 = 0.0 1.0 ;

FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;

PA1 = FBAS POIN 'PROC' A1 ;
PA2 = FBAS POIN 'PROC' A2 ;
P12 = CHAN POI1 FBAS ;
MA1 = P12 ELEM 'CONTENANT' PA1 ;
MA2 = P12 ELEM 'CONTENANT' PA2 ;

SI ( COMPLET ) ;

```

```

* maillage regulier
DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;
SINON ;
* maillage non-structuré
DOMTOT = (FBAS ET FDRO ET FHAU ET FGAU) 'SURF' 'PLAN' ;
FINSI ;

*****
* CREATION MODELE NAVIER-STOKES *
*****

XDOMTOT = CHAN DOMTOT QUAF ;
XFBAS   = CHAN FBAS   QUAF ;
ELIM (XDOMTOT ET XFBAS) 1.E-3 ;

$DOMTOT = MODE XDOMTOT 'NAVIER_STOKES' LINE ;
$FBAS   = MODE XFBAS   'NAVIER_STOKES' LINE ;

*****
* INITIALISATION DU CHAMP DE VITESSE *
*****

VX = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' 0.0 ;
VY = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' 1.0 ;
CHVIT=KCHT $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY') (VX ET VY) ;

*****
* PROFIL DE LA SOLUTION A L'ENTREE *
*****

XBAS = COOR 1 FBAS ;
XBAS = NOMC 'UX' XBAS ;
SOLUTION = 1.5 - (2.0*XBAS) ;
CHP1 = KCHT $FBAS 'SCAL' 'SOMMET' 'COMP' 'UX' SOLUTION ;
CHP1 = NOMC 'CN' CHP1 ;

*****
* TABLE DE RESOLUTION *
*****

RV = EQEX $DOMTOT 'ITMA' NITER 'ALFA' CFL
      'ZONE' $DOMTOT
      'OPER' CALCUL
      'OPTI' 'CONS'
      'OPER' TSCAL DIF 'VITESSE' 0.0 'CN' 'INCO' 'CN'
      'OPTI' 'CENTREE'
      'OPER' 'DFDT' 1. 'CN' 'DELTAT' 'INCO' 'CN'
      'CLIM' 'CN' 'TIMP' FBAS CHP1
      'CLIM' 'CN' 'TIMP' FGAU 1.5
      'CLIM' 'CN' 'TIMP' FDRO (-0.5)
      'CLIM' 'CN' 'TIMP' MA1 (-1.5)
      'CLIM' 'CN' 'TIMP' MA2 (0.5) ;

```

```

RV.INCO = TABLE 'INCO' ;
RV.INCO.'CN' = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
RV.INCO.'VITESSE'= CHVIT ;

RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' 1. ;
RV.INCO.'IT' = PROG 1 ;
RV.INCO.'ER' = PROG 0. ;

EXEC RV ;

*****
* ANALYSE DES RESULTATS *
*****

MESSAGE 'MAX = ' (MAXI RV.INCO.'CN') 'MAXTHEORIQUE = 1.5 ' ;
MESSAGE 'MIN = ' (MINI RV.INCO.'CN') 'MINTHEORIQUE = -0.5 ' ;

SI ( (MINI RV.INCO.'ER') > -3.0 ) ;
      ERREUR 5 ;
FINSI ;
SI ( (MAXI RV.INCO.'CN') > 1.65 ) ;
      ERREUR 5 ;
FINSI ;
SI ( (MINI RV.INCO.'CN') < -0.60 ) ;
      ERREUR 5 ;
FINSI ;

*****
* TRACE DES RESULTATS *
*****

SI ( 'EGA' graph '0' ) ;

UEXACT = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
XX YY = 'COOR' (DOMA $DOMTOT SOMMET) ;
REPETER BLOC1 (NBNO DOMTOT) ;
P1 = (DOMA $DOMTOT SOMMET) POIN &BLOC1 ;
X1 = 'EXTR' XX 'SCAL' P1 ;
Y1 = 'EXTR' YY 'SCAL' P1 ;
D1 = Y1 - (0.5*X1/0.75) ;
D2 = Y1 - (2.0*(1.0-X1)) ;
D3 = Y1 - 1.0 - (2.0*(X1-1.0)) ;
BO1 = ( D1 > 0. ) ;
BO2 = ( D2 > 0. ) ;
BO3 = ( D3 > 0. ) ;
SI ( BO1 ET BO3 ) ;
U1 = 1.5 ;
SINON ;
SI ( (NON BO1) ET (NON BO2) ) ;
U1 = (1.5-(2.0*X1))/(1.-(2.0*Y1)) ;
SINON ;
U1 = -0.5 ;
FINSI ;

```

```

FINSI ;
C1 = MANU 'CHPO' P1 1 SCAL U1 ;
C2 = KCHT $DOMTOT 'SCAL' 'SOMMET' C1 ;
UEXACT = (UEXACT ET C2) ;
FIN BLOC1 ;

TRACE DOMTOT ;
MESSAGE 'MAX = ' (MAXI RV.INCO.'CN')
        'MIN = ' (MINI RV.INCO.'CN') ;
TRAC DOMTOT (RV.INCO.'CN') (CONT DOMTOT) ;

SI ( COMPLET ) ;
P1 = DOMTOT POIN 'PROC' (0.0 0.7) ;
P2 = DOMTOT POIN 'PROC' (1.0 0.7) ;
P1P2 = DOMTOT POIN 'DROI' P1 P2 0.01 ;
LI12 = TRCE P1P2 ;

XX = COOR 1 LI12 ;

EVOL1 = EVOL 'CHPO' (RV.INCO.'CN') SCAL LI12 ;
EVOL2 = EVOL 'CHPO' XX SCAL LI12 ;

LIX = 'EXTR' EVOL2 ORDO ;
LIU = 'EXTR' EVOL1 ORDO ;
EVOL3 = EVOL 'MANU' 'X' LIX 'U(X)' LIU ;

EVOL4 = EVOL 'CHPO' UEXACT SCAL LI12 ;
LIUEX = 'EXTR' EVOL4 ORDO ;
EVOL5 = EVOL 'MANU' 'X' LIX 'UEXACT(X)' LIUEX ;

TAB1 = TABLE ;
TAB1.1 = 'MARQ LOSA' ;
DESS (EVOL3 ET EVOL5) 'XBOR' 0.0 1.0 'YBOR' -1. 2.
        'TITR' 'Coupe a y=0.7' TAB1 ;
FINSI ;

EVOL4 = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
        (RV.INCO.'ER') ;
DESS EVOL4 'XBOR' 0. 2000. 'YBOR' -20.0 0.0 ;

FINSI ;

FIN ;

```

1.4.2 Formulation Eléments Finis Explicite Non-Conservative

<i>Opérateurs utilisés</i>	<i>TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation P1 ou Q1, schéma SUPG</i>
<i>Nature du problème</i>	<i>Résolution d'une équation de convection non-linéaire, avec formation d'une solution discontinue (choc), traitée en formulation non-conservative.</i>
<i>Classification</i>	<i>2D plan, convection, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/burgerNC.dgibi</i>

Jeu de données :

```

COMPLET = FAUX ;
SI ( COMPLET ) ;
    NX = 30 ; NY = 30 ;
    NITER = 500 ;
    CFL = 0.8 ;
SINON ;
    NX = 10 ; NY = 10 ;
    NITER = 180 ;
    CFL = 0.8 ;
FINSI ;
GRAPH = 'N' ;

```

```

*****
* EQUATION DE CONVECTION NON-LINEAIRE  $U_t + \text{div}(F(U)) = 0$  *
* RESOLUE SOUS FORME NON-CONSERVATIVE *
*  $dU/dt + \lambda \cdot \text{nabla} U = 0$  avec  $\lambda = dF/dU$  *
* AVEC OPTION SUPGDC (par défaut) *
*****

```

```

*****
* PROCEDURE POUR TRACER COUPES *
*****

```

```

DEBPROC TRCE ;
ARGU M*'MAILLAGE' ;
nb = nbel M ;
R = (M poin 1) droi 1 (M poin 2) ;
i=2 ; itma = nb-2 ;
repeter bloc1 itma ;
R = R et ((M poin i) droi 1 (M poin (i+1))) ;
i=i+1 ;
fin bloc1 ;
FINPROC R ;

```

```

*****
* PROCEDURE POUR CALCULER CHAMP DE VITESSE ET *
* TESTER LA CONVERGENCE *

```

```

*****

DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;

RV = RVX.'EQEX' ;
CN = RV.INCO.'CN' ;

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;

ERR = KOPS (RV.INCO.'CN') - (RV.INCO.'CNM1') ;

ELI = MAXI ERR 'ABS' ;
ELI = (LOG (ELI + 1.0E-20))/(LOG 10.) ;
MESSAGE 'ITER ' RV.PASDETPS.'NUPASDT' ' ERREUR LINF ' ELI ;
IT = PROG RV.PASDETPS.'NUPASDT' ;
ER = PROG ELI ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER ;
FINSI ;

VY = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' 1.0 ;
VX = NOMC 'UX' CN ;
VV = KCHT $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY')
      (VX ET VY) ;
RV.INCO.'VITESSE' = VV ;

RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' (RV.INCO.'CN') ;
as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

*****

opti dime 2 ;
opti elem tri3 ;
opti isov lign ;
titre 'Equation de Burger' ;

*****
* DIFFUSION *
*****
DIF = 1.0E-10 ;

*****
* MAILLAGE *
*****

A1 = 0.0 0.0 ;
A2 = 1.0 0.0 ;
A3 = 1.0 1.0 ;

```

```

A4 = 0.0 1.0 ;

FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;

PA1 = FBAS POIN 'PROC' A1 ;
PA2 = FBAS POIN 'PROC' A2 ;
P12 = CHAN POI1 FBAS ;
MA1 = P12 ELEM 'CONTENANT' PA1 ;
MA2 = P12 ELEM 'CONTENANT' PA2 ;

SI ( COMPLET ) ;
* maillage regulier
DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;
SINON ;
* maillage non-structuré
DOMTOT = (FBAS ET FDRO ET FHAU ET FGAU) 'SURF' 'PLAN' ;
FINSI ;

*****
* CREATION MODELES NAVIERS_STOKES *
*****

XDOMTOT= DOMTOT CHAN QUAF ;
XFBAS = FBAS CHAN QUAF ;
ELIM XDOMTOT XFBAS 1.E-3 ;

$DOMTOT = MODE XDOMTOT 'NAVIER_STOKES' LINE ;
$FBAS = MODE XFBAS 'NAVIER_STOKES' LINE ;

*****
* INITIALISATION DU CHAMP DE VITESSE *
*****

VX = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UX' 0.0 ;
VY = KCHT $DOMTOT 'SCAL' 'SOMMET' 'COMP' 'UY' 1.0 ;
CHVIT=KCHT $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY') (VX ET VY) ;

*****
* PROFIL DE LA SOLUTION A L'ENTREE *
*****

XBAS = COOR 1 FBAS ;
XBAS = NOMC 'UX' XBAS ;
SOLUTION = 1.5 - (2.0*XBAS) ;
CHP1 = KCHT $FBAS 'SCAL' 'SOMMET' 'COMP' 'UX' SOLUTION ;
CHP1 = NOMC 'CN' CHP1 ;

*****
* TABLE DE RESOLUTION *
*****

```



```

RV = EQEX $DOMTOT 'ITMA' NITER 'ALFA' CFL
      'ZONE' $DOMTOT
      'OPER' CALCUL
      'OPER' TSCAL DIF 'VITESSE' 0.0 'INCO' 'CN'
      'OPTI' 'CENTREE'
      'OPER' 'DFDT' 1. 'CN' 'DELTAT' 'INCO' 'CN'
      'CLIM' 'CN' 'TIMP' FBAS CHP1
      'CLIM' 'CN' 'TIMP' FGAU 1.5
      'CLIM' 'CN' 'TIMP' FDRO (-0.5)
      'CLIM' 'CN' 'TIMP' MA1 (-1.5)
      'CLIM' 'CN' 'TIMP' MA2 (0.5) ;

RV.INCO = TABLE 'INCO' ;
RV.INCO.'CN' = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
RV.INCO.'VITESSE' = CHVIT ;

RV.INCO.'CNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' 1. ;
RV.INCO.'IT' = PROG 1 ;
RV.INCO.'ER' = PROG 0. ;

EXEC RV ;

*****
* ANALYSE DES RESULTATS *
*****

MESSAGE 'MAX = ' (MAXI RV.INCO.'CN') 'MAXTHEORIQUE = 1.5 ' ;
MESSAGE 'MIN = ' (MINI RV.INCO.'CN') 'MINTHEORIQUE = -0.5 ' ;

SI ( (MINI RV.INCO.'ER') > -3.0 ) ;
      ERREUR 5 ;
FINSI ;
SI ( (MAXI RV.INCO.'CN') > 1.75 ) ;
      ERREUR 5 ;
FINSI ;
SI ( (MINI RV.INCO.'CN') < -0.60 ) ;
      ERREUR 5 ;
FINSI ;

*****
* TRACE DES RESULTATS *
*****

SI ( 'EGA' graph '0' ) ;

UEXACT = KCHT $DOMTOT 'SCAL' 'SOMMET' 0. ;
XX YY = 'COOR' (DOMA $DOMTOT SOMMET) ;
REPETER BLOC1 (NBNO DOMTOT) ;
P1 = (DOMA $DOMTOT SOMMET) POIN &BLOC1 ;
X1 = 'EXTR' XX 'SCAL' P1 ;
Y1 = 'EXTR' YY 'SCAL' P1 ;
D1 = Y1 - (0.5*X1/0.75) ;

```

```

D2 = Y1 - (2.0*(1.0-X1)) ;
D3 = Y1 - 1.0 - (2.0*(X1-1.0)) ;
B01 = ( D1 > 0.) ;
B02 = ( D2 > 0.) ;
B03 = ( D3 > 0.) ;
SI ( B01 ET B03 ) ;
U1 = 1.5 ;
SINON ;
SI ( (NON B01) ET (NON B02) );
U1 = (1.5-(2.0*X1))/(1.-(2.0*Y1)) ;
SINON ;
U1 = -0.5 ;
FINSI ;
FINSI ;
C1 = MANU 'CHPO' P1 1 SCAL U1 ;
C2 = KCHT $DOMTOT 'SCAL' 'SOMMET' C1 ;
UEXACT = (UEXACT ET C2) ;
FIN BLOC1 ;

TRACE DOMTOT ;
MESSAGE 'MAX = ' (MAXI RV.INCO.'CN')
        'MIN = ' (MINI RV.INCO.'CN') ;
TRAC DOMTOT (RV.INCO.'CN') (CONT DOMTOT) ;

SI ( COMPLET ) ;
P1 = DOMTOT POIN 'PROC' (0.0 0.7) ;
P2 = DOMTOT POIN 'PROC' (1.0 0.7) ;
P1P2 = DOMTOT POIN 'DROI' P1 P2 0.01 ;
LI12 = TRCE P1P2 ;

XX = COOR 1 LI12 ;

EVOL1 = EVOL 'CHPO' (RV.INCO.'CN') SCAL LI12 ;
EVOL2 = EVOL 'CHPO' XX SCAL LI12 ;

LIX = 'EXTR' EVOL2 ORDO ;
LIU = 'EXTR' EVOL1 ORDO ;
EVOL3 = EVOL 'MANU' 'X' LIX 'U(X)' LIU ;

EVOL4 = EVOL 'CHPO' UEXACT SCAL LI12 ;
LIUEX = 'EXTR' EVOL4 ORDO ;
EVOL5 = EVOL 'MANU' 'X' LIX 'UEXACT(X)' LIUEX ;

TAB1 = TABLE ;
TAB1.1 = 'MARQ LOSA' ;
DESS (EVOL3 ET EVOL5) 'XBOR' 0.0 1.0 'YBOR' -1. 2.
        'TITR' 'Coupe a y=0.7' TAB1 ;
FINSI ;

EVOL4 = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
        (RV.INCO.'ER') ;
DESS EVOL4 'XBOR' 0. 2000. 'YBOR' -20.0 0.0 ;

```

FINSI ;

FIN ;

1.5 Equation de conservation non-linéaire : cas d'une solution sans discontinuité

1.5.1 Formulation Eléments Finis Explicite Non-Conservative

<i>Opérateurs utilisés</i>	<i>TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation Q1, schéma SUPG</i>
<i>Nature du problème</i>	<i>Résolution d'une équation de convection non-linéaire et comparaison avec solution analytique.</i>
<i>Classification</i>	<i>2D plan, convection, instationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/convnonlin1.dgibi</i>

Description du problème :

Ce cas-test est inspiré d'un cas-test proposé dans [1]. Il concerne la résolution d'une équation de conservation non-linéaire, qui sous forme conservative s'écrit :

$$\frac{\partial u}{\partial t} + \nabla \cdot \underline{F}(u) = 0, \quad \underline{F}(u) = \begin{pmatrix} \frac{u^2}{2} \\ \ln u \end{pmatrix}$$

et sous forme non-conservative,

$$\frac{\partial u}{\partial t} + \lambda \cdot \nabla u = 0, \quad \lambda = \begin{pmatrix} u \\ 1/u \end{pmatrix}$$

Ce problème admet comme solution une solution analytique de la forme :

$$u_{ex}(x, y, t) = \frac{x+t}{y+t}$$

dans toute région de l'espace où $y+t \neq 0, \forall t \geq 0$.

On considère ici le domaine carré $\Omega = \{(x, y) \in [1, 2]^2\}$, avec comme solution initiale à $t = 0$, $u_{ex}(x, y, t = 0) = \frac{x}{y}$, et les conditions aux limites sur la frontière entrante $\partial\Omega_e$ (voir figure 1.5.1) :

$$u(x, y, t) = u_{ex}(x, y, t), \quad \forall t \geq 0, \forall (x, y) \in \partial\Omega_e$$




FIG. 1.5.1 – Géométrie et frontière entrante sur lesquelles on impose la solution à chaque instant. Le champ de vitesse représenté sur la figure correspond à la solution à $t = 0$.

Modélisation :

On résout le problème sous forme non-conservative avec l'opérateur **TSCAL**, et avec l'option de décentrement 'SUPG' (schéma linéaire). Le schéma est explicite en temps, avec une matrice masse diagonale non-consistante. A chaque pas de temps, les conditions aux limites sont réactualisées, et le champ de vitesse transportant, à savoir λ , évalué à partir de la solution numérique à ce pas de temps.

On considère 3 maillages de plus en plus fins, le premier constitué de 10×10 mailles, le deuxième de 20×20 mailles et le troisième de 40×40 mailles. On effectue dans chaque cas le calcul jusqu'au temps $t_f = 2.0$, avec le même pas de temps Δt , égal à 5×10^{-3} , qui est plus petit que le plus petit pas de stabilité du schéma sur le maillage le plus fin. Ainsi, on peut comparer la précision du schéma en fonction seulement du pas d'espace $\Delta x = \Delta y = 1/N$ où N est le nombre de points dans l'une ou l'autre direction.

Résultats numériques

La solution numérique correspondant au maillage fin et aux 4 instants $t = 0$, $t = 0.5$, $t = 1$ et $t = 1.5$ est représentée sur la figure 1.5.2. On observe que cette solution est constituée de caractéristiques divergentes, c'est à dire qu'elle représente une solution régulière, sans discontinuité, analogue à une détente.

Les courbes donnant l'erreur entre la solution numérique et la solution exacte (erreur évaluée en norme L_2 discrète) sont tracées pour les 3 maillages sur la courbe 1.5.3. On remarque que lorsque le pas d'espace du maillage est divisé par 2, l'erreur est réduite environ de moitié, ce qui suggère qu'en instationnaire et en explicite (avec une matrice masse non consistante), le schéma SUPG n'est que d'ordre 1 en espace (alors qu'en stationnaire, il est d'ordre 2 en espace).



FIG. 1.5.2 – Evolution de la solution de $t = 0$ à $t = 1.5$.

FIG. 1.5.3 – Evolution de l'erreur entre solution numérique et solution exacte pour les 3 maillages.

Référence :

- 1– D.B. Fairlie, “Formal Solutions of an Evolution Equations of Riemann Type”, in *Studies in Applied Mathematics*, vol. 98, p.203-205.

Jeu de données :

```

COMPLET = FAUX ;
GRAPH = FAUX ;
NX = 10 ;
NY = 10 ;
CFL = 0.7 ;
'SI' ( COMPLET ) ;
      TFIN = 10. ;
'SINON' ;
      TFIN = 1. ;
'FINSI' ;

*****
* EQUATION DE CONVECTION NON-LINEAIRE  $U_t + \text{div} (F(U)) = 0$  *
* AVEC  $F(U) = U*U/2 \mathbf{1}_x + \ln U \mathbf{1}_y$   $U > 0$  *
* RESOLUE SOUS FORME NON-CONSERVATIVE *
*  $dU/dt + U dU/dx + 1/U dU/dy = 0$  *
* AVEC OPTION SUPGDC (par défaut) *
* SOLUTION ANALYTIQUE  $U_{ex}(x,y,t) = (x-t)/(y-t)$  *
* Probleme hyperbolique avec conditions aux limites *
* variant dans le temps *
* H. PAILLERE AVRIL 1997 *
*****

*****
* PROCEDURE POUR CALCULER CHAMP DE VITESSE, METTRE A JOUR LES C.L *
* TESTER LA CONVERGENCE VERS L'ETAT STATIONNAIRE ET L'ERREUR *
* ENTRE SOLUTION NUMERIQUE ET SOLUTION ANALYTIQUE *
*****

'DEBPROC' CALCUL ;
'ARGU' RVX*'TABLE' ;

RV = RVX.'EQEX' ;
CN = RV.INCO.'CN' ;

* calcul du champ de vitesse

VX1 = CN ;
VX1 = 'NOMC' 'UX' VX1 ;
VY1 = 'KOPS' 1. '/' CN ;
VY1 = 'NOMC' 'UY' VY1 ;
VV='KCHT' $DOMTOT 'VECT' 'SOMMET' 'COMP' (MOTS 'UX' 'UY')
      (VX1 'ET' VY1) ;
RV.INCO.'VITESSE' = VV ;

* mise a jour des conditions aux limites

CL_CN = 'EXCO' 'CN' RV.'CLIM' 'CN' ;
UEXAC = 'KOPS' ('KOPS' XX '-' RV.PASDETPS.'TPS') '/'
      ('KOPS' YY '-' RV.PASDETPS.'TPS') ;
CL_NOUV = 'KCHT' $CL 'SCAL' 'SOMMET' UEXAC ;
CL_NOUV = 'NOMC' 'CN' CL_NOUV 'NATU' 'DISCRET' ;

```



```

RV.'CLIM' = 0.DO * CL_CN + CL_NOUV ;

* calcul de l'erreur de convergence vers l'etat stationnaire
* et de l'erreur entre solution numerique et solution analytique

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) 'EGA' 0 ;
'SI' ( LO ) ;

ERR = 'KOPS' (RV.INCO.'CN') '-' (RV.INCO.'CNM1') ;
ELI = 'MAXI' ERR 'ABS' ;
ELI = ('LOG' (ELI + 1.0E-20))/(LOG 10.) ;
'MESSAGE' 'TEMPS = ' RV.PASDETPS.'TPS'
          'ITER ' RV.PASDETPS.'NUPASDT' ' ERREUR LINF (dUdt) ' ELI ;
IT = 'PROG' RV.PASDETPS.'NUPASDT' ;
ER = 'PROG' ELI ;
RV.INCO.'IT' = (RV.INCO.'IT') 'ET' IT ;
RV.INCO.'ERCONV' = (RV.INCO.'ERCONV') 'ET' ER ;

UEXAC = 'KCHT' $DOMTOT 'SCAL' 'SOMMET' UEXAC ;
UEXAC = 'NOMC' 'CN' UEXAC 'NATU' 'DISCRET' ;
ERRP = 'KOPS' UEXAC '-' RV.INCO.'CN' ;
ERRP = 'KOPS' ERRP '*' ERRP ;
ERR2 = 0. ;
'REPETER' BLOC1 (NBNO DOMTOT) ;
P1 = (DOMA $DOMTOT SOMMET) 'POIN' &BLOC1 ;
ERR1 = 'EXTR' ERRP 'SCAL' P1 ;
ERR2 = ERR2 + ERR1 ;
'FIN' BLOC1 ;
ERR2 = ERR2/(NBNO DOMTOT) ;
ERR2 = ERR2 ** 0.5 ;
ER = 'PROG' ERR2 ;
RV.INCO.'ERSOLU' = (RV.INCO.'ERSOLU') 'ET' ER ;
'MESSAGE' 'ERREUR |Uex(x,y,t) - Unum(x,y,t)| EN NORME L2 = ' ERR2 ;

'FINSI' ;

RV.INCO.'CNM1' = 'KCHT' $DOMTOT 'SCAL' 'SOMMET' (RV.INCO.'CN') ;

as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

'OPTI' 'DIME' 2 ;
'OPTI' 'ELEM' 'QUA4' ;

*****
* DIFFUSION *
*****
DIF = 1.0E-10 ;

*****
* MAILLAGE *
*****

```

```

A1 = -2.0 -2.0 ;
A2 = -1.0 -2.0 ;
A3 = -1.0 -1.0 ;
A4 = -2.0 -1.0 ;

FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;

DOMTOT = 'DALL' FBAS FDRO FHAU FGAU 'PLAN' ;

*****
* CREATION DU MODELE NAVIER_STOKES *
*****

GEOCL = FBAS 'ET' FGAU ;
'ELIM' GEOCL 1.E-3 ;

MDOMTOT = CHAN DOMTOT QUAF ;
$DOMTOT = MODE MDOMTOT 'NAVIER_STOKES' LINE ; DOMA $DOMTOT 'IMPR' ;

MGEOCL = CHAN GEOCL QUAF ;
ELIM MDOMTOT MGEOCL 1.E-3 ;
$CL      = MODE MGEOCL 'NAVIER_STOKES' LINE ;

*****
* SOLUTION INITIALE *
*****

XX YY = 'COOR' (DOMA $DOMTOT MAILLAGE) ;
SOLINI = 'KOPS' XX '/' YY ;
SOLCL = 'KCHT' $CL 'SCAL' 'SOMMET' SOLINI ;

*****
* TABLE DE RESOLUTION *
*****

RV = 'EQEX' $DOMTOT 'ITMA' 1000 'ALFA' CFL 'TFINAL' (TFIN/2.)
      'ZONE' $DOMTOT
      'OPER' CALCUL
      'OPTI' 'SUPGDC'
      'OPER' TSCAL DIF 'VITESSE' 0.0 'INCO' 'CN'
      'OPTI' 'CENTREE'
      'OPER' DFDT 1. 'CN' 'DELTAT'          'INCO' 'CN'
      'CLIM' 'CN' 'TIMP' GEOCL SOLCL ;

RV.INCO = 'TABLE' 'INCO' ;
RV.INCO.'CN' = 'KCHT' $DOMTOT 'SCAL' 'SOMMET' SOLINI ;
RV.INCO.'VITESSE' = CHVIT ;

RV.INCO.'CNM1' = 'KCHT' $DOMTOT 'SCAL' 'SOMMET' SOLINI ;
RV.INCO.'IT' = 'PROG' 1 ;
RV.INCO.'ERCONV' = 'PROG' 0. ;
RV.INCO.'ERSOLU' = 'PROG' 0. ;

```

```
* ON EXECUTE JUSQU'A TFIN/2. *  
  
EXEC RV ;  
  
* ON TESTE L'OPTION SUPG (IDCEN=3) DE TFIN/2. A TFIN *  
  
RV.'2TSCAL'.'KOPT'.'IDCEN' = 3 ;  
RV.'TFINAL' = TFIN ;  
EXEC RV ;  
  
'SI' ( GRAPH ) ;  
'OPTI' 'ISOV' 'SULI' ;  
'TRACE' RV.INCO.'CN' DOMTOT ('CONT' DOMTOT) 20 ;  
EVOL4 = 'EVOL' 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'  
      (RV.INCO.'ERCONV') ;  
'DESS' EVOL4 'XBOR' 0. 2000. 'YBOR' -20.0 0.0 ;  
'FINSI' ;  
  
'SI' ( 'MINI' ( RV.INCO.'ERCONV' ) > -2.0 ) ;  
      'ERREUR' 5 ;  
'FINSI' ;  
  
'SI' ( 'MAXI' ( RV.INCO.'ERSOLU' ) > 0.09 ) ;  
      'ERREUR' 5 ;  
'FINSI' ;  
  
'FIN' ;
```

1.6 Transport diffusif : équation de la chaleur

1.6.1 Formulation Eléments Finis Implicite

<i>Opérateurs utilisés</i>	<i>DFDT, LAPN</i>
<i>Méthodes numériques</i>	<i>EF, approximation P1 ou Q1</i>
<i>Nature du problème</i>	<i>Propagation de la chaleur dans une plaque plane avec imposition d'une solution bilinéaire sur les côtés.</i>
<i>Classification</i>	<i>2D plan, diffusion, stationnaire</i>
<i>Jeu de données</i>	

Description du problème :

On s'intéresse à la solution stationnaire de l'équation de la chaleur (sans dimension),

$$\frac{\partial T}{\partial t} = \alpha \Delta T, \quad \forall \underline{x} \in \Omega,$$

avec $\alpha = 1$ et les conditions aux limites de Dirichlet,

$$T(\underline{x}) = T_o(\underline{x}), \quad \forall \underline{x} \in \partial\Omega$$

Le domaine considéré est le carré $\Omega = \{(x, y), 0 \leq x \leq 1, 0 \leq y \leq 1\}$, et l'on impose aux frontières de Ω la solution bilinéaire :

$$T_o(x, y) = \alpha x + \beta y + \gamma xy + \delta, \quad \forall (x, y) \in \partial\Omega$$

avec $\alpha = \beta = 1$, $\gamma = 2$ et $\delta = 3$.

La solution stationnaire sur le domaine Ω est évidemment égale à :

$$T_{ex}(x, y) = \alpha x + \beta y + \gamma xy + \delta, \quad \forall (x, y) \in \Omega$$

Modélisation :

On résoud le problème instationnaire grâce aux opérateurs DFDT et LAPN, afin de converger vers la solution stationnaire. Notons bien que l'on peut résoudre directement l'état permanent sans discrétiser le terme de dérivée en temps, la solution stationnaire étant évidemment identique à la solution permanente issue du transitoire.

La discrétisation spatiale de ces opérateurs repose sur une formulation Eléments Finis implicite. Puisque la solution exacte est bilinéaire, on s'attend à ce que les schémas appliqués à un maillage de quadrangles (éléments bilinéaire Q1) donnent la solution exacte. Pour des maillages de triangles (éléments linéaires P1), une erreur sera commise, proportionnelle au carré du pas d'espace caractéristique du maillage (ordre 2 en espace du schéma de Galerkin pour un problème de diffusion).

Pour le terme de dérivée en temps, on choisit un pas de temps Δt aussi grand que l'on veut. Ici $\Delta t = 1000$. Le problème étant linéaire, aucune relaxation n'est nécessaire ($\omega = 1$).

Synthèse des résultats :

On effectue 2 calculs, l'un sur un maillage constitué de quadrangles (figure 1.6.1), l'autre de triangles (figure 1.6.2). Les solutions stationnaires sont tracées sur les figures 1.6.3 et 1.6.4. Plus intéressante est l'erreur relative entre la solution numérique stationnaire et la solution exacte (figures 1.6.5 et 1.6.6). On trouve que l'erreur dans le cas du maillage de triangles est de l'ordre de 10^{-4} , alors qu'elle est de l'ordre de 10^{-15} (c'est à dire du zéro machine) dans le cas des quadrangles.




FIG. 1.6.1 – Maillage de triangles




FIG. 1.6.2 – Maillage de quadrangles




FIG. 1.6.3 – Solution numérique sur triangles






FIG. 1.6.4 – Solution numérique sur quadrangles

FIG. 1.6.5 – Erreur $\epsilon = (T_{num} - T_{ex})/T_{ex}$ sur triangles : Max $|\epsilon| = 4. \times 10^{-4}$ FIG. 1.6.6 – Erreur $\epsilon = (T_{num} - T_{ex})/T_{ex}$ sur quadrangles : Max $|\epsilon| = 6. \times 10^{-15}$

Jeu de données :

```

*****
* EQUATION DE LAPLACE : THERMIQUE TRAITEE AVEC L'OPERATEUR LAPN      *
* EN IMPLICITE                *
*****

*****
* MAILLAGE *
*****

option dime 2 elem tri3 ;
option isov sul1 ;

NX = 20 ;
NY = 20 ;

A1 = 0.0 0.0 ;
A2 = 1.0 0.0 ;
A3 = 1.0 1.0 ;
A4 = 0.0 1.0 ;

FBAS = A1 'DROI' NX A2 ;
FDRO = A2 'DROI' NY A3 ;
FHAU = A3 'DROI' NX A4 ;
FGAU = A4 'DROI' NY A1 ;

CNT = FBAS ET FDRO ET FHAU ET FGAU ;

*MT = 'DALLER' FBAS FDRO FHAU FGAU 'PLAN' ;
MT = CNT 'SURF' 'PLAN' ;

'TASS' MT ;
'ORIE' MT ;

$MT = 'DOMA' MT 1.e-3 'IMPR' ;
MT = $MT.MAILLAGE ;
CNT = 'CONT' MT ;

*****
* SOLUTION EXACTE BILINEAIRE *
*****

alpha = 1. ;
beta = 1. ;
gamma = 2. ;
delta = 3. ;
XX YY = COOR MT ;
SOLEX = KOPS (KOPS XX '*' alpha) '+' (KOPS YY '*' beta) ;
SOLEX = KOPS SOLEX '+' (KOPS (KOPS XX '*' YY) '*' gamma) ;
SOLEX = KOPS SOLEX '+' delta ;
SOLEX = KCHT $MT 'SCAL' 'SOMMET' SOLEX ;

TCNT = 'REDU' SOLEX CNT ;

```

```

*****
* TABLE DE RESOLUTION *
*****

ALPHA = 1. ;

RV = EQEX $MT 'DUMP' 'OPTI' 'EF' 'IMPL'
      ZONE $MT 'OPER' DFDT 1. 'TNM' 'DT' 'UN' ALPHA INCO 'TN'
      ZONE $MT 'OPER' LAPN ALPHA INCO 'TN' ;

RV = EQEX RV
      'CLIM' 'TN' 'TIMP' CNT TCNT ;

RV.INCO = 'TABLE' INCO ;
RV.INCO.'TN' = 'KCHT' $MT 'SCAL' 'SOMMET' 0. ;
RV.INCO.'UN' = 'KCHT' $MT 'VECT' 'SOMMET' (0. 0.) ;
RV.INCO.'TNM' = 'KCHT' $MT 'SCAL' 'SOMMET' 0. ;

rv.'NITER' = 2 ;
rv.'OMEGA' = 1. ;
rv.'IMPR' = 1 ;
rv.'ITMA' = 10 ;
rv.'EPS' = 1.e-15 ;
rv.'DT' = 1000.0 ;
rv.INCO.'DT' = rv.'DT' ;

EXEC RV ;

TRACE MT 'TITR' 'MAILLAGE' ;
TRACE RV.INCO.'TN' MT CNT 'TITR' 'SOLUTION NUMERIQUE' ;
ERO = KOPS RV.INCO.'TN' '-' SOLEX ;
ER1 = KOPS ERO '/' SOLEX ;
TRACE ER1 MT CNT 'TITR' 'ERREUR RELATIVE' ;

ER2 = KOPS ERO '*' ERO ;
ERR = 0. ;
REPETER BLOC1 $MT.NPTD ;
P1 = $MT.SOMMET POIN &BLOC1 ;
ERR = ERR + ('EXTR' ER2 'SCAL' P1) ;
FIN BLOC1 ;
ERR = ERR/$MT.NPTD ;
ERR = ERR ** 0.5 ;

MESSAGE '===== ' ;
MESSAGE 'ERREUR EN NORME LINF = ' (MAXI (ABS ERO)) ;
MESSAGE 'ERREUR EN NORME L2 = ' ERR ;
MESSAGE '===== ' ;

```

Chapitre 2

Écoulements incompressibles laminaires

2.1 Couche limite sur plaque plane

2.1.1 Formulation Eléments Finis Semi-Implicite

<i>Opérateurs utilisés</i>	<i>NS, PRESSION, TOIMP</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation Q1-P0, schéma SUPG</i>
<i>Nature du problème</i>	<i>développement d'une couche limite laminaire sur plaque plane et comparaison avec solution exacte de Blasius.</i>
<i>Classification</i>	<i>2D plan, incompressible, laminaire, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/blasius.dgibi</i>

Schéma du problème :




FIG. 2.1.1 – Développement d'une couche limite 2D sur une plaque plane, représentée par le segment BC .

Modélisation :

On modélise le développement d'une couche limite laminaire, sur une plaque plane de longueur L . Le fluide est incompressible, de viscosité cinématique ν , et de vitesse U_∞ en amont de la plaque. On cherche la solution stationnaire comme limite d'un transitoire. Sous forme adimensionnée, les équations de Navier-Stokes instationnaires s'écrivent :

$$\begin{aligned} \nabla \cdot \underline{u} &= 0 \\ \frac{\partial \underline{u}}{\partial t} + (\underline{u} \cdot \nabla) \underline{u} &= -\nabla p + \frac{1}{Re} \Delta \underline{u} \end{aligned}$$

où le nombre de Reynolds, $Re = \frac{U_\infty L}{\nu}$, est fixé à 2000. Les conditions aux limites sont les suivantes :

$v = 0$	sur Γ_{AB}	condition de symétrie
$u = v = 0$	sur Γ_{BC}	condition d'adhérence à la paroi
$\underline{\sigma} \cdot \underline{n} = (0., 0.)$	sur $\Gamma_{CD} \cup \Gamma_{ED}$	condition de sortie
$u = 1, v = 0$	sur Γ_{EA}	condition d'entrée

Ces équations sont discrétisées grâce aux opérateurs **NS**, **PRESSION** et **TOIM** de CASTEM2000. L'option 'SUPG' a été choisie pour l'opérateur **NS** car moins diffusive que l'option par défaut. Le maillage, constitué de quadrangles (QUA8 avec l'option 'MACRO') est resserré près de la paroi (environ 15 points dans la couche limite au niveau de la sortie) et du bord d'attaque de la plaque.

Synthèse des résultats :

- La solution stationnaire recherchée est obtenue après 5000 itérations. L'histoire de convergence, $\log \|v^{n+1} - v^n\|_{L^\infty}$ en fonction du nombre d'itérations, montre que le résidu a été réduit de plus de 7 ordres de grandeur, et que la solution peut être considérée comme convergée.




FIG. 2.1.2 – Histoire de convergence

- La solution numérique est comparée à la solution théorique de Blasius [1], p. 265. Le profil de la composante $u(y)$ en sortie montre une très bonne concordance avec le profil théorique. La concordance est moins bonne en ce qui concerne la composante $v(y)$ de la vitesse, résultat très répandu parmi les codes numériques.

Référence :

- 1– Frank M. White. *Viscous Fluid Flow*. Mc Graw-Hill, 1974.




FIG. 2.1.3 – Maillage constitué de 3600 éléments et 3751 nœuds





FIG. 2.1.4 – Isolignes de la norme de la vitesse

FIG. 2.1.5 – Profils des composantes u et v à la sortie

Jeu de données :

```

COMPLET = FAUX ;
SI ( COMPLET ) ;
    d1= 0.5 ; d2= 0.02 ;
    nbit = 5000 ;
    err1 = 7.5E-3 ;
SINON ;
    d1 = 1.0 ; d2 = 0.1 ;
    nbit = 150 ;
    err1 = 7.E-2 ;
FINSI ;
GRAPH = 'N' ;

*****
* COUCHE LIMITE LAMINAIRE SUR PLAQUE PLANE - COMPARAISON AVEC SOLUTION *
* THEORIQUE DE BLASIUS *
*****

DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;

RV = RVX.'EQEX' ;

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;

UN = RV.INCO.'UN' ;
UNM1 = RV.INCO.'UNM1' ;

unx= kcht $MT scal sommet (exco 'UX' un) ;
unm1x = kcht $MT scal sommet (exco 'UX' unm1) ;
uny= kcht $MT scal sommet (exco 'UY' un) ;
unm1y = kcht $MT scal sommet (exco 'UY' unm1) ;

ERRX = KOPS unx '-' unm1x ;
ERRY = KOPS uny '-' unm1y ;
ELIX = MAXI ERRX 'ABS' ;
ELIY = MAXI ERRY 'ABS' ;
ELIX = (LOG (ELIX + 1.0E-10))/(LOG 10.) ;
ELIY = (LOG (ELIY + 1.0E-10))/(LOG 10.) ;
MESSAGE 'ITER ' RV.PASDETPS.'NUPASDT' ' ERREUR LINF ' ELIX ELIY ;
RV.INCO.'UNM1' = KCHT $MT 'VECT' 'SOMMET' (RV.INCO.'UN') ;
IT = PROG RV.PASDETPS.'NUPASDT' ;
ER = PROG ELIY ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER ;
FINSI ;

as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

*****
* MAILLAGE *

```

```

*****

option dime 2 elem qua8 ;

p1 = -1.5 0.0 ;
p2 = 0.0 0.0 ;
p3 = 1.0 0.0 ;
p4 = 1.0 0.2 ;
p5 = 1.0 1.0 ;
p6 = 0.0 1.0 ;
p7 = -1.5 1.0 ;
p8 = -1.5 0.2 ;

bas1 = p1 d dini d1 dfin d2 p2 ;
bas2 = p2 d dini d2 dfin d2 p3 ;
bas = bas1 et bas2 ;
dro = p3 d dini d2 dfin d2 p4 d dini d2 dfin d1 p5 ;
hau = p5 d dini d2 dfin d2 p6 d dini d2 dfin d1 p7 ;
gau = p7 d dini d1 dfin d2 p8 d dini d2 dfin d2 p1 ;

mt= bas dro hau gau daller ;

tass MT ;
orienter mt ;

mt = chan mt quaf ;
dro= chan dro quaf ;
bas = chan bas quaf ;
gau = chan gau quaf ;
hau = chan hau quaf ;

elim (mt et dro et bas et gau et hau) 1.e-3 ;
$mt = mode mt 'NAVIER_STOKES' MACRO ; doma $mt 'IMPR' ;
$dro = mode dro 'NAVIER_STOKES' MACRO ;

mt2 = mt ;
mt = doma $mt maillage ;

Nu = 1./2000. ;
tos = 0. 0. ;

RV = EQEX $MT 'ITMA' nbit 'ALFA' 0.9
'OPTI' 'SUPG'
'ZONE' $MT 'OPER' CALCUL
'ZONE' $MT 'OPER' 'NS' NU 'INCO' 'UN'
'OPTI' 'CENTREE' 'EFM1'
'ZONE' $MT 'OPER' 'DFDT' 1. 'UN' 'DELTAT' 'INCO' 'UN'
'ZONE' $DRO 'OPER' 'TOIM' tos 'INCO' 'UN' ;

RV = EQEX RV
'CLIM' 'UN' 'UIMP' bas2 0.
'UN' 'UIMP' gau 1.0
'UN' 'VIMP' bas 0.0 ;

RVP = EQPR $MT KTYPI 1
ZONE $MT 'OPER' 'PRESSION' 0. ;

```

```

RV.INCO=TABLE INCO ;
RV.INCO.'UN'=kcht $MT VECT SOMMET (1. 0.);
rv.inco.'UNM1' = kcht $MT 'VECT' 'SOMMET' (1.E-5 1.E-5 ) ;
rv.inco.'IT' = PROG 1 ;
rv.inco.'ER' = PROG 0. ;
rv.inco.'NBIT' = 0 ;
RV.PRESSION =RVP ;

EXEC RV ;

* SOLUTION THEORIQUE BLASIVUS *

LIYT = PROG
  0.000000 0.003162 0.006325 0.009487 0.012649 0.015811 0.018974
  0.022136 0.025298 0.028460 0.031623 0.034785 0.037947 0.041110
  0.044272 0.047434 0.050596 0.053759 0.056921 0.060083 0.063246
  0.069570 0.075895 0.082219 0.088544 0.094868 0.101193 0.107517
  0.113842 0.120167 0.126491 0.132816 0.139140 0.145465 0.151789
  0.158114 0.164438 0.170763 0.177088 0.183412 0.189737 1.000000 ;
LIUT = PROG
  0.000000 0.046960 0.093910 0.140810 0.187610 0.234230 0.280580
  0.326530 0.371960 0.416720 0.460630 0.503540 0.545250 0.585590
  0.624390 0.661470 0.696700 0.729930 0.761060 0.790000 0.816690
  0.863300 0.901070 0.930600 0.952880 0.969050 0.980370 0.987970
  0.992890 0.995940 0.997770 0.998820 0.999400 0.999700 0.999860
  0.999940 0.999971 0.999988 0.999995 0.999998 0.999999 1.000000 ;
LIVT = PROG
  0.000000 0.000037 0.000149 0.000334 0.000593 0.000925 0.001327
  0.001800 0.002338 0.002940 0.003599 0.004312 0.005070 0.005867
  0.006695 0.007545 0.008408 0.009275 0.010136 0.010982 0.011804
  0.013350 0.014721 0.015886 0.016836 0.017576 0.018129 0.018525
  0.018797 0.018975 0.019088 0.019155 0.019195 0.019216 0.019228
  0.019234 0.019237 0.019238 0.019239 0.019239 0.019239 0.019239 ;

EVOLU = EVOL 'MANU' 'Y' LIYT 'U(1.,y)' LIUT ;
EVLV = EVOL 'MANU' 'Y' LIYT 'V(1.,y)' LIVT ;

YY = COOR 2 (DOMA $MT SOMMET) ;
EVOL1 = EVOL 'CHPO' RV.INCO.'UN' UX DRO ;
EVOL2 = EVOL 'CHPO' RV.INCO.'UN' UY DRO ;
EVOL3 = EVOL 'CHPO' YY SCAL DRO ;
LIY = 'EXTR' EVOL3 ORDO ;
LIU = 'EXTR' EVOL1 ORDO ;
LIV = 'EXTR' EVOL2 ORDO ;

LIUC = IPOL LIYT LIY LIU ;
NP = DIME LIYT ;
ERRO = 0. ;
REPETER BLOC1 NP ;
  UCAL = EXTRAIRE LIUC &BLOC1 ;
  UREF = EXTRAIRE LIUT &BLOC1 ;
  ERRO = ERRO + ((UCAL-UREF)*(UCAL-UREF)) ;
FIN BLOC1 ;
ERRO = ERRO/NP ;
ERRO = ERRO '**' 0.5 ;

```



```
MESSAGE 'ERREUR ' ERRO ;
SI ( ERRO > ERR1 ) ;
    ERREUR 5 ;
FINSI ;

SI ('EGA' graph '0') ;
opti isov lign ;
pn = elno $MT RVP.PRESSIO n ;
TRACE Pn mt2 ;
unch = vect RV.INCO.'UN' 0.1 UX UY JAUNE ;
TRACE UNCH MT2 ;
TAB1 = TABLE ;
* TAB1.1 = 'MOT' 'TITR SOLUTION_CASTEM2000' ;
TAB1.2 = 'MARQ LOSA NOLI ' ;
TAB1.'TITRE' = TABLE ;
TAB1.'TITRE' . 1 = 'MOT' 'SOLUTION_CASTEM2000' ;
TAB1.'TITRE' . 2 = MOT 'SOLUTION_EXACTE' ;
DESS (EVOL1 ET EVOLU) 'LEGE' 'TITY' 'UX' 'TITR' 'U EN SORTIE' TAB1 ;
DESS (EVOL2 ET EVOLV) 'LEGE' 'TITY' 'UY' 'TITR' 'V EN SORTIE' TAB1 ;
FINSI ;

fin ;
```

2.2 Ecoulement autour d'une marche descendante

2.2.1 Formulation Eléments Finis Semi-Implicite

Opérateurs utilisés	NS, PRESSION, TOIMP
Méthodes numériques	EFM1, approximation Q1-P0, schéma SUPG
Nature du problème	écoulement laminaire autour d'une marche descendante et prédiction de la zone de recirculation. Comparaison avec résultats expérimentaux et autres codes numériques.
Classification	2D plan, incompressible, laminaire, stationnaire
Jeu de données	

Schéma du problème :




FIG. 2.2.1 – Ecoulement autour d'une marche descendante

Modélisation :

Ce problème, qui a fait l'objet d'un 'workshop' de validation de solveurs Navier-Stokes incompressibles [1], consiste à modéliser l'écoulement laminaire (stationnaire) dans un canal autour d'une marche descendante. h représente la hauteur du canal en amont de la marche, H la hauteur en aval de la marche, l la longueur du canal en amont de la marche et D la longueur du canal en aval de la marche. Un profil de vitesse parabolique est imposé à l'entrée, la vitesse maximale étant notée U_{\max} . L'écoulement se sépare au niveau de la marche, et se réattache en aval de la marche à une abscisse à déterminer.

Le cas-test considéré ici correspond aux valeurs $D = 18$ m, $l = 3$ m, $H = 1.5$ m et $h = 1$ m. Les équations de Navier-Stokes, adimensionnées par les échelles de longueur et de vitesse $(H - h)$ et U_{\max} , s'écrivent :

$$\begin{aligned} \nabla \cdot \underline{u} &= 0 \\ \frac{\partial \underline{u}}{\partial t} + (\underline{u} \cdot \nabla) \underline{u} &= -\nabla p + \frac{1}{Re} \Delta \underline{u} \end{aligned}$$

où le nombre de Reynolds, $Re = \frac{U_{\max}(H-h)}{\nu}$, est fixé à 150. On impose les conditions limites suivantes :

$$\begin{aligned} v = 0, u(y) = Ay^2 + By + C & \quad \text{à l'entrée} \\ u = v = 0 & \quad \text{sur les parois} \\ \underline{\sigma} \cdot \underline{n} = (0., 0.) & \quad \text{en sortie} \end{aligned}$$

Le coefficient de frottement à la paroi, $C_f(x)$, est défini sous forme adimensionnelle par :

$$C_f(x) = \frac{\left(\frac{\partial u}{\partial y}\right)_{y_{paroi}}}{\frac{1}{2}Re}$$

Le point de réattachement correspond à l'abscisse où $C_f(x)$ change de signe.

Ces équations sont discrétisées grâce aux opérateurs **NS**, **PRESSION** et **TOIM** de CASTEM2000. L'option 'SUPG' a été choisie pour l'opérateur **NS** car moins diffusive que l'option par défaut. Le maillage, constitué de quadrangles (QUA8 avec l'option 'MACRO') est resserré près le long de la paroi inférieure ainsi qu'autour de la marche descendante.

Synthèse des résultats :

- La solution stationnaire recherchée est obtenue après 10000 itérations. L'histoire de convergence, $\log \|v^{n+1} - v^n\|_{L^\infty}$ en fonction du nombre d'itérations, montre que le résidu a été réduit de plus de 7 ordres de grandeur, et que la solution peut être considérée comme convergée.




FIG. 2.2.2 – Histoire de convergence

- La zone de recirculation en aval de la marche est bien définie. Le point de réattachement, déterminé à partir du tracé de $C_f(x)$, se situe à $x \approx 5.83$, proche de la valeur expérimentale de 6. [1].

Référence :

- 1 K. Morgan, J. Périaux, and F. Thomasset, editors. *Analysis of Laminar Flow over a Backward Facing Step*, volume 9 of *Notes on Numerical Fluid Mechanics*. Vieweg, 1984.




FIG. 2.2.3 – détail du maillage constitué de 2928 éléments et 3123 nœuds





FIG. 2.2.4 – zone de recirculation en aval de la marche : champ de vitesse

FIG. 2.2.5 – Coefficient de frottement à la paroi : réattachement à $x \approx 5.83$. La valeur expérimentale et les autres résultats numériques sont tirés de [1]

Jeu de données :

```
*****
* LAMINAR FLOW OVER BACKWARD FACING STEP *
*****
```

```
opti dime 2 ;
opti elem qua8 ;
```

```
*****
* GEOMETRY OF PROBLEM *
*****
```

```
L1 = 3.0 ;
L2 = 22.0 ;
H1 = 1.0 ;
H2 = 1.5 ;
```

```
d1 = 0.1 ;
d2 = 0.1 ;
```

```
HDIM = H2-H1 ;
L1 = L1/HDIM ;
L2 = L2/HDIM ;
H1 = H1/HDIM ;
H2 = H2/HDIM ;
d1 = d1/HDIM ;
d2 = d2/HDIM ;
```

```
p1 = 0. (H2-H1) ;
p2 = L1 (H2-H1) ;
p3 = L1 0. ;
p4 = L2 0. ;
p5 = L2 H2 ;
p6 = L1 H2 ;
p7 = 0. H2 ;
p8 = L2 (H2-H1) ;
```

```
p1p2 = p1 d dini d1 dfin d2 p2 ;
p2p6 = p2 d dini d2 dfin d1 p6 ;
p6p7 = p6 d dini d2 dfin d1 p7 ;
p7p1 = p7 d dini d1 dfin d2 p1 ;
```

```
mesh1 = p1p2 p2p6 p6p7 p7p1 'DALL' 'PLAN' ;
'TASS' mesh1 ;
'ORIE' mesh1 ;
```

```
p6p2 = p6 d dini d1 dfin d2 p2 ;
p2p3 = p2 d dini d2 dfin d2 p3 ;
p3p4 = p3 d dini d2 dfin d1 p4 ;
p4p8 = p4 d dini d2 dfin d2 p8 ;
p8p5 = p8 d dini d2 dfin d1 p5 ;
p5p6 = p5 d dini d1 dfin d2 p6 ;
```

```
mesh2 = (p6p2 ET p2p3) p3p4 (p4p8 ET p8p5) p5p6 'DALL' 'PLAN' ;
'TASS' mesh2 ;
```

```

'TASS' mesh2 ;

'ELIM' 1.E-3 mesh1 mesh2 ;

domtot = mesh1 ET mesh2 ;

$DOMTOT = 'DOMA' DOMTOT 1.E-3 'MACRO' ;
$INLET = 'DOMA' p7p1 'MACRO' 'INCL' $DOMTOT 1.E-3 ;
$OUTLET = 'DOMA' (p4p8 ET p8p5) 'MACRO' 'INCL' $DOMTOT 1.E-3 ;

wall = p1p2 ET p2p3 ET p3p4 ET p5p6 ET p6p7 ;

*****
* PHYSICAL DATA *
*****

Umax = 1.0 ;
Re = 500. ;
Nu = 1/Re ;
tos = 0. 0. ;

*****
* PARABOLIC VELOCITY PROFILE AT INLET *
*****

YINLET = 'COOR' 2 $INLET.MAILLAGE ;
YMAX = 'MAXI' YINLET ;
YMIN = 'MINI' YINLET ;
UIN = KOPS (KOPS YINLET '-' YMAX) '*' (KOPS YINLET '-' YMIN) ;
UIN = KOPS UIN '*' (-4.0*Umax/((YMAX-YMIN)*(YMAX-YMIN))) ;
UIN = NOMC 'UX' UIN 'NATU' 'DISCRET' ;
VIN = KCHT $INLET 'SCAL' 'SOMMET' 0. ;
VIN = NOMC 'UY' VIN 'NATU' 'DISCRET' ;

RV = EQEX $DOMTOT 'ITMA' 30000 'ALFA' 0.8
VIN = KCHT $INLET 'SCAL' 'SOMMET' 0. ;
VIN = NOMC 'UY' VIN 'NATU' 'DISCRET' ;

RV = EQEX $DOMTOT 'ITMA' 30000 'ALFA' 0.8
      'OPTI' 'SUPG'
      'ZONE' $DOMTOT 'OPER' 'NS' NU 'INCO' 'UN'
      'ZONE' $OUTLET 'OPER' 'TOIM' tos 'INCO' 'UN' ;

RV = EQEX RV
      'CLIM' 'UN' 'UIMP' wall 0. 'UN' 'VIMP' wall 0.
      'UN' 'UIMP' $INLET.MAILLAGE UIN
      'UN' 'VIMP' $INLET.MAILLAGE VIN ;

RVP = EQPR $DOMTOT KTYPI 1
      ZONE $DOMTOT 'OPER' 'PRESSION' 0. ;

RV.INCO = TABLE INCO ;
RV.INCO.'UN' = KCHT $DOMTOT VECT SOMMET (1.E-3 1.E-3) ;
RV.PRESSION = RVP ;

EXEC RV ;

```


2.3 Convection naturelle dans une cavité carrée

2.3.1 Formulation Eléments Finis Semi-Implicite

<i>Opérateurs utilisés</i>	<i>NS, PRESSION, TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation Q1-P0, schéma SUPG</i>
<i>Nature du problème</i>	<i>Ecoulement de convection naturelle dans une cavité carrée et comparaison avec autres solutions numériques</i>
<i>Classification</i>	<i>2D plan, incompressible, Boussinesq, laminaire, stationnaire</i>
<i>Jeu de données</i>	

Schéma du problème :




FIG. 2.3.1 – Définition du problème de la cavité carrée.

Modélisation :

On considère une cavité carrée de côté L , remplie d'un fluide incompressible de viscosité cinématique ν , de coefficient de dilation thermique β et de diffusivité thermique α . La paroi gauche de la cavité carrée est maintenue à la température 'chaude' T_h , et la paroi droite à la température 'froide' T_c . Les parois horizontales sont supposées adiabatiques. On définit une vitesse de référence $V_o = \sqrt{g\beta(T_h - T_c)L}$, une température de référence T_o , prise égale à $(T_h + T_c)/2$, ainsi qu'une température adimensionnée

$\theta = (T - T_c)/(T_h - T_c)$. Les équations de Boussinesq décrivant les mouvements de convection naturelle, s'écrivent en variables adimensionnées,

$$\begin{aligned}\frac{\partial u}{\partial x} + \frac{\partial v}{\partial y} &= 0 \\ \frac{\partial u}{\partial t} + u \frac{\partial u}{\partial x} + v \frac{\partial u}{\partial y} &= -\frac{\partial p'}{\partial x} + \frac{1}{\sqrt{Gr}} \left[\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} \right] \\ \frac{\partial v}{\partial t} + u \frac{\partial v}{\partial x} + v \frac{\partial v}{\partial y} &= -\frac{\partial p'}{\partial y} + \theta - \frac{1}{2} + \frac{1}{\sqrt{Gr}} \left[\frac{\partial^2 v}{\partial x^2} + \frac{\partial^2 v}{\partial y^2} \right] \\ \frac{\partial \theta}{\partial t} + u \frac{\partial \theta}{\partial x} + v \frac{\partial \theta}{\partial y} &= \frac{1}{Pr\sqrt{Gr}} \left[\frac{\partial^2 \theta}{\partial x^2} + \frac{\partial^2 \theta}{\partial y^2} \right]\end{aligned}$$

où Gr est le nombre de Grashof, $Gr = g\beta(T_h - T_c)L^3/\nu^2$, et Pr le nombre de Prandtl, $Pr = \nu/\alpha$. On peut également utiliser le nombre de Rayleigh, Ra , égal au produit $GrPr$.

Dans ce type d'écoulement, il est intéressant de prédire les transferts thermiques en paroi. Le nombre adimensionnel représentatif de ces transferts est le nombre de Nusselt,

$$Nu_{paroi} = \left. \frac{\partial T}{\partial x} \right|_{x=0 \text{ ou } x=1}$$

où $x = 0$ représente la frontière "chaude" et $x = 1$ la frontière "froide". On définit également le Nusselt moyen,

$$\overline{Nu}_{paroi} = \int_0^1 \left. \frac{\partial T}{\partial x} \right|_{x=0 \text{ ou } x=1} dy$$

Il est également intéressant de visualiser les lignes de courant $\Psi = \text{Cte}$ de l'écoulement (voir chapitre 7).

Les équations de Navier-Stokes sont discrétisées à l'aide de l'opérateur NS, et l'équation d'énergie exprimée en température, est discrétisée à l'aide de l'opérateur TSCAL. Pour ces deux opérateurs, l'option 'SUPG' a été choisie afin de minimiser la diffusion numérique (par rapport à l'option par défaut 'SUPG-DC').

Résultats :

On calcule l'écoulement pour quatre nombres de Rayleigh différents, $Ra = 10^3$, $Ra = 10^4$, $Ra = 10^5$ et $Ra = 10^6$, qui correspondent aux cas-tests proposés par Vahl Davis et Jones [1].

Deux types de maillages ont été considérés : le premier est un maillage régulier de quadrangles, contenant $40 \times 40 = 1600$ éléments, soit 1681 nœuds ; le second est un maillage non-structuré, constitué de triangles contenant 3832 éléments, soit 1997 nœuds.

Les résultats sont reportés dans le tableau suivant :

Ra	éléments	$ \Psi_{\max} $	u_{\max}	v_{\max}	\overline{Nu}	Nu_{\max}	Nu_{\min}
10^3	QUA8	1.1952	3.7254	3.7566	1.0798	1.3919	0.7393
10^3	TRI6	0.9552	3.0337	3.0272	1.0119	1.1300	0.8898
10^3	référence [1]	—	3.649	3.697	1.118	1.505	0.692
10^4	QUA8	4.7716	15.563	19.036	2.2649	3.5428	0.6108
10^4	TRI6	5.2337	16.681	20.045	2.3682	3.6441	0.5627
10^4	référence [1]	—	16.178	19.617	2.243	3.528	0.586
10^5	QUA8	10.992	47.953	66.168	4.7334	7.8260	1.2306
10^5	TRI6	9.0343	41.402	66.206	4.4175	7.3230	0.7507
10^5	référence [1]	9.612	34.73	68.59	4.519	7.717	0.729
10^6	QUA8	17.382	121.73	219.96	8.5630	16.645	1.0467
10^6	TRI6	16.667	113.96	213.64	8.2850	15.536	0.9690
10^6	référence [1]	16.750	64.63	219.36	8.800	17.925	0.989

Dans l'ensemble, on remarque que les résultats obtenus avec CASTEM2000 sont en bon accord avec les solutions de référence, notamment en ce qui concerne les transferts de chaleur, les maxima de la fonction de courant ainsi que de la composante verticale de la vitesse. L'accord est moins bon pour ce qui est de la composante horizontale de la vitesse, puisque CASTEM2000 tend à surestimer ces vitesses.

Référence :

- 1 G. De Vahl Davis and I.P. Jones. Natural Convection in a Square Cavity : a Comparison Exercise. *International Journal for Numerical Methods in Fluids*, 3 :227-248, 1983.




FIG. 2.3.2 – $Ra = 10^6$, maillage de quadrangles




FIG. 2.3.3 – $Ra = 10^6$, iso-températures




FIG. 2.3.4 – $Ra = 10^6$, champ de vitesse




FIG. 2.3.5 – $Ra = 10^6$, lignes de courant




FIG. 2.3.6 – $Ra = 10^6$, histoire de convergence




FIG. 2.3.7 – $Ra = 10^6$, Nusselt à la paroi




FIG. 2.3.8 – $Ra = 10^6$, maillage de triangles




FIG. 2.3.9 – $Ra = 10^6$, iso-températures




FIG. 2.3.10 – $Ra = 10^6$, champ de vitesse




FIG. 2.3.11 – $Ra = 10^6$, lignes de courant




FIG. 2.3.12 – $Ra = 10^6$, histoire de convergence




FIG. 2.3.13 – $Ra = 10^6$, Nusselt à la paroi

Jeu de données :

```

COMPLET = FAUX ;
SI ( COMPLET ) ;
    d1 = 0.02 ;
    d2 = 0.1;
    NITER = 5000 ;
    CFL = 1.0 ;
SINON ;
    d1 = 0.05 ;
    d2 = 0.5 ;
    NITER = 600 ;
    CFL = 1.0 ;
FINSI ;
GRAPH = FAUX ;

*****
*                CAVITE CARREE - VAHL-DAVIS                *
* A. Chene/H. Paillere TTMF Aout 1997                       *
*****

*****
*ESTIMATION DE LA CONVERGENCE*
*****

DEBPROC CALCUL;
ARGU RVX*'TABLE';

RV = RVX.'EQEX';

DD = RV.PASDETPS.'NUPASDT';
NN = DD/5;
LO = (DD-(5*NN)) EGA 0;

SI (LO);

UN  = RV.INCO.'UN';
UNM1 = RV.INCO.'UNM1';

unx = kcht $MT scal sommet (exco 'UX' un);
unm1x = kcht $MT scal sommet (exco 'UX' unm1);
uny = kcht $MT scal sommet (exco 'UY' un);
unm1y = kcht $MT scal sommet (exco 'UY' unm1);

ERRX = KOPS unx '- ' unm1x;
ERRY = KOPS uny '- ' unm1y;
ELIX = MAXI ERRX 'ABS';
ELIY = MAXI ERRY 'ABS';
ELIX = (LOG (ELIX + 1.0E-10))/(LOG 10.);
ELIY = (LOG (ELIY + 1.0E-10))/(LOG 10.);
MESSAGE 'ITER' RV.PASDETPS.'NUPASDT' 'ERREUR LINF' ELIX ELIY;
RV.INCO.'UNM1' = KCHT $MT 'VECT' 'SOMMET' (RV.INCO.'UN');
IT = PROG RV.PASDETPS.'NUPASDT';
ER = PROG ELIY;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER;

```

```

FINSI;
as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

*****
* MAILLAGE *
*****

OPTI DIME 2;
OPTI ELEM QUA8;

p1 = 0. 0.;
p15 = 0.5 0.;
p2 = 1. 0.;
p25 = 1. 0.5;
p3 = 1. 1.;
p35 = 0.5 1.;
p4 = 0. 1.;
p45 = 0. 0.5;

bas = p1 d dini d1 dfin d2 p15 d dini d2 dfin d1 p2;
cdro = p2 d dini d1 dfin d2 p25 d dini d2 dfin d1 p3;
haut = p3 d dini d1 dfin d2 p35 d dini d2 dfin d1 p4;
cgau = p4 d dini d1 dfin d2 p45 d dini d2 dfin d1 p1;

cnt = bas et cdro et haut et cgau;
mt = bas cdro haut cgau daller;

*****
* MODE *
*****

mt= chan mt quaf ;
$mt = MODE mt 'NAVIER_STOKES' 'MACRO' ;
doma $mt 'IMPR' ;
mt = doma $mt 'MAILLAGE' ;

*****
* PARAMETRES *
*****

Pr = 0.71;
Ra = 1.e6;
Gr = Ra/Pr;
NU = 1/(Gr**0.5);
ALF= NU/Pr;
gb = 0 -1;
uref = 1;

*****
* CREATION DES TABLES *
*****

RV = EQEX $MT 'ITMA' NITER 'ALFA' CFL
'ZONE' $MT 'OPER' CALCUL

```

```

'OPTI' 'SUPG'
'ZONE' $MT 'OPER' 'NS' NU GB 'TN' 0.5 'INCO' 'UN'
'OPTI' 'SUPG' 'MMPG'
'ZONE' $MT 'OPER' 'TSCAL' ALF 'UN' 0. 'INCO' 'TN'
'OPTI' 'CENTREE'
ZONE $MT 'OPER' 'DFDT' 1. 'UN' 'DELTAT' INCO 'UN'
ZONE $MT 'OPER' 'DFDT' 1. 'TN' 'DELTAT' INCO 'TN'
;

RV = EQEX RV
'CLIM' 'UN' 'UIMP' cnt 0.
      'UN' 'VIMP' cnt 0.
      'TN' 'TIMP' cgau 1.
      'TN' 'TIMP' cdro 0.;

RVP = EQPR $MT KTYPI 1
      'ZONE' $MT 'OPER' 'PRESSION' 0.
      PIMP 0. ;

RV.INCO = TABLE INCO;
RV.INCO.'UN' = kcht $MT VECT SOMMET (0. 0.);
corx = coor 1 mt;
RV.INCO.'TN' = kcht $MT SCAL SOMMET (1.-corx);

RV.INCO.'UNM1' = kcht $MT VECT SOMMET (1.E-5 1.E-5);
RV.INCO.'IT' = PROG 1;
RV.INCO.'ER' = PROG 0.;
RV.PRESSION = RVP;

EXEC RV;

*****
* RESULTATS *
*****

cdro=chan cdro quaf ;
elim (mt et cdro) 1.e-3 ;
$cdro = mode cdro 'NAVIER_STOKES' ;
DY = DOMA $cdro 'VOLUME';
DYT = SOMT DY;

GRADT = KOPS RV.INCO.'TN' 'GRAD' $MT;
DTDX = KCHT $MT 'SCAL' 'CENTRE' (EXCO 'UX' GRADT);
DTDX = ELNO $MT DTDX;

DTDXp = KCHT $CDRO 'SCAL' 'SOMMET' DTDX;
DTDXp = KOPS DTDXp '*' (-1.);

EVOL1 = EVOL 'CHPO' DTDXp SCAL CDRO;
EVOL1 = EVOL1 COUL ROUG ;

DTDXpc = NOEL $CDRO DTDXp;
num = KOPS DTDXpc '*' DY;
num = SOMT NUM;
num = (-1)*NUM/DYT;

```



```

MESSAGE 'NUSSELT MOYEN' NUM;
MESSAGE 'NUSSELT MAX' (MAXI DTDXpc);
MESSAGE 'NUSSELT MIN' (MINI DTDXpc);

SI ( (MAXI DTDXp) < 12.00 ) ;
    ERREUR 5 ;
FINSI ;
SI ( ABS (NUM) < 7.55 ) ;
    ERREUR 5 ;
FINSI ;
SI ( (MINI RV.INCO.'ER') > -4.95 ) ;
    ERREUR 5 ;
FINSI ;

SI ( GRAPH ) ;
OPTI ISOV SULI;
trace mt 'TITR' 'MAILLAGE' ;
trac RV.INCO.'TN' mt (cont mt) 14 'TITR' 'TEMPERATURE' ;

unch = vect RV.INCO.'UN' 1. UX UY JAUNE;
trace unch mt (cont mt) 'TITR' 'CHAMP DE VITESSE' ;

EVOL4 = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
    (RV.INCO.'ER') ;
dess EVOL4 'XBOR' 0. 10000. 'YBOR' -10.0 0.0
    'TITR' 'CONVERGENCE VERS LE STATIONNAIRE' ;

TAB1 = TABLE ;
TAB1.1 = 'MARQ LOSA';
DESS EVOL1 'XBOR' 0. 1. 'GRIL' TAB1 'TITR' 'NUSSELT A LA PAROI' ;

un = RV.INCO.'UN';
un2=KOPS un '*' (Pr*(Gr**0.5));
sw = (-1.) * (KOPS un 'ROT' $mt) ;
rk = EQEX $mt 'OPTI' 'EF' 'IMPL'
ZONE $mt OPER LAPN 1. INCO 'PSI'
ZONE $mt OPER FIMP sw INCO 'PSI'
'CLIM' 'PSI' 'TIMP' (cgau et haut et bas et cdro) 0.;
rk.'INCO'=table 'INCO' ;
rk.'INCO'.'PSI'=kcht $mt scal sommet 0. ;
EXEC rk ;
psi=rk.'INCO'.'PSI';
psi2=kops psi '*' (Pr*(Gr**0.5)) ;
LISTE (MAXI PSI) ;
LISTE (MINI PSI) ;
trac psi mt CNT 14 'TITR' 'FONCTION DE COURANT' ;
FINSI ;

FIN ;

```

2.4 Tourbillon dans une cavité carrée

2.4.1 Formulation Eléments Finis Semi-Implicite

<i>Opérateurs utilisés</i>	<i>NS, PRESSION</i>
<i>Méthodes numériques</i>	<i>EFM1, approximations P1-P0 ou Q1-P0, schéma SUPG</i>
<i>Nature du problème</i>	<i>écoulement tourbillonnaire dans une cavité carrée : comparaison avec une solution analytique des équations de Navier-Stokes pour déterminer l'ordre de convergence des schémas</i>
<i>Classification</i>	<i>2D plan, incompressible, laminaire, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/vortex.dgibi</i>

Description du problème

Ce cas-test de validation concerne la simulation d'un écoulement tourbillonnaire dans une cavité carrée, pour lequel on peut exhiber une solution analytique des équations de Navier-Stokes [1]. Cela permet d'effectuer une étude de convergence pour mesurer l'ordre de convergence de l'opérateur de discrétisation Navier-Stokes **NS**, pour des éléments P1-P0 (triangles) ou Q1-P0 (quadrangles), représentés sur la figure 2.4.1.




FIG. 2.4.1 – Eléments vitesse-pressure Q1-P0 et P1-P0 de CASTEM2000

On considère les équations de Navier-Stokes incompressibles, stationnaires, en présence de forces de volume \vec{f} , de composantes (f_1, f_2) ,

$$\begin{aligned} u_x + v_y &= 0 \\ uu_x + vv_y &= -p_x + f_1 + \nu(u_{xx} + u_{yy}) \\ uv_x + vv_y &= -p_y + f_2 + \nu(v_{xx} + v_{yy}) \end{aligned}$$

sur le domaine $\Omega = \{(x, y) \in [0, 1] \times [-\frac{1}{2}, \frac{1}{2}]\}$, avec les conditions aux limites suivantes :

$$\begin{aligned} \text{sur } x = 0, & \quad u = 0 & \quad v = b \cos \pi y \\ \text{sur } x = 1, & \quad u = 0 & \quad v = -b \cos \pi y \\ \text{sur } y = -\frac{1}{2}, & \quad u = -b \sin \pi x & \quad v = 0 \\ \text{sur } y = \frac{1}{2}, & \quad u = b \sin \pi x & \quad v = 0 \end{aligned}$$

avec $b = 1/\nu$. Pour (f_1, f_2) de la forme :


$$\begin{pmatrix} f_1 \\ f_2 \end{pmatrix} = \begin{pmatrix} \pi b^2 \sin \pi x \cos \pi y + 2\pi^2(1 + \alpha) \sin \pi x \sin \pi y \\ -\pi b^2 \sin \pi y \cos \pi x + 2\pi^2(1 - \alpha) \cos \pi x \cos \pi y \end{pmatrix}$$

où α est un paramètre, la solution analytique (u, v, p) est donnée par :

$$\begin{aligned} u_{ex} &= b \sin \pi x \sin \pi y \\ v_{ex} &= b \cos \pi x \cos \pi y \\ p_{ex} &= -2\pi\alpha \cos \pi x \sin \pi y \end{aligned}$$

où la pression p est bien sûr définie à une constante additive près. On a en outre,

$$\delta p_{ex} = p_{\max} - p_{\min} = 4\pi \approx 12.57$$




```

VAL = 150
>= 6.28E+00
< 6.28E+00

A -5.9
B -5.0
C -4.2
D -3.4
E -2.5
F -1.7
G -.84
H 0.00E+00
I .84
J 1.7
K 2.5
L 3.4
M 4.2
N 5.0
O 5.9

```



```

VAL = 150
>= 1.00E+01
< 1.00E+01

A -9.3
B -8.0
C -6.7
D -5.3
E -4.0
F -2.7
G -1.3
H 0.00E+00
I 1.3
J 2.7
K 4.0
L 5.3
M 6.7
N 8.0
O 9.3

```

FIG. 2.4.2 – Pression théorique

FIG. 2.4.3 – Composante u théorique




FIG. 2.4.4 – Champ de vitesse théorique

On définit les normes L_2 discrètes de l'erreur sur la vitesse et la pression de la façon suivante :

$$\begin{aligned} E(u) &= \|u - u_{ex}\|_{L_2} = \sqrt{\frac{1}{N_s} \sum_{i=1}^{N_s} (u_i - u_{ex}(x_i, y_i))^2} \\ E(p) &= \|p - p_{ex}\|_{L_2} = \sqrt{\frac{1}{N_c} \sum_{i=1}^{N_c} (p_i - p_{ex}(x_i, y_i))^2} \end{aligned}$$

où N_s et N_c sont respectivement le nombre de points sommet (où la vitesse est définie) et le nombre de points centre (où la pression est définie).

Etude de convergence

On considère deux types de maillages, l'un constitué de quadrangles avec des éléments Q1-P0, l'autre de triangles avec des éléments P1-P0. On évalue les erreurs $E(u)$ et $E(p)$ pour des maillages de plus en plus fins. Les résultats sont reportés dans les tables 2.4.1, et tracés sur les courbes 2.4.5-2.4.6 et 2.4.7-2.4.8 en échelle logarithmique. Une régression linéaire sur ces courbes permet d'estimer l'ordre de convergence de la vitesse et de la pression. On trouve que sur des maillages de quadrangles, la vitesse converge en $\mathcal{O}(h^{1.44})$ et la pression en $\mathcal{O}(h^{1.21})$, et sur des triangles, la vitesse converge en $\mathcal{O}(h^{1.52})$ et la pression en $\mathcal{O}(h^{1.08})$. Ces résultats, légèrement en deçà des prévisions théoriques pour la vitesse (on attend un ordre proche de 2, voir l'étude de convergence de l'opérateur de convection-diffusion scalaire TSCAL) et légèrement au dessus de l'ordre 1 anticipé pour la pression (supposée constante par élément), restent néanmoins satisfaisants.

Les figures 2.4.9-2.4.12 et 2.4.13-2.4.16 montrent les solutions numériques obtenues sur des maillages de quadrangles et de triangles de longueur caractéristique de maille $h = 1/10$.

\square	δp	E_u	E_p
N=10	62.97	1.74	15.61
N=16	55.33	1.43	13.57
N=20	49.26	1.22	12.03
N=24	43.85	1.05	10.54
N=30	37.12	0.84	8.59
N=36	31.81	0.68	7.01
N=40	28.90	0.59	6.16
N=50	23.32	0.44	4.57
N=56	20.83	0.37	3.91
N=60	19.43	0.33	3.57
N=70	16.67	0.26	2.99
N=84	14.00	0.19	2.62
\longrightarrow <i>Ordre</i> =		1.44	1.21

Δ	δp	E_u	E_p
N=10	63.64	1.36	13.46
N=16	48.79	1.05	10.31
N=20	40.15	0.83	8.18
N=24	33.67	0.66	6.87
N=30	26.48	0.49	5.14
N=36	21.86	0.38	4.02
N=40	19.71	0.32	3.51
N=50	15.86	0.23	2.87
N=56	14.23	0.18	2.73
N=60	13.39	0.16	2.67
N=70	13.30	0.12	2.04
\longrightarrow <i>Ordre</i> =		1.52	1.08

TAB. 2.4.1 – Ecart de pression $\delta p = p_{\max} - p_{\min}$ et erreurs E_u et E_p en fonction de la longueur caractéristique de maille $h = 1/N$, pour l'opérateur Navier-Stokes NS (option SUPG) dans le cas de maillages de quadrangles et de triangles.

Référence :

- 1– K. Boukir, Y. Maday, B. Métivet and E. Razafindrakoto, A High Order Characteristics/Finite Element Method for the Incompressible Navier-Stokes Equations. Submitted to Int. J. Num. Meth. Fluids. Rapport EDF :HI-72/7622

Simulation d'un tourbillon : convergence de la vitesse




FIG. 2.4.5 – Convergence en vitesse sur quadrangles

Simulation d'un tourbillon : convergence de la pression




FIG. 2.4.6 – Convergence en pression sur quadrangles

Simulation d'un tourbillon : convergence de la vitesse




FIG. 2.4.7 – Convergence en vitesse sur triangles

Simulation d'un tourbillon : convergence de la pression






FIG. 2.4.8 – Convergence en pression sur triangles




GIBI PRCT

FIG. 2.4.9 – Maillage de quadrangles ($h = 1/10$)



GIBI PRCT


FIG. 2.4.10 – Historique de convergence pour atteindre l'état stationnaire



PRCT

FIG. 2.4.11 – Champ de pression calculé


VAL - 100
 >= 2.00E+01
 < 2.00E+01
 A -10.
 B -18.
 C -12.
 D -8.9
 E -5.7
 F -2.4
 G 0.
 H 4.2
 I 7.5
 J 11.
 K 14.
 L 17.
 M 21.
 N 24.
 O 27.



UX

FIG. 2.4.12 – Champ de vitesse (composante u) calculé

VAL - 100
 >= 1.00E+01
 < 1.00E+01
 A -9.3
 B -8.0
 C -6.7
 D -5.3
 E -4.0
 F -2.7
 G -1.3
 H 0.00E+00
 I 1.3
 J 2.7
 K 4.0
 L 5.3
 M 6.7
 N 8.0
 O 9.3



GIBI PFCST

FIG. 2.4.13 – Maillage de triangles ($h = 1/10$)






FIG. 2.4.14 – Historique de convergence pour atteindre l'état stationnaire



PRESSI001

FIG. 2.4.15 – Champ de pression calculé

VAL - 100
 > 1.220E+01
 < 2.140E+01
 A -11.
 B -8.5
 C -6.0
 D -3.4
 E -0.93
 F 1.6
 G 4.1
 H 6.6
 I 9.1
 J 12.
 K 14.
 L 17.
 M 19.
 N 22.
 O 24.



UX

VAL - 100
 > 1.000E+01
 < 1.000E+01
 A -9.3
 B -8.0
 C -6.7
 D -5.3
 E -4.0
 F -2.7
 G -1.3
 H 0.000E+00
 I 1.3
 J 2.7
 K 4.0
 L 5.3
 M 6.7
 N 8.0
 O 9.3

FIG. 2.4.16 – Champ de vitesse (composante u) calculé

Jeu de données :

```

COMPLET = FAUX ;
SI ( COMPLET ) ;
    NX = 20 ;
    NITER = 2000 ;
    CFL = 1.0 ;
SINON ;
    NX = 5 ;
    NITER = 250 ;
    CFL = 1.0 ;
FINSI ;
GRAPH = 'N' ;

*****
* SIMULATION D'UN TOURBILLON - COMPARAISON AVEC SOLUTION ANALYTIQUE *
* DES EQUATIONS DE NAVIER-STOKES *
* OPERATEUR NS - OPTION SUPG *
* H. Paillere - Avril 1997 *
*****

*****
* PROCEDURE POUR ESTIMER LA CONVERGENCE VERS L'ETAT STATIONNAIRE *
*****

DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;

RV = RVX.'EQEX' ;

UN = RV.INCO.'UN' ;
UNM1 = RV.INCO.'UNM1' ;

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;

unx= kcht $MT scal sommet (exco 'UX' un) ;
unm1x = kcht $MT scal sommet (exco 'UX' unm1) ;
uny= kcht $MT scal sommet (exco 'UY' un) ;
unm1y = kcht $MT scal sommet (exco 'UY' unm1) ;

ERRX = KOPS unx '-' unm1x ;
ERRY = KOPS uny '-' unm1y ;
ELIX = MAXI ERRX 'ABS' ;
ELIY = MAXI ERRY 'ABS' ;
ELIX = (LOG (ELIX + 1.0E-20))/(LOG 20.) ;
ELIY = (LOG (ELIY + 1.0E-20))/(LOG 20.) ;

PMAX = MAXI RVP.PRESSION ;
PMIN = MINI RVP.PRESSION ;
DP = PMAX - PMIN ;
MESSAGE 'ITER ' RV.PASDETPS.'NUPASDT' ' ERREUR LINF ' ELIY
        'PMAX = ' PMAX 'PMIN = ' PMIN 'DP = ' DP ;
IT = PROG RV.PASDETPS.'NUPASDT' ;

```



```

ER = PROG ELIY ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER ;

FINSI;

RV.INCO.'UNM1' = KCHT $MT 'VECT' 'SOMMET' (RV.INCO.'UN') ;
as2 ama1 = 'KOPS' 'MATRIK' ;
FINPROC as2 ama1 ;

*****
* DEFINITION DU MAILLAGE *
*****

option dime 2 elem qua8 ;

P1 = 0.0 (0.5*(-1.)) ;
p2 = 1.0 (0.5*(-1.)) ;
p3 = 1.0 0.5 ;
p4 = 0.0 0.5 ;

NY = NX ;

bas = p1 d NX p2 ;
cdro = p2 d NY p3 ;
haut = p3 d NX p4 ;
cgau = p4 d NY p1 ;
cnt = bas et cdro et haut et cgau ;
mt = bas cdro haut cgau daller ;

tass MT ;
orienter mt ;
mt= chan mt quaf ;

$mt = MODE mt 'NAVIER_STOKES' 'MACRO' ;

mt2 = mt ;
mt = DOMA $mt maillage ;

*****
* SOLUTION ANALYTIQUE *
*****

PI = 3.141592654 ;
B = 10. ;
NU = 1.0/B ;
ALPHA = -1.0 ;
XX YY = COOR (DOMA $MT SOMMET) ;

CX = COS ((KOPS XX '*' 180.)) ;
CY = COS ((KOPS YY '*' 180.)) ;
SX = SIN ((KOPS XX '*' 180.)) ;
SY = SIN ((KOPS YY '*' 180.)) ;

UEX = KOPS ( KOPS SX '*' SY ) '*' B ;
VEX = KOPS ( KOPS CX '*' CY ) '*' B ;

```

```

PEX = KOPS ( KOPS CX '*' SY ) '*' (PI*(-2.)*ALPHA) ;

UEX1 = NOMC 'UX' UEX 'NATU' 'DISCRET' ;
VEX1 = NOMC 'UY' VEX 'NATU' 'DISCRET' ;
UUX = KCHT $MT VECT SOMMET (UEX1 ET VEX1) ;

B1 = B * CY ;
B2 = B * SX ;

XXC YYC = COOR (DOMA $MT CENTRE) ;
CX = COS ((KOPS XXC '*' 180.)) ;
CY = COS ((KOPS YYC '*' 180.)) ;
SX = SIN ((KOPS XXC '*' 180.)) ;
SY = SIN ((KOPS YYC '*' 180.)) ;

F1_1 = KOPS ( KOPS SX '*' CX ) '*' (PI*B*B) ;
F1_2 = KOPS ( KOPS SX '*' SY ) '*' (2.0*PI*PI*(ALPHA+1.0)) ;
F1 = KOPS F1_1 '+' F1_2 ;
F2_1 = KOPS ( KOPS SY '*' CY ) '*' (PI*B*B*(-1.0)) ;
F2_2 = KOPS ( KOPS CX '*' CY ) '*' (2.0*PI*PI*(1.0-ALPHA)) ;
F2 = KOPS F2_1 '+' F2_2 ;

F1 = nomc 'UX' F1 'NATU' 'DISCRET' ;
F2 = nomc 'UY' F2 'NATU' 'DISCRET' ;
SS = KCHT $MT VECT CENTRE (F1 ET F2) ;

*****
* TABLES EQEX ET PRESSION *
*****
RV = EQEX $MT 'ITMA' NITER 'ALFA' CFL
'ZONE' $MT 'OPER' CALCUL
'OPTI' 'SUPG' 'MMPG'
'ZONE' $MT 'OPER' 'NS' NU SS 'INCO' 'UN'
'OPTI' 'CENTREE' EFM1
ZONE $MT 'OPER' 'DFDT' 1. 'UN' 'DELTAT' INCO 'UN'
;

RV = EQEX RV
'CLIM' 'UN' 'UIMP' cgau 0. 'UN' 'VIMP' cgau B1
'UN' 'UIMP' cdro 0. 'UN' 'VIMP' cdro (B1*(-1.))
'UN' 'UIMP' bas (B2*(-1.)) 'UN' 'VIMP' bas 0.
'UN' 'UIMP' haut B2 'UN' 'VIMP' haut 0. ;

RVP = EQPR $MT KTYPI 1
ZONE $MT 'OPER' 'PRESSION' 0.
'PIMP' 0. ;

*****
* TABLE INCO *
*****
RV.INCO=TABLE INCO ;
RV.INCO.'UN'=kcht $MT VECT SOMMET (0. 0.) ;
rv.inco.'UNM1' = kcht $MT 'VECT' 'SOMMET' (0. 0.) ;
rv.inco.'IT' = PROG 1 ;
rv.inco.'ER' = PROG 0. ;
RV.PRESSION =RVP ;

```

```

* pour triangles:
* RVP.BETA = 1000. ;

EXEC RV ;

SI ( 'EGA' graph '0' ) ;
EVOL4 = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
      (RV.INCO.'ER') ;
DESS EVOL4 'XBOR' 0. 2000. 'YBOR' -20.0 0.0 ;
TRACE PEX MT CNT 15 'TITR' 'PRESSION THEORIQUE' ;
TRACE RVP.PN MT CNT 15 'TITR' 'PRESSION' ;
unx= kcht $MT scal sommet (exco 'UX' RV.INCO.'UN') ;
TRACE UEX MT CNT 15 'TITR' 'UX THEORIQUE' ;
TRACE UNX MT CNT 15 'TITR' 'UX' ;
UNCH = VECT RV.INCO.'UN' 0.01 UX UY JAUNE ;
TRACE UNCH MT CNT ;
FINSI ;

un = RV.INCO.'UN' ;
unx= kcht $MT scal sommet (exco 'UX' un) ;
uny= kcht $MT scal sommet (exco 'UY' un) ;

ERRX = KOPS UNX '-' UEX1 ;
ERRX = KOPS ERRX '*' ERRX ;
ERR2 = 0. ;
NPTD= NBNO (DOMA $MT SOMMET) ;
REPETER BLOC1 NPTD ;
P1 = (DOMA $MT SOMMET) POIN &BLOC1 ;
ERR1 = 'EXTR' ERRX 'SCAL' P1 ;
ERR2 = ERR2 + ERR1 ;
FIN BLOC1 ;
ERR2 = ERR2/NPTD ;
ERR2 = ERR2 ** 0.5 ;
MESSAGE 'ERREUR VITESSE UX EN NORME L2 = ' ERR2 ;

SI ( ERR2 > 6.0E-2 ) ; ERREUR 5 ; FINSI ;

PEX = NOEL $MT PEX ;
PPP = ((MAXI RVP.PRESSION) + (MINI RVP.PRESSION))/2. ;
PN = KOPS RVP.PRESSION '-' PPP ;
ERRP = KOPS PEX '-' PN ;
ERRP = KOPS ERRP '*' ERRP ;
ERR2 = 0. ;
NELD= NBEL (DOMA $MT CENTRE) ;
REPETER BLOC1 NELD ;
P1 = (DOMA $MT CENTRE) POIN &BLOC1 ;
ERR1 = 'EXTR' ERRP 'SCAL' P1 ;
ERR2 = ERR2 + ERR1 ;
FIN BLOC1 ;
ERR2 = ERR2/NELD ;
ERR2 = ERR2 ** 0.5 ;
MESSAGE 'ERREUR PRESSION EN NORME L2 = ' ERR2 ;

SI ( ERR2 > 0.80 ) ; ERREUR 5 ; FINSI ;

```

```
SI ( (MINI RV.INCO.'ER') > -4.0 ) ; ERREUR 5 ; FINSI ;  
FIN ;
```

2.5 Convection naturelle dans un fluide à nombre de Prandtl nul

2.5.1 Formulation Eléments Finis Semi-Implicite

<i>Opérateurs utilisés</i>	<i>NS, PRESSION</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation Q1-P0, schéma CENTREE</i>
<i>Nature du problème</i>	<i>Convection naturelle en cavité pour un fluide de $Pr=0$</i>
<i>Classification</i>	<i>2D plan, Boussinesq, laminaire, stationnaire et instationnaire périodique</i>
<i>Jeu de données</i>	

Schéma du problème :




FIG. 2.5.1 – Modèle de la cavité rectangulaire.

Modélisation :

Le cas étudié, proposé dans le cadre du GAMM-Workshop ([1] et [2]) avait servi de validation au code TRIO-EF. C'est un cas test simple de convection naturelle pour un fluide de nombre de Prandtl nul dans une cavité fermée de rapport d'aspect $L/H = 4$ (figure 2.5.1).

Les équations de Navier-Stokes instationnaires dans le cadre de l'approximation de Boussinesq s'écrivent sous la forme adimensionnée :

$$\nabla \cdot \underline{u} = 0$$

$$\frac{\partial \underline{u}}{\partial t} + Gr^{1/2} (\underline{u} \cdot \nabla) \underline{u} = -\nabla p + \Delta \underline{u} - Gr^{1/2} T \underline{\eta}$$

avec : $Gr = \frac{g\beta\gamma H^4}{\nu^2}$; $\gamma = \frac{T_2 - T_1}{L}$ et $\underline{\eta} = (0, -1, 0)$.

β est le coefficient d'expansion volumique et ν la viscosité cinématique du fluide.

Conditions aux limites :

- $u = 0$ sur les bords de la cellule ;
- $T(x, y, t) = x$ car l'équation de l'énergie à Prandtl nul se réduit à $\Delta T = 0$ et on a flux nul sur les parois horizontales et température imposée sur les parois verticales.

Discretisation :

Les équations précédentes sont discrétisées grâce aux opérateurs **NS** et **PRESSION** de CASTEM2000. L'option 'CENTREE' a été choisie car c'est la plus précise, et elle s'est avérée suffisamment stable pour la résolution. Le maillage est régulier, constitué de 120x30 quadrangles (QUA8 avec l'option 'MACRO').

Synthèse des résultats :

Solution à obtenir

L'important diagramme de stabilité (figure 2.5.2) résume les résultats obtenus avec TRIO-EF comparés à ceux d'un autre participant à la session de travail du GAMM-Workshop. On distingue sur ce diagramme différentes formes d'écoulements :




FIG. 2.5.2 – Diagramme de stabilité (d'après Le Garrec, S. & Magnaud, J-P. [2]).

- : Pulicani et al.
- et lignes de courant : TRIO-EF.

1. Écoulements stationnaires :

- Pour les faibles nombres de Grashof ($Gr = 7500 - 10000$), le fluide est entraîné le long des parois de la cavité ; l'écoulement de base est monocellulaire et stationnaire.

- Pour des nombres de Grashof plus élevé ($Gr = 20000$), on observe un autre type d'écoulement stationnaire. A l'écoulement de base se superpose un écoulement composé d'une cellule centrale et de deux rouleaux périphériques.
2. Ecoulements instationnaires périodiques :
 - Pour des nombres de Grashof compris entre 25000 et 35000 environ, on observe des écoulements instationnaires périodiques. Au cours d'une période, une configuration multicellulaire succède à une configuration monocellulaire. Pour d'autres caractéristiques (période, vitesses,...) de ce type d'écoulements, se reporter à [2].
 3. Transition vers un état bicellulaire stationnaire, phénomène d'hystérésis :
 - Pour $Gr = 40000$, l'écoulement est composé de deux rouleaux stationnaires. Cette solution est obtenue après une phase transitoire périodique durant laquelle l'écoulement se dissymétrise progressivement et tend brutalement vers la configuration à deux rouleaux (dite «yeux de chat»).
 - Partant de cette solution bicellulaire comme condition initiale, on obtient le même type d'écoulement pour $Gr = 30000$. Il existe donc un domaine dans lequel, pour une même valeur du nombre de Grashof, il est possible d'obtenir deux solutions différentes en fonction de la condition initiale choisie.

Résultats obtenus avec CASTEM2000

Les résultats que l'on obtient sont très similaires à ceux obtenus au cours du GAMM-Workshop (tableau 2.5.1).

Gr	CI	état atteint	Nb cellules	Allure (figure 2.5.3)
10000	$\underline{u} = 0$	stationnaire	1	(a)
20000	$\underline{u} = 0$	stationnaire	3	(b)
30000	$\underline{u} = 0$	périodique	1-3	(c)
40000 ¹	$\underline{u} = 0$	stationnaire	2	(d)
30000	$Gr = 40000$	stationnaire	2	(d)
10000	$Gr = 40000$	périodique	1-3	(c)

TAB. 2.5.1 – Résumé des propriétés des écoulements obtenus avec CASTEM2000

Comparaison CASTEM2000 - TRIO-EF

Les tableaux qui suivent (2.5.2) permettent de comparer quelques grandeurs caractéristiques obtenues avec CASTEM2000 et TRIO-EF. Pour des comparaisons avec d'autres participants à la session de travail du GAMM-WORKSHOP, voir [2], p. 24-25]. On obtient des résultats numériques assez proches (écarts de l'ordre de 5%) même si le passage de l'état tricellulaire stationnaire à l'état instationnaire s'effectue pour un nombre de Grashof plus élevé pour CASTEM2000 (supérieur à 25000) que pour TRIO-EF ($Gr = 24500$).

- 1 LE GARREC, S. 1988 CONVECTION NATURELLE EN CAVITÉ POUR DES FLUIDES À FAIBLE NOMBRE DE PRANDTL. *Rapport CEA DMT/88/216*
- 2 LE GARREC, S. & MAGNAUD, J-P. 1988 NUMERICAL SIMULATION OF OSCILLATORY CONVECTION IN LOW PRANDTL FLUIDS WITH TRIO-EF. GAMM-WORKSHOP. IN PP. 189-199. VIEWEG. *Numerical Simulation of Oscillatory Convection in Low-Pr Fluids* (ED. B. ROUX),




FIG. 2.5.3 – Lignes de courant obtenues avec CASTEM2000 pour : (a) $Gr = 10000$; (b) $Gr = 20000$; (c) $Gr = 30000$ à différents instants et (d) $Gr = 40000$ avec $T_{ref} = T_1$.

(a)

	Etat	Umax en $x=A/4$; $y=$		Vmax en $y=1/2$; $x=$	
CASTEM2000	stable	0.671	0.167	0.496	1.53
TRIO-EF	stable	0.6882	0.168	0.5121	1.567

(b)

	Etat	Umax en $x=A/4$	Vmax en $y=1/2$	f
CASTEM2000	périodique	0.943	0.895	18.0
TRIO-EF	périodique	0.9738	0.9549	17.6

(c)

	Etat	Umax en $x=A/4$ et $y=$		Vmax en $y=1/2$ et $x=$		f (régime transitoire)
CASTEM2000	stable	1.23	0.833	0.811	0.633	21.8
TRIO-EF	stable	1.1910	0.800	0.8567	0.7667	21.0

TAB. 2.5.2 – Comparaison CASTEM2000 - TRIO-EF dans les cas : (a) $Gr = 20000$; (b) $Gr = 30000$ et (c) $Gr = 40000$.

Jeu de données

Le jeu de données présenté est à $Gr=40000$ et démarre d'une situation au repos. La structure de l'écoulement passe rapidement d'une cellule stationnaire à trois cellules instationnaires puis au bout d'un certain nombre de cycles bifurque brutalement vers la solution stationnaire à 2 cellules. Comptez 40 minutes et 15000 itérations pour obtenir les yeux de chat avec un maillage régulier 40×10 .

```

***** OPTIONS *****
OPTI DIME 2 ELEM QUA8 ;

***** DONNEES *****
a=4. ; nu=1. ; l=a ; h=1. ; t1=0. ; t2=a ;

***** MAILLAGE *****
nh=5 ; nl=nh*4 ;
p1=0. 0. ; p2=1 0. ; p3=1 h ; p4= 0. h ;
bas = p1 D nl p2 ; rwall=p2 D nh p3 ;
haut= p3 D nl p4 ; lwall=p4 D nh p1 ;
mt=DALLER bas rwall haut lwall ;
TASSER mt ; mt= ORIE mt ;
$mt=DOMA mt 'MACRO' 'IMPR' ;
* On doit remettre le maillage fin dans mt pour 'CLIM'
mt=($mt.maillage) ;

***** CREATION DES TABLES *****
rv = EQEX $mt 'ALFA' 0.5
OPTI CENTREE
'ZONE' $mt 'OPER' 'NS' 'NU' 'GB' 'TN' 'TREF' 'INCO' 'UN'
'ZONE' $mt 'OPER' 'CALCPSI'
'CLIM' 'UN' 'UIMP' (bas et lwall et rwall et haut) 0.
'UN' 'VIMP' (bas et lwall et rwall et haut) 0.
;
rvp = EQPR $mt KTYPI 1
ZONE $mt 'OPER' 'PRESSION' 0.
PIMP 0.
;

***** CONDITIONS INITIALES *****
rv.'INCO'=TABLE 'INCO' ;
rv.'INCO'.'UN' = KCHT $mt VECT SOMMET (0. 0.) ;
rv.'INCO'.'PSI' = KCHT $mt SCAL SOMMET 0. ;
xmt = COOR 1 mt;
rv.'INCO'.'TN'=kcht $mt SCAL SOMMET (t1+ ((xmt/L)*(t2-t1)) ) ;
rv.'PRESSION'=rvp ;
* Les deux listes qui suivent contiennent t et psi(t)
rv.'INCO'.'LTPS' = PROG 0.;
rv.'INCO'.'LPSIMAX' = PROG 0.;
nbiter=15000;
Gr=40000 ;
rv.'INCO'.'NU'=nu ;
rv.'INCO'.'GB'= (0. (-1.*Gr));
* Le choix de TREF conditionne l'apparition des bifurcations
* rv.'INCO'.'TREF' = (t1+t2)/2.;
rv.'INCO'.'TREF'= t1;
rv.'FIDT'=100;

```

```

rv.'ITMA'=nbiter;
uref = 0.2* (1/(Gr ** 0.5)) ;

***** PROCEDURE DE CALCUL DE LA FONCTION DE COURANT *****
*****      Laplacien(psi) = - Rotationnel2D (vitesse) *****
DEBPROC CALCPSI ;
ARGU RX*TABLE ;
rv=rx.'EQEX' ;
pasdt=rv.'PASDETPS'.'NUPASDT';
vitesse = rv.'INCO'.'UN';
tps = rv.'PASDETPS'.'TPS';
rt2d= KOPS vitesse 'ROT' $mt ;
sw = (-1.)* rt2d ;
rk = EQEX $mT 'OPTI' 'EF' 'IMPL'
ZONE $mt OPER LAPN 1.      INCO 'PSI'
ZONE $mt OPER FIMP sw      INCO 'PSI'
'CLIM' 'PSI' 'TIMP' (bas et lwall et rwall et haut) 0.
;
rk.'INCO'=table 'INCO' ;
rk.'INCO'.'PSI'=kcht $mt scal sommet 0. ;
EXIC rk ;
psi=rk.'INCO'.'PSI';
rv.'INCO'.'PSI'= psi;
psimax = MAXI (ABS psi);
ltps   = rv.'INCO'.'LTPS';
lpsimax = rv.'INCO'.'LPSIMAX';
rv.'INCO'.'LTPS'   = ltps   ET (PROG tps   );
rv.'INCO'.'LPSIMAX' = lpsimax ET (PROG psimax);
FINPROC;

***** CALCUL DE LA SOLUTION *****
EXEC rv ;

```

2.6 Convection Rayleigh-Marangoni

2.6.1 Formulation Eléments Finis Semi-Implicite

<i>Opérateurs utilisés</i>	<i>NS, TSCAL, TOIMP, PRESSION</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation Q1-P0, schéma SUPG-SUPGCC</i>
<i>Nature du problème</i>	<i>Convection couplée Rayleigh-Marangoni dans l'acétone.</i>
<i>Classification</i>	<i>2D plan, Boussinesq, laminaire, stationnaire.</i>
<i>Jeu de données</i>	

Schéma du problème :




FIG. 2.6.1 – Modèle de la cavité utilisée.

Modélisation :

Le cas étudié s'appuie sur un article de 1991 paru dans le *Journal of Fluid Mechanics* (voir [1]) dont les auteurs sont D. Villers et J.K. Platten. Dans la couche de fluide soumise à un gradient de température horizontal (figure 2.6.1), la convection est due à deux phénomènes :

- influence des différences de densité (convection naturelle) ;
- forces de tension de surface qui proviennent de la variation de σ avec la température

$$T : \frac{\partial \sigma}{\partial x} = \frac{\partial \sigma}{\partial T} \frac{\partial T}{\partial x}.$$

Nous effectuerons la comparaison entre expérience, simulation numérique Villers & Platten, et simulation CASTEM2000 sur deux cas.

Equations

Les équations de Navier-Stokes instationnaires dans le cadre de l'approximation de Boussinesq s'écrivent sous la forme adimensionnée :

$$\begin{aligned}\nabla \cdot \underline{u} &= 0 \\ \frac{\partial \underline{u}}{\partial t} + Gr^{1/2} (\underline{u} \cdot \nabla) \underline{u} &= -\nabla p + \Delta \underline{u} - Gr^{1/2} T \underline{\eta} \\ \frac{\partial T}{\partial t} + Gr^{1/2} (\underline{u} \cdot \nabla) T &= Pr^{-1} \Delta T\end{aligned}$$

avec : $Gr = \frac{g\beta\gamma H^4}{\nu^2}$; $Pr = \frac{\nu}{\alpha}$; $\gamma = \frac{T_2 - T_1}{A}$; $A = L/H$ et $\underline{\eta} = (0, -1, 0)$.

β est le coefficient d'expansion volumique, α la diffusivité thermique et ν la viscosité cinématique du fluide.

Conditions aux limites

- $\underline{u} = 0$ sur les bords inférieur et latéraux de la cellule ;
- sur la surface libre : $u_z = 0$ et l'effet thermocapillaire (pour une interface plane) se traduit par la condition aux limites de Levich (1962, voir [2]) $\rho_0 \nu \frac{\partial u_x}{\partial z} = \frac{\partial \sigma}{\partial T} \frac{\partial T}{\partial x}$, soit en adimensionnel $Gr^{1/2} \frac{\partial \tilde{u}_x}{\partial \tilde{z}} = -Ma^* \frac{\partial \tilde{T}}{\partial \tilde{x}}$.
- $T(0, z) = T_1 = 0$; $T(A, z) = T_2 = A$ pour les cotés verticaux ;
- $T(x, 0) = T(x, 1) = x$ pour le bas de la cavité (qui est conducteur).

Conditions initiales

Pour les calculs, on choisit les conditions initiales : $\underline{u} = 0$ et $\nabla T = 1$. Les nombres adimensionnels choisis par Villers & Platten pour décrire l'écoulement différent légèrement de ceux que nous avons utilisés :

- nombre de Rayleigh : $Ra = \frac{g\beta(T_2 - T_1)h^3}{\alpha\nu} = \frac{Gr}{A \cdot Pr}$
- nombre de Marangoni : $Ma = -\frac{(\partial\sigma/\partial T)(T_2 - T_1)h}{\rho_0\alpha\nu} = \frac{Ma^*}{A \cdot Pr}$

Discretisation

Les équations précédentes sont discrétisées grâce aux opérateurs **NS**, **TSCAL** et **PRESSION** de CASTEM2000. **TOIMP** est utilisé pour discrétiser la condition de Levich. Les options 'SUPG' et 'SUPGCC' ont été choisies, l'option CENTREE n'étant pas assez stable pour la résolution de ce problème. Le maillage est régulier, constitué de 120x30 quadrangles (QUA8 avec l'option 'MACRO').

Dans les calculs CASTEM2000, la condition de Levich n'a pas été appliquée sur toute la surface (on "enlève" artificiellement le point du maillage le plus proche du mur froid) pour des raisons de stabilité.

Synthèse des résultats

Convection monocellulaire stationnaire




FIG. 2.6.2 – Modèle de la cavité utilisée.

Description sommaire des conditions expérimentales La cavité expérimentale est schématisée sur la figure 2.6.2. Les profils de vitesse sont mesurés par vélocimétrie laser à effet Doppler. Le fluide choisi est l'acétone :

$$\begin{cases}
 - Pr \approx 4.24 \text{ à } 20^\circ\text{C} \\
 - \frac{(\partial\sigma/\partial T) \Delta T}{\rho_0 \alpha \nu} \approx 38 \times 10^3 \text{ cm}^{-1} \\
 - \frac{g\beta\Delta T}{\alpha\nu} \approx 344 \times 10^3 \text{ cm}^{-3}
 \end{cases}
 \quad \text{pour } \Delta T = 1 \text{ K et à } T = 20^\circ\text{C}.$$

Résultats expérimentaux La figure 2.6.3 montre le profil horizontal de vitesse $u_x(z)$ obtenu sur la médiane verticale de la cavité dans les conditions suivantes :

$$\begin{cases}
 h = 2.5 \text{ mm} \\
 \Delta T = 0.8 \text{ K}
 \end{cases}
 \Rightarrow
 \begin{cases}
 Ra = 4300 \\
 Ma = 7600 \\
 A = 12^2
 \end{cases}$$

Résultats numériques et comparaison avec l'expérience Le tableau 2.6.1 permet de comparer les résultats des calculs numériques (figure 2.6.4 avec l'expérience. On s'aperçoit que les valeurs trouvés par CASTEM 2000 sont systématiquement supérieures à l'expérience, ce qui peut s'expliquer par le fait que l'on n'a pas tenu compte de l'influence des murs avant et arrière de la cavité dans la simulation, introduisant de la diffusion. Pour cette raison, on estime que les résultats obtenus sont corrects.




FIG. 2.6.3 – Profil de vitesse horizontale $u_x(z)$ sur la médiane verticale de la cavité pour : $h=2.5$ mm, $\Delta T = 0.8$ K (d'après Villers, D. & Platten, J.K. [1]).




FIG. 2.6.4 – (a) Lignes de courant et isothermes d'une simulation numérique de l'écoulement avec $Pr=4.24$, $Ra=4300$, $Ma=7600$ et $A=12$; ces conditions correspondent à l'expérience de la figure 2.6.3. (b) Profil horizontal de vitesse $u_x(z)$ obtenu par simulation et converti sous une forme dimensionnelle. A gauche, les résultats obtenus par Villers & Platten, à droite, à l'aide de CASTEM2000.

Vitesses en mm/s	Expérience (Villers et Platten)	Simulation numérique (Villers et Platten)	Simulation numérique (CASTEM 2000)
Maximum de $u_x(z)$ en $x = A/2$.	1.1	1.15	1.39
Minimum de $u_x(z)$ en $x = A/2$.	-3.6	-3.5	-3.98

TAB. 2.6.1 – Quelques valeurs numériques obtenues avec $Pr=4.24$, $Ra=4300$, $Ma=7600$ et $A=12$; ces conditions correspondent à l'expérience de la figure 2.6.3.

Convection multicellulaire stationnaire

On se place désormais dans les conditions suivantes :

$$\begin{cases} h = 2.5 \text{ mm} \\ \Delta T = 1.6 \text{ K} \end{cases} \Rightarrow \begin{cases} Ra = 8600 \\ Ma = 15200 \\ A = 12 \end{cases}$$

La figure 2.6.1 résume les résultats expérimentaux comparés aux résultats numériques de Villers et Platten. Avec CASTEM2000, on a testé les trois options de discrétisation de termes de convection : CENTREE, SUPG, SUPGCC. L'option CENTREE n'a pas totalement convergé. Les courbes obtenues sont du même type que celles de la figure 2.6.1. Quelques valeurs caractéristiques de ces courbes sont données dans le tableau 2.6.2.

Vitesses en mm/s	VP Exp.	VP Num.	CENTREE	SUPG	SUPGCC
Maximum de $u_x(z)$ en $x = A/2$.	1.4	2.0	2.26	1.70	1.42
Minimum de $u_x(z)$ en $x = A/2$.	-5.31	-6.0	-7.67	-4.79	-3.45
Maximum de $u_z(x)$ en $z = 1.35 \text{ mm}$.	1.4	1.6	1.43	1.47	1.44
Minimum de $u_z(x)$ en $z = 1.35 \text{ mm}$.	-1.3	-1.3	-1.13	-1.21	-1.36
Minimum de $u_x(x)$ sur la surface libre	-7.5	-8.8	-8.08	-8.79	-8.17

TAB. 2.6.2 – Quelques valeurs numériques obtenues avec $Pr=4.24$, $Ra=8600$, $Ma=15200$ et $A=12$.

On s'aperçoit que les valeurs numériques trouvées tant par Villers et Platten que par CASTEM 2000 sont quasiment toujours supérieures à l'expérience, ce qui est conforme à la remarque déjà faite concernant le freinage de l'écoulement par les parois latérales de la cavité expérimentale. Par ailleurs, dans ce cas, il se trouve que l'option SUPGCC, la plus diffuse et la moins précise est justement celle qui permet d'obtenir des valeurs extrêmes de vitesses se rapprochant le plus des résultats expérimentaux.

Références

- 1– D. Villers and J.K. Platten. Coupled buoyancy and Marangoni convection in acetone : experiments and comparison with numerical simulations. *J. Fluid Mech.*, vol. 234, pp.487–510.




FIG. 2.6.5 – (a) Profil de vitesse horizontale $u_x(z)$ sur la médiane verticale de la cavité, (b) composante verticale de la vitesse $u_z(x)$ sur une ligne horizontale à $z = 1.35$ mm, et (c) composante horizontale de la vitesse $u_x(x)$ sur la surface. A droite, la simulation numérique de l'écoulement avec $Pr=4.24$, $Ra=8600$, $Ma=15200$ et $A=12$; ces conditions correspondent à l'expérience, à gauche (d'après Villers, D. & Platten, J.K. [1]).

–2– V.G. Levich. *Physicochemical Hydrodynamics*. Prentice Hall, 1962 (first Russian edn. 1952).

Jeu de données

```

***** OPTIONS *****
OPTI DIME 2 ELEM QUA8 ;

***** DONNEES *****
a=12. ; nu=1. ; l=a ; h=1. ; t1=0. ; t2=a ;
Pr = 4.24 ; Ra = 4300. ; Ma = 7600. ;
Gr = Ra / (A*Pr) ; Ma2 = Ma / (A*Pr) ;
alph=1. / Pr ;
*****

***** MAILLAGE *****
nh=15 ; nl=4 * nh ;
p1=0. 0. ; p2=1 0. ; p3=1 h ; p4= 0. h ;
bas = p1 D nl p2 ; rwall=p2 D nh p3 ;
haut= p3 D nl p4 ; lwall=p4 D nh p1 ;
mt=DALLER bas rwall haut lwall ;
TASSER mt ; mt= ORIE mt ;
$mt=DOMA mt 'MACRO' 'IMPR' ;
mt=($mt.maillage) ;

$haut = DOMA haut 'MACRO' 'IMPR' 'INCL' $mt 1.e-4 ;
$bas = DOMA bas 'MACRO' 'IMPR' 'INCL' $mt 1.e-4 ;
$rwall =DOMA rwall 'IMPR' 'INCL' $mt 1.e-4;
* Haut2 = Haut moins l'élément le plus proche du mur froid
haut2 = CHANGER haut SEG2;
haut3 = haut2 ELEM (ABS (2*nl));
haut2 = DIFF haut2 haut3;
$haut2= DOMA haut2 'IMPR' 'INCL' $mt 1.e-4 ;
xmt = COOR 1 mt;
grat= KCHT $bas SCAL SOMMET (xmt) ;

***** CREATION DES TABLES *****
rv = EQEX $mt 'ALFA' 0.5
* OPTI CENTREE
'ZONE' $mt 'OPER' 'NS' 'NU' 'GB' 'TN' 'TREF' 'INCO' 'UN'
'ZONE' $mt 'OPER' 'TSCA' alph 'UN' 0. 'INCO' 'TN'
'ZONE' $mt 'OPER' 'CALCMAR'
'ZONE' $haut2 'OPER' 'TOIMP' 'TSURF' 'INCO' 'UN'

rv = EQEX rv
'CLIM' 'UN' 'UIMP' (bas et lwall et rwall) 0.
'UN' 'VIMP' (bas et lwall et rwall et haut) 0.
'TN' 'TIMP' lwall t1
'TN' 'TIMP' rwall2 t2
'TN' 'TIMP' bas grat
;

rvp = EQPR $mt KTYPI 1
ZONE $mt 'OPER' 'PRESSION' 0.
PIMP 0.
;

***** CONDITIONS INITIALES *****
rv.'INCO'=TABLE 'INCO' ;

```

```

rv.'INCO'.'UN' = KCHT $mt VECT SOMMET (0. 0.) ;
rv.'INCO'.'PSI' = KCHT $mt SCAL SOMMET 0. ;
rv.'INCO'.'TN' = KCHT $mt SCAL SOMMET (t1+ ((xmt/L)*(t2-t1)) ) ;
rv.'INCO'.'TSURF'=KCHT $haut2 VECT CENTRE (-1.*Ma2 0.) ;
rv.'PRESSION'=rvp ;
nbiter=50000;
rv.'INCO'.'NU'=nu ;
rv.'INCO'.'GB' = (0. (-1.*Gr));
rv.'INCO'.'TREF'=( (t2+t1) / 2.);
rv.'FIDT'=100;
rv.'ITMA'=nbiter;

***** PROCEDURE DE CALCUL DE LA TENSION DE SURFACE *****
***** (EFFET MARANGONI) *****
DEBPROC CALCMAR ;
ARGU RX*TABLE ;
rv=rx.'EQEX' ;
tn=rv.'INCO'.'TN';
tnn=KCHT $haut2 SCAL SOMMET tn;
rv.'INCO'.'TSURF'=(KOPS tnn 'GRAD' $haut2) * Ma2 * -1.;
FINPROC;

***** CALCUL DE LA SOLUTION *****
EXEC rv ;
FIN;

```

Chapitre 3

Écoulements incompressibles turbulents

3.1 Décroissance de la turbulence homogène

3.1.1 Formulation Eléments Finis Semi-Implicite

<i>Opérateurs utilisés</i>	<i>NSKE</i>
<i>Méthodes numériques</i>	<i>EFM1, schéma SUPG</i>
<i>Nature du problème</i>	<i>Simulation de la décroissance de la turbulence homogène convectée par un champ de vitesse uniforme.</i>
<i>Classification</i>	<i>2D plan, incompressible, turbulent, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/gridturb.dgibi</i>

Description du problème :

Ce cas-test, référencé dans [1], simule la décroissance de la turbulence homogène ($k = k^0$ et $\varepsilon = \varepsilon^0$ en entrée) dans un champ de vitesse uniforme, et sur le domaine $x \in [0, 1]$.

La solution analytique correspondant à l'intégration des équations du modèle $k - \varepsilon$, est donnée par :

$$\begin{aligned}k &= k^0 \left(1 + (c_2 - 1)x \frac{\varepsilon^0}{k^0} \right)^{1/(1-c_2)} \\ \varepsilon &= \varepsilon^0 \left(1 + (c_2 - 1)x \frac{\varepsilon^0}{k^0} \right)^{c_2/(1-c_2)}\end{aligned}$$

où c_2 est une des constantes du modèle, égale à 1.92. Même si le cas est unidimensionnel, nous avons choisi de le résoudre sur un maillage bidimensionnel, de 8×80 éléments quadrangulaires. Les équations du modèle $k - \varepsilon$ sont résolues grâce à l'opérateur **NSKE** (schéma de convection-diffusion SUPG avec l'option SUPG-DC). Puisqu'on ne souhaite pas ici coupler les équations du modèle de turbulence aux équations de Navier-Stokes, mais simplement évaluer l'intégration des équations sur k et ε dans un champ de vitesse

uniforme, on impose à chaque itération et en chaque point du domaine, le champ de vitesse $\mathbf{u} = (1., 0.)$. Les valeurs de k et de ε sont quant à elles fixées en entrée à 10^{-2} . Cette valeur est également prise comme valeur initiale pour ces deux variables.

Résultats numériques :

Le maillage utilisé, ainsi que les isovaleurs de la solution (k, ε) sont représentés sur la figure 3.1.1. La solution stationnaire, obtenue après quelques 600 itérations, est convergée comme l'atteste la figure 3.1.2. La comparaison des profils numériques et analytiques sur k et ε , représentés sur les figures 3.1.3 et 3.1.4, montre un très bon accord. Les résultats obtenus avec CASTEM semblent être meilleurs que ceux obtenus avec le code NSC2KE de l'INRIA, décrits pour ce même cas-test dans [1].

Référence :

- 1- Analysis of the K-Epsilon TURBULENCE MODEL. B. Mohammadi and O. Pironneau, RAM Research in Applied Mathematics. Wiley, 1994.




FIG. 3.1.1 – Maillage 8×80 utilisé pour simuler la décroissance de la turbulence. L'entrée du domaine est constituée par le bord gauche du maillage. Isovaleurs de k et de ε pour la solution convergée.




FIG. 3.1.2 – Convergence de la solution numérique vers l'état stationnaire, représentée par le logarithme de $\|k^n - k^{n-1}\|_{L_\infty}$ et de $\|\varepsilon^n - \varepsilon^{n-1}\|_{L_\infty}$ en fonction de l'itération n .




FIG. 3.1.3 – Comparaison sur k entre solutions numérique et analytique.




FIG. 3.1.4 – Comparaison sur ε entre solutions numérique et analytique.

Jeu de données :

```
*****
* GRID TURBULENCE : convection of homogeneous turbulence *
* Analysis of the K-Epsilon TURBULENCE MODEL *
* Mohammadi/Pironneau p. 74 (Wiley) *
* H. PAILLÈRE/TTMF/AVRIL 1997 *
* J.P. MAGNAUD/LTMF/MARS 2000 *
* Teste K-Epsilon en implicite methode de projection *
*****
```

```
GRAPH = FAUX ;
*GRAPH = VRAI ;
dt=1. ;
KPRES='MSOMMET';
*DISCR='QUAF';
*DISCR='MACRO';
DISCR='LINE';
KSUPG='CENTREE';
*KSUPG='SUPG';
* KSUPG='SUPGDC' ne marche pas sur ce cas
* Si KSUPG='SUPG' et D= 4 la composante verticale de u
* ne converge pas neanmoins la solution sur K et Epsilon
* est correcte et converge
*
*      D = 4 ;
*      D = 1 ;
*      NITMA =40;
```

```
NX = D*10 ;
NY = D ;
```

```
opti dime 2 ;
opti elem qua8 ;
opti isov sul1 ;
```

```
*****
* PROCEDURE POUR ESTIMER LA CONVERGENCE VERS L'ETAT STATIONNAIRE *
*****
```

```
DEBPROC CALCUL ;
ARGU RVX*'TABLE' ;
```

```
RV = RVX.'EQEX' ;
```

```
DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/10 ;
LO = (DD-(10*NN)) EGA 0 ;
SI ( LO ) ;
```

```
EN = RV.INCO.'EN' ;
ENM1 = RV.INCO.'ENM1' ;
KN = RV.INCO.'KN' ;
KNM1 = RV.INCO.'KNM1' ;
```

```

ERRE = KOPS EN '-' ENM1 ;
ELIE = MAXI ERRE 'ABS' ;
ELIE = (LOG (ELIE + 1.0E-20))/(LOG 20.) ;
ERRK = KOPS KN '-' KNM1 ;
ELIK = MAXI ERRK 'ABS' ;
ELIK = (LOG (ELIK + 1.0E-20))/(LOG 20.) ;
MESSAGE 'ITER' RV.PASDETPS.'NUPASDT' ' ERREUR LINF ' ELIK ELIE ;
RV.INCO.'ENM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' (RV.INCO.'EN') ;
RV.INCO.'KNM1' = KCHT $DOMTOT 'SCAL' 'SOMMET' (RV.INCO.'KN') ;
IT = PROG RV.PASDETPS.'NUPASDT' ;
ERK = PROG ELIK ;
ERE = PROG ELIE ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ERK' = (RV.INCO.'ERK') ET ERK ;
RV.INCO.'ERE' = (RV.INCO.'ERE') ET ERE ;
FINSI ;
  as2 ama1 = 'KOPS' 'MATRIK' ;
  FINPROC as2 ama1 ;

*****
* GEOMETRY OF PROBLEM *
*****

A1 = 0.0 0.0 ;
A2 = 1.0 0.0 ;
A3 = 1.0 0.1 ;
A4 = 0.0 0.1 ;

A1A2 = A1 'DROI' NX A2 ;
A2A3 = A2 'DROI' NY A3 ;
A3A4 = A3 'DROI' NX A4 ;
A4A1 = A4 'DROI' NY A1 ;

DOMTOT = 'DALLER' A1A2 A2A3 A3A4 A4A1 'PLAN' ;
DOMTOT = CHAN DOMTOT QUAF ;
A4A1 = CHAN A4A1 QUAF ;
A2A3 = CHAN A2A3 QUAF ;
A1A2 = CHAN A1A2 QUAF ;
A3A4 = CHAN A3A4 QUAF ;
ELIM (DOMTOT et A4A1 et A2A3 et A3A4 et A1A2) 1.E-3 ;
MCOTE = A1A2 et A3A4;

$DOMTOT = 'MODE' DOMTOT 'NAVIER_STOKES' DISCR ;
DOMA $DOMTOT IMPR ;
$INLET = 'MODE' A4A1 'NAVIER_STOKES' DISCR ;
  INLET = 'DOMA' $INLET 'MAILLAGE' ;
$OUTLET = 'MODE' A2A3 'NAVIER_STOKES' DISCR ;
  OUTLET = 'DOMA' $OUTLET 'MAILLAGE' ;
  poutlet= 'DOMA' $OUTLET 'MSOMMET' ;
$COTE = 'MODE' MCOTE 'NAVIER_STOKES' DISCR ;
  COTE = DOMA $COTE MAILLAGE ;

DOMTOT = DOMA $DOMTOT MAILLAGE ;
CNT = CONT DOMTOT ;

*****

```

```

* PHYSICAL DATA *
*****

Uinlet = 1.71;
Kinlet = 0.013;
Einlet = 0.029;

Nu = 1.E-6 ;
tos = 0. 0. ;

*****
* ANALYTICAL SOLUTION FOR K(x) AND E(x) *
*****

C2 = 1.92 ;
X1 = COOR 1 (DOMA $DOMTOT SOMMET) ;
TINLET = KINLET/EINLET;
TAN=((C2 - 1.)*X1) + (UINLET*TINLET)/UINLET;
M=1. - C2 ;
KAN=KINLET * (( 1. - ((M*X1)/UINLET/TINLET))**(1./M));
EAN=KAN*(INVE TAN) ;

*****
* SET OF EQUATIONS *
*****

RV = EQEX $DOMTOT 'ITMA' NITMA 'OMEGA' 1. 'NITER' 1 'FIDT' 20
      'ZONE' $DOMTOT 'OPER' CALCUL
      'OPTI' KSUPG 'IMPL' EF
      'ZONE' $DOMTOT 'OPER' 'NS'          'NUT' 'INCO' 'UN'
      'ZONE' $DOMTOT 'OPER' 'KEPSILON' NU 'UN' 'DT' 'INCO' 'KN' 'EN'
      'ZONE' $DOMTOT 'OPER' 'DFDT' 1. 'UN' 'DT' 'UN' 'NUT' 'INCO' 'UN'
      ;

RV = EQEX RV
      'CLIM' 'UN' 'UIMP' INLET Uinlet
      'UN' 'VIMP' INLET 0.
      'UN' 'VIMP' COTE 0.
      'CLIM' 'KN' 'TIMP' INLET Kinlet
      'EN' 'TIMP' INLET Einlet
      ;

RVP = EQEX 'OPTI' 'EF' KPRES
      ZONE $DOMTOT 'OPER' KBBT (-1.)      INCO 'UN' 'PRES'
      CLIM 'PRES' TIMP poutlet 0.
      ;

rvp.'METHINV' . 'FCPRECT'=300 ;

RV.INCO = TABLE INCO ;
RV.INCO.'UN' = KCHT $DOMTOT VECT SOMMET (uinlet 0.) ;
RV.INCO.'KN' = KCHT $DOMTOT SCAL SOMMET Kinlet;
RV.INCO.'EN' = KCHT $DOMTOT SCAL SOMMET 1.e-3 ;

```

```

RV.INCO.'ENM1' = KCHT $DOMTOT SCAL SOMMET Einlet ;
RV.INCO.'KNM1' = KCHT $DOMTOT SCAL SOMMET Kinlet ;
RV.INCO.'NUT' = KCHT $DOMTOT SCAL CENTRE 1.E-10 ;
RV.INCO.'PRES' = KCHT $DOMTOT SCAL KPRES 0. ;
RV.INCO.'DT' = dt;
RV.INCO.'IT' = PROG 1 ;
RV.INCO.'ERK' = PROG 0.;
RV.INCO.'ERE' = PROG 0.;
RV.INCO.'EAN'=EAN;
RV.INCO.'KAN'=KAN;

RV.'PROJ' = RVP ;

EXEC RV ;

SI ( GRAPH ) ;
TRACE DOMTOT ;
EVOLK = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
      (RV.INCO.'ERK') ;
EVOLE = EVOL 'MANU' 'ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
      (RV.INCO.'ERE') ;
TAB1 = TABLE ;
TAB1.1 = 'MARQ CROI ' ;
TAB1.2 = 'MARQ TRIA ' ;
TAB1.'TITRE' = TABLE ;
TAB1.'TITRE'. 1 = MOT 'CONVERGENCE_K' ;
TAB1.'TITRE'. 2 = MOT 'CONVERGENCE_E' ;
DESS (EVOLK ET EVOLE) 'XBOR' 0. (2.*NITMA) 'YBOR' -15.0 0.0
      LEGE TAB1 ;

TRACE RV.INCO.'KN' DOMTOT CNT 'TITR' 'KN' 20 ;
TRACE RV.INCO.'EN' DOMTOT CNT 'TITR' 'EN' 20 ;

EVOLKN = EVOL (ROUG) 'CHPO' RV.INCO.'KN' 'SCAL' A1A2 ;
EVOLKA = EVOL (BLEU) 'CHPO' KAN 'SCAL' A1A2 ;
EVOLEN = EVOL (ROUG) 'CHPO' RV.INCO.'EN' 'SCAL' A1A2 ;
EVOLEA = EVOL (BLEU) 'CHPO' EAN 'SCAL' A1A2 ;

TAB1 = TABLE ;
TAB1.1 = 'MARQ CROI ' ;
TAB1.2 = ' ' ;
TAB1.'TITRE' = TABLE ;
TAB1.'TITRE'. 1 = MOT 'NUMERIQUE' ;
TAB1.'TITRE'. 2 = 'MOT' 'ANALYTIQUE' ;

DESS (EVOLKN ET EVOLKA) LEGE TAB1 'TITR' 'Grid Turbulence K' ;
DESS (EVOLEN ET EVOLEA) LEGE TAB1 'TITR' 'Grid Turbulence E' ;
FINSI ;

ERRK = KOPS RV.INCO.'KN' '-' KAN ;
ERRK = KOPS ERRK '*' ERRK ;
ERR2 = 0. ;
REPETER BLOC1 (NBNO (DOMA $DOMTOT MAILLAGE)) ;
P1 = (doma $DOMTOT SOMMET) POIN &BLOC1 ;
ERR1 = 'EXTR' ERRK 'SCAL' P1 ;
ERR2 = ERR2 + ERR1 ;

```

```
FIN BLOC1 ;
ERR2 = ERR2/(nbno (doma $DOMTOT MAILLAGE)) ;
ERR2 = ERR2 ** 0.5 ;
ERR2K = ERR2/Kinlet ;
MESSAGE 'ERREUR SUR K EN NORME L2 = ' ERR2K ;

ERRE = KOPS RV.INCO.'EN' '-' EAN ;
ERRE = KOPS ERRE '*' ERRE ;
ERR2 = 0. ;
REPETER BLOC1 (NBNO (DOMA $DOMTOT MAILLAGE)) ;
P1 = (DOMA $DOMTOT SOMMET) POIN &BLOC1 ;
ERR1 = 'EXTR' ERRE 'SCAL' P1 ;
ERR2 = ERR2 + ERR1 ;
FIN BLOC1 ;
ERR2 = ERR2/(NBNO (DOMA $DOMTOT MAILLAGE)) ;
ERR2 = ERR2 ** 0.5 ;
ERR2E = ERR2/Einlet ;
MESSAGE 'ERREUR SUR E EN NORME L2 = ' ERR2E ;

* TESTS D'ERREUR *

SI ( (MINI RV.INCO.'ERE') > -5.0 ) ;
    ERREUR 5 ;
FINSI ;
SI ( (MINI RV.INCO.'ERK') > -5.0 ) ;
    ERREUR 5 ;
FINSI ;

SI ( ERR2K > 1.5D-2 ) ;
    ERREUR 5 ;
FINSI ;

SI ( ERR2E > 3.0D-2 ) ;
    ERREUR 5 ;
FINSI ;

FIN ;
```

3.2 Écoulement dans une conduite cylindrique

3.2.1 Formulation Eléments Finis Semi-Implicite

<i>opérateurs utilisés</i>	<i>NSKE, FPU, PRESSION</i>
<i>Méthodes numériques</i>	<i>EFMI, approximation Q1-P0 ou P1-P0, schéma SUPG</i>
<i>Modèles physiques</i>	<i>équations de Navier-Stokes en incompressible avec modèle de turbulence K-ϵ et fonction de paroi standard</i>
<i>Nature du problème</i>	<i>Écoulement turbulent d'un fluide incompressible newtonien dans une conduite lisse cylindrique</i>
<i>classification</i>	<i>2D axisymétrique, incompressible, turbulent, stationnaire</i>
<i>Jeu de donnée</i>	<i>/u2/castem/dgibi/tubturb.dgibi</i>

Modélisation :

On considère l'écoulement d'un fluide incompressible newtonien dans une conduite lisse cylindrique. Les simulations numériques sont menées pour des régimes de convection forcée à fort nombre de Reynolds. 4 valeurs du Reynolds sont utilisées : 5.10^4 ; 10^5 ; 5.10^5 ; 10^6 .

Domaine fluide :

L'écoulement est décrit par les équations de Navier-Stokes en incompressible. Les contraintes de Reynolds sont représentées par un modèle de viscosité turbulente. La turbulence est décrite par les équations du modèle K- ϵ standard.

Les flux des variables de la turbulence et les contraintes du fluides sont discrétisés par l'opérateur **NSKE** . L'option de décentrement 'SUPG' est choisie pour ces propriétés de non diffusivité numérique transversale. Les forces de pression sont prises en compte par l'opérateur **PRESSION** . L'algorithme de résolution est géré par la procédure **EXEC** .

La représentation spatiale du domaine fluide est à 2 dimensions en axisymétrique. Le domaine est limité transversalement par l'axe et une ligne fluide située à la distance de la paroi de 0,05 fois le rayon de la conduite.

Les calculs sont menés avec trois maillages :

- le maillage 'Q24' comprend 24 quadrangles suivant le rayon ;
- le maillage 'Q48' comprend 48 quadrangles suivant le rayon ;
- le maillage 'T24' comprend 24 triangles suivant le rayon ;

Les mailles ont un rapport d'allongement de 6.

Conditions aux limites :

On impose une condition de symétrie sur l'axe. Vitesse, énergie cinétique turbulente k et taux de dissipation ϵ sont imposés en entrée. Les interactions avec la paroi sont gérées par des fonctions de paroi standard. Elles sont discrétisées par l'opérateur **FPU**. Les conditions aux limites en sortie sont naturelles (au sens formulation élément fini).

Mode opérationnel :

Le calcul est effectué sur des tronçons successifs de conduite pour des raisons d'économie de temps calcul. L'entrée et la sortie sont distantes de 3 diamètres. Les variables sont initialisées et imposées en entrée aux valeurs expérimentales. On effectue une simulation de l'écoulement en transitoire sur un premier tronçon de conduite. Après obtention du permanent, on recherche le régime permanent sur le tronçon suivant. (le sens de l'écoulement est inversé et les entrée sortie permutées pour utiliser le même maillage). L'opération est renouvelée jusqu'à l'obtention d'un résultat invariant selon l'axe à une erreur relative de 2.10^{-5} près.

Résultats :

Perte de charge :

Les coefficients de perte de charge Λ obtenus par simulation sont comparés aux valeurs de la loi universelle de frottement [1] :

$$\frac{1}{\sqrt{\Lambda}} = 2 \cdot \log(Re\sqrt{\Lambda}) - 0,8$$

Les coefficients Λ issus des calculs sont déduits de la perte de charge :

$$\Lambda = 2 \cdot \frac{\Delta P}{\rho L} \cdot \frac{2 \cdot D_h}{\bar{U}^2}$$

D_h diamètre hydraulique ; ΔP perte de charge sur la longueur L ; \bar{U} vitesse débitante.

vitesse

Les profils de vitesses sont comparés aux profils proposés par la théorie des couches limites (figures (e)) :

$$\frac{U}{u_*} = \frac{1}{\kappa} \text{Log}(y_+) + 5,5$$

$y_+ = \frac{y\nu}{u_*}$ distance adimensionnelle à la paroi ; y distance réelle à la paroi ; $\kappa = 0,41$.

A noter que cette comparaison n'est pas valide à proximité de l'axe. On observe une bonne cohérence des résultats.




FIG. 3.2.1 – Coefficient de perte de charge en fonction du Reynolds ;

turbulence

On constate que, sur la gamme de Reynolds testés, les variables descriptives de la turbulence (k , ϵ , ν_t et lm) sont en similitudes car les profils radiaux des variables adimensionnées obtenus pour les différents régimes hydrauliques sont confondus.

Les profils radiaux de viscosité turbulente ν_t et de longueur de mélange sont comparés à des formulations analytiques des profils observés par Nikuradse [2] dans la gamme de Reynolds ($10^5 < \text{Reynolds} < 3.10^6$) :

$$\frac{\nu_t}{u_* \cdot R} = 0,08 \cdot \chi_1 + 0,015 \cdot \chi_2$$

$$\frac{lm}{R} = 0,14 - 0,08 \cdot \left(\frac{r}{R}\right)^2 - 0,06 \cdot \left(\frac{r}{R}\right)^4$$


y distance réelle à la paroi ; r abscisse radial ; R rayon du tube ; $\chi_1 = \kappa \cdot \frac{r \cdot y}{R^2}$; $\chi_2 = \left(\frac{y}{R}\right)^4$;

Ces comparaisons montrent que, pour les forts Reynolds, les ordres de grandeur de la viscosité turbulente et de la longueur de mélange sont respectés (figures (c) et (d) des planches 3.2.2, 3.2.3 et 3.2.4). Elles montrent une différence qualitative des résultats avec ceux obtenus par Nikuradse (1932) [2] concernant la viscosité turbulente à proximité de l'axe. Les nombreux résultats numériques et expérimentaux [3] obtenues depuis les expériences de Nikuradse n'ont pas confirmé les profils de viscosité turbulente obtenus par ce dernier. Ces contradictions peuvent s'expliquer par la difficulté d'obtenir expérimentalement une description précise de ce paramètre au voisinage de l'axe. Celui-ci se déduit en effet du profil radial des forces de cisaillement (profil linéaire) et de la pente du profil radial de la vitesse locale moyenne. Cette pente, faible au voisinage de l'axe est susceptible d'être affectée par le niveau de précision des mesures de la vitesse locale moyenne. Les profils de viscosité turbulente obtenus par CASTEM sont en revanche proches de ces derniers résultats [3].


Il faut enfin remarquer que la similitude des résultats pour les bas Reynolds n'existe pas en réalité [3]. Une description correcte de l'écoulement à bas Reynolds nécessiterait l'utilisation d'un modèle de turbulence adéquat dit "bas Reynolds".

Références


- 1- Mécanique expérimentale des fluides - Tome II 3^{ième} édition - R.COMOLET - Édition MASSON (1982)
- 2- Boundary Layer Theory ; Dr H. SCHLICHTING - Mc GRAW-HILL BOOK CO. INC
- 3- Comparison of low Reynolds number k- ϵ turbulent models in predicting fully developed pipe flow - C.M.HRENYA, E.J.BOLIO, D.CHAKRABARTI, J.L.SINCLAIR - Vol.50, No. 12 pp. 1923-1941, 1995




(a) vitesse




(b) viscosité turbulente



(c) énergie turbulente




(d) taux de dissipation




(e) longueur de mélange


FIG. 3.2.2 – Profils radiaux des variables addimensionnées pour le maillage Q24




(a) vitesse




(b) viscosité turbulente



(c) énergie turbulente




(d) taux de dissipation




(e) longueur de mélange


FIG. 3.2.3 – Profils radiaux des variables addimensionnées pour le maillage T24




(a) vitesse




(b) viscosité turbulente



(c) énergie turbulente



(d) taux de dissipation



(e) longueur de mélange

FIG. 3.2.4 – Profils radiaux des variables addimensionnées pour le maillage Q48

Jeu de données :

```
COMPLET = FAUX ;
GRAPH = 'N' ;
```

```
SI ( COMPLET ) ;
  along = 3 ; n = 24 ; nlong = 6 ;
  SINON ;
  along = 3 ; n = 24 ; nlong = 15 ;
  FINSI ;
```

```
*****
* TESTE DE STABILITÉ DU MODELE K-E ET DES FONCTIONS DE PAROI EN *
* CONVECTION FORCE - ÉCOULEMENT EN CONDUITE LISSE CYLINDRIQUE *
* LES PROFILS RADIAUX DES VARIABLES V,K,E OBTENUS À 3 DIAMÈTRES *
* DE L'ENTRÉE SONT COMPARÉS AU PROFIL OBTENU LORS DE L'ÉTABLISSE *
* DE LA FICHE DE VALIDATION CORRESPONDANTE *
*****
```

```
'OPTI' 'DIME' 2 'ELEM' 'QUA4' 'MODE' 'AXIS' ;
'OPTI' 'ISOV' 'SURF' ;
```

```
** dimension
Rh = 0.05 ; DH = 2. * RH ; LH = along*DH ;
```

```
** maillage
A = 0.0 0. ; B = RH 0. ; C = RH LH ; D = 0. LH ;
bas = a DROIT n b ;
paroi = b DROIT (2*n*along/nlong) c ;
haut = c droit n d ;
axe = d droit (2*n*along/nlong) a ;
mgeo = dalle bas paroi haut axe plan ;
mgeo = orien mgeo ;
cgeo = contour mgeo ;
```

```
*-----
**** donnée hydraulique de l écoulement
NU = 6.53e-4 / 992.2D0 ;
CNU = 0.09 ;
AKER = 0.41 ;
REY = 100000. ;
UI = REY*NU/DH ;
```

```
*-----
**** calcul de la perte de charge

LAM = LOG REY / (LOG 10.) * 1.8 - 1.64 ** (-2.) ;
mess 'Lambda = ' LAM ;
REPE BL1 5 ;
  lam = lam ** 0.5 * REY log / (log 10) * 2. - 0.8 ** (-2.) ;
  mess 'Lambda = ' LAM ;
FIN bl1 ;

DPSRDZ = LAM*0.5/DH*UI*UI ;
```

```

*-----
**** calcul de la vitesse de glissement à partir des pertes de charge
T00      = DPSRDZ * Rh * 0.5 ;
T00_IMP  = (-1.*t00) 0. ;
UETH     = T00 ** 0.5 ;

*-----
**** restriction du domaine à la zone logarithmique (Y>40)
R = MGEO COOR 1 ;
Y = RH - R ;
YCL= RH * 0.05 ;
DEPLA MGEO AFFI (RH - YCL/RH) A B ;
R = MGEO COOR 1 ;
Y = RH - R ;

*-----
**** construction des objets modeles
Mmgeo = chan mgeo QUAF ;
Mbas  = chan bas  QUAF ;
ELIM Mmgeo Mbas 0.0001 ;
DOMT  = 'MODE' Mmgeo 'NAVIER_STOKES' LINE ;
DBAS  = 'MODE' Mbas  'NAVIER_STOKES' LINE ;
DOMA  DOMT 'IMPR' ;
VTBAS = PI * RH * RH ;
Mhaut = chan haut  QUAF ;
nbp   = nbel paroi ;
paroi = elem paroi (lect 2 pas 1 nbp) ;
Mparoi = chan paroi QUAF ;
ELIM (Mmgeo et Mbas et Mhaut et Mparoi) 0.0001 ;
DHAU  = 'MODE' Mhaut 'NAVIER_STOKES' LINE ;
DPAU  = 'MODE' Mparoi 'NAVIER_STOKES' LINE ;
YPLUS = Y * UETH / NU ;

*-----
**** estimation du profil de vitesse (hypothèse de couche limite log.)
VZ = YPLUS LOG / AKER + 5.5 * UETH ;
VZ = NOMC VZ 'UY' ;
VZ1 = KCHT DOMT VECT SOMMET VZ ;
DZ = DBIT DBAS (VZ1 + (VZ1 mini * PI * YCL* RH)) ;
UZ = DZ / VTBAS ;
**** correction du profil de vitesse pour le respect du REYNOLDS
MESS 'FACTEUR DE CORRECTION DES VITESSES ' (UI/UZ) ;
VZ = KCHT DOMT VECT SOMMET (VZ1/UZ*UI) ;

*-----
**** estimation de la viscosité turbulente
NUT = AKER * Y * UETH * (R / RH) + (Y/RH*nu) ;
NUTET = NUT/UETH/RH*0.75+(Y/RH*0.015) ;
NI = kcht domt scal sommet (NUTET*UETH*RH) ;

*-----
**** calcul de la longueur de mélange
R2 = R/RH ** 2 ;
LP = R2 * 0.06 + 0.08 * R2 * (-1.) + 0.14 ;
LP = LP * RH ;

```



```

*-----
**** estimation de l'énergie cinétique turbulente à partir
**** de la longueur de mélange
KI = NI / LP ** 2. / (CNU**0.5) ;

*-----
**** estimation de la dissipation visqueuse
EI = CNU * KI * KI / NI ;

*-----
**** conditions en entrée
VEN = exco (REDU VZ BAS) 'UY' '2UN' ;
KEN = exco (REDU KI BAS) 'SCAL' 'KN' ;
EEN = exco (REDU EI BAS) 'SCAL' 'EN' ;

MT = table ;
MT.1 = 'MARQ TRIA REGU TITR SORTIE' ;
MT.2 = 'TIRR TITR ENTREE' ;
HAUT = INVE HAUT ;
ZC = MGEO COOR 2 ; ZC = (MAXI ZC) - ZC ;

RV = EQEX DOMT 'ITMA' 160 'ALFA' 0.9
'OPTI' 'SUPG'
'ZONE' DOMT 'OPER' 'FILTREKE' UI Dh 'INCO' 'KN' 'EN'
'ZONE' DOMT 'OPER' 'NSKE' NU 'NUT' 'INCO' 'UN' 'KN' 'EN'
'ZONE' DPAR 'OPER' 'FPU' NU 'UET' YCL 'INCO' 'UN' 'KN' 'EN' ;

RV = EQEX RV 'OPTI' 'CENTREE'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'UN' 'DELTAT' INCO 'UN'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'KN' 'DELTAT' INCO 'KN'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'EN' 'DELTAT' INCO 'EN'
;

RV = EQEX RV
'CLIM' 'UN' 'VIMP' BAS VEN
'CLIM' 'KN' 'TIMP' BAS KEN
'CLIM' 'EN' 'TIMP' BAS EEN
'CLIM' 'UN' 'UIMP' CGEO 0. ;

RVP = 'EQPR' DOMT 'ZONE' DOMT 'OPER' 'PRESSION' ;
RV.'PRESSION' = table RVP ;

RV.'INCO' = table 'INCO' ;
RV.INCO.'UN' = VZ ;
RV.INCO.'NUT' = NOEL DOMT NI ;
RV.INCO.'KN' = KCHT DOMT SCAL SOMMET KI ;
RV.INCO.'EN' = KCHT DOMT SCAL SOMMET EI ;
RV.'INCO'.'UET' = KCHT DPAR SCAL CENTRE UETH ;

*-----EXEC
EXEC RV ;
*-----EXEC

SI ( COMPLET ) ;
** On recherche le régime permanent :

```

```

repe bl2 15 ;

K = RV.'INCO'.'KN' ;
E = RV.'INCO'.'EN' ;
N = RV.'INCO'.'NUT' ;
U = kops (RV.'INCO'.'UN') '*' (-1.) ;
UT = RV.'INCO'.'UET' ;

RV = EQEX DOMT 'ITMA' 400 'ALFA' 0.8
'OPTI' 'SUPG'
'ZONE' DOMT 'OPER' 'FILTREKE' UI Dh 'INCO' 'KN' 'EN'
'ZONE' DOMT 'OPER' 'NSKE' NU 'NUT' 'INCO' 'UN' 'KN' 'EN'
'ZONE' DPAR 'OPER' 'FPU' NU 'UET' YCL 'INCO' 'UN' 'KN' 'EN' ;

RV = EQEX RV 'OPTI' 'CENTREE'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'UN' 'DELTAT' INCO 'UN'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'KN' 'DELTAT' INCO 'KN'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'EN' 'DELTAT' INCO 'EN'
;

RVP = 'EQPR' DOMT 'ZONE' DOMT 'OPER' 'PRESSION' ;
RV.'PRESSION' = table RVP ;

RV.'INCO' = table 'INCO' ;
RV.INCO.'UN' = U ;
RV.INCO.'NUT' = N ;
RV.INCO.'KN' = K ;
RV.INCO.'EN' = E ;
RV.'INCO'.'UET' = UT ;

VEN = exco (redu HAUT U) 'UY' 'SCAL' ;
KEN = redu HAUT K ;
EEN = redu HAUT E ;

RV.CLIM = 0 ;
RV = EQEX RV
'CLIM' 'UN' 'VIMP' HAUT VEN
'CLIM' 'KN' 'TIMP' HAUT KEN
'CLIM' 'EN' 'TIMP' HAUT EEN
'CLIM' 'UN' 'UIMP' CGEO 0. ;

*-----EXEC aller
EXEC RV ;
*-----EXEC aller

K2 = copi RV.'INCO'.'KN' ;
K = RV.'INCO'.'KN' ;
E = RV.'INCO'.'EN' ;
N = RV.'INCO'.'NUT' ;
U = kops (RV.'INCO'.'UN') '*' (-1.) ;
UT = RV.'INCO'.'UET' ;

RV = EQEX DOMT 'ITMA' 400 'ALFA' 0.8
'OPTI' 'SUPG'
'ZONE' DOMT 'OPER' 'FILTREKE' UI Dh 'INCO' 'KN' 'EN'

```

```

'ZONE' DOMT 'OPER' 'NSKE' NU 'NUT' 'INCO' 'UN' 'KN' 'EN'
'ZONE' DPAR 'OPER' 'FPU' NU 'UET' YCL 'INCO' 'UN' 'KN' 'EN' ;

RV = EQEX RV 'OPTI' 'CENTREE'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'UN' 'DELTAT' INCO 'UN'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'KN' 'DELTAT' INCO 'KN'
'ZONE' DOMT 'OPER' 'DFDT' 1. 'EN' 'DELTAT' INCO 'EN'
;

RVP = 'EQPR' DOMT 'ZONE' DOMT 'OPER' 'PRESSION' ;
RV.'PRESSION' = table RVP ;

RV.'INCO' = table 'INCO' ;
RV.INCO.'UN' = U ;
RV.INCO.'NUT' = N ;
RV.INCO.'KN' = K ;
RV.INCO.'EN' = E ;
RV.'INCO'.'UET' = UT ;

VEN          = exco (redu BAS U) 'UY' 'SCAL' ;
KEN          = redu BAS K ;
EEN          = redu BAS E ;

RV.CLIM = 0 ;
RV = EQEX RV
'CLIM' 'UN' 'VIMP' BAS VEN
'CLIM' 'KN' 'TIMP' BAS KEN
'CLIM' 'EN' 'TIMP' BAS EEN
'CLIM' 'UN' 'UIMP' CGEO 0. ;

*-----EXEC retour
EXEC RV ;
*-----EXEC retour

erm = k - k2 / (k maxi) abs maxi ;
mess 'maxi (k - k2 / (k maxi)) : ' erm ;
si ( erm < 2.e-5) ;
  quitter bl2 ;
fini ;
menage ;
FIN bl2 ;

*****
*          *
*   POSTRAITEMENT DES RÉSULTATS   *
*          *
*****

SI ( 'EGA' graph '0' ) ;

Evv = EVOL CHPO (RV.INCO.UN/UETH) 'UY' (DOMA DBAS MAILLAGE) ;
EVVO = EVOL VERT CHPO (VZ/UETH) 'UY' (DOMA DBAS MAILLAGE) ;
RAY = EXTR evv 'ABSC' ; RAY = RAY / RH ;
YY = EXTR evv 'ORDO' ; YYO = EXTR evvO 'ORDO' ;
Evv = EVOL MANU RAY YY 'UY' ;
EVVO = EVOL MANU RAY YYO 'UY' ;

```

```

RV.INCO.'NNT' = ELNO DOMT RV.INCO.'NUT' ;
EVN = EVOL CHPO (RV.INCO.'NNT'/(UETH*RH)) 'SCAL' (DOMA DBAS MAILLAGE);
EVNO = EVOL VERT CHPO NUTET 'SCAL' (DOMA DBAS MAILLAGE) ;
YY = EXTR evN 'ORDO' ; YYO = EXTR evNO 'ORDO' ;
EVN = EVOL MANU RAY YY 'SCAL' ;
EVNO = EVOL MANU RAY YYO 'SCAL' ;

EVK = EVOL CHPO (RV.INCO.'KN'/(UETH*UETH)) 'SCAL' (DOMA DBAS MAILLAGE);
EVKO = EVOL VERT CHPO (KI/UETH/UETH) 'SCAL' (DOMA DBAS MAILLAGE) ;
YY = EXTR evK 'ORDO' ; YYO = EXTR evKO 'ORDO' ;
EVK = EVOL MANU RAY YY 'SCAL' ;
EVKO = EVOL MANU RAY YYO 'SCAL' ;

EVE = EVOL CHPO (RV.INCO.'EN'/(UETH*UETH*UETH/RH))
'SCAL' (DOMA DBAS MAILLAGE) ;
EVEO = EVOL VERT CHPO (EI/(UETH*UETH*UETH/RH))
'SCAL' (DOMA DBAS MAILLAGE) ;
YY = EXTR evE 'ORDO' ; YYO = EXTR evEO 'ORDO' ;
EVE = EVOL MANU RAY YY 'SCAL' ;
EVEO = EVOL MANU RAY YYO 'SCAL' ;

RV.INCO.'LN' = KOPS (0.09**(-0.25)) '*'
(KOPS RV.INCO.'NNT' '/' (KOPS RV.INCO.'KN' '**' 0.5)) ;
EVL = EVOL CHPO ((RV.INCO.'LN')/RH) 'SCAL' (DOMA DBAS MAILLAGE) ;
EVLO = EVOL VERT CHPO (LP/RH) 'SCAL' (DOMA DBAS MAILLAGE) ;
YY = EXTR evL 'ORDO' ; YYO = EXTR evLO 'ORDO' ;
EVL = EVOL MANU RAY YY 'SCAL' ;
EVLO = EVOL MANU RAY YYO 'SCAL' ;

nn = nbel dbas.maillage ;
EVTO = EVOL MANU RAY RAY 'SCAL' ;
nn = nbel dPAR.maillage ;
X = (dbas.maillage COOR 1) MAXI / RH ;
TOC = SOMT RV.INCO.'UET' / nn / UETH ** 2 ;
EVT = EVOL MANU (PROG X) (PROG TOC) 'SCAL' ;

MT = TABLE ;
MT.1 = 'MARQ CROI REGU TIRR' ;
MT.2 = 'MARQ CARR ' ;

DESS MT (EVTO ET EVT)
titr 'PROFIL RADIAL DU CISAILLEMENT ADIMENSIONNELLE'
TITX 'RAYON/RH' TITY 'TO/UETH/UETH'

DESS (EVN ET EVNO)
titr 'PROFIL RADIAL DE LA VISCOSITE ADIMENSIONNELLE'
TITX 'RAYON/RH' TITY 'NUT/UETH/RH';

DESS (EVE ET EVEO)
titr 'PROFIL RADIAL DU TAUX DE DISS. ADIMENSIONNEL'
TITX 'RAYON/RH' TITY 'E/UETH*RH';

DESS (EVV ET EVVO)
titr 'PROFIL RADIAL DE LA VITESSE ADIMENSIONNELLE'
TITX 'RAYON/RH' TITY 'V/UETH';

```

```

DESS (EVK ET EVKO) titr
'PROFIL RADIAL DE L ENERGIE TURBULENTE ADIMENSIONNELLE'
TITX 'RAYON/RH' TITY 'K/UETH/UETH';

DESS (EVL ET EVLO) titr
'PROFIL RADIAL DE LA LONGUEUR DE MELANGE ADIMENSIONNELLE'
TITX 'RAYON/RH' TITY 'L/RH';

FINSI ;

SINON ;
EVV = EVOL CHPO (RV.INCO.UN/UETH) 'UY' (DOMA DBAS MAILLAGE) ;
v = EXTR evv 'ORDO' ;
list v ;
v0 = prog
26.642 26.535 26.423 26.306 26.184 26.056 25.921 25.780 25.630
25.472 25.303 25.123 24.930 24.721 24.495 24.248 23.976 23.672
23.330 22.936 22.474 21.914 21.204 20.231 18.682 ;
VDIFO = V - V0 / V0 abs maxi ; list VDIFO ;
SI ( VDIFO > 5.E-5 ) ;
ERREUR 5 ;

FINSI ;

EVK = EVOL CHPO (RV.INCO.'KN'/(UETH*UETH)) 'SCAL' (DOMA DBAS MAILLAGE);
K = EXTR evK 'ORDO' ;
list k ;
KO = PROG
3.98961E-02 .11874 .22847 .36068 .50826 .66518 .82637 .98756 1.1453
1.2966 1.4394 1.5718 1.6926 1.8010 1.8965 1.9789 2.0483 2.1050
2.1495 2.1822 2.2041 2.2158 2.2181 2.2120 2.1982 ;
KDIFO = K - KO / KO ABS MAXI ; list KDIFO ;
SI (KDIFO > 5.E-5 ) ;
ERREUR 5 ;

FINSI ;

EVE = EVOL CHPO (RV.INCO.'EN'/(UETH*UETH*UETH/RH))
'SCAL' (DOMA DBAS MAILLAGE) ;
E = EXTR eve 'ORDO' ;
list e ;
EO = PROG
9.35296E-03 4.80687E-02 .12863 .25632 .43158 .65176 .91234 1.2079
1.5329 1.8824 2.2521 2.6394 3.0434 3.4653 3.9095 4.3841 4.9025
5.4863 6.1700 7.0111 8.1107 9.6651 12.110 16.648 28.290 ;
EDIFO = E - EO / EO ABS MAXI ; list EDIFO ;
SI ( EDIFO > 5.E-5 ) ;
ERREUR 5 ;

FINSI ;

FINSI ;

FIN;

```


3.3 Convection naturelle turbulente dans une cavité carrée

3.3.1 Formulation Eléments Finis Semi-Implicite

<i>Opérateurs utilisés</i>	<i>NSKE, PRESSION, TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation Q1-P0, schéma SUPG</i>
<i>Nature du problème</i>	<i>Écoulement de convection naturelle turbulente (modèle $k-\varepsilon$) dans une cavité carrée et comparaison avec autres solutions numériques</i>
<i>Classification</i>	<i>2D plan, incompressible, Boussinesq, turbulent, stationnaire</i>
<i>Jeu de données</i>	

Schéma du problème :

On s'intéresse dans cet exemple à l'écoulement de convection naturelle dans une cavité carrée en régime turbulent pour un nombre de Rayleigh donné, égal à 5×10^{10} . Ce cas-test a fait l'objet d'un benchmark [1]. Le schéma du problème est identique aux cas-tests de Vahl-Davis (écoulements laminaires à des nombres de Rayleigh compris entre 10^3 et 10^6) décrits précédemment.




FIG. 3.3.1 – Définition du problème de la cavité carrée.

Modélisation :

L'écoulement faiblement turbulent à $Ra = 5 \times 10^{10}$ est modélisé grâce à l'opérateur NSKE, qui discrétise les équations de Navier-Stokes moyennées, couplées au modèle

de turbulence $k - \varepsilon$. Les effets de flottabilité sont pris en compte dans le cadre de l'hypothèse de Boussinesq (voir le cas laminaire). L'équation de transport sur la température moyenne est discrétisée par l'opérateur TSCAL.

Le système d'équations résolues est donc le suivant :

$$\begin{aligned} \frac{\partial \bar{u}}{\partial x} + \frac{\partial \bar{v}}{\partial y} &= 0 \\ \frac{\partial \bar{u}}{\partial t} + \bar{u} \frac{\partial \bar{u}}{\partial x} + \bar{v} \frac{\partial \bar{u}}{\partial y} &= -\frac{1}{\rho_o} \frac{\partial \bar{p}}{\partial x} + (\nu + \nu_t) \left(\frac{\partial^2 \bar{u}}{\partial x^2} + \frac{\partial^2 \bar{u}}{\partial y^2} \right) \\ \frac{\partial \bar{v}}{\partial t} + \bar{u} \frac{\partial \bar{v}}{\partial x} + \bar{v} \frac{\partial \bar{v}}{\partial y} &= -\frac{1}{\rho_o} \frac{\partial \bar{p}}{\partial y} + g\beta(\bar{T} - T_o) + (\nu + \nu_t) \left(\frac{\partial^2 \bar{v}}{\partial x^2} + \frac{\partial^2 \bar{v}}{\partial y^2} \right) \\ \frac{\partial \bar{T}}{\partial t} + \bar{u} \frac{\partial \bar{T}}{\partial x} + \bar{v} \frac{\partial \bar{T}}{\partial y} &= \left(\frac{\nu}{Pr} + \frac{\nu_t}{Pr_t} \right) \left(\frac{\partial^2 \bar{T}}{\partial x^2} + \frac{\partial^2 \bar{T}}{\partial y^2} \right) \\ \frac{\partial k}{\partial t} + \bar{u} \frac{\partial k}{\partial x} + \bar{v} \frac{\partial k}{\partial y} &= \left(\nu + \frac{\nu_t}{\sigma_k} \right) \left(\frac{\partial^2 k}{\partial x^2} + \frac{\partial^2 k}{\partial y^2} \right) + P_k + G_k - \varepsilon \\ \frac{\partial \varepsilon}{\partial t} + \bar{u} \frac{\partial \varepsilon}{\partial x} + \bar{v} \frac{\partial \varepsilon}{\partial y} &= \left(\nu + \frac{\nu_t}{\sigma_\varepsilon} \right) \left(\frac{\partial^2 \varepsilon}{\partial x^2} + \frac{\partial^2 \varepsilon}{\partial y^2} \right) + (C_{\varepsilon,1}(P_k + C_{\varepsilon,3}G_k) - C_{\varepsilon,2}\varepsilon) \frac{\varepsilon}{k} \end{aligned}$$

où les termes sources dans les équations de k et de ε sont donnés par :

$$\begin{aligned} P_k &= \nu_t \left[2 \left(\frac{\partial \bar{u}}{\partial x} \right)^2 + 2 \left(\frac{\partial \bar{v}}{\partial y} \right)^2 + \left(\frac{\partial \bar{u}}{\partial y} + \frac{\partial \bar{v}}{\partial x} \right)^2 \right] \\ G_k &= -\frac{\nu_t}{Pr_t} g\beta \frac{\partial \bar{T}}{\partial y} \end{aligned}$$

La viscosité turbulente ν_t est donnée par :

$$\nu_t = C_\mu \frac{k^2}{\varepsilon}$$

et les constantes ont les valeurs suivantes :

$$Pr = 0.71, \quad Pr_t = 0.9, \quad C_\mu = 0.09, \quad C_{\varepsilon,1} = 1.44, \quad C_{\varepsilon,2} = 1.92, \quad C_{\varepsilon,3} = 0.8, \quad \sigma_k = 1.0, \quad \sigma_\varepsilon = 1.3$$

Le système est complété par les conditions aux limites suivantes :

$$\begin{aligned} \bar{u} = \bar{v} = 0 & \quad \text{sur les parois} \\ \bar{T} = T_c & \quad \text{sur la paroi verticale droite} \\ \bar{T} = T_h & \quad \text{sur la paroi verticale gauche} \\ \frac{\partial \bar{T}}{\partial y} = 0 & \quad \text{sur les parois horizontales} \\ k = 0, \quad \varepsilon = \infty & \quad \text{sur les parois} \end{aligned}$$

Résultats :

On a effectué la simulation sur un maillage constitué de 60×60 éléments quadrangulaires, resserré près des parois. En effet, puisqu'aucune fonction de paroi n'est utilisée,

il est nécessaire de résoudre les couches limites qui se développent le long des parois chaudes et froides. L'épaisseur des éléments près des parois verticales est environ $\Delta x_1 \approx 2 \times 10^{-4}$, ce qui est encore insuffisant pour résoudre correctement les gradients thermiques importants qui caractérisent l'écoulement à ce nombre de Rayleigh.

L'état permanent est à peu près atteint après 200000 itérations, bien que la solution ne soit pas convergée. Le champ de température est représenté sur la figure 3.3.4, montrant une couche limite thermique très fine, et des isothermes à peu près horizontales dans la partie centrale de la cavité. Un détail du champ de vitesse est représenté sur la figure 3.3.5.

La figure 3.3.6 représente le rapport entre la viscosité turbulente et la viscosité laminaire. Enfin, la figure 3.3.7 représente la distribution de Nusselt à la paroi (froide).

Les résultats quantitatifs de la simulation sont reportés dans le tableau suivant, qui comporte également un résumé des résultats obtenus par les 10 participants au workshop [1], dont certains avec les codes ASTEC (AEA), FLUENT, PHOENICS. Les résultats obtenus avec CASTEM2000 sont donc globalement satisfaisants, puisqu'ils se situent dans la fourchette des différentes contributions au workshop [1], à l'exception de la grandeur $(\nu_t/\nu)_{\max}$ qui est légèrement trop élevée. Cela peut provenir du fait que la solution n'est pas suffisamment convergée.

Code	formulation	maillage	Δx_1	\overline{Nu}	Nu_{\max}	$(\nu_t/\nu)_{\max}$
CASTEM2000	EF	60×60	2×10^{-4}	284	594	72.5
ASTEC	VF	80×80	2×10^{-4}	256	604	53.5
FLUENT	VF	150×150	1×10^{-4}	246	675	53.7
PHOENICS	VF	60×60	1×10^{-3}	323	573	59.6
autres	VF/EF	48×48	1×10^{-3}	223	573	33.7
		189×189	6.3×10^{-6}	323	786	59.6

TAB. 3.3.1 – Tableau récapitulatif des résultats CASTEM2000/workshop [1].

Référence :

- 1– R.A.W.M. Henkes and C.J. Hoogendoorn, Eds. Turbulent Natural Convection in Enclosures : a computational and experimental benchmark study. Proceedings of the Eurotherm Seminar n° 22, March 25-27, 1992, Delft, The Netherlands. Editions Européennes Thermique et Industrie.




FIG. 3.3.2 – Maillage 61 × 61




FIG. 3.3.3 – Détail du maillage (coin supérieur droit)




FIG. 3.3.4 – Champ de température




FIG. 3.3.5 – Détail du champ de vitesse (coin supérieur gauche)




FIG. 3.3.6 – Turbulence : ν_t/ν (minimum=0, maximum=72.5)




FIG. 3.3.7 – Distribution de Nusselt à la paroi

Jeu de données :

```

*****
* CAVITE CARREE TURBULENTE SANS FONCTION DE PAROI
* CL a la paroi : u=v=0
*                   k=0
*                   e=infinity)
*****
opti trac ps;

DEBPROC CALCUL1 ;
  ARGU RVX*'TABLE' ;
RV = RVX.'EQEX' ; RV.INCO.'DT' = RV.PASDETPS.'DELTAT' ;
  FINPROC ;
*****
* PROCEDURE POUR ESTIMER LA CONVERGENCE VERS L'ETAT *
* STATIONNAIRE : LE CRITERE EST BASE SUR LES 2      *
* COMPOSANTES DE LA VITESSE                        *
*****)*****
DEBPROC CALCUL ;
  ARGU RVX*'TABLE' ;
RV = RVX.'EQEX' ;
DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;
UN = RV.INCO.'UN' ;
UNM1 = RV.INCO.'UNM1' ;
unx= kcht (rv.'DOMAINE') scal sommet (exco 'UX' un) ;
unmx = kcht (rv.'DOMAINE') scal sommet (exco)g('UX' unm1) ;
uny= kcht (rv.'DOMAINE') scal sommet (exco 'UY' un) ;
unmy = kcht (rv.'DOMAINE') scal sommet (exco 'UY' unm1) ;
ERRX = KOPS unx '-' unmx ;
ERRY = KOPS uny '-' unmy ;
ELIX = MAXI ERRX 'ABS' ;
ELIY = MAXI ERRY 'ABS' ;
ELIX = ELIX '/' RV.INCO.'DT' ;
ELIY = ELIY '/' RV.INCO.'DT' ;
ELIX = (LOG (ELIX + 1.0E-20))/(LOG 20.) ;
ELIY = (LOG (ELIY + 1.0E-20))/(LOG 20.) ;
MESSAGE 'ITER ' RV.PASDETPS.'NUPASDT' 'ERREUR LINF ' ELIX ELIY
'MAX NUT/NU = '((MAXI RV.INCO.'NUT')/NU) ;
RV.INCO.'UNM1' = KCHT $MT 'VECT' 'SOMMET' (RV.INCO.'UN') ;
IT = PROG RV.PASDETPS.'NUPASDT' ;
ER = PROG ELIY ;
RV.INCO.'IT' = (RV.INCO.'IT') ET IT ;
RV.INCO.'ER' = (RV.INCO.'ER') ET ER ;
  FINSI ;
  FINPROC ;
*****
* MAILLAGE *
*****
option dime 2 ;
  opti elem qua8 ;
p1 = 0. 0. ;
p15 = 0.5 0. ;

```

```

p2 = 1. 0. ;
p25 = 1. 0.5 ;
p3 = 1. 1. ;
p35 = 0.5 1. ;
p4 = 0. 1. ;
p45 = 0. 0.5 ;
d1 = 0.01 ; d2 = 0.05 ;
d1 = 0.005 ;
d2 = 0.1 ;
d1 = 0.1 ;
d2 = 0.1 ;
  d1 = 0.00003 ; d2 = 0.22 ;
d1 = 0.0003 ; d2 = 0.22 ;
bas=p1 d dini d1 dfin d2 p15 d dini d2 dfin d1 p2 ;
cdro=p2 d dini d1 dfin d2 p25 d dini d2 dfin d1 p3 ;
  haut=p3 d dini d1 dfin d2 p35 d dini d2 dfin d1 p4 ;
cgau=p4 d dini d1 dfin d2 p45 d dini d2 dfin d1 p1 ;
cnt= bas et cdro et haut et cgau ;
mt= bas cdro haut cgau daller ;
$mt = DOMA mt 'MACRO' impr ;
$wall = DOMA cnt 'MACRO' 'INCL' $mt ;
LISTE $MT ;
LISTE (COOR 1 BAS) ;

mt = $mt.maillage ;
trace mt;
Pr = 0.71 ;
Prt = 0.9 ;
Ra = 5.0E10 ;
Gr= Ra/Pr ;
NU=1/(Gr**0.5) ;
ALF=NU/Pr ;
gb=0 -1 ;
RV = EQEX $MT 'ITMA' 200000 'ALFA' 1.2 'ZONE' $MT 'OPER'
'FILTREKE' 'UN' 1.0 NU 'INCO' 'KN' 'EN'
'ZONE' $MT 'OPER' CALCUL 'OPTI' 'SUPGDC'
'ZONE' $MT 'OPER' 'NSKE' NU 'NUT' GB 'TN' 0.5 'INCO' 'UN' 'KN' 'EN'
'OPTI' 'SUPGDC' 'ZONE' $MT 'OPER' 'TSCAL' ALF 'UN' 0. 'NUT'
Prt 'INCO' 'TN' 'ZONE' $MT 'OPER' CALCUL1 ;
RV = EQEX RV
      'CLIM' 'UN' 'UIMP' cnt 0. 'UN' 'VIMP' cnt 0.
'TN' 'TIMP' cgau 1.
'TN' 'TIMP' cdro 0.
'KN' 'TIMP' cnt 1E-10 'EN' 'TIMP' cnt 1E30 ;
RVP = EQPR $MT KTYPI 1 ZONE $MT 'OPER' 'PRESSION' 0. ;
RV.INCO=TABLE INCO ;
RV.INCO.'UN'=kcht $MT VECT SOMMET (1.E-5 1.E-5) ;
cory=coor 2 mt ;
RV.INCO.'TN'=kcht $MT SCAL SOMMET (cory ) ;
RV.INCO.'KN'=kcht $MT SCAL SOMMET 1.E-5 ;
RV.INCO.'EN'=kcht $MT SCAL SOMMET 1.E-10 ;
RV.INCO.'NUT'=kcht $MT SCAL CENTRE (20.*NU) ;
rv.inco.'UNM1' = kcht $MT 'VECT' 'SOMMET' (1.E-5 1.E-5 ) ;
rv.inco.'IT' = PROG 1 ; rv.inco.'ER' =PROG 0. ;
rv.inco.'DT' = 0.DO ;
RV.PRESSION =RVP ;

```

```

TRACE RV.INCO.'TN' MT ;
EXEC RV ;
repix rv ;
*opti sauv '/test4/paillere/caviteke.sol';

*sauv ;
opti isov suli ;
trace RV.INCO.'TN' mt (cont mt) 14 'TITR' 'TEMPERATURE' ;
unch = vect RV.INCO.'UN' 0.1 UX UY JAUNE ;
TRACE UNCH MT (CONT MT) 'TITR' 'CHAMP DE VITESSE' ;
ZZ = KOPS (ELNO $MT RV.INCO.'NUT') '/' NU ;
TRACE ZZ MT CNT 14 'TITR' 'Nut/Nu' ;
EVOL4 = EVOL 'MANU' ('ITERATIONS' (RV.INCO.'IT') 'LOG|E|inf'
(RV.INCO.'ER') ;
DESS EVOL4 'XBOR' 0. 20000. 'YBOR' -10.0 0.0 'TITR' 'CONVERGENCE' ;
$CDRO = 'DOMA' CDRO 'MACRO' 'INCL' $MT ;
DY = 'DOMA' $CDRO 'VOLUME' ;
DYT = SOMT DY ;
MESSAGE 'LONGUEUR DE LA PAROI = ' DYT ;
GRADT = KOPS RV.INCO.'TN' ('GRAD' $MT ;
DTDX = KCHT $MT 'SCAL' 'CENTRE' (EXCO 'UX' GRADT) ;
DTDX = ELNO $MT DTDX ;
DTDXp = KCHT $CDRO 'SCAL' 'SOMMET' DTDX ;
EVOL1 = EVOL 'CHPO' DTDXp SCAL CDRO ;
TAB1 = TABLE ; TAB1.1 = 'MARQ LOSA' ;

DESS EVOL1 'GRIL' TAB1 ;
DTDXpc = NOEL $CDRO DTDXp ;
num = KOPS DTDXpc '*' DY;
num = SOMT NUM ; num = NUM / DYT;
mESSAGE 'NUSELT MOYEN ' NUM ;
MESSAGE 'NUSELT MAX ' (MAXI DTDXpc) ;
MESSAGE 'NUSEL MIN ' (MINI DTDXpc) ;
un=rv.inco.'UN' ;
sw = (kops un 'ROT' $mt) ;
rk = EQEX $mt 'OPTI' 'EF' 'IMPL'
ZONE $mt OPER LAPN 1.0 INCO 'PSI'
ZONE $mt OPER FIMP sw)120 b(INCO 'PSI'
'CLIM' 'PSI' 'TIMP' (cgau et haut et bas et cdro) 0.;
rk.'INCO'.'PSI'='kcht $mt scal sommet 0.;
exic rk ;
psi=rk.'INCO'.'PSI';
LISTE (MAXI PSI) (MINI PSI) ;
trace psi mt CNT 14 'TITR' 'FONCTION DE COURANT' ;
fin ;

```

Chapitre 4

Écoulements bi-fluides à phase dispersée

4.1 Motion of a particle in a fluid at rest

4.1.1 Formulation Eléments Finis Semi-Implicite

<i>Operators :</i>	<i>NS (NOCONS), FROT, BIF</i>
<i>Numerical method :</i>	<i>EFM1, SUPG</i>
<i>Physical problem</i>	<i>Motion in air at rest of a spherical glass particle under gravity force</i>
<i>Classification</i>	<i>2D plan, convection, steady</i>
<i>Jeu de données</i>	

Problem description :

Motion in air at rest of a particle under the gravity force (assumed that the particle motion has negligible effects on the air).


$$V \frac{\partial V}{\partial x} = g \frac{\Delta \rho}{\rho_p} + I(U - V)$$

where

$$\left\{ \begin{array}{l} I = 18 \frac{\nu}{d^2} \frac{\rho_a}{\rho_p} F_D \\ F_D = 1 + 0.15 Re^{0.687} \\ Re = \frac{|U-V|d}{\nu} \\ U = 0 \end{array} \right.$$

V increases from 0 to V_{lim} . The exact value of V_{lim} is known and in the present case is $V_{lim}=3.7\text{m/s}$.

Results :



Particles velocity along the vertical axe

FIG. 4.1.1 – Particle velocity along the vertical axis compared with the limit velocity.

Input data :

```

* mesh
  opti dime 2 elem qua8 ;
  L      = 5.000 ;
  d1     = 0.100 ;
  d1     = 0.500 ;
  d2     = 0.020 ;
  p1     = -0.100 0.000 ;
  p2     = 0.100 0.000 ;
  bot    = droi p1 p2 1 ;
  top    = plus bot (0.000 L) ;
  mt     = regl bot top dini d1 dfin d2 ;
  right  = cote 2 mt ;
  left   = cote 4 mt ;
  cnt    = cont mt ;
  elim   mt 1.0e-5 ;
  tass   mt ;

* Data

  Vi     = 0.16 ;
  Dp     = 5.00e-4 ;

* Dominions

  $mt    = doma mt macro impr ;
  $bot   = doma bot macro incl $mt 1.0e-5 ;
  $top   = doma top macro incl $mt 1.0e-5 ;

  mt     = $mt.maillage ;
  bot    = $bot.maillage ;
  top    = $top.maillage ;

* Physical properties

  condf  = 2.62e-2 ;
  nuf    = 1.50e-5 ;
  rof    = 1.20e-0 ;
  cpf    = 1.00e+3 ;
  diff   = condf/(rof*cpf) ;
  nup    = 1.00e-20 ;
  rop    = 2.50e+3 ;
  cpp    = 8.40e+2 ;
  difp   = nup ;
  rrho   = rof/rop ;
  qvol   = 0.00 ;
  gx     = 0.00e+0 ;
  gy     = -9.81e+0 ;
  gdr    = 0.0 (-9.81*(rop-rof)/rop) ;
  zero   = 0.0 0.0 ;
  beta   = 1.0 1.0 ;
  alpha  = 1.0 ;

* Table for bif

```

```

tabbif          = table          ;
tabbif.'RHOF'   = rof            ;
tabbif.'RHOP'   = rop            ;
tabbif.'DPART'  = Dp             ;
tabbif.'NUF'    = nuf            ;
tabbif.'ALPHA'  = mot 'A'        ;
tabbif.'UFLUID' = mot 'UN'       ;
tabbif.'UPART'  = mot 'VN'       ;
tabbif.'KFLUID' = mot 'KFL'      ;
tabbif.'KPART'  = mot 'KPR'      ;

* Points for time histories

lh              = (poin mt proc (0.000 (0.1*L))) et
                (poin mt proc (0.000 (0.3*L))) et
                (poin mt proc (0.000 (0.5*L))) et
                (poin mt proc (0.000 (0.7*L))) et
                (poin mt proc (0.000 (0.9*L))) ;

tran = khis          'VN' 2 lh ;

* Equations + boudary conditions

* Navier-Stokes + K-epsilon + particle mass conservation

rv = EQEX $mt ITMA 500 ALFA 0.5
     OPTI 'NOCONS' 'SUPG'
     ZONE $mt OPER NS nuf zero INCO UN
     ZONE $mt OPER NS nup gdr INCO VN
;

* Coupling terms on q.d.m equations

rv = EQEX rv
     OPTI 'NOCONS'
     ZONE $mt OPER FROT 'KPR' beta 'UN' INCO VN
     ZONE $mt OPER PPRE 'VN' rrho
     ZONE $mt OPER BIF tabbif
;

* Boundary coditions

rv = EQEX rv
     CLIM UN UIMP mt 0.0 UN VIMP mt 0.0
     CLIM VN UIMP left 0.0 VN UIMP right 0.0
     CLIM VN VIMP top (-1*Vi) Vn UIMP top 0.0
;

* Pressure equation

rp = EQPR $mt KTYPI 1
     ZONE $mt OPER PRESSION 0.0 ;

* Initial conditions

```

```

rv.titulo      = mot 'tubo' ;
rv.INCO        = table INCO ;
rv.INCO.'UN'    = kcht $mt vect sommet (1.0e-5 1.0e-5) ;
rv.INCO.'VN'    = kcht $mt vect sommet (1.0e-5 (-1*Vi)) ;
rv.INCO.'A'     = kcht $mt scal sommet (alpha ) ;
rv.INCO.'KPR'   = kcht $mt vect centre (0.0 0.0 ) ;
rv.PRESSION    = rp ;
rv.'HIST'      = tran ;
rv.titotlo     = titulo ;
rv.tdomc       = $cnt ;

* Executing

exec rv ;

* post processing

* Procedura visu

debproc visu rv*table ;
tdom = rv.domaine ;
mesh = rv.domaine.maillage ;

pt1 = poin mesh proc (0.0 1.000) ;
pt2 = poin mesh proc (0.0 L ) ;

pt12 = poin mesh droit pt2 pt1 0.0001 ;
mesh1 = chan mesh lign ;
line1 = elem mesh1 appuye pt12 ;
line1 = line1 comp pt2 pt1 ;
pro1 = prog 0.0 4.0 ;
pro2 = prog -3.7 -3.7 ;
vt = evol manu ' ' pro1 ' ' pro2 ;

vn = rv.inco.'VN' ;
v2ln1 = evol chpo vn uy line1 ;
tx1 = mot '(m)' ;
tx2 = mot 'time (s)' ;
ty = mot 'V (m/s)' ;
tt1 = mot 'Particles velocity along the vertical axe' ;
tt2 = mot 'Particles velocity v.s. time' ;
dess (v2ln1 et vt) titx tx1 tity ty titr tt1 ;
dess (rv.hist.'2VN') titx tx2 tity ty titr tt2 ;

finproc ;

visu rv ;

fin ;

```

4.2 Gas flow induced by a dispersed phase

4.2.1 Formulation Eléments Finis Semi-Implicite

<i>Operators :</i>	<i>NSKE, NS (NOCONS), TSCAL (CONS), FROT, BIF, PPRE, FPU, PRESSION</i>
<i>Numerical method :</i>	<i>EFM1, SUPG approximation Q1-P0</i>
<i>Physical problem</i>	<i>Air flow in a closed chamber induced by a particle jet falling under gravity.</i>
<i>Classification</i>	<i>2D plan, convection, steady as asymptotic limit of a transient</i>
<i>Input data</i>	<i>/u2/castem/dgibi/dynasp.dgibi</i>

Problem description :

The problem is a schematic representation of the DYNASP experiment where a glass particle jet falls under gravity in the central region of a closed air-filled chamber inducing an air motion.




FIG. 4.2.1 – Schematic representation of the Dynasp experiment

Equations :

The two phases are treated as two separate but inter-penetrating continua. The flow is described in term of the gas velocity U_i , the particle velocity V_i and the mass fraction of the dispersed phase γ . Turbulence effects in the particles are neglected due to their inertia. The conservation equations read :

$$\begin{aligned}
U_{j,j} &= 0 \\
\gamma_{,t} + (\gamma V_j)_{,j} &= 0 \\
U_{i,t} + U_j U_{i,j} &= \nu U_{i,jj} - P_{,i} / \rho_g - (u_i u_j)_{,j} + \gamma I (V_i - U_i) \\
V_{i,t} + V_j V_{i,j} &= g_i (\rho_p - \rho_g) / \rho_p - P_{,i} / \rho_p + I (U_i - V_i)
\end{aligned}$$

where

$$\left\{ \begin{array}{l} I = 18 \frac{\nu}{d^2} \frac{\rho_g}{\rho_p} F_D \\ F_D = 1 + 0.15 Re^{0.687} \\ Re = \frac{|U-V|d}{\nu} \end{array} \right.$$

Discretization :

The interaction term between the two phases is computed via *BIF* and assembled via *FROT*. The correction of the pressure term in the particle equation is done via *PPRE*. Turbulence is modeled by a $\bar{K} - \epsilon$ model.

Results :

The results concern the analysis of the *DYNASP* test cases 35/500 and 60/500 (see [1]). The computed results are compared with the experimental measurements.

Reference :

- 1– Magnaud J.P., Caroli C., Cornet P., Coche J.C. : *Numerical and experimental study of spray dynamics. DYNASP program*. Proceedings of NURETH 8 - 30/9-4/10 97 Kyoto, Japan






FIG. 4.2.2 – Test case 35/500. Computed air and particles velocity





Input data :

* Test DYNASP K-epsilon

```
titolo_g = chaine 'Dynasp-35 ' ;
titolo   = chaine titolo_g   ;
titr     = titre             ;
```

* Geometry

```
opti dime 2 elem qua4 ;

n1 = 4 ; n2 = 4*n1 ; n3 = -10*n1 ;
n4 = -1*n1 ; n5 = -1*n2 ;
dmed = 2.000/n3 ; dmin = dmed/1.5 ; dmax = dmed*1.5 ;

pa0 = 0.000 0.000 ; pb0 = 0.0435 0.000 ; pc0 = 0.200 0.000 ;
pa1 = 0.000 1.000 ; pb1 = 0.0435 1.000 ; pc1 = 0.200 1.000 ;
pa2 = 0.000 2.000 ; pb2 = 0.0435 2.000 ; pc2 = 0.200 2.000 ;
pa3 = 0.000 0.500 ; pb3 = 0.0435 0.500 ; pc3 = 0.200 0.500 ;
pa4 = 0.000 1.500 ; pb4 = 0.0435 1.500 ; pc4 = 0.200 1.500 ;

axei = pa0 droi n3 pa1 dini dmax dfin dmin ;
verti = pb0 droi n3 pb1 dini dmax dfin dmin ;
sidei = pc0 droi n3 pc1 dini dmax dfin dmin ;
axes = pa1 droi n3 pa2 dini dmin dfin dmax ;
verts = pb1 droi n3 pb2 dini dmin dfin dmax ;
sides = pc1 droi n3 pc2 dini dmin dfin dmax ;

weti = axei regl n1 verti ;
dryi = verti regl n2 sidei ;
wets = axes regl n1 verts ;
drys = verts regl n2 sides ;

bot = (cote 4 weti) et (cote 4 dryi) ;
hori = (cote 4 wets) et (cote 4 drys) ;
in = (cote 2 wets) ;
dtop = (cote 2 drys) ;
top = in et dtop ;

wet = weti et wets ;
dry = dryi et drys ;
axe = axei et axes ;
side = sidei et sides ;
mt = dry et wet ;
elim mt 0.00001 ;
mt = orie mt ;
cnt = cont mt ;
wall = bot et side et dtop ;
ppa3 = mt poin proc pa3 ;
ppb3 = mt poin proc pb3 ;
ppc3 = mt poin proc pc3 ;
ppa4 = mt poin proc pa4 ;
ppb4 = mt poin proc pb4 ;
ppc4 = mt poin proc pc4 ;
```

```

sez3 = ppa3 droi n1 ppb3 droi n2 ppc3 ;
sez4 = ppa4 droi n1 ppb4 droi n2 ppc4 ;
elim (mt et sez3 et sez4) 0.00001 ;
tass      mt ;

C      =      3.50e-2 ;
Vi     =      0.16 ;
Dp     =      4.70e-4 ;
S      =      0.400*0.087 ;
C_s    =      C/S ;

* Options

opti isov lign ;

* Reading mesh

yp      =      0.010 ;
rapp    =      (0.400 - yp)/0.400 ;
depl mt affi rapp (0.0 0.0) (1.0 0.0) ;

* Dominions

$mt     = doma mt      impr ;
$bot    = doma bot     incl $mt 1.0e-4 ;
$axe    = doma axe     incl $mt 1.0e-4 ;
$side   = doma side    incl $mt 1.0e-4 ;
$top    = doma top     incl $mt 1.0e-4 ;
$in     = doma in      incl $mt 1.0e-4 ;
$cnt    = doma cnt     incl $mt 1.0e-4 ;
$wet    = doma wet     incl $mt 1.0e-4 ;
$dry    = doma dry     incl $mt 1.0e-4 ;
$wall   = doma wall    incl $mt 1.0e-4 ;
$dtop   = doma dtop    incl $mt 1.0e-4 ;
volmt   = doma $mt     volume ;

* Physical properties

condf   =      2.62e-2 ;
nuf     =      1.50e-5 ;
rof     =      1.20e-0 ;
cpf     =      1.00e+3 ;
diff    =      condf/(rof*cpf) ;
nup     =      1.00e-20 ;
rop     =      2.50e+3 ;
cpp     =      8.40e+2 ;
difp    =      nup ;
rrho    =      rof/rop ;
qvol    =      0.00 ;
gx      =      0.00e+0 ;
gy      =      -9.81e+0 ;
gdr     =      0.0      (-9.81*(rop-rof)/rop) ;
zero    =      0.0      0.0 ;
beta    =      1.0      1.0 ;
* alpha =      (3.14/6.0)*n_dot*(Dp**3.0)/Vi ;

```

```

alpha      = C_s/rop/Vi ;
n_dot     = (6.0/3.14)*C_s/(rop*(Dp**3.0)) ;
ustella   = 30.0*nuf/yp ;

L0        = 0.30 ;
K0        = 1.0e-4 ;
Rec       = 50.0 ;

* mess 'Cs' C_s 'Dp' Dp 'n_dot' n_dot ; opti donn 5 ;

tabbil    = table ;
tabbil.'TF' = mot 'TF' ;
tabbil.'TP' = mot 'TP' ;
tabbil.'VP' = mot 'VN' ;
tabbil.'ALPHA' = mot 'A' ;
tabbil.'ROCPF' = (rof*cpf) ;
tabbil.'ROCPP' = (rop*cpp) ;
tabbil.'VOL' = volmt ;
tabbil.'CNT' = $cnt ;
tabbil.'TOP' = $top ;
tabbil.'BOT' = $bot ;

* Table for bif

tabbif    = table ;
tabbif.'RHOF' = rof ;
tabbif.'RHOP' = rop ;
tabbif.'DPART' = Dp ;
tabbif.'NUF' = nuf ;
tabbif.'ALPHA' = mot 'A' ;
tabbif.'UFLUID' = mot 'UN' ;
tabbif.'UPART' = mot 'VN' ;
tabbif.'KFLUID' = mot 'KFL' ;
tabbif.'KPART' = mot 'KPR' ;

* Points for time histories

nc        = 10 ;
npti     = nbno mt ;
ps       = npti / nc ;
pf       = ps * nc ;
lh       = elem ($mt.sommet) 'POI1' (lect ps pas ps pf) ;
tran = khis
          'UN' 1 lh 'UN' 2 lh
          'VN' 1 lh 'VN' 2 lh
          'KN' 1 lh 'EN' 1 lh
*         'TP' 1 lh 'TF' 1 lh
          'A' 1 lh
;

* Equations + boudary conditions

* Navier-Stokes + K-epsilon + particle mass conservation

rv = EQEX $mt ITMA 10000 ALFA 0.70
     OPTI 'NOCONS' 'SUPG'

```

```

ZONE $mt OPER NSKE nuf 'NT' zero INCO UN KN EN
ZONE $wall OPER FPU nuf 'USTAR' yp INCO UN KN EN
ZONE $mt OPER NS nup gdr INCO VN
OPTI 'CONS' 'SUPG'
ZONE $mt OPER TSCA nup 'VN' qvol 'TN' INCO A
;

* Coupling terms on q.d.m equations

rv = EQEX rv
OPTI 'NOCONS'
ZONE $mt OPER FROT 'KFL' beta 'VN' INCO UN
ZONE $mt OPER FROT 'KPR' beta 'UN' INCO VN
;

rv = EQEX rv
ZONE $mt OPER FILTREKE 'UN' LO nuf INCO KN EN
ZONE $mt OPER PPRE 'VN' rrho
ZONE $mt OPER BIF tabbif
;

* Boundary coditions

rv = EQEX rv
CLIM UN VIMP top 0.0 UN VIMP bot 0.0
CLIM UN UIMP axe 0.0 UN UIMP side 0.0
CLIM VN UIMP axe 0.0 VN UIMP side 0.0
CLIM VN VIMP in (-1*Vi) VN VIMP dtop 0.0
CLIM A TIMP in alpha A TIMP dtop 0.0
;

* Pressure equation

rp = EQPR $mt KTYPI 1
ZONE $mt OPER PRESSION 0.0 ;

* Initial conditions

alphaw = kcht $wet scal sommet (alpha ) ;
tpwet = kcht $wet scal sommet (20.0 ) ;
rv.titulo = titulo ;
rv.INCO = table INCO ;
rv.INCO.'UN' = kcht $mt vect sommet (1.0e-5 1.0e-5) ;
rv.INCO.'VN' = kcht $mt vect sommet (1.0e-5 (-1*Vi)) ;
rv.INCO.'KN' = kcht $mt scal sommet (1.0e-5 ) ;
rv.INCO.'EN' = kcht $mt scal sommet (1.0e-5 ) ;
rv.INCO.'NT' = kcht $mt scal centre nuf ;
rv.INCO.'A' = kcht $mt scal sommet (0.0 ) alphaw ;
rv.INCO.'TN' = kcht $mt scal sommet (0.0 ) ;
rv.INCO.'KFL' = kcht $mt vect centre (0.0 0.0 ) ;
rv.INCO.'KPR' = kcht $mt vect centre (0.0 0.0 ) ;
rv.INCO.'USTAR' = kcht $wall scal centre ustella ;
rv.PRESSION = rp ;
rv.'HIST' = tran ;
rv.'YP' = yp ;
rv.'NUF' = nuf ;

```

* Executing

```
exec rv ;
```

* Saving results

```
repix rv ;  
sauv rv ;  
fin ;
```

Chapitre 5

Écoulements compressibles

5.1 Calcul d'un Tube à Choc

5.1.1 Formulation Volumes Finis Explicite Conservative

<i>Opérateurs utilisés</i>	<i>PENT PRET KONV AVCT</i>
<i>Méthodes numériques</i>	<i>Godounov étendu au second ordre par une approche de type M.U.S.C.L</i>
<i>Nature du problème</i>	<i>Problème de Riemann (Tube à Choc pour un gaz parfait)</i>
<i>Classification</i>	<i>2D plan, compressible, instationnaire</i>
<i>Jeu de données</i>	

Description du problème :

On résout le système d'équations d'Euler suivant :

$$\begin{cases} \rho_t + \operatorname{div}(\rho \vec{V}) = 0 \\ (\rho \vec{V})_t + \operatorname{div}(\rho \vec{V} \otimes \vec{V} + p \vec{I}) = 0 \\ (\rho E)_t + \operatorname{div}(\rho H \vec{V}) = 0 \end{cases} \quad (5.1.1)$$

où ρ désigne la densité, \vec{V} la vitesse, p la pression, E l'énergie spécifique totale et H l'enthalpie spécifique totale.

On cherche à déterminer la pression, la masse volumique et le débit le long d'un tube à choc de longueur $L = 2$ m. Cet tube est composé d'une chambre à haute pression remplie d'un gaz au repos, dont la pression initiale est P_G et la masse volumique ρ_G . La partie droite du cylindre est remplie du même gaz mais à la pression P_D et de masse volumique ρ_D .

Solution analytique :

Le gaz est supposé parfait ; de plus, on a imposé que la température initiale est la même et que $P_G > P_D$. A l'instant $t = 0^+$ le diaphragme qui separe les deux milieux est enlevé. On cherche alors à déterminer les variations des inconnues en fonction du temps et de l'espace (on sait que les inconnues sont fonctions de x/t).

On observe une onde de choc S, une discontinuité de contact C et un onde de détente centrée. En imposant les conditions de continuité pour le choc et pour la discontinuité de contact, en sachant que pour une onde de détente centrée rentrante "C+" l'entropie et l'invariant de Riemann r sont constants, on trouve la solution analytique du problème.

Discrétisation du problème :

Le problème est résolu par les schémas explicites - Volumes Finis de type Godounov étendus au second ordre par une approche de type de M.U.S.C.L . On a utilisé les opérateurs suivants :




FIG. 5.1.1 – Conditions initiales du tube à choc.




FIG. 5.1.2 – Diagramme espace-temps pour le tube à choc.




FIG. 5.1.3 – Pression et masse volumique à un instant donné le long du tube à choc.

- l'opérateur **KONV**, qui discrétise les termes de convection par des schémas volumes finis ;
- l'opérateur **PENT**, qui calcule les pentes associées à la masse volumique, la quantité de mouvement et l'énergie totale ;
- l'opérateur **PRET**, qui calcule les incréments associés aux variables si on utilise un schéma du second ordre en temps (de Type Prediction / Correction) ;
- l'opérateur **AVCT**, qui réalise un incrément en temps ;

Résultats :

On a comparé les solutions numériques obtenues pour les schémas de premier et second ordre en espace et en temps avec la solution analytique. On a créé un maillage constitué d'éléments de type carré de dimension $\Delta x = 0.1$. Le pas temporel est calculé en imposant la condition de stabilité de la méthode.


$$\Delta t = \alpha \times \min \left(\frac{\Delta x}{\max(|u| + c)} \right) \quad (5.1.2)$$

où α est un facteur de sécurité ($\alpha < 1$).


On a considéré les données adimensionnées suivantes :

$P_g = 5.0$, $P_d = 1.0$, $\rho_g = 1.0$ et $\rho_d = (P_d/P_g)\rho_g = 0.2$. On a pris $\alpha = 0.2$ et, après 300 itérations, on a calculé l'erreur relative (norme 2) pour la masse volumique, la pression, le débit, l'enthalpie spécifique totale et l'entropie spécifique. Dans la table suivante on a montré les résultats pour les différents ordres en espace (IE) et en temps (IT).

ERREURS			
	IT 1 IE 1	IT 1 IE 2	IT 2 IE 2
ρ	3.174×10^{-2}	1.870×10^{-2}	1.862×10^{-2}
p	3.236×10^{-2}	1.836×10^{-2}	1.818×10^{-2}
ρu	1.085×10^{-1}	7.373×10^{-2}	7.328×10^{-2}
H	3.821×10^{-2}	2.525×10^{-2}	2.509×10^{-2}
S	4.788×10^{-2}	2.903×10^{-2}	2.898×10^{-2}




2D Van Leer : Masse Volumique tmps .13515




2D Van Leer : Masse Volumique tmps .13515

FIG. 5.1.4 – Masse volumique le long du tube et dans la zone de détente.



2D Van Leer : Masse Volumique tmps .13515



2D Van Leer : Masse Volumique tmps .13515

FIG. 5.1.5 – Masse volumique sur la surface de discontinuité et sur le choc.





FIG. 5.1.6 – Pression le long du tube et dans la zone de détente.




FIG. 5.1.7 – Pression au niveau du choc.




FIG. 5.1.8 – Débit le long du tube.




FIG. 5.1.9 – Débit dans la zone de détente et sur la surface de discontinuité.




FIG. 5.1.10 – Débit au niveau du choc.






FIG. 5.1.11 – Enthalpie totale.




2D Van Leer : HTOT tmps .13515



2D Van Leer : HTOT tmps .13515

FIG. 5.1.12 – Enthalpie dans la zone de détente et sur la surface de discontinuité.



2D Van Leer : HTOT tmps .13515

FIG. 5.1.13 – Enthalpie totale sur le choc.






FIG. 5.1.14 – Entropie.



2D Van Leer : S tmps .13515



2D Van Leer : S tmps .13515

FIG. 5.1.15 – Entropie sur la surface de discontinuité et sur le choc.

Jeu de données :

```

*
DEBPROC CALPRAP;

* FICHER DE VALIDATION AVEC SOLUTION ANALITIQUE

* Procédure pour le calcul de P- $\text{rap}$  = P-shock/P-d

* HP Ficher de validation

* 1) meme gas;
* 2) gas parfait;
* 3)  $E_g = E_d$ ;
* 4)  $V_x(t=0)=0$ ;
* 5)  $P_g > P_d$  (Pas limitative);
*

ARGU Rho_g*FLOTTANT Rho_d*FLOTTANT gam_g*FLOTTANT;

RHO_rap=Rho_g/Rho_d;
NP_rap = 100;
EpsP_rap=1.0E-6;
P_rap=1.0;
REPETER BP_rap NP_rap;
P_rap0=P_rap;
B1 = (P_rap/Rho_rap)**((gam_g-1.0)/2.0/gam_g);
B1=1-B1;
B1=B1/(gam_g-1.0);
B2=gam_g*(P_rap+1);
B2=B2+P_rap-1;
B2=B2*2.0*gam_g;
B2=B2**(0.5);
P_rap=1+(B1*B2);
Eps_P=abs ((P_rap-P_rap0)/P_rap0);
SI ( Eps_P < EpsP_rap );
  MESSAGE 'Iterations pour calculer P_rap=' &BP_rap ;
  QUITTER BP_rap;
FINSI;
FIN BP_rap;
FINPROC P_rap;

DEBPROC SOLEX;
*
* FICHER DE VALIDATION
*
* HP Ficher de validation
* 1) meme gas;
* 2) gas parfait;
* 3)  $E_g = E_d$ ;
* 4)  $V_x(t=0)=0$ ;
* 5)  $P_g > P_d$  (Pas limitative);
*
* 4.12.95

```



```

*
* zone g = zone gauche;
* zone z3 = zone de l'onde de détente centré ("backward centre
*   rarefaction wave")
* zone z2 = zone entre la surface de discontinuité et le choc;
* zone d = zone droite
*
ARGU Rho_g*FLOTTANT P_g*FLOTTANT P_d*FLOTTANT gam_g*FLOTTANT
      tmps*FLOTTANT xx*LISTREEL LONG*FLOTTANT;

Rho_d= Rho_g/P_g*P_d;
P_rap= CALPRAP Rho_g Rho_d gam_g;
Rho_d= Rho_g/P_g*P_d;
P_rap= CALPRAP Rho_g Rho_d gam_d;
P_z3=P_rap*P_d;
P_z2=P_z3;
B1=gam_d*(P_rap+1)+P_rap-1;
B2=gam_d*(P_rap+1)-P_rap+1;
Rho_z2=Rho_d*B1/B2;
Rho_z3=(P_z3/P_g)**(1/gam_d)*Rho_g;
* message 'P-rap=' P_rap;
* message 'P-z3=' P_z3;
* message 'Rho-z2=' Rho_z2;
* message 'Rho-z3=' Rho_z3;
B1 = (P_z3/P_g)**((gam_g-1.0)/2.0/gam_g);
B1=1-B1;
B2=(gam_g*P_d/Rho_d)**0.5;
B2=B2/(gam_g-1)*2.0;
U_sur=B1*B2;
U_choc=Rho_z2/(Rho_z2-Rho_d)*U_sur;
message 'U_sur=' U_sur;
message 'U_choc' U_choc;
* Calcul de la distance de la surface de discontinuité a t=0;

x_s=LONG;

* l_choc= U_choc*(temps passe);
* l-sur = U-sur*(temps passe);
* vl_rar = (u-c)[Rho_z3];
* l-rar = vl_rar*(temps passe);
* l_g= (u-c)[Rho_g]*(temps passe)=-c[Rho_g]*(temps passe);
* NB:   u[Rho-z3]=u[Rho-z2]=U_sur;
*       u[Rho_g]=0;
*       c[Rho_g]=c_g;
A1=gam_g*P_g/(Rho_g**gam_g);
A1=A1**0.5;
celesp=(gam_g-1)/2.0;
c_g=A1*(Rho_g**celesp);
A2=1/celesp*A1*(Rho_g**celesp);
A3=(gam_g+1.0)/(gam_g-1.0)*A1;
vl_rar = A2 -(A3* (Rho_z3**celesp));
vl_g= A2 -(A3* (Rho_g**celesp));
message 'c_g=' c_g;
message 'vl_g=' vl_g;
message 'vl_rar =' vl_rar;
l_choc= U_choc*tmps;

```

```

l_sur = U_sur*tmps;
l_rar =vl_rar*tmps;
l_g=vl_g*tmps;
x_s=LONG;
x_choc=x_s+l_choc;
x_sur=x_s+l_sur;
x_rar=x_s+l_rar;
x_g=x_s+l_g;
* message 'l_choc=' l_choc;
* message 'l_sur =' l_sur;
* message 'l_rar =' l_rar;
* message 'l_g =' l_g;
message 'x_choc=' x_choc;
message 'x_sur =' x_sur;
message 'x_rar =' x_rar;
message 'x_g =' x_g;

* Creation des LISTREEL des solutions exactes definies sur COURB;

RHOEX= PROG;
PEX = PROG;
UEX = PROG;
REPETER BLOC1 (DIME XX);
  XCEL = EXTR XX &BLOC1;
  SI(XCEL < x_g);
  RHOCEL=Rho_g;
  PCEL = P_g;
  UCEL = 0.0;
  SINON;
  SI(XCEL < x_rar);
  XCEL1=XCEL-x_s;
  RHOCEL=XCEL1*(-1)/tmps+A2;
  RHOCEL=RHOCEL/A3;
  RHOCEL=RHOCEL**(1.0/celesp);
  PCEL=(RHOCEL/Rho_g)**gam_g;
  PCEL=PCEL*P_g;
  UCEL=A1/celesp;
  UCEL=UCEL*((Rho_g**celesp)-(RHOCEL**celesp));
  SINON;
  SI(XCEL < x_sur);
  RHOCEL=Rho_z3;
  PCEL=P_z3;
  UCEL=U_sur;
  SINON;
  SI(XCEL < x_choc);
  RHOCEL=Rho_z2;
  PCEL=P_z2;
  UCEL=U_sur;
  SINON;
  RHOCEL=Rho_d;
  PCEL = P_d;
  UCEL = 0.0;
  FINSI;
  FINSI;
  FINSI;
  FINSI;

```



```

ENCEL = NOMC 'SCAL' RV.INCO.'EN';
ETCEL = KOPS ENCEL '-' CEL;
CEL = KOPS CEL1 '/' RHOCEL;
UCEL = KOPS CEL '**' 0.5;
GAMCEL = NOMC 'SCAL' RV.INCO.'GAMMA';
CEL = KOPS GAMCEL '-' 1.0;
PCEL = KOPS ETCEL '*' CEL;
C_CEL = KOPS PCEL '/' RHOCEL;
C_CEL = KOPS C_CEL '*' GAMCEL;
C_CEL = KOPS C_CEL '**' 0.5;
CEL = KOPS UCEL '+' C_CEL;
CEL = MAXI CEL;
dmin = RV.INCO. 'DMIN';
DTI = dmin/cel;
RV.INCO.'PASCRI' = RV.INCO.'PASCRI' ET (PROG DTI);
*
* ----- Valeur du Pas de Temps initial
* -----
TPSI = rv.'TPSI' ;
*
* ----- On modifie la table PASDETPS
* -----
rv.PASDETPS.'DELTAT' = DTI;
rv.PASDETPS.'DELTAT-1' = DTI;
rv.PASDETPS.TPS = TPSI;
*
* ----- Valeur de la CFL
* -----
CFL = rv.'ALFA' ;
*
* ----- Nombre de pas de temps maxi pour le calcul
* -----
ITMA = rv.'ITMA' ;
*
* ----- Frequence d'impression
* -----
FIDT = rv.'FIDT' ;
*
* ----- Ordre en Temps et en Espace (1 par défaut ou 2)
* -----
ord_tps = 1 ;
ord_esp = 1 ;
*
si ((exist rv 'ORDRETPS')) ;
ord_tps = RV.ORDRETPS ;
finsi;
*
si ((exist rv 'ORDREESP')) ;
ord_esp = RV.ORDREESP ;
finsi;
*
* ----- Table contenant les inconnues
* -----
KIZT = rv.'INCO' ;
*
* ----- Table Domaine

```

```

* -----
KDOMA = rv.'DOMAINE' ;
*
* ----- Création d'une table pour les historiques
* -----
si (non(exist rv 'HIST')) ;
rv.'HIST' = table 'KHIS' ;
finsi ;
HIST = rv.'HIST';
*
* ----- si KIZD n'existe pas on crée alors les CHPO contenant les
* matrices "masse" diagonales (en fait un chpo centre contenant
* le volume des éléments)
*
si (non(exist rv kizd)) ;
kdia rv ;
finsi ;
KD = rv.'KIZD' ;
*
* ----- si KIZG n'existe pas on crée alors la table contenant les
* "incréments" (Flux)
*
si (non(exist rv kizg)) ;
rv.kizg = table 'KIZG' ;
finsi ;
KIZG = rv.'KIZG' ;
*
I = 0 ;
*
* ----- Boucle Sur les Pas de Temps -----
* =====
REPETER BLOC1 ITMA ;
*-----
*
I = &BLOC1 ;
*
si ( I EGA 1) ;
MESS '-----';
MESS 'Ordre en Espace : ' ord_esp;
MESS 'Ordre en Temps : ' ord_tps;
MESS '-----';
sinon;

*
***** Recherche du plus petit pas de temps *****
* -----
RHOCEL = NOMC 'SCAL' RV.INCO.'RN';
GCEL = RV.INCO.'GN';
GCELSQ = KOPS GCEL 'PSCA' GCEL;
CEL1 = KOPS GCELSQ '/' RHOCEL;
CEL = KOPS CEL1 '/' 2.0;
ENCEL = NOMC 'SCAL' RV.INCO.'EN';
ETCEL = KOPS ENCEL '-' CEL;
CEL = KOPS CEL1 '/' RHOCEL;
UCEL = KOPS CEL '**' 0.5;
GAMCEL = NOMC 'SCAL' RV.INCO.'GAMMA';

```

```

CEL = KOPS GAMCEL '-' 1.0;
PCEL = KOPS ETCEL '*' CEL;
C_CEL = KOPS PCEL '/' RHOCEL;
C_CEL = KOPS C_CEL '*' GAMCEL;
C_CEL = KOPS C_CEL '**' 0.5;
CEL = KOPS UCEL '+' C_CEL;
CEL = MAXI CEL;
dmin = RV.INCO. 'DMIN';
DTI = dmin/cel;
RV.INCO.'PASCRI' = RV.INCO.'PASCRI' ET (PROG DTI);
rv.PASDETPS.'DELTAT' = DTI;

    finsi;
*
* ----- Si l'on est d'ordre 2 en espace on calcule les pentes
*
* si (ord_esp EGA 2) ;
*
*     PT = PENT 'MINMOD' KDOMA KIZT.'RN' KIZT.'GN' KIZT.'EN' ;
*
*
* ----- Si l'on est d'ordre 2 en Temps on calcule les matrices
*          Jacobiennes ==> calcul de l'estimation DU au pas n+1/2
*
*     si (ord_tps EGA 2) ;
*     si (i > 1) ;
*         PT DU = PRET 'ENERGIE' ( DTI / 2. ) KDOMA
*                 KIZT.'RN' KIZT.'GN' KIZT.'EN' KIZT.'GAMMA' PT ;
*
*
*     xrn = KIZT.'RN';
*     xgn = KIZT.'GN';
*     xen = KIZT.'EN';
*
* ---- On change les noms des composantes pour avoir les memes
*       noms pour l'incrément et U sinon il n y a pas addition
*       possible composantes à composantes
*
*     DRO1 = exco 'DRO' DU 'SCAL';
*     DROE1 = exco 'DROE' DU 'SCAL';
*
*     si ( (VALE DIME) EGA 2) ;
*     listm1 = MOTS 'DROU' 'DROV' ;
*     listm2 = MOTS 'UX ' 'UY ' ;
*     finsi ;
*
*     si ( (VALE DIME) EGA 3) ;
*     listm1 = MOTS 'DROU' 'DROV' 'DROW';
*     listm2 = MOTS 'UX ' 'UY ' 'UZ ' ;
*     finsi ;
*
*     DROV1 = exco listm1 DU listm2 ;
*
*     KIZT.'RN' = (KIZT.'RN' + DRO1) ;
*     KIZT.'GN' = (KIZT.'GN' + DROV1) ;
*     KIZT.'EN' = (KIZT.'EN' + DROE1) ;
*

```

```

    finsi      ;
    finsi      ;
    PT = PENT 'MINMOD' KDOMA KIZT.'RN' KIZT.'GN' KIZT.'EN' ;
*
* ----- On range dans la Table Inco les Nouvelles Pentes
* -----
*           rv.inco.'PENT' = PT ;
*
* finsi      ;
*
*
* nbop = dime (rv.listoper) ;
*
* ----- Boucle sur les opérateurs
* =====
*           repeter bloc2 nbop ;
*
* ****      I = I + 1 ;
*
*           nomper = extr &bloc2 rv.LISTOPER ;
*           notable= mot (text (chai &bloc2 nomper) ) ;
*           mess ' exex oper ' notable ;
*           (text nomper) (rv.notable) ;
*
*           fin bloc2 ;
*
* ----- Fin de La Boucle sur Les Opérateurs -----
*
* ----- On fait l'incrément en Temps
* -----
*list ITMA;
*list I      ;
*
* ----- On IMPOSE LE PAS DE TEMPS
* -----
*           rv.PASDETPS.'DELTAT' = DTI;
*           rv.PASDETPS.'DELTAT-1' = DTI;
*
* ----- on avance au pas de temps suivant
* -----
*           avct KIZT KDOMA (rv.pasdetps) KIZG RV KD HIST CFL 'IMPR' FIDT ;
*
* mess ' PAS NUMERO : ' I      ;
*
* ----- On écrase la valeur de début de Pas avec la valeur calculée
* -----
*
*           KIZT.'RN' = KIZT.'RNI'      ;
*           KIZT.'GN' = KIZT.'GNI'      ;
*           KIZT.'EN' = KIZT.'ENI'      ;
*           Si (ord_tps EGA 2) ;
*           si (i > 1) ;
*           KIZT.'RN' = KIZT.'RNI' - DRO1 ;
*           KIZT.'GN' = KIZT.'GNI' - DROV1 ;
*           KIZT.'EN' = KIZT.'ENI' - DROE1 ;

```

```

        finsi;
        finsi;
*
* Impression (provisoire Pour debug)
*
        menage ;
* -----
FIN BLOC1 ;
*
* ----- FIN de Boucle Sur les Pas de Temps -----
*
* =====
*
FINPROC ;
*****
***** MAQUETTE *****
***** TEST DETO H2 *****
*****
*****
*
opti dime 2 elem qua4;
*
*
* -----
* = DONNEES INITIALES =
* -----

        ax = 100 ;
        ay = 1 ;
*
NITER = 300;
FAC_SIC = 0.2;
*
LONG = 1.0 ;
EPAI = 0.01;

message 'NX =' (2*ax);
message 'NITER =' NITER;
*
* NB : Xs = LONG;
*

p1 = 0 0;
p2 = LONG 0 ;
p3 = LONG EPAI ;
p4 = 0 EPAI ;
l12 =p1 d ax p2 ;
l23 =p2 d ay p3 ;
l34 =p3 d ax p4 ;
l41 =p4 d ay p1 ;
zone1 = l12 l23 l34 l41 daller plan ;
zone2 = (l12 l23 l34 l41 daller plan) plus (LONG 0);
*
zone1 = orient zone1 ;

```



```

zone2 = orient zone2 ;
angle1 = 0.0;
zone1 = zone1 tourner angle1 p1 ;
zone2 = zone2 tourner angle1 p1 ;
s = zone1 et zone2;
elim 0.0001 s;
    trac s;
*
* ----- Creation de la ligne Utilisée pour le Post-Traitement
*         reliant les points centres
*
    xinit = (LONG / (ax * 2.)) ;
    yinit = EPAI / (2. * ay) ;
    xfin = ( 2 * LONG) - xinit ;
    yfin = yinit ;
    PINI = xinit yinit;
    PFIN = xfin yfin ;
    iaux = (2 * ax) - 1 ;
    Courb = PINI d iaux PFIN;
    Courb = Courb tourner angle1 p1;
*
*   domaine pour la zone complete
*   -----
*
TOUT = 'DOMA' s 0.0001          ;
CHSUR = 'DOMA' TOUT 'SURFACE'  ;
CHNOR = 'DOMA' TOUT 'NORMALE'   ;
MCORI = 'DOMA' TOUT 'ORIENTAT' ;
CHdia = 'DOMA' TOUT 'DIAMIN'   ;
CHvol = 'DOMA' TOUT 'VOLUME'   ;
* diamin = mini CHdia;
diamin = LONG / ax;

*
*
*   domaine pour la zone 1
*   -----
*
T1 = 'DOMA' zone1 'INCL' TOUT 0.0001;
T1_CHSUR = 'DOMA' T1 'SURFACE'      ;
T1_CHNOR = 'DOMA' T1 'NORMALE'     ;
T1_MCORI = 'DOMA' T1 'ORIENTAT'    ;
T1_CHdia = 'DOMA' T1 'DIAMIN'     ;
T1_CHvol = 'DOMA' T1 'VOLUME'     ;
*
*
*   domaine pour la zone 2
*   -----
*
T2 = 'DOMA' zone2 'INCL' TOUT 0.0001;
T2_CHSUR = 'DOMA' T2 'SURFACE'      ;
T2_CHNOR = 'DOMA' T2 'NORMALE'     ;
T2_MCORi = 'DOMA' T2 'ORIENTAT'    ;
T2_CHdia = 'DOMA' T2 'DIAMIN'     ;
T2_CHval = 'DOMA' T2 'VOLUME'     ;
*

```

```

* opti donn 5 ;
*
*

*          -----
*          = DONNEES INITIALES =   Z O N E  N° 1
*          -----
*          =====
*
*  Fiche de validation
*
*  HP.
*  1) meme gas;
*  2) gas parfait;
*  3) Eg = Ed;
*  4) Vx(t=0)=0;
*  5) P_g>P_d(Pas limitative);
*
*
*  CONSTANTES
*  -----
*
P_g = 5.0E0;
Rho_g = 1.0;
P_d= 1.0E0;
Rho_d= Rho_g/P_g*P_d;
gam_g = 1.4;
gam_d = gam_g;

*c_g=c_d=2.65

debx_d=0.0;
deby_d=0.0;
debx_g=0.0;
deby_g=0.0;

P_INI = P_g ;
RO_INI = Rho_g ;
gam1 = gam_g ;
*
*
DEBX_INI = debx_g ;
DEBY_INI = deby_g ;
* RO * Energie
Eint0ini = (P_INI/(gam1 - 1d0) )+(DEBX_INI**2.0/2.0/RO_INI )
          + (DEBY_INI**2.0/2.0/RO_INI );
*
*
*  CREATION DES CHAMPOINTS
*  -----
*
*  Masse Volumique

```

```

* -----
ROO_1 = kcht t1 scal centre RO_INI ;
*
*
* debits
* -----
GN_1 = kcht t1 vect centre (DEBX_INI DEBY_INI);
*
* energie interne du gaz
* -----
E_1 = kcht t1 scal centre eint0ini ;
*
*
* -----
* = DONNEES INITIALES = Z O N E N° 2
* -----
*
* CONSTANTES
* -----
*
P_INI = P_d ;
RO_INI = Rho_d ;
gam2 = gam_d ;
*
*
DEBX_INI = debx_d ;
DEBY_INI = deby_d ;
* Ro * Energie
Eint0ini = (P_INI/(gam2 - 1d0) )+(DEBX_INI**2.0/2.0/RO_INI )
+ (DEBY_INI**2.0/2.0/RO_INI );
*
TAU = 1.E10 ;
Q = 0. ;
Tseuil = 350. ;
*
*
* CREATION DES CHAMPOINTS
* -----
*
* Masse Volumique
* -----
ROO_2 = kcht t2 scal centre RO_INI ;
*
*
* debits
* -----
GN_2 = kcht t2 vect centre (DEBX_INI DEBY_INI);
*
* énergie interne du gaz
* -----
E_2 = kcht t2 scal centre eint0ini ;
*
*
*****
*

```

```

*   on assemble les differents champoints pour les zones 1 et 2   *
*   -----                                                       *
*                                                                 *
*****
*
RN   = KCHT TOUT scal centre ( R00_1 et R00_2 ) ;
*
*
GN   =KCHT TOUT vect centre ( GN_1 et GN_2 );
*
*
EN = KCHT TOUT scal centre ( E_1 et E_2 );
*
*
*
*****
*
*           CREATION DE LA TABLE      "RV"
*
*****
*
RV = EQEX      TOUT          'ITMA' NITER          'ALFA' FAC_SIC
*
*           ^-----          ^-----          ^-----
*           table          nbre itération max          tolérance sur
*           domaine          le pas de temps?
*
*           'TPSI' 0.      'NOMVI' 'GNI'
*           ^-----          ^-----
*           Instant initial      Nom du CHPO
*           pour la reprise      de vitesse
*
OPTI          'VF' 'CONS'
*
*           ^-----          ^-----
*           on utilise les Eq. sous      type de schema ?
*           forme conservative en VF      Godunov : 1er ordre
*                                           VanLeer : 2eme ordre
**
** -----> operateur KONV : - estimation des valeurs aux faces
**           - chgt Rep. - reso. Riemann - chgt rep.
**           ==> Flux dans KIZG
*
*           'ZONE'      TOUT          'OPER'      'KONV'
*
*           ^-----
*           table domaine
*           de l'operateur
*
*           'RN'          'GN'          'EN'          'GAMMA'          'PENT'
*           ^-----          ^-----          ^-----          ^-----
*           masse          quantité de          énergie          gamma          Pentes
*           volumique          mouvement
*
*           'INCO'          RNi          GNi          ENi ;
*           ^-----          ^-----          ^-----
*           masse          quantite de          energie
*           volumique          mouvement

```

```

*           Pb de Riemann
*
*
*
* OPTION DE CALCUL POUR LA PROCEDURE EXEX
* -----
*
*   ----- Ordre  en temps et en espace
*   -----
*   IT      = 2 ;
*   IE      = 2 ;
*   RV.ORDRETPS = IT      ;
*   RV.ORDREESP = IE      ;
*
*
*   RV.INCO = TABLE  INCO ;
*
*
* ON "ASSEMBLE" LA TABLE GENERALE INCO
* -----
*
*   masse volumique
*
*   RV.INCO . 'RN' = RN      ;
*   RNI      = KCHT TOUT scal centre ( ROO_1 et ROO_2 ) ;
*   RV.INCO . 'RNI' = RNI    ;
*
*
*   gamma du mélange
*
*   GAMMA = kcht tout scal centre gam_d ;
*   RV.INCO . 'GAMMA' = GAMMA ;
*
*   les debits
*
*   RV.INCO . 'GN' = GN      ;
*   GNi      =KCHT TOUT vect centre ( GN_1 et GN_2 );
*   RV.INCO . 'GNI' = GNI    ;
*
*   énergie TOTALE
*
*   RV.INCO . 'EN' = EN      ;
*   ENi      = KCHT TOUT scal centre ( E_1 et E_2 );
*   RV.INCO . 'ENI' = ENI    ;
*
*   diamin
*
*   RV.INCO. 'DMIN' = diamin;
*
*   Pas critiques
*
*   RV.INCO.'PASCRI' = PROG;
*
*   NB tmps = alfa * ( resu RV.INCO.'PASCRI' );
*
* ***** Execution de la Procédure de Calcul Explicite *****

```

```

* -----
  Temps ;
                                EXEX RV ;
  Temps ;
*
*****
*
* -----  Masse Volumique et Energie Totale
* -----
RHO = rv.inco.'RNI';
RHOE= rv.inco.'ENI';
*
* -----  Vitesses suivant x et y et Energie
* -----
RHOVIT = rv.inco.'GNI';
RHOV  = exco 'UX' RHOVIT;
RHOY  = exco 'UY' RHOVIT;
ECIN  = 0.5d0 * ( (RHOV * RHOV) + (RHOY*RHOY) )/RHO ;
RHOEther=RHOE - ECIN;
*
* -----  Pression
* -----
PRESSION = (gam_d-1)*RHOEther ;
*
* -----  Enthalpie
* -----
ETHER= kops RHOETHER '/' RHO;
ENTH=gam_d*ETHER;
ENTHTOT=ENTH+(ECIN/RHO);
*
*-----  Entropie
* -----
ENTR=KOPS RHO '**' gam_d;
ENTR=KOPS PRESSION '/' ENTR;
*
  elim 0.001 Courb rv.domaine.centre ;
  evrho = evol 'CHPO' RHO Courb ;
  evrho = evol 'CHPO' RHO Courb ;
  evP   = evol 'CHPO' PRESSION Courb ;
  evS   = evol 'CHPO' ENTR Courb ;
  evHTOT = evol 'CHPO' ENTHTOT Courb ;

*****
***** SOLUTION EXACTE *****
*****

  tmps = rv.pasdetPS.'TPS' - rv.tpsi;
  mess 'tmps =' tmps;
  xx = EXTR evrho ABSC;

*
* SOLX = Table des solutions exactes.
*

```

```

SOLX = SOLEX Rho_g P_g P_d gam_g tmps xx LONG;

DEB_EX=SOLX.RHO_EX*SOLX.U_EX;
H_EX=gam_d/(gam_d-1)*(SOLX.P_EX/SOLX.RHO_EX);
HTOT_EX=SOLX.U_EX**2/2.0+H_EX;
S_EX=SOLX.RHO_EX**gam_d;
S_EX=SOLX.P_EX/S_EX;
*
evRHOex = EVOL ROUG 'MANU' 'X' xx 'RHO' SOLX.RHO_EX;
evPex = EVOL ROUG 'MANU' 'X' xx 'RHO' SOLX.P_EX;
evDEBex=EVOL ROUG 'MANU' 'X' xx 'RHO' DEB_EX;
evSex=EVOL ROUG 'MANU' 'X' xx 'P/RHO**g' S_EX;
evHTOTex = EVOL ROUG 'MANU' 'X' xx 'HTOT' HTOT_EX;

titre '2D Van Leer : Masse Volumique' 'temps' IT      'Espace' IE
      ' tmps' tmps ;
dess (evrho et evRHOex);
* dess (evrho et evRHOex) 'XBOR' .98 1.04 'YBOR' 5.1 5.40;
titre '2D Van Leer : Pression ' 'temps' IT      'Espace' IE
      ' tmps' tmps ;
dess (evP et evPex );
titre '2D Van Leer : RHO * U ' 'temps' IT      'Espace' IE
      ' tmps' tmps ;
dess (evrhov et evDEBex);
titre '2D Van Leer : HTOT ' 'temps' IT      'Espace' IE
      ' tmps' tmps ;
dess (evHTOT et evHTOTex );
titre '2D Van Leer : S ' 'temps' IT      'Espace' IE
      ' tmps' tmps ;
dess (evS et evSex );

*
* Erreurs (norme 2);
*
RHO_AP=EXTR evrho 'ORDO';
P_AP=EXTR evP 'ORDO';
DEB_AP=EXTR evrhov 'ORDO';
S_AP=EXTR evS 'ORDO';
HTOT_AP=EXTR evHTOT 'ORDO';
RHO_ER= SOLX.RHO_EX - RHO_AP;
RHO_ER=RHO_ER**2;
P_ER=SOLX.P_EX-P_AP;
P_ER=P_ER**2;
DEB_ER=DEB_EX-DEB_AP;
DEB_ER=DEB_ER**2;
S_ER=S_EX-S_AP;
S_ER=S_ER**2;
HTOT_ER=HTOT_EX-HTOT_AP;
HTOT_ER=HTOT_ER**2;

*
RHO_C=(RESU (SOLX.RHO_EX**2))**0.5;
P_C=(RESU (SOLX.P_EX**2))**0.5;
DEB_C=(RESU (DEB_EX**2))**0.5;
S_C=(RESU (S_EX**2))**0.5;
HTOT_C=(RESU (HTOT_EX**2))**0.5;

```

```
*
RHO_N2=RESU RHO_ER;
RHO_N2=RHO_N2**0.5/RHO_C;
P_N2=RESU P_ER;
P_N2=P_N2**0.5/P_C;
DEB_N2=RESU DEB_ER;
DEB_N2=DEB_N2**0.5/DEB_C;
S_N2=RESU S_ER;
S_N2=S_N2**0.5/S_C;
HTOT_N2=RESU HTOT_ER;
HTOT_N2=HTOT_N2**0.5/HTOT_C;

message 'ERREURS';
message 'RHO_N2 =' RHO_N2;
message 'P_N2   =' P_N2;
message 'DEB_N2 =' DEB_N2;
message 'HTOT_N2 =' HTOT_N2;
message 'S_N2   =' S_N2;
```


5.2 Ecoulement supersonique dans un canal avec rampe à 15°

5.2.1 Formulation Volumes Finis Explicite Conservative

<i>Opérateurs utilisés</i>	<i>PENT KONV AVCT</i>
<i>Méthodes numériques</i>	<i>Godounov étendu au second ordre par une approche de type M.U.S.C.L</i>
<i>Nature du problème</i>	<i>Ecoulement supersonique stationnaire à Mach 2 dans un canal avec contraction créée par une rampe inclinée à 15°. Chocs obliques, réflexions multiples, détente de Prandtl-Meyer, interaction détente/choc.</i>
<i>Classification</i>	<i>2D plan, compressible, stationnaire</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/15wedge.dgibi</i>

Description du problème :

Dans cet exemple [1], on s'intéresse à l'écoulement supersonique (Mach 2) stationnaire dans un canal plan dont la géométrie est représentée sur la figure 5.2.1.




FIG. 5.2.1 – Ecoulement supersonique ($M_\infty = 2.0$) dans un canal avec rampe à 15°

La présence d'une rampe inclinée à 15° induit la formation d'un choc oblique qui se réfléchit sur le mur supérieur. Vu les conditions de l'écoulement, il y a formation d'un choc droit (ou *Mach stem*) derrière lequel l'écoulement est subsonique. Une couche de cisaillement et d'entropie est également créée au niveau du point triple.

Modélisation :

Cet écoulement, gouverné par les équations d'Euler compressibles, est modélisé dans CASTEM2000 grâce à l'opérateur KONV. Cet opérateur discrétise les équations d'Euler dans une formulation Volumes Finis où les flux sont calculés à l'aide d'un solveur de

Riemann exact (solveur de Godounov). Une approche de type MUSCL (avec limiteur MinMod) permet d'augmenter la précision spatiale du schéma.

Les conditions aux limites de type paroi sont prises en compte directement dans le solveur. Pour ce qui est des entrées et des sorties de fluide, une approche classique basée sur la technique des 'cellules fantômes' permet d'approximer les flux à ces interfaces.

Résultats :

On effectue le calcul sur un maillage de quadrangles contenant 180×60 éléments, soit 10800 éléments. Le maillage est représenté sur la figure 5.2.3.

L'histoire de convergence est représentée sur la figure 5.2.2 et montre une décroissance du résidu de 3 ordres de grandeur suivie d'un plateau. Des non-linéarités du schéma (méthode de Newton pour le solveur de Riemann, limiteurs sur les pentes) peuvent expliquer ce phénomène.

Les figures 5.2.4 et 5.2.5 montrent les iso-valeurs de pression et du nombre de Mach pour l'écoulement. Les chocs sont bien capturés, sans aucune oscillation. On remarque également que si le maillage reste trop grossier au niveau du point triple et de la couche de cisaillement, le choc droit est également bien simulé puisque c'est à cet endroit que l'on retrouve une poche subsonique (nombre de Mach égal à 0.80). Pour trouver une valeur plus proche de la valeur théorique,

$$M_{aval} = \sqrt{\frac{1 + \frac{\gamma-1}{2} M_{amont}^2}{\gamma M_{amont}^2 - \frac{\gamma-1}{2}}} \approx 0.58$$

il faudrait évidemment pouvoir raffiner localement le maillage. Pour cela, on peut baser un critère de raffinement sur les gradients de vitesse, et notamment la divergence et le rotationnel de la vitesse [2]. Ces deux critères ont en effet l'avantage de détecter à la fois les zones de chocs et de détente, ainsi que les zones de cisaillement. La figure 5.2.6 montre ainsi les zones détectées par ces 2 critères, calculés à l'aide de l'opérateur PENT.

Références :

- 1– D. De Zeeuw and K.G. Powell. An Adaptively-Refined Cartesian Mesh Solver for the Euler Equations. *AIAA 10th Computational Fluid Dynamics Conference, Honolulu, Hawaii, 1991*. Paper AIAA-91-1542-CP.
- 2– H. Paillère, K.G. Powell and D. De Zeeuw. A Wave-Model-Based Refinement Criterion for Adaptive-Grid Computation of Compressible Flows. *AIAA 30th Aerospace Sciences Meeting & Exhibit, Reno, NV, 1992*. Paper AIAA-92-0322.






FIG. 5.2.2 – Histoire de convergence




MAILLAGE 180x60

FIG. 5.2.3 – Maillage de 10800 éléments




PRESSION : Min=0.176 Max=0.801

FIG. 5.2.4 – Iso-contours de la pression



MACH : Min=0.80 Max=2.00

FIG. 5.2.5 – Iso-contours du nombre de Mach



CRITERE DIV/ROT DE LA VITESSE

Jeu de données :

```

*****
* CALCUL DE L'ECOULEMENT SUPERSONIQUE STATIONNAIRE DANS UN CANAL *
* AVEC RAMPE INCLINEE A 15deg. *
* POWELL AND DEZEEUW, AIAA-91-1542-CP *
* FORMULATION VF COMPRESSIBLE EXPLICITE/IMPLICITE *
* *
* H. PAILLERE/P. GALON TTMF AOUT 1997 *
* *
* MODIF BECCANTINI MARS 97 (reconstruction lineaire exacte) *
* MODIF BECCANTINI MARS 01 (implicite) *
* MODIF, 10/07/01, syntaxe de PENT changée *
*****

'MESSAGE' 'A mettre a jours' ;
'FIN' ;

TYEL = 'QUA4' ;

'OPTION' 'DIME' 2 'ELEM' TYEL 'ISOV' 'SULI'
'ECHO' 1 'TRAC' 'X' ;

*****
*****
***** PROCEDURE PROLIM *****
*****
*****
*
*
**** Cas Euler mono-espece
*

'DEBPROC' PROLIM ;
'ARGUMENT' RV * TABLE ;

UN = RV . 'UN' ;
UNLIM = RV . 'UNLIM' ;
MAILLIM = 'EXTRAIRE' UNLIM 'MAILLAGE' ;
UNLEV = 'REDU' UN MAILLIM '*' (-1.0) ;

UN = UN '+' UNLEV '+' UNLIM ;

'FINPROC' UN ;

*****
**** Procedure CALC *****
*****
*
* Cette procédure, utilisée comme un opérateur, calcule la différence
* entre deux pas de temps toutes les (RVX . 'FCALC') itérations.
* L'évolution de cette différence (erreur absolue) au cours du temps
* est conservée en vue d'un post-traitement.
*
*

```

```

'DEBP' CALC ;
'ARGU' RVX*'TABLE' ;
*
RV = RVX . 'EQEX' ;
UN = RV . 'UN' ;
NOMDEN = 'EXTRAIRE' (RV . 'LISTCONS') 1 ;
MOMMOM = 'EXTRAIRE' (RV . 'LISTCONS') ('LECT' 2 3) ;
RN = 'EXCO' UN NOMDEN ;
GN = 'EXCO' UN NOMMOM ('MOTS' 'UX' 'UY') ;

*
'SI' ( 'EXIS' RVX 'COMPT') ;
  RVX . 'COMPT' = RVX . 'COMPT' + 1 ;
'SINON' ;
  RVX . 'COMPT' = 1 ;
  RVX . 'IT' = 'PROG' ;
  RVX . 'TPS' = 'PROG' ;
  RVX . 'ER1' = 'PROG' ;
  RVX . 'ERINF' = 'PROG' ;
  RVX . 'GNO' = 'COPI' GN ;
  RVX . 'CFLX' = 'PROG' ;
  RVX . 'CFLY' = 'PROG' ;
  RVX . 'RESLINY' = 'PROG' ;
'FINSI' ;

*
DD = RVX . 'COMPT' ;
COMBIEN = RVX . 'FCALC' ;
NN = DD '/' COMBIEN ;
LO = (DD '-' (COMBIEN '*' NN)) 'EGA' 0 ;
'SI' LO ;
  ERR = GN '-' (RVX . 'GNO') ;
  ERRINF = 'MAXIMUM' ERR 'ABS' ;
  ERR = ('XTX' ERR) '**' 0.5 ;
  ERR = ERR '/' ('NBNO' ('EXTRAIRE' GN 'MAILLAGE')) ;
  RVX . 'GNO' = 'COPI' GN ;
  EL1 = ('LOG' (ERR + 1.0E-50))/('LOG' 10.) ;
  ELINF = ('LOG' (ERRINF + 1.0E-50))/('LOG' 10.) ;
'MESSAGE' ;
'MESSAGE' 'Erreur de convergence' ;
'MESS' 'ITERATION' (RVX . 'COMPT') ' LOG10 ERR_1(GN)' EL1 ;
'MESS' 'ITERATION' (RVX . 'COMPT') ' LOG10 ERR_I(GN)' ELINF ;
'MESSAGE' ;
IT = 'PROG' RVX . 'COMPT' ;
ER1 = 'PROG' EL1 ;
ERINF = 'PROG' ELINF ;
TPS = 'PROG' (RV . 'PASDETPS' . 'TPS') ;
RVX . 'IT' = (RVX . 'IT') 'ET' IT ;
RVX . 'ER1' = (RVX . 'ER1') 'ET' ER1 ;
RVX . 'ERINF' = (RVX . 'ERINF') 'ET' ERINF ;
RVX . 'TPS' = (RVX . 'TPS') 'ET' TPS ;
'FINS' ;

*
* On peut augmunter le CFL (et MATTAB . 'RESID')
*
'SI' VRAI ;
  COMBIEN = RVX . 'FCFL' ;

```

```

NN = DD '/' COMBIEN ;
LO = (DD '-' (COMBIEN '*' NN)) 'EGA' 0 ;
'SI' LO ;
  CFL = PKONTAB . 'ALPHA' ;
  CFL = CFL '*' 2. ;
  CFL = 'MINIMUM' ('PROG' CFL (RVX . 'ALPMAX')) ;
  PKONTAB . 'ALPHA' = CFL ;
  'MESSAGE' ;
  'MESSAGE' ('CHAINE' 'CFL =' ((PKONTAB . 'ALPHA') '/' 2)) ;
* We decrease the linear iterations error
  MATTAB . 'RESID' = 'MAXIMUM' ('PROG'
    (MATTAB . 'RESID' '/' 3.1623) 1.0D-14) ;
  'MESSAGE' ('CHAINE' 'MATTAB . RESID = ' MATTAB . 'RESID') ;
  'FINSI' ;
  RVX . 'CFLX' = RVX . 'CFLX' 'ET' ('PROG' DD) ;
  RVX . 'CFLY' = RVX . 'CFLY' 'ET' ('PROG'
    ((PKONTAB . 'ALPHA') '/' 2)) ;
  RVX . 'RESLINY' = (RVX . 'RESLINY') 'ET'
    ('PROG' MATTAB . 'RESID') ;
'FINSI' ;

IRESU IJACO ='KOPS' 'MATRIK' ;
IALPDT = 'PROG' ;

'MENAGE' ;

'RESPRO' IRESU IALPDT ;

'FINP' ;

*****
*****
***** FIN PARTIE PROCEDURES *****
*****
*****

GRAPH = VRAI ;
* GRAPH = FAUX ;

*****
***** MAILLAGE *****
*****

NRAFF = 8 ;

NX1 = 1 '*' NRAFF ;
NX2 = 1 '*' NRAFF ;
NX3 = 4 '*' NRAFF ;
NY = 2 '*' NRAFF ;
DY = 2.0 '/' NY ;

H1 = 0.268 ;

A1 = (1. '-' DY) 0.0 ;
A2 = 1.0 0.0 ;
A3 = 2.0 0.0 ;

```

```

A4 = 3.0 H1 ;
A5 = 7.0 H1 ;
A6 = (7.0 '+' DY) H1 ;

B1 = (1.0 '-' DY) 2.0 ;
B2 = 1.0 2.0 ;
B3 = 2.0 2.0 ;
B4 = 3.0 2.0 ;
B5 = 7.0 2.0 ;
B6 = (7.0 '+' DY) 2.0 ;

BAS1 = A2 'DROI' NX1 A3 ;
BAS2 = A3 'DROI' NX2 A4 ;
BAS3 = A4 'DROI' NX3 A5 ;
HAU3 = B5 'DROI' NX3 B4 ;
HAU2 = B4 'DROI' NX2 B3 ;
HAU1 = B3 'DROI' NX1 B2 ;

COTEG = B2 'DROI' NY A2 ;
COTED = A5 'DROI' NY B5 ;

FRONTG = 'DALLER' (A1 'DROI' 1 A2) ('INVE' COTEG)
          (B2 'DROI' 1 B1) (B1 'DROI' NY A1) ;
FRONTG = FRONTG 'COUL' ROUG ;

FRONTD = 'DALLER' (A5 'DROI' 1 A6) (A6 'DROI' NY B6)
          (B6 'DROI' 1 B5) ('INVE' COTED) ;
FRONTD = FRONTD 'COUL' VERT ;

FRONT = FRONTG 'ET' FRONTD ;
'ELIMINATION' FRONT 1D-6;

DOMINT = 'DALLER' (BAS1 ET BAS2 ET BAS3) COTED
          (HAU3 ET HAU2 ET HAU1) COTEG 'PLAN' ;

DOMTOT = DOMINT ET FRONT ;
'ELIMINATION' DOMTOT (1D-6 '/' NRAFF) ;

$DOMTOT = 'DOMA' DOMTOT ;
$DOMINT = 'DOMA' DOMINT 'INCL' $DOMTOT (1D-6 '/' NRAFF) ;
$FRONT = 'DOMA' FRONT 'INCL' $DOMTOT (1D-6 '/' NRAFF) ;

'DOMA' $DOMINT 'XXNORMAF' ;
CHPONOR = $DOMINT . 'XXNORMAF' ;

*****
* CONDITIONS INITIALES ET LIMITES : ECOULEMENT SUPERSONIQUE M=2 *
* Adimensionalisation cf. These HP. p. 152 *
*****

gamscal = 1.4 ;
GAMN = 'KCHT' $DOMTOT 'SCAL' 'CENTRE' gamscal;

Minf = 2.0D0 ;
alpha = 0.D0 ;

```



```

'SI' ('NEG' ALPHA 0.0) ;
  'MESSAGE' 'Attention aux C.L.' ;
  'ERREUR' 21 ;
'FINSI' ;

rhoinf = 1.0D0 ;
rhouinf = 'COS' alpha ;
rhovinf = 'SIN' alpha ;
rhoEtinf = 2.0D0 '+' (gamscal '*' Minf '*' Minf
  '*' (gamscal '-' 1.0D0)) ;
rhoEtinf = rhoEtinf '/' (2.D0 '*' gamscal '*' Minf
  '*' Minf '*' (gamscal '-' 1.0D0)) ;

RNO = 'MANUEL' 'CHPO' ($DOMTOT . 'CENTRE') 1 'SCAL' rhoinf ;
GNO = 'MANUEL' 'CHPO' ($DOMTOT . 'CENTRE') 2 'UX' rhouinf
  'UY' rhovinf ;
RETNO = 'MANUEL' 'CHPO' ($DOMTOT . 'CENTRE') 1 'SCAL' rhoEtinf ;

VNO PNO = 'PRIM' 'PERFMONO' RNO GNO RETNO GAMN ;

VN2 = 'PSCAL' VNO VNO ('MOTS' 'UX' 'UY') ('MOTS' 'UX' 'UY') ;
C2 = GAMN '*' ( PNO '/' RNO) ;

MACH2 = VN2 '/' C2;
MACHNO = MACH2 '**' 0.5;

MOD1 = 'MODELISER' ($DOMTOT . 'MAILLAGE' ) 'THERMIQUE';

'SI' GRAPH ;

  CHM_RN = 'KCHA' $DOMTOT 'CHAM' RNO ;
  CHM_PN = 'KCHA' $DOMTOT 'CHAM' PNO ;
  CHM_VN = 'KCHA' $DOMTOT 'CHAM' VNO ;
  CHM_MACH = 'KCHA' $DOMTOT 'CHAM' MACHNO ;

  'TRACER' CHM_RN MOD1
    'TITR' ('CHAINE' 'RN at t=' 0.0);
  'TRACER' CHM_PN MOD1
    'TITR' ('CHAINE' 'PN at t=' 0.0);
  'TRACER' CHM_MACH MOD1
    'TITR' ('CHAINE' 'MACHN at t=' 0.0);
  'TRACER' CHM_VN MOD1
    'TITR' ('CHAINE' 'VN at t=' 0.0);

'FINSI' ;

NOMDEN = 'RN' ;
NOMMOM = 'MOTS' 'RUX' 'RUY' ;
NOMENE = 'RETN' ;
LISTCONS = ('MOTS' NOMDEN) 'ET' NOMMOM 'ET' ('MOTS' NOMENE) ;

UN =
  ('NOMC' NOMDEN RNO 'NATURE' 'DISCRET') 'ET'
  ('EXCO' ('MOTS' 'UX' 'UY') GNO NOMMOM
    'NATURE' 'DISCRET') 'ET'
  ('NOMC' NOMENE RETNO 'NATURE' 'DISCRET') ;

```

```

*****
*****
*****
***** La table RV *****
*****
*****
*****
RV = 'TABLE' ;
* Constant time step
* RV . 'DTIMP' = 1.191 ;
*
RV . 'UN' = UN ;
* Primitive formulation
RV . 'FORMPRIM' = FAUX ;
RV . 'LISTCONS' = LISTCONS ;
RV . 'LISTPRIM' = LISTCONS ;
**** Le domaine interne
RV . 'DOMINT' = $DOMINT . 'CENTRE' ;
*
RV . 'CLIM' = VRAI ;
RV . 'MAIFAN' = $FRONT . 'CENTRE' ;
*
RV . 'INST' = VRAI ;
RV . 'CONS' = FAUX ;
*
'SI' (RV . 'INST') ;
    RV . 'TFINAL' = 10.0D5 ;
    RV . 'NITMAEX' = 100 ;
'FINSI' ;
*
RV . 'NITMAIN' = 1 ;
RV . 'NITMIIN' = 1 ;
RV . 'EPSINT' = 3.0D-10 ;
*
RV . 'FEXT' = 20 ;
RV . 'FINT' = 1 ;
*
RV . 'LISTOPER' = 'MOTS' 'CALC' 'PKON' ;
RV . '1CALC' = 'TABLE' ;
CALCTAB = RV . '1CALC' ;
RV . '2PKON' = 'TABLE' ;
PKONTAB = RV . '2PKON' ;
*
RV . 'MATIDE' = 'KOPS' 'MATIDE' LISTCONS ($DOMTOT . 'CENTRE') ;
*
**** Parametres de la procedure PROLIM
*
RV . 'UNLIM' = 'REDU' (RV . 'UN') (RV . 'MAIFAN') ;
* Gas model (Procedure EXEXPRIM, EXEXCONV)
RV . 'GAMMA' = GAMN ;
* Bas Mach (Procedures EXEXDT, PKON)
RV . 'ORDTPS' = 1 ;
RV . 'PRECTEMP' = FAUX ;

```

```

* RV . DUALTEXP = VRAI ;
* RV . DUALTIMP = VRAI ;
RV . CFLDTAU = 1.0 ;
RV . 'DIAM' = ('DOMA' $DOMTOT 'DIAMIN') '*' 1.0 ;
* RV . 'CO1' = 'MANUEL' 'CHPO' ($DOMTOT . 'CENTRE')
*           1 'SCAL' (0.1 * U_G) '+'
*           ('MANUEL' 'CHPO' ($FRONT . 'CENTRE'))
*           1 'SCAL' (0. * U_G) ;
* RV . 'CO2' = 'MANUEL' 'CHPO' ($DOMTOT . 'CENTRE')
*           1 'SCAL' (0. * U_G) ;
*****
** CALCUL *****
*****
CALCTAB . 'EQEX' = RV ;
CALCTAB . 'FCALC' = 5 ;
CALCTAB . 'IMPL' = FAUX ;
* Doubling frequency for CFL
CALCTAB . 'FCFL' = 5 ;
* Maximum of CFL
CALCTAB . 'ALPMAX' = 10000.0 ;

*****
* PKON *****
*****
PKONTAB . 'EQEX' = RV ;
PKONTAB . 'GAZ' = 'PERFMONO' ;
PKONTAB . 'GAMN' = GAMN ;
PKONTAB . 'DOMA' = $DOMTOT ;
PKONTAB . 'LISTCONS' = LISTCONS ;
* PKONTAB . 'LISTPRIM' = LISTPRIM ;
PKONTAB . 'METHODE' = 'VLH' ;
*
*
PKONTAB . 'IMPL' = VRAI ;
'SI' (PKONTAB . 'IMPL') ;
    PKONTAB . 'TYPEJACO' = 'VLH' ;
'SI' ;
*
PKONTAB . 'ORDREESP' = 2 ;
'SI' ((PKONTAB . 'ORDREESP') 'EGA' 2) ;
    PKONTAB . 'LIMITEUR' = 'LIMITEUR' ;
    CHPVID MATVID = 'KOPS' 'MATRIK' ;
    PKONTAB . 'VLIM' = CHPVID ;
    CACCA PIPI GRADSCAL = 'PENTE' $DOMTOT 'CENTRE' 'EULESCAL'
        'NOLIMITE' ('EXCO' NOMDEN (RV . 'UN')) ;
    CACCA PIPI GRADVEC = 'PENTE' $DOMTOT 'CENTRE' 'EULEVECT'
        'NOLIMITE' ('EXCO' NOMMOM (RV . 'UN'))
        ('MOTS' 'UX' 'UY')) 'CLIM' (PKONTAB . 'VLIM') ;
    PKONTAB . 'GRADRN' = GRADSCAL ;
    PKONTAB . 'GRADPN' = GRADSCAL ;
    PKONTAB . 'GRADVN' = GRADVEC ;
'SI' ;
*
ELP1 = 'MANUEL' 'POI1' (123.56 124.87) ;
PKONTAB . 'FACELIM' = 'DIFF' ELP1 ELP1 ;
PKONTAB . 'ALPHA' = 10. ;

```

```

*****
* Inversion de la matrice *****
*****
* RV . 'LINEHIST' = historique de convergence des iterations internes
* RV . 'MATUPDAT' updating
* RV . 'MUPINI'   we update de matrix if the external iterations
*                are less than (RV . 'MUPINI')
* RV . 'MUPEXT'   external iterations updating
* RV . 'MUPINT'   internal iterations updating
* RV . 'MATAACC'  Broyden method?
* RV . 'MUPLIN'   We update the matrix if in the previous
*                internal iteration the number of linear
*                iterations to solve the system were bigger
*                than (RV . 'MUPLIN')
RV . 'LINEHIST' = 'PROG' ;
RV . 'MATUPDAT' = VRAI ;
'SI' ('NON' ( RV . 'MATUPDAT')) ;
  RV . 'MUPEXT' = 4000 ;
  RV . 'MUPINT' = 200 ;
  RV . 'MATAACC' = FAUX ;
  RV . 'MUPINI' = 0 ;
  RV . 'MUPLIN' = 80 ;
'FINSI' ;
RV . 'MATINV' = 'TABLE' 'METHINV' ;
MATTAB = RV . 'MATINV' ;
MATTAB . 'TYPINV' = 5 ;
MATTAB . 'IMPINV' = 0 ;
*
* Matrice pour assurer que la matrice à inverser est correctement
* assemblé
*
* MATTAB . 'MATASS'  definie en EXEXIM
* MATTAB . 'MAPREC'   "
* MATTAB . 'XINIT'   "
*
* Methode de numerotation de DDL
MATTAB . 'TYRENU' = 'RIEN' ;
MATTAB . 'PCMLAG' = 'APR2' ;
MATTAB . 'GMRESTR' = 500 ;
* ILU 3
* ILUT (dual truncation) 5
* Dans le cas ILUT il faut definir
* ILUTLFIL
* ILUTDTOL
MATTAB . 'PRECOND' = 3 ;
* MATTAB . 'ILUTLFIL' = 32. ;
* MATTAB . 'ILUTDTOL' = 0. ;
MATTAB . 'NITMAX' = 1000 ;
MATTAB . 'RESID' = 1.D-3 ;
* bcgstab
MATTAB . 'BCGSBTOL' = 1.0d-40 ;

'OPTION' DONN 5 ;

'TEMPS' 'ZERO' ;

```

```

EXEXIM RV ;
'TEMPS' ;

*
**** Les variables conservatives
*

'SI' (RV . 'INST') ;
    TFIN = RV . 'PASDETPS' . 'TPS' ;
'SINO' ;
    TFIN = 'INF' ;
'FINSI' ;

RN = 'EXCO' 'RN' (RV . 'UN') ;
GN = 'EXCO' ('MOTS' 'RUX' 'RUY') (RV . 'UN') ('MOTS' 'UX' 'UY') ;
RETN = 'EXCO' 'RETN' (RV . 'UN') ;

*
**** Les variables primitives
*

VN PN = 'PRIM' 'PERFMONO'
        RN GN RETN GAMN ;

CSON2 = (GAMN '*' PN) '/' RN ;
VN2 = 'PSCAL' VN VN ('MOTS' 'UX' 'UY') ('MOTS' 'UX' 'UY') ;
MACHN2 = VN2 '/' CSON2 ;
MACHN = MACHN2 '**' 0.5 ;

*
*** GRAPHIQUE DES SOLUTIONS
*

MOD1 = 'MODELISER' $DOMINT 'THERMIQUE';

'SI' GRAPH ;

CHM_RN = 'KCHA' $DOMINT 'CHAM' ('REDU' RN ($DOMINT . 'CENTRE')) ;
CHM_PN = 'KCHA' $DOMINT 'CHAM' ('REDU' PN ($DOMINT . 'CENTRE')) ;
CHM_VN = 'KCHA' $DOMINT 'CHAM' ('REDU' VN ($DOMINT . 'CENTRE')) ;
CHM_MACH= 'KCHA' $DOMINT 'CHAM' ('REDU' MACHN ($DOMINT . 'CENTRE')) ;

'TRACER' CHM_RN MOD1 ('CONTOUR' DOMINT)
'TITR' ('CHAINE' 'METO =' METO ' NBEL =' NTOTEL
' RN at t=' TFIN ' OE =' 2);
'TRACER' CHM_PN MOD1 ('CONTOUR' DOMINT)
'TITR' ('CHAINE' 'METO =' METO ' NBEL =' NTOTEL
' PN at t=' TFIN ' OE =' 2);
'TRACER' CHM_MACH MOD1 ('CONTOUR' DOMINT)
'TITR' ('CHAINE' 'METO =' METO ' NBEL =' NTOTEL
' MACHN at t=' TFIN ' OE =' 2);
'TRACER' CHM_VN MOD1 ('CONTOUR' DOMINT)
'TITR' ('CHAINE' 'METO =' METO ' NBEL =' NTOTEL
' VN at t=' TFIN ' OE =' 2);

```

```

RNC = 'ELNO' $DOMINT ('REDU' RN ($DOMINT . 'CENTRE'));
'TRACER' DOMINT RNC ('CONTOUR' DOMINT)
  'TITR' ('CHAINE' 'METO =' METO ' NBEL =' NTOTEL
  ' RN at t=' TFIN ' OE =' 2) 15 ;

'FINSI' ;

*
*** Test
*

PCEN1 = ($DOMTOT . 'CENTRE') 'POIN' 'PROC' (2.5 (H1 '/' 2.0)) ;
'SI' GRAPH ;
  'TRACER' (DOMTOT 'ET' PCEN1) 'TITRE'
  'Test 1: solution derrier le premier choc' ;
'FINSI' ;
ro1 = 'EXTRAIRE' RN 'SCAL' PCEN1 ;

erro = ((ro1 '-' 1.7289) 'ABS') '/' 1.7289 ;
'SI' (erro > 1.0D-1) ;
  'ERREUR' 5 ;
'FINSI' ;

p1 = 'EXTRAIRE' PN 'SCAL' PCEN1 ;

erro = ((p1 '-' 0.3919) 'ABS') '/' 0.3919 ;
'SI' (erro > 1.0D-1) ;
  'ERREUR' 5 ;
'FINSI' ;

m1 = 'EXTRAIRE' MACHN 'SCAL' PCEN1 ;

erro = ((m1 '-' 1.44572) 'ABS') '/' 1.44572 ;
'SI' (erro > 1.0D-1) ;
  'ERREUR' 5 ;
'FINSI' ;

* Derrier le mach stem on a le mach min
* MMIN = 0.577350
* Il faut avoir un maillage assez fin pour l'avoir

MMIN = 'MINIMUM' MACHN ;
'MESSAGE' ('CHAINE' 'MMIN_ex =' 0.577350) ;
'MESSAGE' ('CHAINE' 'MMIN =' MMIN) ;

'SI' (MMIN > 1.5) ;
  'ERREUR' 5 ;
'FINSI' ;

'FIN' ;

```


5.3 Injection d'un mélange de gaz dans une enceinte fermée

5.3.1 Formulation Eléments Finis Semi-Implicite Non-Conservative

<i>Opérateurs utilisés</i>	<i>NSKE, PRESSION, TSCAL</i>
<i>Méthodes numériques</i>	<i>EFM1, approximation P1 ou Q1, schéma SUPG, bas Mach</i>
<i>Nature du problème</i>	<i>Injection d'un mélange de gaz dans une enceinte fermée. Calcul de l'évolution de la masse totale et de la pression, et comparaison avec théorie.</i>
<i>classification</i>	<i>2D plan, compressible, bas Mach</i>
<i>jeux de données</i>	<i>/u2/castem/dgibi/basmachQ.dgibi basmachT.dgibi</i>

Description du problème :

Cet exemple concerne l'injection d'un mélange de gaz chauds ($T_{jet} = 650$ K) (air, hydrogène et vapeur) dans une enceinte fermée, composée de 2 compartiments interconnectés et initialement remplie d'air à 400 K.




FIG. 5.3.1 – Injection d'un mélange de gaz chauds dans une enceinte fermée

Cet écoulement est un écoulement compressible caractérisé par des nombres de Mach très inférieur à un. Pour le traiter, un algorithme “faibles Mach” a été développé, lié à une discrétisation par éléments finis, dans lequel l'équation de continuité est remplacée par une équation de Poisson sur la pression dynamique. La masse est quant à elle évaluée à partir de l'équation d'état des gaz parfaits [1,2].

Cette modélisation, basée sur l'approximation “faibles Mach” des équations de Navier-

Stokes, couplées au modèle de turbulence $k-\varepsilon$, comporte donc les équations suivantes :

$$\begin{aligned} \frac{\partial Y_\ell}{\partial t} + \underline{u} \cdot \nabla Y_\ell &= \left(D_\ell + \frac{\nu_t}{Sc_t} \right) \Delta Y_\ell + \dot{r}_\ell \\ \frac{\partial \rho \underline{u}}{\partial t} + \nabla \cdot (\rho \underline{u} \otimes \underline{u}) &= -\nabla p' + (\rho - \bar{\rho}) \underline{g} + \nabla \cdot \left(\underline{\tau} + \underline{\tau}_t \right) \\ \frac{\partial T}{\partial t} + \underline{u} \cdot \nabla T - \frac{1}{\rho c_p} \frac{dP}{dt} &= \left(\frac{\lambda}{\rho c_p} + \frac{\nu_t}{Pr_t} \right) \Delta T + \frac{1}{\rho c_p} \Omega \\ \rho &= P / (RT) \\ \frac{1}{\gamma - 1} V \frac{dP}{dt} &= (c_p T Q)_{\text{jet}} - \oint_{\partial V} \underline{q} \cdot d\underline{n} + \iint_V \Omega \, dV \\ \Delta p' &= \frac{1}{\Delta t} \left[\rho \left(\frac{1}{P} \frac{dP}{dt} - \frac{1}{RT} \frac{dRT}{dt} \right) + \nabla \cdot (\rho \underline{u}) \right] \\ &\quad + \nabla \cdot \left((\rho - \bar{\rho}) \underline{g} + \nabla \cdot \left(\underline{\tau} + \underline{\tau}_t \right) - \nabla \cdot (\rho \underline{u} \otimes \underline{u}) \right) \end{aligned}$$

Ce système est résolu grâce aux opérateurs Eléments Finis NSKE (option conservative) portant sur l'équation de quantité de mouvement et les équations de la turbulence, et TSCAL (option non-conservative) portant sur l'équation d'énergie et de transport des espèces chimiques.

Analyse des résultats :

On s'intéresse à une injection de 10 kg/s d'un mélange constitué d'air (60% en vol.), d'hydrogène (10%), le reste étant de la vapeur d'eau surchauffée. On effectue la simulation sur 2.8 secondes, et on s'intéresse à l'évolution de la masse contenue dans l'enceinte.

La figure 5.3.2 montre la comparaison entre la masse estimée et la masse réellement injectée. L'accord entre les 2 courbes est excellent. La figure 5.3.3 montre les iso-valeurs des courbes de concentration d'hydrogène dans l'enceinte, et la figure 5.3.4 le champ de vitesse correspondant.

Références :

- 1- H. Paillère, L. Dada, F. Dabbene, J.-P. Magnaud and J. Gauvain, "Development of Hydrogen Distribution and Combustion Models for the Multi-Dimensional/Lumped-Parameter TONUS code", Proc. 8th Int. Topical Meeting on Nuclear Reactor Thermal-Hydraulics, Kyoto, Japan, Sept.30-Oct.4, 1997.
- 2- H. Paillère and J.-P. Magnaud, "A Finite Element Elliptic Flow Solver for Low Mach Number Compressible Flow", 10th International Conference on Finite Elements in Fluids, Tucson AZ, USA, January 5-8, 1998.




FIG. 5.3.2 – Evolution de la masse dans l'enceinte : prédiction numérique et théorie




FIG. 5.3.3 – Iso-valeurs du champ de concentration d'hydrogène au temps $t = 2.8$ s, comprises entre 10% (partie sombre) et 0% (partie claire)




FIG. 5.3.4 – Champ de vitesse au temps $t = 2.8$ s (vitesse max. égale à 5.2 m/s)

Jeu de données : basmachQ.dgibi

```
*****
* JEU DE DONNEES POUR TESTER LE BON FONCTIONNEMENT DES OPERATEURS *
* NSKE, PRESSION, TSCAL. INJECTION D'UN MELANGE DE GAZ DANS UN *
* VOLUME FERME ET PRESSURISATION. ALGORITHME 'FAIBLES MACH' POUR *
* LES EQUATIONS NAVIER-STOKES FAIBLEMENT COMPRESSIBLES + MODELE *
* DE TURBULENCE K-E. ON TESTE SUR LA CONSERVATION DE LA MASSE *
* H. PAILLERE SEMT/TMF AVRIL 1997 *
* FORMULATION EF ELEMENTS QUA8 *
*****
```

```
*****
* PROCEDURE PROCINJ *
*****
```

```
DEBPROC PROCINJ ;
ARGU RVX*'TABLE' ;
```

```
RV = RVX.'EQEX' ;
$MT = RV.'DOMAINE' ;
```

* recuperation des anciennes conditions aux limites :

```
c1_YH2 = 'EXCO' 'YH2' RV.'CLIM' 'YH2' ;
c1_YO2 = 'EXCO' 'YO2' RV.'CLIM' 'YO2' ;
c1_YH20 = 'EXCO' 'YH20' RV.'CLIM' 'YH20' ;
c1_GNX = 'EXCO' '1GN' RV.'CLIM' '1GN' ;
c1_GNY = 'EXCO' '2GN' RV.'CLIM' '2GN' ;
c1_KN = 'EXCO' 'KN' RV.'CLIM' 'KN' ;
c1_EN = 'EXCO' 'EN' RV.'CLIM' 'EN' ;
c1_PSI = 'EXCO' 'PSIN' RV.'CLIM' 'PSIN' ;
```

* modification des conditions aux limites :

```
RHOJet = RV.INCO.'PM'/PSIJET ;
UJet = GJet/RHOJet ;
```

```
KJet = 0.05*UJet*UJet ;
EJet = 0.02*(UJet**3.)/DINJ ;
```

```
c1_KN = 0.D0 * c1_KN + KJet ;
c1_EN = 0.D0 * c1_EN + EJet ;
```

* reimposition des conditions aux limites :

```
RV.'CLIM' = c1_YH2 + c1_YO2 + c1_YH20 + c1_GNX + c1_GNY +
            c1_KN + c1_EN + c1_PSI ;
```

```
detr c1_YH2 ;
detr c1_YO2 ;
detr c1_YH20 ;
detr c1_GNX ;
detr c1_GNY ;
```

```

detr cl_KN ;
detr cl_EN ;
detr cl_PSI ;
FINSI ;

FINPROC ;

*****
* PROCEDURE CALCUL1
*****
DEBPROC CALCUL1 ;
ARGU RVX*'TABLE' ;
RV = RVX.'EQEX' ;
RVP = RV.'PRESSION' ;
$MT = RV.'DOMAINE' ;

RV.PASDETPS.'DELTAT' = 0.01 ;

* filtres sur fractions massiques et calcul de la fraction massique d'azote

YH2 = RV.INCO.'YH2' ;
YO2 = RV.INCO.'YO2' ;
YH2O = RV.INCO.'YH2O' ;
YH2 = KOPS YH2 '|<' 0. ;
YO2 = KOPS YO2 '|<' 0. ;
YH2O = KOPS YH2O '|<' 0. ;

RV.INCO.'YH2' = YH2 ;
RV.INCO.'YO2' = YO2 ;
RV.INCO.'YH2O' = YH2O ;
RV.INCO.'YN2' = KOPS 1.0 '-' RV.INCO.'YH2O' ;
RV.INCO.'YN2' = KOPS RV.INCO.'YN2' '-' RV.INCO.'YH2' ;
RV.INCO.'YN2' = KOPS RV.INCO.'YN2' '-' RV.INCO.'YO2' ;

* calcul de rho à partir de la loi d'état des gaz parfaits

RV.INCO.'RHO' = KOPS (RV.INCO.'PM') '/' (RV.INCO.'PSIN') ;
RV.INCO.'RHOC' = NOEL $MT RV.INCO.'RHO' ;
RV.INCO.'RHOC' = KCHT $MT 'SCAL' 'CENTRE' (RV.INCO.'RHOC') ;
RHOM = (SOMT (KOPS RV.INCO.'RHOC' '*' VOL))/VOLT ;
RV.INCO.'RHOM' = RHOM ;

* calcul du terme (rho-rhom)*g de la QDM

DRHO = KOPS (RV.INCO.'RHOC') '-' (RV.INCO.'RHOM') ;
rogx = KCHT (RV.'DOMAINE') SCAL CENTRE 0. ;
rogy = KCHT (RV.'DOMAINE') SCAL CENTRE (-9.81*DRHO) ;
rogx = nomc 'UX' rogx 'NATU' 'DISCRET' ;
rogy = nomc 'UY' rogy 'NATU' 'DISCRET' ;
RV.INCO.'ROG' = KCHT (RV.'DOMAINE') VECT CENTRE (rogx ET rogy) ;

* calcul la vitesse à partir de la qdm

GN = RV.INCO.'GN' ;
gnx= kcht (rv.'DOMAINE') scal sommet (exco 'UX' gn) ;
gny= kcht (rv.'DOMAINE') scal sommet (exco 'UY' gn) ;

```

```

unx = kops gnx '/' RV.INCO.'RHO' ;
uny = kops gny '/' RV.INCO.'RHO' ;
unx = nomc 'UX' unx 'NATU' 'DISCRET' ;
uny = nomc 'UY' uny 'NATU' 'DISCRET' ;
un = kcht (RV.'DOMAINE') vect sommet (unx et uny) ;
RV.INCO.'UN' = un ;

* calcul de nut à partir de mut et de rho

RV.INCO.'NUT' = KOPS (RV.INCO.'MUT') '/' (RV.INCO.'RHOC') ;

FINPROC ;

*****
* PROCEDURE CALCUL2
*****
DEBPROC CALCUL2 ;
ARGU RVX*'TABLE' ;

RV = RVX.'EQEX' ;
RVP = RV.'PRESSION' ;
$MT = RV.'DOMAINE' ;

* on récupère le pas de temps

Dt = RV.PASDETPS.'DELTAT' ;
Dt = Dt * RV.'ALFA' ;

* calcul l'intégrale se trouvant dans le terme Dp/Dt.

ZT = KOPS (RV.INCO.'PSIN') '-' (RV.INCO.'PSI1') ;
ZT = KOPS ZT '/' Dt ;
ZT = KOPS ZT '/' (RV.INCO.'PSIN') ;

ZZ = KOPS ZT '*' RV.INCO.'RHO' ;
ZZ = NOEL $MT ZZ ;
ZZ = KCHT $MT 'SCAL' 'CENTRE' ZZ ;
ZZT = SOMT ( KOPS ZZ '*' VOL ) ;

RV.INCO.'PSI1' = KCHT $MT 'SCAL' 'SOMMET' (RV.INCO.'PSIN') ;

* calcul de dPdt

q0 = 'DBIT' RV.INCO.'GN' $INJ ;
RV.INCO.'DEBIT' = q0 ;
dPdt = ZZT + q0 ;
dPdt = dPdt*(RV.INCO.'PM')/VOLT/RV.INCO.'RHOM' ;

* terme source de l'équation d'énergie

zuzu = (gamma-1.)/gamma ;

QC = NOEL $MT RV.INCO.'Q' ;
QC = KCHT $MT 'SCAL' 'CENTRE' QC ;
TOTO = KOPS QC '+' dPdt ;
TOTO = KOPS TOTO '/' RV.INCO.'RHOC' ;

```

```

TOTO = KOPS TOTO '*' zuzu ;
RV.INCO.'S' = KCHT $MT 'SCAL' 'CENTRE' TOTO ;

* terme source équation de Poisson

Z1 = dPdt/RV.INCO.'PM' ;
ZP = KOPS RV.INCO.'RHOC' '*' Z1 ;
ZS = KOPS ZP '-' ZZ ;
ZZS = KOPS ZS '*' VOL ;
ZZS = ZZS * (-1.0) ;
RV.INCO.'ZS' = KCHT $MT 'SCAL' 'CENTRE' ZZS ;

* calcul de l'évolution de la pression

RV.INCO.'PM' = RV.INCO.'PM' + (dPdt*dt) ;

* affichage informations toutes les 5 iterations

DD = RV.PASDETPS.'NUPASDT' ;
NN = DD/5 ;
LO = (DD-(5*NN)) EGA 0 ;
SI ( LO ) ;
MESSAGE '===== ' ;
MESSAGE 'TEMPS = ' RV.PASDETPS.'TPS' ' PRESSION = ' RV.INCO.'PM' ;
MESSAGE 'MASSE INI = ' MASINI ' MASSE TOTALE = ' (RV.INCO.'RHOM'*VOLT) ;
PT = PROG RV.'PASDETPS'.'TPS' ;
PP = PROG RV.INCO.'PM' ;
PMTO = PROG (RV.INCO.'RHOM'*VOLT) ;
RV.INCO.'TEMPS' = RV.INCO.'TEMPS' ET PT ;
RV.INCO.'PRESS' = RV.INCO.'PRESS' ET PP ;
RV.INCO.'MTOTA' = RV.INCO.'MTOTA' ET PMTO ;
MESSAGE '===== ' ;
FINSI ;

FINPROC ;

*****
* JEU DE DONNEES NURETH-8 : CALCUL DE DISTRIBUTION/COMBUSTION *
* ALGORITHME 'COMPRESSIBLE FAIBLE MACH' SEMI-IMPLICITE *
*          === DISTRIBUTION === *
*****

opti dime 2 ;
opti elem qua8 ;
opti isov sul1 ;

*=====*
*          GENERATION DU MAILLAGE          *
*=====*

L_A = 4.0 ;
H_A = 4.0 ;
L_B = 4.0 ;
H_B = H_A ;

HINJ = 1.0 ;

```

```

DINJ = 1.0 ;

HAB = 1. ;
DAB = 2.0 ;
LAB = 1. ;

DX = 1. ;
DY = 1. ;

epsi = (DX/10.) ;

*****
* COMPARTIMENT A *
*****

A7 = 0. 0. ;
A8 = HINJ 0. ;
A1 = (HINJ+DINJ) 0. ;
A2 = L_A 0.0 ;
A3 = L_A HAB ;
A4 = L_A (HAB+DAB) ;
A5 = L_A H_A ;
A6 = 0. H_A ;

N78 = 'ENTIER' (HINJ/DX) ;
N81 = 'ENTIER' (DINJ/DX) ;
N12 = 'ENTIER' ((L_A-HINJ-DINJ)/DX) ;
N23 = 'ENTIER' (HAB/DY) ;
N34 = 'ENTIER' (DAB/DY) ;
N45 = 'ENTIER' ((H_A-HAB-DAB)/DY) ;
N56 = N78 + N81 + N12 ;
N67 = N23 + N34 + N45 ;

A7A8 = A7 'DROI' N78 A8 ;
A8A1 = A8 'DROI' N81 A1 ;
A1A2 = A1 'DROI' N12 A2 ;
A2A3 = A2 'DROI' N23 A3 ;
A3A4 = A3 'DROI' N34 A4 ;
A4A5 = A4 'DROI' N45 A5 ;
A5A6 = A5 'DROI' N56 A6 ;
A6A7 = A6 'DROI' N67 A7 ;

BAS_A = A7A8 ET A8A1 ET A1A2 ;
DRO_A = A2A3 ET A3A4 ET A4A5 ;
HAU_A = A5A6 ;
GAU_A = A6A7 ;

COMP_A = 'DALLER' BAS_A DRO_A HAU_A GAU_A 'PLAN' ;

*****
* COMPARTIMENT B *
*****

B1 = (L_A+LAB) 0. ;
B2 = (L_A+LAB+L_B) 0. ;
B3 = (L_A+LAB+L_B) H_B ;

```

```

B6 = (L_A+LAB) H_B ;
B7 = (L_A+LAB) (HAB+DAB) ;
B8 = (L_A+LAB) HAB ;

N23 = 'ENTIER' (H_B/DY) ;
N36 = 'ENTIER' (L_B/DX) ;
N67 = 'ENTIER' ((H_B-HAB-DAB)/DY) ;
N78 = 'ENTIER' (DAB/DY) ;
N81 = N23 - N67 - N78 ;
N12 = N36 ;

B1B2 = B1 'DROI' N12 B2 ;
B2B3 = B2 'DROI' N23 B3 ;
B3B6 = B3 'DROI' N36 B6 ;
B6B7 = B6 'DROI' N67 B7 ;
B7B8 = B7 'DROI' N78 B8 ;
B8B1 = B8 'DROI' N81 B1 ;

BAS_B = B1B2 ;
DRO_B = B2B3 ;
HAU_B = B3B6 ;
GAU_B = B6B7 ET B7B8 ET B8B1 ;

COMP_B = 'DALLER' BAS_B DRO_B HAU_B GAU_B 'PLAN' ;

*****
* JONCTION AB *
*****

NX = 'ENTIER' (LAB/DX) ;
NY = 'ENTIER' (DAB/DY) ;

BAS_AB = A3 'DROI' NX B8 ;
DRO_AB = B8 'DROI' NY B7 ;
HAU_AB = B7 'DROI' NX A4 ;
GAU_AB = A4 'DROI' NY A3 ;

J_AB = 'DALLER' BAS_AB DRO_AB HAU_AB GAU_AB 'PLAN' ;

*****
* MAILLAGE COMPLET *
*****

COMP_A = COMP_A 'COUL' 'ROUG' ;
COMP_B = COMP_B 'COUL' 'VERT' ;

J_AB = J_AB 'COUL' 'BLAN' ;

MT = COMP_A ET COMP_B ET J_AB ;
'ELIM' epsi MT ;

'ORIE' MT ;
CNT = 'CONT' MT ;

*****
* TABLES DOMAINE *

```



```

*****

$MT = 'DOMA' MT epsi 'MACRO' ;

$INJ = 'DOMA' BAS_A 'MACRO' 'INCL' $MT epsi ;
$COMP_A = DOMA COMP_A 'MACRO' 'INCL' $MT epsi ;
$COMP_B = DOMA COMP_B 'MACRO' 'INCL' $MT epsi ;
$J_AB = DOMA J_AB 'MACRO' 'INCL' $MT epsi ;

MT = $MT.MAILLAGE ;

VOL = 'DOMA' $MT 'VOLUME' ;
VOLT = 'SOMT' VOL ;
VOL = KCHT $MT 'SCAL' 'CENTRE' VOL ;

'MESSAGE' 'VOLUME TOTAL = ' VOLT ;

*****
* FRONTIERES POUR CONDITIONS AUX LIMITES *
*****

MUR_HOR = A7A8 ET A1A2 ET BAS_AB ET B1B2 ET B3B6
          ET HAU_AB ET A5A6 ;
MUR_VER = A2A3 ET B8B1 ET B2B3 ET B6B7 ET A4A5 ET A6A7 ;
BRECHE = A8A1 ;

MUR_HOR = MUR_HOR 'COUL' 'VERT' ;
MUR_VER = MUR_VER 'COUL' 'ROUG' ;
BRECHE = BRECHE 'COUL' 'JAUNE' ;

*====*
*                DONNEES DU CALCUL DE DISTRIBUTION                *
*====*

*-----*
* masses molaires et constantes des gaz
*-----*
MH2 = 2.0E-3 ;
MO2 = 32.0E-3 ;
MN2 = 28.0E-3 ;
MH2O = 18.0E-3 ;

Rg = 8.313 ;

RH2 = Rg/MH2 ;
RO2 = Rg/MO2 ;
RN2 = Rg/MN2 ;
RH2O = Rg/MH2O ;

*-----*
* fractions molaires des differents constituants du melange
*-----*
XH2 = 0.0 ;
XH2O = 0.1 ;
XO2 = 0.21*(1.0 - XH2 - XH2O) ;
XN2 = 1.0 - XH2 - XH2O - XO2 ;

```

```

*-----
* masse molaire du melange et constante du gaz
*-----
M = (XH2*MH2) + (XO2*MO2) + (XN2*MN2) + (XH20*MH20) ;
R = Rg/M ;

*-----
* pression, temperature et densite du melange (loi d'etat GP)
*-----
P = 1.0E5 ;
T = 400.0 ;
PSI = R*T ;
RHO = P/PSI ;

*-----
* Caractéristiques du jet
*-----

XH2Jet = 0.1 ;
XH20Jet = 0.3 ;
XO2Jet = 0.21*( 1.0 - XH2Jet - XH20Jet ) ;
XN2Jet = 1.0 - XH2Jet - XH20Jet - XO2Jet ;
MJet = (XH2Jet*MH2) + (XH20Jet*MH20) + (XO2Jet*MO2) +
      (XN2Jet*MN2) ;
RJet = Rg/MJet ;
TJet = 650.0 ;
PSIJet = Rjet*Tjet ;
RhoJet = P/PSIJet ;
GJet = 10.*DINJ/(DINJ+(DX*0.5)) ;
UJet = GJet/RhoJet ;
KJet = 0.05*UJet*UJet ;
EJet = 0.02*(UJet**3.)/DINJ ;

*-----
* fractions massiques des differents constituants
*-----
YH2 = XH2*MH2/M ;
YO2 = XO2*MO2/M ;
YN2 = XN2*MN2/M ;
YH20 = XH20*MH20/M ;
YH2Jet = XH2Jet*MH2/MJet ;
YH20Jet =XH20Jet*MH20/MJet ;
YO2Jet = XO2Jet*MO2/MJet ;
YN2Jet =XN2Jet*MN2/MJet ;

*-----
* Cp des différents constituants et du mélange (supposés constants)
*-----
CPH2 = 13514.04 + (1.684537*T) ;
CPH20 = 1715.632 + (0.552805*T) ;
CPN2 = 1006.15 + (0.1387166*T) ;
CP02 = 907.580 + (0.1420522*T) ;

CP = (YH2*CPH2) + (YH20*CPH20) + (YO2*CP02) + (YN2*CPN2) ;

```

```

gamma = CP/(CP-R) ;

*-----
* calcul d'une vitesse de référence (pour les tracés)
*-----

uref = gamma*R*T ;
uref = uref**0.5 ;
uref = 0.01*uref ;
ampl = 1./uref ;

*-----
* Prandtl, Schmidt, viscosité (loi de Sutherland), conductivité thermique
* coefficients de diffusion moléculaire
*-----
Pr = 0.7 ;
Prt = 1.0 ;
Sct = 1.0 ;

Mu = 1.716E-5*((T/273.0)**1.5)*(273.0+110.5)/(T+110.5) ;
Nu = Mu/RHO ;
Lambda = Mu*CP/Pr ;

alpha = Lambda/RHO/CP ;

DH20 = 2.55E-5 ;
DH2 = 7.12E-5 ;
DO2 = 2.06E-5 ;

*=====
* TABLE EQEX CONTENANT LES EQUATIONS A RESOUDRE :
* 0) FILTRE K-E
* 1) QUANTITE DE MOUVEMENT --- OPERATEUR NSKE
* 2) EQUATION DE L'ENERGIE --- OPERATEUR TSCAL
* 3) TRANSPORT DES ESPECES --- OPERATEUR TSCAL
* AINSI QUE 2 PROCEDURES (CALCUL1 ET CALCUL2) POUR L'ALGORITHME
* 'FAIBLE MACH'
* 4) CONDITIONS AUX LIMITES
*=====

RV = 'EQEX' $MT 'ITMA' 30 'ALFA' 0.9
'ZONE' $MT 'OPER' 'PROCINJ'
'ZONE' $MT 'OPER' 'CALCUL1'
'ZONE' $MT 'OPER' 'FILTREKE' UJet DINJ 'INCO' 'KN' 'EN'
'OPTI' 'CONS'
'ZONE' $MT 'OPER' 'NSKE' 'RHOC' MU 'MUT' 'UN' 'ROG'
'INCO' 'GN' 'KN' 'EN'
'OPTI' 'NOCONS'
'ZONE' $MT 'OPER' 'TSCAL' alpha 'UN' 'S' 'NUT' Prt
'INCO' 'PSIN' ;

RV = 'EQEX' RV
'ZONE' $MT 'OPER' 'TSCAL' DH2 'UN' 0.0 'NUT' SCT
'INCO' 'YH2'
'ZONE' $MT 'OPER' 'TSCAL' DO2 'UN' 0.0 'NUT' SCT
'INCO' 'YO2'

```

```

'ZONE' $MT 'OPER'          'TSCAL' DH20 'UN' 0.0 'NUT' SCT
                          'INCO' 'YH20'
'ZONE' $MT 'OPER'          'CALCUL2' ;

RV = 'EQEX' RV
'CLIM' 'GN' 'UIMP' MUR_VER 0.0
      'GN' 'VIMP' MUR_HOR 0.0
      'GN' 'UIMP' BRECHE 0.0
      'GN' 'VIMP' BRECHE GJet
      'KN' 'TIMP' BRECHE KJet
      'EN' 'TIMP' BRECHE EJet ;

RV = 'EQEX' RV
'CLIM' 'PSIN' 'TIMP' BRECHE PSIJet
      'YH2' 'TIMP' BRECHE YH2Jet
      'YO2' 'TIMP' BRECHE YO2Jet
      'YH20' 'TIMP' BRECHE YH20Jet ;

*=====
* TABLE EQPR POUR L'EQUATION DE POISSON (SOLVEUR ELLIPTIQUE)
*=====

RVP = 'EQPR' $MT KTYPI 1
      'ZONE' $MT 'OPER' 'PRESSION' 'ZS'
      'PIMP' 0. ;

*=====
* TABLE INCO CONTENANT LES INCONNUES ET DONNEES DU PROBLEME
*=====

RV.'PRESSION' = RVP ;
RV.'NOMVI' = 'GN' ;

RV.'INCO' = 'TABLE' 'INCO' ;
RVP.'INCO' = RV.'INCO' ;

* vitesse et quantité de mouvement
RV.INCO.'UN' = kcht $MT VECT SOMMET (0. 0.);
RV.INCO.'GN' = kcht $MT VECT SOMMET (0. 0.);
* température
RV.INCO.'PSIN' = kcht $MT SCAL SOMMET PSI ;
RV.INCO.'PSI1' = kcht $MT SCAL SOMMET PSI ;
* variables modèle de turbulence
RV.INCO.'EN' = kcht $MT SCAL SOMMET 1.E-5 ;
RV.INCO.'KN' = kcht $MT SCAL SOMMET 1.E-5 ;
RV.INCO.'MUT' = kcht $MT SCAL CENTRE Mu ;
RV.INCO.'NUT' = kcht $MT SCAL CENTRE Nu ;
* masse volumique
RV.INCO.'RHO' = kcht $MT SCAL SOMMET RHO ;
RV.INCO.'RHOC' = kcht $MT SCAL CENTRE RHO ;
RV.INCO.'RHOM' = RHO ;
* fractions massiques
RV.INCO.'YH2' = kcht $MT SCAL SOMMET YH2 ;
RV.INCO.'YO2' = kcht $MT SCAL SOMMET YO2 ;
RV.INCO.'YN2' = kcht $MT SCAL SOMMET YN2 ;
RV.INCO.'YH20' = kcht $MT SCAL SOMMET YH20 ;
* termes sources (Poisson/énergie/QDM)

```

```
RV.INCO.'S'=kcht $MT SCAL CENTRE 0. ;
RV.INCO.'Q'=kcht $MT SCAL SOMMET 0. ;
RV.INCO.'ROG'=kcht $MT VECT CENTRE (0. 0.) ;
RV.INCO.'ZS' = KCHT $MT SCAL CENTRE 0.0 ;
* pression
RV.INCO.'PM' = P ;
* progressions contenant certaines données
RV.INCO.'TEMPS' = PROG 0. ;
RV.INCO.'PRESS' = PROG P ;
RV.INCO.'MTOTA' = PROG (RHO*VOLT) ;
* debit de masse
RV.INCO.'DEBIT' = 0. ;

MASINI = KOPS RV.INCO.'RHOC' '*' VOL ;
MASINI = SOMT MASINI ;

exec rv ;

MTHEO = MASINI + (RV.INCO.'DEBIT' * RV.PASDETPS.'TPS') ;
MESSAGE MTHEO ;
SI ( (MAXI RV.INCO.'MTOTA') < (0.95*MTHEO) ) ;
    ERREUR 5 ;
FINSI ;

FIN ;
```

Chapitre 6

Hydrogéologie

6.1 Convection naturelle dans un milieu géologique

6.1.1 Formulation Eléments Finis Mixtes avec Hybridation

<i>Méthodes numériques</i>	<i>EFMH, schéma en temps Crank-Nicolson</i>
<i>Nature du problème</i>	<i>Transfert thermique transitoire et circulation hydraulique par convection naturelle dans un milieu géologique saturé dûs à une source thermique sphérique.</i>
<i>opérateurs utilisés</i>	<i>MATP, SMTP, MHYB opt. MASSE, HYBP, HDEB, SQTP</i>
<i>classification</i>	<i>3D, thermique et hydrogéologie, convection naturelle, transitoire</i>

Présentation du problème :

On considère une source de chaleur uniforme, sphérique, décroissant exponentiellement dans le temps et située dans un milieu géologique homogène, isotrope, perméable, saturé et infiniment grand. On se propose d'obtenir par un calcul discret les évolutions spatio-temporelles de la pression et de la température pour en faire la comparaison avec la solution analytique du problème. Ce problème constitue un des benchmarks du projet HYDROCOIN [1].

Modélisation physique :

Les transferts thermiques de nature convective sont négligés étant donnée la prédominance des transferts conductifs. Les variations de la température sont donc indépendantes du régime hydraulique. Étant donnée la faible porosité de la roche, les propriétés thermiques du milieu homogène équivalent (fluide-solide) sont assimilées à celles de la roche. L'équation de la chaleur est prise en compte sous la forme suivante :

$$\rho C_p \frac{\partial T_r}{\partial t} = -\vec{\nabla} \cdot \vec{\phi} + Q$$

$$\vec{\phi} = -\lambda \vec{\nabla} T_r$$

- $\vec{\phi}$ flux diffusif ;
- $T_r = T - T_{ref}$ température relative ;
- T_{ref} température initiale du massif ;
- $Q = \frac{3Q_0}{4\pi A^3} e^{-\alpha t}$ puissance volumique ;
- t temps.

L'eau présente dans le milieu est mise en mouvement par les forces de flottabilité générées par la dilatation thermique locale du fluide. Le régime hydraulique est donc dépendant de l'évolution spatio-temporelle de la température. Les forces de flottabilité sont déduites de l'approximation de Boussinesq. On considère que l'écoulement est gouverné par la loi de Darcy auquel s'ajoute la contrainte d'incompressibilité du fluide :

$$\frac{\mu}{k} \vec{u} + \vec{\nabla} p_r = \rho_{ref} \vec{g} \beta T_r$$

$$\vec{\nabla} \cdot \vec{u} = 0$$

- \vec{u} flux diffusif;
- $p_r = p - p_{stat}$ pression relative;
- $p_{stat} = -\rho_{ref} \cdot |g| \cdot z$ pression hydrostatique;
- \vec{g} accélération de la gravité.

Le milieu géologique infini est représenté par un domaine sphérique de 1500 m de rayon. Conditions initiales et conditions aux limites sont les suivantes :

- $T_r(t = 0 \text{ s}) = 0$;
- $p_r(t = 0 \text{ s}) = 0$;
- $T_r(R = 1500 \text{ m}) = 0$;
- $p_r(R = 1500 \text{ m}) = 0$.

Les paramètres du problème sont les suivants :

Paramètre	Symbole	Valeur	Unité
Rayon de la sphère	A	250	m
puissance initial de la source	Q_0	10	MW
Constante de décroissance	α	$7,3215 \cdot 10^{-10}$	s^{-1}
Conductivité thermique de la roche	λ	2,51	$W \cdot m^{-1} \cdot K^{-1}$
Densité de la roche	ρ	2600	$kg \cdot m^{-3}$
Capacité thermique de la roche	Cp	879	$J \cdot kg^{-1} \cdot K^{-1}$
Perméabilité intrinsèque de la roche	k	10^{-16}	m^2
Référence pour la densité de l'eau	ρ_{ref}	992,2	$kg \cdot m^{-3}$
Coefficient de dilatabilité thermique	β	$3,85 \cdot 10^{-4}$	K^{-1}
Viscosité dynamique de l'eau	μ	$6,529 \cdot 10^{-4}$	$kg \cdot m^{-1} \cdot s^{-1}$
Porosité de la roche	Φ	10^{-1}	

Modélisation numérique :

Le maillage représente un quartier vertical de sphère d'angle 5° . Il ne comprend qu'un élément dans son épaisseur. Le maillage est composé d'éléments cubiques. Le pas d'espace est de 25 m dans la direction radiale en deça de 750 m de rayon. Au delà il passe à 50 m. La discrétisation spatiale des équations est réalisée avec une méthode d'éléments finis mixtes hybrides (EFMH). La discrétisation en temps de l'équation de la chaleur utilise un algorithme de Crank-Nicolson.




FIG. 6.1.1 – Maillage

Résultats :

Pressions et températures relatives obtenues par CASTEM2000 sont comparées aux solutions analytiques tabulées dans le document référencé [1]. Les figures 6.1.2 et 6.1.3 représentent les évolutions spatiales des deux variables selon le rayon vertical ascendant pour quatre temps. Les solutions discrètes sont représentées par des marqueurs. Les solutions analytiques sont représentées par un trait continu.

Références :

- 1- THE INTERNATIONAL HYDROCOIN PROJECT ; Level 1 : Code Verification - OCDE 88
- 2- Résolution des équations de Darcy par une méthode d'éléments finis mixtes hybrides - Rapport DMT.95/613 - F.DABBENE
- 3- Résolution des équations de Darcy exprimées en fonction de la pression et de la vitesse ; prise en compte des effets de densité - Rapport DMT.96/250 - L.V.BÉNET

Jeu de données :

```
OPTI ECHO 0 ;
OPTI TRAC PS ;
```

```
'DEBPROC' DARCYMOD TRANSI*'TABLE' ;
```

```
*-----*
* Cette procedure s'inspire de la procedure DARCYTRA
* Elle conserve l'informatique de l'algorithme transitoire
* calculant les transferts de chaleur.
* A cette partie s'ajoute la résolution du problème hydraulique,
* ce dernier calcul étant renouvelé à chaque pas de temps.
*
*
```




FIG. 6.1.2 – Évolution de la pression relative suivant le rayon vertical ascendant




FIG. 6.1.3 – Évolution de la température relative suivant le rayon vertical ascendant

```

*-----
* Phrase d'appel (en GIBIANE) :
*-----
*
* DARCYMOD TABLE ;
*
*-----
* Opérandes (à mettre dans TABLE) :
*-----
*
*-----
* |
* | Indice      Contenu
* |-----
* |
* |-----
* | Données physiques, géométriques et matérielles :
* |-----
* | 'SOUSTYPE'   'DARCY_TRANSITOIRE' (type MOT)
* |
* | 'MODELE'     Objet modèle (MMODEL créé par MODE)
* |
* | 'CONDUCTIVITE' Données physiques et matérielles :
* |                conductivité thermique (MCHAML créé par MATE)
* |
* | 'CAPACITE'   Capacité thermique (CHAMPOIN de support
* |                support DOMAINE.CENTRE) - Temporaire : Sera mis
* |                dans CONDUCTIVITE - Défaut 1.
* |
* | 'DOMAINE'    Références géométriques (TABLE créée par KDOM)
* |
* |-----
* | Conditions initiales :
* |-----
* |
* | 'TEMPS'      TABLE contenant à l'indice 0 la valeur du temps
* |                initial (FLOTTANT)
* |
* | 'TRACE_TEMPERATURE' TABLE contenant à l'indice 0 la trace de
* |                temperature initiale (CHAMPOIN)
* |
* | 'TEMPERATURE' TABLE contenant à l'indice 0 la temperature
* |                initiale (CHAMPOIN)
* |
* | 'FLUX'       TABLE contenant à l'indice 0 le flux initial
* |                (CHAMPOIN)
* |
* |-----
* | Conditions aux limites / chargements :
* |-----
* |
* | 'BLOCAGE'    Contient les matrices de blocage (RIGIDITE)
* |
*

```

```

* |'TEMPERATURE_IMPOSE' Valeurs des traces de température imposées
* |      (CHARGEMENT - Obligatoire si BLOCAGE)
* |
* |
* |'SOURCE'      source volumique de chaleur
* |      (CHARGEMENT - défaut 0.)
* |
* |-----
* |Algorithme en temps :
* |-----
* |
* |'TEMPS_CALCULES' Valeur des temps calculés (LISTREEL)
* |      Contient obligatoirement le temps final.
* |
* |'TEMPS_SAUVES' Valeur des temps sauvegardés (LISTREEL - défaut
* |      on sauve tout les pas de temps)
* |
* |'THETA'      Coefficient de relaxation compris entre 0. et 1.
* |      (theta-méthode) (FLOTTANT - défaut 1.)
* |      Possibilité de non-convergence lorsque theta<1/2
* |      Valeurs de theta généralement utilisées :
* |      Schéma de Euler explicite : 0.
* |      Schéma de Crank-Nicolson : 1/2
* |      Schéma de Galerkin      : 2/3
* |      Schéma de Euler implicite : 1.
* |-----
* |
* |
* |-----
* | Résultats (stockés dans TABLE) :
* |-----
* |
* |-----
* |
* | Indice      Contenu
* |-----
* |
* |'TEMPS'      TABLE contenant les temps sauvegardés (FLOTTANT)
* |
* |'TRACE_TEMPERATURE' TABLE contenant les traces de température
* |
* |'TEMPERATURE' TABLE contenant les températures (CHAMPOIN)
* |
* |'FLUX'      TABLE contenant les flux thermiques (CHAMPOIN)
* |
* |'TRACE_PRESSION' TABLE contenant les traces de pression
* |
* |'PRESSION'   TABLE contenant les pressions      (CHAMPOIN)
* |
* |'DEBIT'     TABLE contenant les débits (CHAMPOIN)
* |-----
* |
* |

```



```

si (exist TRANSI 'TEMPERATURE_IMPOSE') ;
  CHAIMP = TRANSI . 'TEMPERATURE_IMPOSE' ;
finsi ;
*
* SOURCE
si (exist TRANSI 'SOURCE') ;
  TERSOU = TRANSI . 'SOURCE' ;
finsi ;
*
*-----*
* RECUPERATION DES DONNEES POUR LE SCHEMA EN TEMPS *
*-----*
*
* THETA
'SI' ( 'EXISTE' TRANSI 'THETA' ) ;
  COEF = TRANSI . 'THETA' ;
'SINON' ;
  COEF = 1.DO ;
'FINSI' ;
*
* TEMPS_CALCULES
ICAL = 0 ;
IOK1 = 0 ;
'SI' ( 'EXISTE' TRANSI 'TEMPS_CALCULES' ) ;
  TPCAL = TRANSI . 'TEMPS_CALCULES' ;
  'ORDO' TPCAL ;
  TRANSI . 'TEMPS_CALCULES' = TPCAL ;
  DCAL = 'DIME' TPCAL ;
  TPSFIN = 'EXTR' TPCAL DCAL ;
  TPSO = 'EXTR' TPCAL 1 ;
  EPSO = TPSFIN - TPSO / DCAL * 1.D-6 ;
*
  'REPETER' BOU2 DCAL ;
  ICAL = ICAL + 1 ;
  TEMS = 'EXTRA' TPCAL ICAL ;
  'SI' ( TPSINI '<' TEMS ) ;
  'SI' ( TPSINI 'NEG' TEMS EPSO ) ;
  IOK1 = 1 ;
  'QUITTER' BOU2 ;
  'FINSI' ;
  'FINSI' ;
'FIN' BOU2 ;
'SI' ( IOK1 'EGA' 0 ) ;
'MESS' 'Listreel des temps de calcul inferieur à tini.' ;
'MESS' 'On ne fait donc rien.' ;
'QUITTER' DARCYTRA ;
'FINSI' ;
'SINON' ;
  'ERREUR' ;
  'Indice TEMPS_CALCULES absent de la table de données.' ;
  'QUITTER' DARCYTRA ;
'FINSI' ;
*
* TEMPS_SAUVES
ISOR = 0 ;
IOK1 = 0 ;
'SI' ( 'EXISTE' TRANSI 'TEMPS_SAUVES' ) ;
  TPSOR = TRANSI . 'TEMPS_SAUVES' ;
  'ORDO' TPSOR ;

```

```

TRANSI . 'TEMPS_SAUVES' = TPSOR      ;
DSOR   = 'DIME' TPSOR                ;
'REPETER' BOU3 DSOR                  ;
      ISOR = ISOR + 1                 ;
      TEMS = 'EXTRA' TPSOR ISOR       ;
      'SI' ( TPSINI '<' TEMS )        ;
          'SI' ( TPSINI 'NEG' TEMS EPSO ) ;
              IOK1 = 1                ;
              'QUITTER' BOU3         ;
          'FINSI'                     ;
      'FINSI'                         ;
'FIN' BOU3                           ;
'SI' ( IOK1 'EGA' 0 )                 ;
'ERREUR' 'Listreel des temps de sauvegarde hors tmin,tmax.' ;
'QUITTER' DARCYTRA                   ;
'FINSI'                               ;
'FINSI'                               ;
*-----*
* RAPPEL DES DONNEES ENTREES DANS LA PROCEDURE *
*-----*
*
'SAUTER' 1 'LIGNE'                   ;
'MESS' 'MODELISATION Darcy EFMH en transitoire.' ;
'MESS' '-----' ;
'SAUTER' 1 'LIGNE'                   ;
'MESS' 'Donnees presentes en entree: ' ;
'SAUTER' 1 'LIGNE'                   ;
'SI' ( 'EXISTE' TRANSI 'BLOCAGE' )    ;
      'MESS' 'Calcul avec conditions aux limites sur Th (Dirichlet)' ;
'FINSI'                               ;
'SI' ( 'EXISTE' TRANSI 'FLUX_IMPOSE' ) ;
      'MESS' 'Calcul avec conditions aux limites de flux (Neumann)' ;
'FINSI'                               ;
'SI' ( 'EXISTE' TRANSI 'SOURCE' )     ;
      'MESS' 'Ce probleme comporte un terme source' ;
'FINSI'                               ;
'SAUTER' 1 'LIGNE'                   ;
'MESS' 'Valeur de THETA, parametre du schema numerique : ' COEF ;
'MESS' ' (0:Schema explicite, 0.5:Crank-Nicolson, 1:Implicite)' ;
'SAUTER' 1 'LIGNE'                   ;
'MESS' 'Valeur du temps initial : ' TPSINI ;
'MESS' 'Valeur du temps final : ' TPSFIN ;
TPS1      = 'EXTR' TPCAL ICAL        ;
DELTAT    = TPS1 - TPSINI            ;
'MESS' 'Pas de temps initial : ' DELTAT ;
'SAUTER' 1 'LIGNE' ;
*-----*
* BOUCLE RESOLVANT LE SYSTEME POUR CHAQUE PAS DE TEMPS *
*-----*
*
* Initialisations
*
PRECED = TPSINI      ;
IPAS   = ICAL        ;

```



```

DELOLD = 0.DO ;

MHPR = 'MHYB' MODHYB HYTOT MASTP ;
M1 = 'MHYB' MODHYB HYTOT 'MASSE';
HND1A = 'MATP' MODHYB HYTOT MHPR ;
MHTH = 'MHYB' MODHYB HYTOT MASTT ;
*
* Table utilisée par les operateurs MATP, SMTP, HYBP et HDEBI
*
TAB = 'TABLE' ;
TAB.'SOUSTYPE' = 'DARCY_TRANSITOIRE' ;
TAB.'THETA' = COEF ;
TAB.'SURF' = EMSURF ;
*
*
*=====
'REPETE' BOU5 ;
*=====
*
* Initialisations
*
TPS = 'EXTR' TPCAL IPAS ;
'MESS' 'TEMPS T=' TPS ;
DELTAT = TPS - PRECED ;
TAB.'PAS' = DELTAT ;
TAB.'TRACE' = TRP ;
TAB.'CHARGE' = CHRG ;

menage ;
*
*- Construction des seconds membres et des matrices élémentaires
*- On sépare pour plus de clarté les cas avec et sans convection.
*

'SI' ( 'EXISTE' TRANSI 'SOURCE' ) ;
TERSCE = 'TIRE' TERSOU TPS ;
SMTR = 'SMTP' MODHYB HYTOT MHTH TAB TERSCE ;
'SINON' ;
SMTR = 'SMTP' MODHYB HYTOT MHTH TAB ;
'FINSI' ;
'SI' ( DELOLD 'NEG' DELTAT EPSO ) ;
HNDTR = 'MATP' MODHYB HYTOT MHTH TAB ;
'SI' ( 'EXISTE' TRANSI 'BLOCAGE' ) ;
HNDT1 = HNDTR 'ET' MATBLOC ;
'SINON' ;
HNDT1 = HNDTR ;
'FINSI' ;
'FINSI' ;
*
*- Incorporation des conditions aux limites
*

'SI' ( 'EXISTE' TRANSI 'BLOCAGE' ) ;
CHARIMPO = 'TIRE' CHAIMP TPS ;
CHARIMPO = 'CHAN' 'ATTRIBUT' CHARIMPO 'NATURE' 'DISCRET' ;

```

```

SMTR      = SMTR 'ET' CHARIMPO      ;
'FINSI'   ;
*
*
*- Résolution du système en trace de température
*
TP1 = 'RESO' HNDT1 SMTR      ;
TP2 = 'EXCO' TP1 'TH' 'TH' ;
mess 'resolution therm' ;

*
*- Calcul des inconnues secondaires à partir des trace de température.
*- On sépare pour plus de clarté les cas avec et sans convection.
*
'SI' ( 'EXISTE' TRANSI 'SOURCE' ) ;
      CHA2 = 'HYBP' MODHYB HYTOT MHTH TP2 TAB TERSCE ;
      FLU2 = 'HDEB' MODHYB HYTOT MHTH TP2 MCHYB TAB TERSCE ;
'SINON' ;
      CHA2 = 'HYBP' MODHYB HYTOT MHTH TP2 TAB ;
      FLU2 = 'HDEB' MODHYB HYTOT MHTH TP2 MCHYB TAB ;
'FINSI' ;

*
*- Archivage des resultats
*
'SI' ( 'EXISTE' TRANSI 'TEMPS_SAUVES' ) ;
      'REPETER' BOU4 ;
      TEMS = 'EXTRA' TPSOR ISOR ;
      'SI' ( ( PRECED '<' TEMS ) 'ET' ( TEMS '<EG' TPS ) ) ;
      DTEM = ( TPS - TEMS ) / DELTAT ;
      UNMO = 1. - DTEM ;
      TPBIS = 'COLI' TRP DTEM TP2 UNMO ;
      CHBIS = 'COLI' CHRQ DTEM CHA2 UNMO ;
      FLUBIS = 'COLI' FLUX DTEM FLU2 UNMO ;
      LAST1 = LAST1 + 1 ;
      TRANSI . 'TEMPS' . LAST1 = TEMS ;
      TRANSI . 'TRACE_TEMPERATURE' . LAST1 = TPBIS ;
      TRANSI . 'TEMPERATURE' . LAST1 = CHBIS ;
      TRANSI . 'FLUX' . LAST1 = FLUBIS ;
      ISOR = ISOR + 1 ;

*-----$$
*-----CALCUL DU RÉGIME HYDRAULIQUE INSTANTANÉ-----$$
*-----CALCUL EFFECTUÉ POUR CHAQUE TEMPS SAUVÉE-----$$
*-----$$

GRAVSDT = (NOMC CHBIS 'SCAL') * RGB ;
GRAV1TP = 'SQTP' MODHYB HYTOT HND1A M1 GRAVSDT ;

PIMP = (redu (NOMC TPBIS SCAL) PMAILIM) * PLIMSDT;
PPFRO = 'DEPI' MATBLOC (NOMC PIMP 'TH') ;
FFF1 = GRAV1TP et PPFRO ;
CCC1 = HND1A ET MATBLOC ;
CHTER1 = RESOUDRE CCC1 FFF1 ;
CHTER1 = exco CHTER1 'TH' 'TH';

```

```

mess 'resolution hydro, temps ' (TEMS/tannee) ;

CH = 'HYBP' MODHYB HYTOT MHPR CHTER1 M1 GRAVSDT ;
QFACE1 = HDEBI MODHYB HYTOT MHPR CHTER1 MCHYB M1 GRAVSDT;

TRANS2. 'PRESSION' . LAST1 = CH ;
TRANS2. 'TRACE_PRESSION' . LAST1 = CHTER1 ;
TRANS2. 'DEBIT' . LAST1 = QFACE1 ;

*-----$$
*-----$$
      'SINON' ;
      'QUITTER' BOU4 ;
      'FINSI' ;
      'SI' ( ISOR '>' DSOR ) ;
      ISOR = DSOR ;
      'QUITTER' BOU4 ;
      'FINSI' ;
      'FIN' BOU4 ;
'SINON' ;
      LAST1 = LAST1 + 1 ;
      TRANSI. 'TEMPS' . LAST1 = TPS ;
      TRANSI. 'TRACE_TEMPERATURE' . LAST1 = TP2 ;
      TRANSI. 'TEMPERATURE' . LAST1 = CHA2 ;
      TRANSI. 'FLUX' . LAST1 = FLU2 ;
'FINSI' ;

*
*- Sortie de boucle ou Initialisations pour le pas suivant
*
'SI' ( IPAS 'EGA' DCAL) ;
      TPSDER = TRANSI. 'TEMPS' . LAST1 ;
      'SI' ( TPSDER 'NEG' TPSFIN EPSO ) ;
      LAST1 = LAST1 + 1 ;
      TRANSI . 'TEMPS' . LAST1 = TPS ;
      TRANSI . 'TRACE_TEMPERATURE' . LAST1 = TP2 ;
      TRANSI . 'TEMPERATURE' . LAST1 = CHA2 ;
      TRANSI . 'FLUX' . LAST1 = FLU2 ;

      'FINSI' ;
      'QUITTER' BOU5 ;
'SINON' ;
      IPAS = IPAS + 1 ;
      PRECED = TPS ;
      TRP = TP2 ;
      CHRГ = CHA2 ;
      FLUX = FLU2 ;
      DELOLD = DELTAT ;
'FINSI' ;

*
*=====
'FIN' BOU5 ;
*=====
*
'FINPROC' TRANSI ;

```

```

*
*- Options générales de calcul.
*
TITRE 'EFMH DARCY ISOTROPE : Hydrocoïne : Convection naturelle' ;
OPTI DIME 3 ELEM CUB8 ;
OPTI ISOV SURF ;
OPTI ECHO 1 ;
*OPTI TRAC PSC ;
*
*- INITIALISATIONS pour la découpe des droites support des zones
*

IX0 = 10 ;
IY0 = 10 ;
SXY = IX0 + IY0 ;
IZ2 = 20 ;

*
*- INITIALISATIONS des coordonnées des points support des droites
*
RAY0 = 250.0 ;
RMAX = 750.0 ;
RMAX2 = 1500.0 ;
*
*- Définition des points de base
*
C0 = 0.0 0.0 0. ;
C2 = RAY0 0.0 0. ;
C3 = RMAX 0.0 0. ;
C4 = RMAX2 0.0 0. ;
C5 = 0.0 RAY0 0. ;
C6 = 0.0 RMAX 0. ;
C7 = 0.0 RMAX2 0. ;
*
*- Définition du cercle frontière du domaine
*
C2C3 = C2 D IZ2 C3 ;
c3c4 = C3 D 10 C4 ;
ZONED2 = (C2C3 et c3c4) ROTA SXY 90 C0 (0. 0. 1.);
*
*- Définition de la zone contenant la source de chaleur
*- Utilisation de DEDU - Maillage en TRI3
*
A0 = 0. 0. 0. ;
A1 = 1. 0. 0. ;
A3 = 1. 1. 0. ;
A2 = 0. 1. 0. ;
AOA1 = A0 D IX0 A1 ;
A1A3 = A1 D IY0 A3 ;
A3A2 = A3 D IX0 A2 ;
A2A0 = A2 D IY0 A0 ;
LIE1 = AOA1 ET A1A3 ET A3A2 ET A2A0 ;
*
SEPA1 = COTE 4 ZONED2 ;

```

```

SEPA1 = INVE SEPA1          ;
COC2  = C0 D IX0 C2        ;
C5C0  = C5 D IY0 C0        ;
LIM1  = COC2 ET SEPA1 ET C5C0 ;
OPTI  ELEM QUA4            ;
ZAO   = DALLER AOA1 A1A3 A3A2 A2A0 ;
ZONED1 = DEDU ZAO LIE1 LIM1 ;

*
* Définition des zones et symétrisation
*
ZONEG1 ZONEG2 = SYME ZONED1 ZONED2 DROIT C0 C6 ;
ZONE1  = ZONED1 ET (inve ZONEG1)                ;
ZONE2  = ZONED2 ET (inve ZONEG2)                ;
ZONE1  = coul ZONE1 ROUG ;

*
*- Creation maillage plan
*
ZONE1  = coul ZONE1 ROUG ;
PTOT   = ZONE1 ET ZONE2 ;
ELIM PTOT 0.001 ;

DEPLA PTOT TOUR (-90.DO) C0 (0.DO 1.DO 0.DO) ;
DEPLA PTOT TOUR (-90.DO) C0 (0.DO 0.DO 1.DO) ;

*
*- maillage de la troisième dimension
*

OPTI  ELEM CUB8 ;

*
*- rotation de 5 degré
*
DOMVOL = PTOT VOLU 1 ROTA 5. (-5.D-1 0.DO 0.DO ) (-5.D-1 0.DO 1.DO) ;
noyvol = domvol elem roug ;

*
*- on enlève l'axe de la troisième face pour définir le maillage
*- supportant les CL
*
autour = domvol face 3 ;
x = autour point droit C0 (0.0 0. 1.) 2.e-1 ;
saxe = autour elem appuye strictement x ;
front = diff saxe autour ;

ELI0 = 1.e-6 ;

***** origine de la temperature
TINF  = 0.DO ;

***** conductivité thermique du milieu
LAMED = 2.51D0 ;

***** conductivité thermique du liquide

```

```

LAMLIQ = 0.623D0      ;

***** porosité du milieu
PORO   = 1.D-4        ;

***** fraction volumique du liquide
XVLIQ  = PORO         ;

***** fraction volumique du milieu
XVMED  = 1.D0 - PORO ;

***** masse volumique du liquide
RHOLIQ = 992.2D0      ;

***** capacité calorifique du liquide
CPLIQ  = 4180.D0      ;

***** masse volumique du milieu
RHOMED = 2600.D0      ;

***** capacité calorifique du milieu
CPMED  = 879.D0       ;

***** rayon du noyau contenant la puissance
RAY    = 250.D0       ;

***** coefficient de décroissance de la puissance
DLAM   = -7.3215D-10 ;

***** puissance initiale
Q0     = 1.D7         ;

***** modèle
MODHYB = 'MODL' DOMVOL 'DARCY' 'ISOTROPE' ;

***** construction des domaines et des objets leurs étant attaché
HYTOT  = 'DOMA' DOMVOL ELIO IMPR          ;
MCHYB  = 'DOMA' HYTOT 'ORIENTAT'         ;
HYNOY  = 'DOMA' NOYVOL 'INCL' HYTOT ELIO ;
VKN    = 'DOMA' HYNOY 'VOLUME'           ;
HYFRO  = 'DOMA' front 'INCL' HYTOT ELIO ;
HYAXE  = 'DOMA' saxe 'INCL' HYTOT ELIO  ;

***** champoints d'initialisation
HO     = MANU CHPO HYTOT.CENTRE 1 'H'     TINF 'NATURE' 'DISCRET' ;
THO    = MANU CHPO HYTOT.FACE 1 'TH'     TINF 'NATURE' 'DISCRET' ;
QFACEO = MANU CHPO HYTOT.FACE 1 'FLUX' 0.DO 'NATURE' 'DISCRET' ;

***** conductivité du milieu homogène
*LAMEFF = (XVMED * LAMED) + (XVLIQ * LAMLIQ) ;
MATIO  = 'MATE' MODHYB 'K' LAMED          ;

***** capacité du milieu homogène
*F     = (XVLIQ * RHOLIQ * CPLIQ) + (XVMED * RHOMED * CPMED) ;
COEMM  = 'MANU' 'CHML' HYTOT.'MAILLAGE' 'CK' (RHOMED * CPMED) ;

```



```

TRANS2.'SOUSTYPE'          = 'DARCY_TRANSITOIRE' ;
TRANS2.'MODELE'           = MODHYB ;
TRANS2.'CONDUCTIVITE'     = MATIO ;
TRANS2.'CAPACITE'         = COEMM ;
TRANS2.'DOMAINE'          = HYTOT ;
*
TRANS2.'TEMPS' . 0         = TINI ;
TRANS2.'TRACE_TEMPERATURE' . 0 = THO ;
TRANS2.'TEMPERATURE' . 0   = HO ;
TRANS2.'FLUX' . 0          = QFACEO ;
*
TRANS2.'BLOCAGE'           = BBFRO ;
TRANS2.'TEMPERATURE_IMPOSE' = VALIO ;
TRANS2.'SOURCE'           = CHSRCE ;
*
TRANS2.'THETA'             = 0.5D0 ;
TRANS2.'TEMPS_CALCULES'    = LICALC1 ;
TRANS2.'TEMPS_SAUVES'     = LISAUV ;
*

=====
*          calcul hydraulique
=====

***** dilatabilité thermique de l'eau
BETA = -3.85d-4 ;
***** viscosité dynamique de l'eau
MU = 6.53d-4 ;
***** perméabilité intrinsèque du milieu
KINTR = 1.d-16 ;
***** gravité
G = -9.81 ;

**** force de pesanteur à dT près (approximation de Boussinesc)
RGBETA = MANU CHPO HYTOT.CENTRE
3 'FX' 0. 'FY' 0. 'FZ' (RHOLIQ*beta*g) 'NATURE' 'DISCRET' ;

**** perméabilité : champ par élément
K = kintr /mu ;
MATI1 = 'MATE' MODHYB 'K' K ;

**** variation de pression / à la pression hydrostatique en frontière
**** (approximation de Boussinesc)
PSDT = nomc (hyfro.centre coor 3) 'TH' 'NATURE' 'DISCRET' ;
PSDT = PSDT * (RHOLIQ*beta*g) ;

TRANS2.'PRESSION'          = 'TABLE' ;
TRANS2.'TRACE_PRESSION'    = 'TABLE' ;
TRANS2.'DEBIT'             = 'TABLE' ;

HO = MANU CHPO HYTOT.CENTRE 1 'H' 0.0 'NATURE' 'DISCRET' ;
THO = MANU CHPO HYTOT.FACE 1 'TH' 0.0 'NATURE' 'DISCRET' ;
QFACEO = MANU CHPO HYTOT.FACE 1 'FLUX' 0.DO 'NATURE' 'DISCRET' ;

TRANS2.'PERMEABILITE'      = MATI1 ;

```



```

TRANS2. 'PRESSION'. 0      = HO ;
TRANS2. 'TRACE_PRESSION'. 0 = THO ;
TRANS2. 'DEBIT'. 0        = QFACEO ;
TRANS2. 'PRESSION/DT_IMPOSE' = PSDT ;
TRANS2. 'FORCE/DT_IMPOSE'   = RGBETA ;

*=====
* Resolution transitoire
*=====
*
DARCYMOD TRANS2 ;
*
*- procédure permettant de construire un LISTREEL à partir
*- d'un champ point
*

*****
debproc CHPRO TCHARO*'CHPOINT' MAO*'MAILLAGE' ;
LSTO = PROG ;
REPETER BLO (NBEL MAO); ;
  LSTO = LSTO ET
  (PROG (EXTR TCHARO 'SCAL'(MAO POIN &BLO))) ;
FIN BLO ;
finproc LSTO ;
*****

*
*- progression de l'abscisse correspondant aux solutions calculées
*
BB = CHPRO (HYAXE.CENTRE coor 3) (HYAXE.CENTRE);
*
*- progression de l'abscisse correspondant à la solutions analytiques
*
DD = PROG
0. 25. 50. 75. 100. 125. 150. 175. 200. 225. 250. 275. 300. 325.
350. 375. 400. 425. 450. 475. 500. 525. 550. 575. 600. 625. 650. 675.
700. 725. 750. ;

MT = TABLE ;
MT.1 = 'MARQ CARR NOLI TITR 50_ANS' ;
MT.2 = 'MARQ PLUS NOLI TITR 100_ANS' ;
MT.3 = 'MARQ LOSA NOLI TITR 500_ANS' ;
MT.4 = 'MARQ ETOI NOLI TITR 1000_ANS' ;
MT.5 = 'TITR SOLUTION_ANALYTIQUE' ;
MT.6 = 'TITR SOLUTION_ANALYTIQUE' ;
MT.7 = 'TITR SOLUTION_ANALYTIQUE' ;
MT.8 = 'TITR SOLUTION_ANALYTIQUE' ;

*-----
*
*- Comparaison de l'accroissement de la pression entre CASTEM2000
*-          et la solution analytique
*
*-----

```

```

*----- postraitement castem

X = trans2.'TRACE_PRESSION'. 14 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVP = EVOL MANU bb cc ;

X = trans2.'TRACE_PRESSION'. 17 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVP = EVP ET (EVOL MANU bb cc) ;

X = trans2.'TRACE_PRESSION'. 22 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVP = EVP ET (EVOL MANU bb cc) ;

X = trans2.'TRACE_PRESSION'. 25 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVP = EVP ET (EVOL MANU bb cc) ;

*----- solution analytique

P_50ANS = PROG
0. 1.9525 3.9045 5.8545 7.7977 9.7190 11.579 13.293 14.696 15.539
15.504 14.466 12.936 11.347 9.9019 8.6628 7.6240 6.7558 6.0265
5.4089 4.8816 4.4277 4.0344 3.6912 3.39 3.1242 2.8885 2.6785
2.4906 2.3218 2.1696 ; P_50ANS = P_50ANS * 1000. ;

P_100ANS = PROG
0. 2.5405 5.0649 7.5499 9.9582 12.230 14.279 15.986 17.213 17.819
17.697 16.857 15.564 14.079 12.5874 11.187 9.9385 8.8498 7.9125 7.1086
6.4179 5.8220 5.3050 4.8538 4.578 4.1083 3.7983 3.5222 3.2751 3.0531
2.8530 ; P_100ANS = P_100ANS * 1000. ;

P_500ANS = PROG
0. 1.2077 2.3876 3.5128 4.5589 5.5048 6.3333 7.0320 7.5934 8.0152
8.2997 8.4536 8.4875 8.4146 8.2501 8.0102 7.7112 7.3689 6.9977 6.6106
6.2184 5.8301 5.4527 5.0912 4.7491 4.4285 4.1303 3.8544 3.6004 3.3672
3.1535 ; P_500ANS = P_500ANS * 1000. ;

P_1000AN = PROG
0. 0.5271 1.0468 1.5521 2.0361 2.4927 2.9165 3.3027 3.6478 3.949
4.2047 4.4142 4.5779 4.6969 4.7731 4.8092 4.8084 4.7742 4.7104 4.6211
4.5103 4.3818 4.2395 4.0869 3.9271 3.763 3.5973 3.432 3.269 3.1097
2.9553 ; P_1000AN = P_1000AN * 1000. ;

EVPA = EVOL MANU DD P_50ANS ;
EVPA = EVPA et (EVOL MANU DD P_100ANS) ;
EVPA = EVPA et (EVOL MANU DD P_500ANS) ;
EVPA = EVPA et (EVOL MANU DD P_1000AN) ;

*----- tracé comparatif

DESS MT (EVP et EVPA) LEGE XBOR 0. 750.
TITX 'Z (m)' TITY 'PRESSION (Pa)'
TITRE
'COMPARAISON ENTRE CASTEM2000 (MARQUEURS) ET LES SOLUTIONS ANALYTIQUES
(LIGNES CONTINUES)';

```

```

*-----
*
*- Comparaison de l'accroissement de la pression entre CASTEM2000
*-          et la solution analytique
*
*-----

*-----
X = trans2.'TRACE_TEMPERATURE'. 14 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVT = EVOL MANU bb cc ;

X = trans2.'TRACE_TEMPERATURE'. 17 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVT = EVT ET (EVOL MANU bb cc) ;

X = trans2.'TRACE_TEMPERATURE'. 22 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVT = EVT ET (EVOL MANU bb cc) ;

X = trans2.'TRACE_TEMPERATURE'. 25 EXCO TH SCAL ;
CC = CHPRO (REDU X HYAXE.CENTRE) (HYAXE.CENTRE);
EVT = EVT ET (EVOL MANU bb cc) ;

*----- solution analytique

T_50ANS = PROG
62.547 62.544 62.529 62.477 62.306 61.804 60.506 57.564 51.751 41.77
26.927 13.365 5.8378 2.2325 0.7437 0.2148 0.0535 0.0115 0.0021 0.0003
4.e-5 5e-6 5.e-7 4.e-8 3.e-9 1.e-10 8e-11 3e-13 1e-14 3e-16 9e-18 ;

T_100ANS = PROG
81.463 81.33 80.878 79.954 78.285 75.489 71.105 64.693 55.974 45.014
32.367 20.808 12.474 6.9677 3.6233 1.7522 0.7872 0.3282 0.1268 0.0454
0.0150 0.0046 0.0013 0.0003 8.e-5 2.e-5 4.e-6 7.e-7 1.e-7 2.e-8 3.e-9 ;

T_500ANS = PROG
38.839 38.589 37.847 36.64 35.01 33.012 30.714 28.19 25.52 22.78 20.05
17.394 14.871 12.528 10.397 8.4986 6.8416 5.4233 4.2326 3.2518 2.4591
1.8302 1.3405 0.96614 0.6851 0.4780 0.3281 0.2216 0.1472 0.0962 0.0618 ;

T_1000AN = PROG
16.924 16.859 16.663 16.342 15.903 15.356 14.712 13.986 13.192 12.346
11.464 10.561 9.6533 8.7539 7.8726 7.0294 6.2243 5.4675 4.7644 4.1185
3.5316 3.004 2.5345 2.1211 1.7607 1.4497 1.1838 0.95884 0.7702 0.6136
0.4848 ;

EVTA = EVOL MANU DD T_50ANS ;
EVTA = EVTA et (EVOL MANU DD T_100ANS) ;
EVTA = EVTA et (EVOL MANU DD T_500ANS) ;
EVTA = EVTA et (EVOL MANU DD T_1000AN) ;

*----- tracé comparatif

```

```
DESS MT (EVT et EVTA) LEGE XBOR 0. 750.  
TITX 'Z (m)' TITY 'TEMPERATURE (K)'  
TITRE  
'COMPARAISON ENTRE CASTEM2000 (MARQUEURS) ET LES SOLUTIONS ANALYTIQUES  
(LIGNES CONTINUES)';  
*  
FIN ;
```

6.2 Convection naturelle en milieu poreux dans une cavité carrée

6.2.1 Formulation Eléments Finis Mixtes avec Hybridation

<i>Opérateurs utilisés</i>	<i>MATP, SMTP, MHYB, opt MASSE HYBP HDEB SQTP</i>
<i>Méthodes numériques</i>	<i>EFMH, Schéma en temps explicite et implicite</i>
<i>Nature du problème</i>	<i>Traitement de la convection naturelle en milieu poreux</i>
<i>classification</i>	<i>2D plan, milieu poreux, thermique, hydraulique, convection naturelle, transitoire et permanent</i>

Schéma du problème

La convection naturelle en milieu poreux est le couplage de deux équations (hydraulique et thermique) par une équation d'état liant la densité du fluide à sa température. On présente ci après deux exemples de calculs de convection naturelle dans une cavité carrée. Dans ces deux exemples, les conditions aux limites sont :

Thermique : faces supérieures et inférieures : $T = 0$ et $T1 > 0$ faces latérales, flux thermique nul.

Hydraulique : flux hydraulique nul sur les quatre frontières

Conditions aux limites hydrauliques et thermiques :




FIG. 6.2.1 – Convection naturelle : schéma

Modélisation

L'hydraulique est régie par la loi de Darcy,

$$\vec{U} = -\frac{k \rho_0 g}{\mu} \left(\vec{\nabla} \frac{P}{\rho_0 g} + \frac{\rho}{\rho_0} \vec{\nabla} z \right) \text{ avec :}$$

\vec{U} = vitesse de Darcy

k = perméabilité intrinsèque

ρ = masse volumique, et ρ_0 = masse volumique de référence

μ = viscosité

P = pression

z = altitude

La thermique est régie par une équation de convection diffusion :

$$\rho_m C_m \frac{\partial T}{\partial t} = \lambda \Delta T - \rho_e C_e \vec{U} \vec{\nabla} T \text{ avec :}$$

$\rho_m C_m$ = capacité calorifique moyenne du milieu (eau+roche)

$\rho_e C_e$ = capacité calorifique de l'eau

λ = conductivité thermique moyenne du milieu (eau+roche)

T = température

t = temps

Le couplage entre les deux équations est assuré par l'équation d'état du fluide :

$$\rho = \rho_0 (1 - \beta T)$$

avec : β coefficient de dilatation volumique

Ecriture adimensionnelle

Une écriture adimensionnelle de ces équations aboutit à :

$$\vec{U} = -\vec{\nabla} h^* + T \vec{\nabla} z$$

$$\frac{\partial T}{\partial t} = \Delta T - Ra \vec{U} \vec{\nabla} T$$

$$Ra = \frac{k \rho_0 g \beta T L \rho_e C_e}{\mu \lambda} \text{ est le nombre de Rayleigh en milieu poreux.}$$

Résultat des calculs numériques

Evolution d'une perturbation

L'évolution dans le temps d'une perturbation de faible amplitude peut être décrite par une solution analytique.

Pour des conditions initiales $T = T1 (1 - z)$ pour la température, une perturbation de la forme $A \sin(x) \cos(z)$ évoluera, si son amplitude A est faible, suivant :

$$A \sin(x) \cos(z) e^{\alpha t} \quad \text{avec :} \quad \alpha = -2\pi^2 + \frac{Ra}{2}$$

La perturbation s'amplifie en fonction du temps pour : $Ra > Ra_c$ avec $Ra_c = 4\pi^2$ ($Ra = Ra_c$ conduit à $\alpha = 0$)

Un calcul est effectué avec le jeu de données ci- après, et les conditions suivantes :

Pas de temps $dt = 10^{-3}$, traitement explicite
 50 itérations en temps
 Amplitude de la perturbation initiale : 10^{-3}
 Nombre de Rayleigh = 100.

La valeur obtenue pour α est de 29.6, pour 30 théoriquement.




FIG. 6.2.2 – Convection naturelle : Evolution d'une perturbation en fonction du temps

Obtention d'une cellule stabilisée

Un calcul est effectué avec le jeu de données ci- après, de façon à obtenir une cellule de convection stabilisée :

Pas de temps $dt = 10^{-2}$, traitement implicite
 50 itérations en temps
 Amplitude de la perturbation initiale : 10^{-2}
 Nombre de Rayleigh = 100.




FIG. 6.2.3 – Convection naturelle, Cellule stabilisée : Températures




FIG. 6.2.4 – Convection naturelle, Cellule stabilisée : Vitesses

Références

- 1- Résolution des équations de Darcy par une méthode d'éléments finis mixtes hybrides - Rapport DMT 95/613 - F. DABENNE
- 2- Résolution des équations de Darcy exprimées en fonction de la pression et de la vitesse : prise en compte des effets de densité - Rapport DMT 96/250 - L.V. BENET
- 3- M COMBARNOUS Convection naturelle et mixte en milieux poreux, thèse 1970.

Jeu de données :

```

*
*- Options générales de calcul.
*
GRAPH = '0' ;
SAUT PAGE ;
*OPTI ECHO 0 ;
*OPTI TRAC ps ;
*
OPTI DIME 2 ELEM QUA4 ;
OPTI ISOV LIGN ;
*-----
*- Grandeurs physiques
*
**** Nombre de Rayley
RA = 100. ;
**** Amplitude de la perturbation initiale de température
ampl = 1d-2;
*-----
*- Maillage
*
*- Création des points
A0 = 0. 0. ; B0 = 1. 0. ;
A1 = 0. 1. ; B1 = 1. 1.;
A11 = 0.1 1.;
*- Création des droites
AB0 = DROI 10 A0 B0 ;
AB1 = DROI 10 A1 B1 ;
AA = DROI 1 A1 A11 ;
*- Creation des SURFACES
MASSIFO = ABO regle 10 AB1 ;
ELIM 0.01 (MASSIFO ET AA);
*-----
*- Tables domaines
*
HYTOT = DOMA MASSIFO 0.01 ;
CHYB1 = DOMA HYTOT 'SURFACE' ;
CHYB2 = DOMA HYTOT 'NORMALE' ;
MCHYB = DOMA HYTOT 'ORIENTAT' ;
ALPHA = DOMA HYTOT 'DIAMAX' ;
HYHAU = DOMA AB1 INCL HYTOT 0.01 ;
HYBAS = DOMA ABO INCL HYTOT 0.01 ;
HYAA = DOMA AA INCL HYTOT 0.01 ;
*-----
*- Parametres de l'algorithme
*
NITER = 50 ;
DELTAT = 1.E-2;
*-----
*- Table utilisée par les operateurs MATP, SMTP, HYBP et HDEBI
*
TAB = 'TABLE' ;
TAB.'SOUSTYPE' = 'DARCY_TRANSITOIRE' ;

```

```

TAB.'THETA'      = 1.                ;
TAB.'PAS'       = DELTAT              ;
*-----
*- HYDRAULIQUE
*- Modèle hydraulique
*
MODHYB = MODL HYTOT DARCY ISOTROPE HYQ4          ;
*-----
COEMM = 'MANU' 'CHML' HYTOT.MAILLAGE 'CK' 1.DO ;
EMSURF = 'INTG' 'ELEM' MODHYB COEMM           ;
TAB.'SURF'     = EMSURF                 ;
*-----
*- objets du modèle hydraulique
*
**** Perméabilité : champ par point
LMMA = MANU CHPO HYTOT.CENTRE 1 'K' RA 'NATURE' 'DIFFU' ;
LMMA = KCHA HYTOT 'CHAM' LMMA;
MATI_P = MATERIAU MODHYB 'K' LMMA                ;
**** Matrice de DARCY inversée
CND1A_P = MHYBR MODHYB HYTOT MATI_P              ;
**** Masse hybride
M_P = MHYBR MODHYB HYTOT 'MASSE'                ;
**** Matrice en Trace de pression TP
HND1A_P = MATP MODHYB HYTOT CND1A_P              ;
*
*- Conditions aux limites du modèle hydraulique
**** blocage des degrés de liberté
BBAA = BLOQUE HYAA.CENTRE 'TH' ;
**** valeurs des Traces imposées sur les degrés de liberté bloqués
PIMP = MANU CHPO HYAA.CENTRE 1 'TH' 0. 'NATURE' 'DISCRET' ;
EEAA = DEPI BBAA PIMP ;
*-----
*- Initialisation de la Température et de la Masse volumique
*
XXC YYC = 'COORD' HYTOT.'CENTRE' ;
TEMIO = 1.-YYC;
TEMO = TEMIO + (ampl*(COS(180.*XXC))*(SIN(180.*YYC)));
TEMO = 'NOMC' TEMO 'H' 'NATURE' 'DISCRET';
*
XXT YYT = 'COORD' HYTOT.'FACE' ;
TTEMIO = 1.-(1.*((YYT)));
TTEMO = TTEMIO + (ampl*(COS(180.*XXT))*(SIN(180.*YYT)));
CHTER2=EXCO TTEMO SCAL TH;
CHTER2=CHAN ATTRIBUT CHTER2 NATURE DIFFUS ;
*
RHO =kops 1. - TEMO ;
RHODEBUT = RHO;
RHO = 'EXCO' RHO 'SCAL' ;
RHO = 'CHAN' 'ATTRIBUT' RHO 'NATURE' 'DISCRET';
*
EVOLPERT = prog ampl;
*-----
***** TRANSPORT *****
*- Modèle de transport du sel
*
MODHYBK = MODL HYTOT DARCY ISOTROPE HYQ4          ;

```

```

*-----
*- Conditions aux limites du modèle de transport du sel
*
**** blocage des degrés de liberté
BBAS = BLOQ HYBAS.CENTRE 'TH' ;
BHAU = BLOQ HYHAU.CENTRE 'TH' ;
**** valeurs des Traces imposées sur les degrés de liberté bloqués
SIMPB = MANU CHPO HYBAS.CENTRE 1 'TH' 1. 'NATURE' 'DISCRET' ;
SIMPB = MANU CHPO HYHAU.CENTRE 1 'TH' 0. 'NATURE' 'DISCRET' ;
EBAS = DEPI BBAS SIMPB ;
EHAU = DEPI BHAU SIMPB ;
*-----couplage itératif HYDRAULIQUE / TRANSPORT-----
ITER = 0;
repetier bloc1 NITER ;
ITER = ITER + 1;
*-----
***** HYDRAULIQUE (PARTI VARIABLE)*****
*- contribution des forces de volume au second membre
*
RCH = MANU CHPO HYTOT.CENTRE 2 'FX' 0. 'FY' -1. 'NATURE' 'DISCRET' ;
RCH = RHO * RCH ;
GRAV2TP = SQTP MODHYB HYTOT HND1A_P M_P RCH ;
*-----
*- Assemblage matrice
*
CCC1_P = HND1A_P ET BBAA ;
*-----
*- second membre du système matriciel en trace de charge
*
FFF1_P = GRAV2TP et EEAA ;
*-----
*- Resolution en trace de pression
*
CHTER1_P = RESOUDRE CCC1_P FFF1_P ;
CHTER1_P = exco CHTER1_P 'TH' 'TH';
*-----
*- Calcul des débits
*
QONV1 = HDEBI MODHYB HYTOT CND1A_P CHTER1_P MCHYB M_P RCH;
VCENT = HVIT MODHYB HYTOT QONV1 MCHYB ;
*-----
***** TRANSPORT (PARTI VARIABLE) *****
*-----modification pour le modèle en transitoire
TAB.'TRACE' = CHTER2 ;
TAB.'CHARGE' = TEMO;
TAB.'PAS' = DELTAT;
*-----
*
VCENT1 = HVIT MODHYBK HYTOT (NOMC FLUX QONV1) MCHYB ;
MOT1 = 'MOTS' 'VX' 'VY' ;
VCVC = 'PSCA' VCENT1 VCENT1 MOT1 MOT1 ;
VCVC = VCVC ** 0.5 ;
*-----
LAM = MANU CHPO HYTOT.CENTRE 1 'K' 1. 'NATURE' 'DIFFU' ;
LAM = KCHA HYTOT 'CHAM' LAM ;

```

```

MATI2 = 'MATE' MODHYBK 'K' LAM ;
*-----
*- objets du modèle transport
*
**** Matrice de DARCY inversée
CND2A = MHYB MODHYBK HYTOT MATI2 ;
**** MATrice en Trace de pression salinité
HND2A = MATP MODHYBK HYTOT CND2A TAB ;
**** contribution de la convection au système en trace de salinité
*-----
*- Conditions aux limites du modèle de transport du sel
*
*-----
*- Assemblage matrice
*
CCC2 = HND2A ET BHAU ET BBAS ;
*-----
*- second membre du système matriciel en trace de salinité
*
FFF = EBAS ET EHAU;
*-----
*- complement
*
      'SI' ( (TAB.'THETA') 'NEG' 0.DO 1.D-4 ) ;
            RIGS SM1 = 'SMTP' MODHYBK HYTOT CND2A TAB QONV1 ;
            FFF = FFF et SM1 ;
            CCC2 = CCC2 et RIGS ;
      'SINON' ;
            SM1 = 'SMTP' MODHYBK HYTOT CND2A TAB QONV1 ;
            FFF = FFF et SM1 ;
      'FINSI' ;
*-----
*- Resolution en trace de salinité
*
CHTER2 = RESOUDRE CCC2 FFF ;
CHTER2 = EXCO CHTER2 TH TH ;
THTH1 = EXCO CHTER2 TH SCAL ;
*-----
*- Calcul des flux convectif de salinité
*
CONCO = 'NOMC' 'SCAL' CHTER2 ;
CONC1 = 'NOMC' 'SCAL' (TAB.'TRACE') ;
CONC2 = (TAB.'THETA')*CONCO + ((1.DO-(TAB.'THETA'))*CONC1) ;
CONC3 = (-1.DO) * CONC2 * QONV1 ;
TERSC2 = 'DIVU' MODHYBK HYTOT CONC3 MCHYB ;
TERSC2 = 'NOMC' 'SOUR' TERSC2 ;
*-----
*- Calcul de la salinité au centre des éléments (bilan des flux)
*
TEMO = HYBP MODHYBK HYTOT CND2A CHTER2 TAB TERSC2 ;
TEM = NOMC SCAL TEMO ;
*-----
*- Calcul de la densité de l'eau
*
RH02 = kops 1. - TEMO ;

```

```

DRSR = RHO / RHO2 - 1. ;
RHO = RHO2 ;
*-----
*
mess ' ' ;
mess 'ITERATION N° ' ITER;
  erm = maxi (abs drsr) ;
mess 'Evolution RHO :' erm ;
  DIFFTEM = (TEM - TEMIO ) ;
  MAXTEM = maxi ( abs DIFFTEM ) ;
mess 'max(|TEM - TEMO|)' MAXTEM;
nrmmax = maxi (abs VCVC) ;
mess 'Norme de vitesse maxi :' nrmmax ;
si ( iter > 1);
  rapport = nrmmax/medium;
  mess 'rapport |v2|/|v1|' rapport;
  AM1 = (rapport - 1.)/DELTAT;
  mess 'd/dt (log(|v|) = 'AM1;
finsi;
  medium = nrmmax;
  CFL = (nrmmax * DELTAT)/1.d2;
mess 'C F L = ' CFL;
  ALPHAM = maxi (ALPHA);
  CFO = (3.d-3 * ALPHAM * DELTAT)/(1.d2 * 1.d2);
mess 'C F O = ' CFO;
*----- Increment de la LISTREELS sur CFL-----
si (NITER > 1);
  EVOLPERT = EVOLPERT et (prog MAXTEM);
  SINON;
  EVOLPERT = prog MAXTEM;
finsi;
*-----
* fin de la boucle itérative du  HYDRAULIQUE / TRANSPORT
fin bloc1 ;
***** POST-TRAITEMENT *****
*
*- Calcul des pressions par élément
PCEN1 = HYBP MODHYB HYTOT CND1A_P CHTER1_P M_P RCH  ;
*
SI (NEG GRAPH 'N') ;
ampla=MAXI(ABS VCVC);
ampli=0.1/ampla;
VV = VECT VCENT ampli 'VX' 'VY' VERT ;
titre 'VITESSES AUX CENTRES DES ELEMENTS' ampla  ;
TRAC VV HYTOT.CENTRE ;
*
PSOM1 = ELNO HYTOT PCEN1 ;
TITRE 'PRESSION (Pa)' ;
TRAC PSOM1 HYTOT.MAILLAGE (CONTOUR HYTOT.MAILLAGE) ;
*
RHO = KCHT HYTOT SCAL CENTRE RHO;
RSOM1 = ELNO HYTOT RHO ;
TITRE 'MASSE VOLUMIQUE' ;
TRAC RSOM1 HYTOT.MAILLAGE (CONTOUR HYTOT.MAILLAGE) ;
*
TEM1 = 'NOMC' TEMO 'SCAL' 'NATURE' 'DISCRET';

```

```
TEM1 = 'EXCO' TEM1 'SCAL';
TEM1 = KCHT HYTOT SCAL CENTRE TEM1;
RSOM1 = ELNO HYTOT TEM1 ;
MAS = MAXI (ABS TEMO);
MIS = MINI (ABS TEMO);
TITRE 'TEMPERATURE max = ' MAS ' min = ' MIS;
TRAC RSOM1 HYTOT.MAILLAGE (CONTOUR HYTOT.MAILLAGE) ;
liter = prog 0. pas DELTAT (niter*DELTAT);
titre ' evolution de max|T-Tlin| en fonction du temps';
evPER = evol manu 'iteration' liter 'perturbation max' EVOLPERT;
dess evPER;
*
finsi ;
*
fin;
```

6.3 Hydraulique en milieu poreux : influence de la Salinité

6.3.1 Formulation Eléments Finis Mixtes avec Hybridation

<i>Opérateurs utilisés</i>	<i>MATP, SMTP, MHYB, opt MASSE HYBP HDEB SQTP</i>
<i>Méthodes numériques</i>	<i>EFMH, Schéma en temps explicite et implicite</i>
<i>Nature du problème</i>	<i>Traitement de l'influence de la salinité sur les circulations hydrauliques</i>
<i>classification</i>	<i>2D plan, milieu poreux, salinité, hydraulique, transitoire et permanent</i>

Schéma du problème

Les équations sont voisines de celles de convection naturelle. Les deux équations (hydraulique et salinité) sont couplées par une équation d'état liant la densité du fluide à sa salinité. Les équations utilisées (en particulier la présence d'un terme de dispersion proportionnel à la vitesse) et les conditions aux limites rencontrées amènent fréquemment une stratification, avec une interface fortement marquée entre l'eau douce et l'eau salée.

On présente ci après un exemple de calcul représentant, au voisinage d'une côte, une nappe d'eau salée (alimentée par l'océan) sous une nape d'eau douce.

Les conditions aux limites sont :

Salinité : frontières supérieure et latérale gauche : $S = 0$ (eau douce).
frontières inférieure et latérale droite : $S = 1$ (eau salée).
Hydraulique : charge hydraulique imposée sur les quatre frontières
Frontière supérieure : $P = 0$ (pression atmosphérique).
Frontières latérale droite et inférieure : pression hydrostatique de l'eau salée
Frontière latérale gauche : evolution linéaire : entre les frontières inférieures et supérieures.

Ces conditions aux limites sont choisies, en liaison avec les paramètres du problème, de façon telle que l'interface eau douce - eau salée soit rectiligne, joignant le coin inférieur gauche du domaine considéré au coin supérieur droit.

Modélisation

L'hydraulique est régie par la loi de Darcy (voir exemple précédent) :




FIG. 6.3.1 – Salinité : schéma

$$\vec{U} = -\frac{k \rho_0 g}{\mu} \left(\vec{\nabla} \frac{P}{\rho_0 g} + \frac{\rho}{\rho_0} \vec{\nabla} z \right)$$

Avec :

- \vec{U} = vitesse de Darcy
- k = perméabilité intrinsèque
- ρ = masse volumique
- ρ_0 = masse volumique de référence
- μ = viscosité
- P = pression
- z = altitude

La salinité est régie par une équation de convection diffusion, dispersion :

$$\varepsilon \frac{\partial S}{\partial t} = \nabla D \vec{\nabla} S - \vec{U} \vec{\nabla} S$$

Avec :

- ε = porosité du milieu
- D = tenseur de convection dispersion
- $D = \varepsilon D_0 + \alpha U$, avec
- D_0 coefficient de diffusion moléculaire
- α la dispersivité (α_L longitudinalement et α_T transversalement)
- S = salinité
- t = temps

Le couplage entre les deux équations est assuré par l'équation d'état du fluide :

$$\rho = \rho_0 (1 + \beta S)$$

avec : β coefficient de dilatation volumique

Résultat des calculs numériques

Le calcul est effectué en régime transitoire, jusqu'à obtention d'un régime permanent. Les conditions initiales sont $S = 0$ sur l'ensemble du domaine. Les conditions sont les suivantes :

- Pas de temps $dt = 5 \cdot 10^{-1}$, traitement implicite
- 10 itérations en temps
- Dispersion longitudinale : α_L 0.1
- Dispersion transversale : α_T 0.01

Analyse des résultats numériques

L'interface eau douce - eau salée est relativement marquée.

Le champ de vitesse dans la zone $S = 0$ est pratiquement uniforme, tangentiel à l'interface avec la zone $S = 1$.

Le champ de vitesse est pratiquement nul dans la zone $S = 1$




FIG. 6.3.2 – Salinité : Champ de salinité




FIG. 6.3.3 – Salinité : Champ de vitesse

Références

- 1- Résolution des équations de Darcy par une méthode d'éléments finis mixtes hybrides - Rapport DMT 95/613 - F. DABENNE
- 2- Résolution des équations de Darcy exprimées en fonction de la pression et de la vitesse : prise en compte des effets de densité - Rapport DMT 96/250 - L.V. BENET
- 3- G. de MARSILY, Hydrogéologie quantitative, Masson 1981.

Jeu de données :

```

GRAPH='N' ;
SAUT PAGE ;
*OPTI ECHO 0 ;
*OPTI TRAC ps ;
*
*- Options générales de calcul.
*
OPTI DIME 2 ELEM QUA4 ;
OPTI ISOV LIGN ;
*-----
*- grandeurs physiques
*
*** Coefficient de diffusion
DIFF=1.E-8;
*** Dispersivité
displ=1.0E-0;
dispt=1.0E-2;
*-----
*- Initialisation maillage
*
*- Création des points
*
A0 = 0.d0 0.d0 ;          B0 = 2. 0.d3 ;
A1 = 0.d0 1.2 ;          B1 = 1. 1.;
*-----
*- Création des droites
*
BAS = DROI 20 A0 B0 ;
HAU = DROI 20 A1 B1 ;
DRO = DROI 20 B0 B1 ;
GAU = DROI 20 A0 A1 ;
*-----
*- Creation des SURFACES
*
MASSIFO = BAS regle 20 HAU ;
ELIM 0.01 (MASSIFO ET DRO ET GAU);
*-----
*- Tables domaines
*
HYTOT = DOMA MASSIFO 0.01 ;
CHYB1 = DOMA HYTOT 'SURFACE';
CHYB2 = DOMA HYTOT 'NORMALE';
MCHYB = DOMA HYTOT 'ORIENTAT';
ALPHA = DOMA HYTOT 'DIAMAX';
HYHAU = DOMA HAU INCL HYTOT 0.01 ;
HYBAS = DOMA BAS INCL HYTOT 0.01 ;
HYDRO = DOMA DRO INCL HYTOT 0.01 ;
HYGAU = DOMA GAU INCL HYTOT 0.01 ;
*-----
*
VOLELMT = DOMA HYTOT 'VOLUME';
VOLELMT = 'CHAN' 'ATTRIBUT' VOLELMT 'NATURE' 'DIFFUS';
*- coefficient de diffusion et dispersion

```

```

*
ALFL = MANU CHPO HYTOT.CENTRE 1 'SCAL' displ 'NATURE' 'INDETRMINE' ;
ALFT = MANU CHPO HYTOT.CENTRE 1 'SCAL' dispt 'NATURE' 'INDETRMINE' ;
*-----
*- Algorithmme
*
NITER = 10 ;
*-----Modification pour le transitoire-----
DELTAT = 5.E-1;
*
* Table utilisée par les operateurs MATP, SMTP, HYBP et HDEBI
*
TAB = 'TABLE' ;
TAB.'SOUSTYPE' = 'DARCY_TRANSITOIRE' ;
TAB.'THETA' = 1. ;
TAB.'PAS' = DELTAT ;
*-----
***** HYDRAULIQUE *****
*
*-----
*- Modèle hydraulique
*
MODHYB = MODL HYTOT DARCY ISOTROPE HYQ4 ;
*-----
*- objets du modèle hydraulique
*
**** perméabilité : champ par point
LMMA = MANU CHPO HYTOT.CENTRE 1 'K' 1. 'NATURE' 'DIFFU' ;
LMMA = KCHA HYTOT 'CHAM' LMMA;
MATI_P = MATERIAU MODHYB 'K' LMMA ;
**** Matrice de DARCY inversée
CND1A_P = MHYBR MODHYB HYTOT MATI_P ;
**** Masse hybride
M_P = MHYBR MODHYB HYTOT 'MASSE' ;
**** MATrice en Trace de pression TP
HND1A_P = MATP MODHYB HYTOT CND1A_P ;
*-----
*- Conditions aux limites du modèle hydraulique
*
**** blocage des degrés de liberté
BBAH = BLOQUE HYHAU.CENTRE 'TH' ;
BBAB = BLOQUE HYBAS.CENTRE 'TH' ;
BBAD = BLOQUE HYDRO.CENTRE 'TH' ;
BBAG = BLOQUE HYGAU.CENTRE 'TH' ;
**** valeurs des Traces imposées sur les degrés de liberté bloqués
XXT YYT = 'COORD' HYTOT.'FACE' ;
PIMH = (0.*XXT) + (0.*YYT) ;
PIMB = 1.5*(1.-YYT);
PIMD = 1.5*(1.-YYT);
PIMG = 1.25*(1.2-YYT);
PIMH = NOMC PIMH TH ;
PIMB = NOMC PIMB TH ;
PIMD = NOMC PIMD TH ;
PIMG = NOMC PIMG TH ;
EEAH = DEPI BBAH PIMH ;
EEAB = DEPI BBAB PIMB ;

```

```

EEAD = DEPI BBAD PIMD ;
EEAG = DEPI BBAG PIMG ;
*-----
*- Initialisation de la salinité à 0
TEMIO = MANU CHPO HYTOT.CENTRE 1 'H' 0. 'NATURE' 'DISCRET' ;
TEMO = TEMIO;
*
EVOLPERT = prog (0.);
*
TTEMO = MANU CHPO HYTOT.FACE 1 'SCAL' 0. 'NATURE' 'DIFFUS' ;
CHTER2=EXCO TTEMO SCAL TH;
CHTER2=CHAN ATTRIBUT CHTER2 NATURE DIFFUS ;
*
*
*- Masse volumique
*-----
RHO =(1. + (kops 0.5 '*' TEMO) );
RHO = 'EXCO' RHO 'SCAL' ;
RHO = 'CHAN' 'ATTRIBUT' RHO 'NATURE' 'DISCRET';
*-----
      COEMM = 'MANU' 'CHML' HYTOT.MAILLAGE 'CK' 1.DO ;
      EMSURF = 'INTG' 'ELEM' MODHYB COEMM          ;
TAB.'SURF'      = EMSURF          ;
*
***** TRANSPORT *****
*-----
*- Modèle de transport du sel
*
MODHYBK = MODL HYTOT DARCY ANISOTROPE HYQ4          ;
*-----
*- Conditions aux limites du modèle de transport du sel
*
**** blocage des degrés de liberté
BBAS = BLOQ HYBAS.CENTRE 'TH' ;
BHAU = BLOQ HYHAU.CENTRE 'TH' ;
BDRO = BLOQ HYDRO.CENTRE 'TH' ;
BGAU = BLOQ HYGAU.CENTRE 'TH' ;
**** valeurs des Traces imposées sur les degrés de liberté bloqués
SIMPB = MANU CHPO HYBAS.CENTRE 1 'TH' 1. 'NATURE' 'DISCRET' ;
SIMPH = MANU CHPO HYHAU.CENTRE 1 'TH' 0. 'NATURE' 'DISCRET' ;
SIMPD = MANU CHPO HYDRO.CENTRE 1 'TH' 1. 'NATURE' 'DISCRET' ;
SIMPG = MANU CHPO HYGAU.CENTRE 1 'TH' 0. 'NATURE' 'DISCRET' ;
EBAS = DEPI BBAS SIMPB ;
EHAU = DEPI BHAU SIMPH ;
EDRO = DEPI BDRO SIMPD ;
EGAU = DEPI BGAU SIMPD ;
*-----couplage itératif HYDRAULIQUE / TRANSPORT-----
ITER = 0;
repeter bloc1 NITER ;
ITER = ITER + 1;
*
***** HYDRAULIQUE (PARTI VARIABLE)*****
*-----
*- contribution des forces de volume au second membre
*
RCH = MANU CHPO HYTOT.CENTRE 2 'FX' 0. 'FY' -1. 'NATURE' 'DISCRET' ;

```

```

RCH = RHO * RCH ;
GRAV2TP = SQTP MODHYB HYTOT HND1A_P M_P RCH ;
*-----
*- Assemblage matrice
*
CCC1_P = HND1A_P ET BBAH ET BBAB ET BBAD ET BBAG      ;
*-----
*- second membre du système matriciel en trace de charge
*
FFF1_P = GRAV2TP et EEAH ET EEAB ET EEAD ET EEAG      ;
*-----
*- Resolution en trace de pression
*
CHTER1_P = RESOUDRE CCC1_P FFF1_P ;
CHTER1_P = exco CHTER1_P 'TH' 'TH';
*-----
*- Calcul des débits
*
QONV1 = HDEBI MODHYB HYTOT CND1A_P CHTER1_P MCHYB M_P RCH;
VCENT = HVIT  MODHYB HYTOT QONV1 MCHYB ;
*-----
***** TRANSPORT (PARTIE VARIABLE) *****

TAB.'TRACE' = CHTER2      ;
TAB.'CHARGE' = TEMO;
TAB.'PAS' = DELTAT;
*-----
* calcul des coefficient de dispersion
*
VCENT1 = HVIT  MODHYBK HYTOT (NOMC FLUX QONV1) MCHYB ;
MOT1 = 'MOTS' 'VX' 'VY' ;
VCVC = 'PSCA' VCENT1 VCENT1 MOT1 MOT1 ;
VCVC = VCVC ** 0.5 ;
*-----
V1 = EXCO VCENT1 'VX' ;
V2 = EXCO VCENT1 'VY' ;
D11P = ALFT*VCVC + ( (ALFL - ALFT)*V1*V1/VCVC ) ;
D22P = ALFT*VCVC + ( (ALFL - ALFT)*V2*V2/VCVC ) ;
D21P = ( (ALFL - ALFT)*V2*V1/VCVC ) ;
DIR1 = 1.DO 0.DO ;
CHD11 = KCHA HYTOT CHAM D11p ;
CHD22 = KCHA HYTOT CHAM D22p ;
CHD21 = KCHA HYTOT CHAM D21p ;
MATI2 =
'MATE' MODHYBK DIRECTION DIR1 'K11' CHD11 'K21' CHD21 'K22' CHD22 ;
*-----
*- objets du modèle transport
*
**** Matrice de DARCY inversée
CND2A = MHYB MODHYBK HYTOT MATI2 ;
**** MATrice en Trace de pression salinité
HND2A = MATP MODHYBK HYTOT CND2A TAB ;
**** contribution de la convection au système en trace de salinité
*-----
*- Conditions aux limites du modèle de transport du sel
*

```

```

*-----
*- Assemblage matrice
*
CCC2 = HND2A ET BHAU ET BBAS ET BDRO ET BGAU;
*-----
*- second membre du système matriciel en trace de salinité
*
FFF = EBAS ET EHAU ET EDRO ET EGAU;
*-----
*- complement
  'SI' ( (TAB.'THETA') 'NEG' 0.DO 1.D-4 ) ;
        RIGS SM1 = 'SMTP' MODHYBK HYTOT CND2A TAB QONV1 ;
        FFF = FFF et SM1 ;
        CCC2 = CCC2 et RIGS ;
  'SINON' ;
        SM1 = 'SMTP' MODHYBK HYTOT CND2A TAB QONV1 ;
        FFF = FFF et SM1 ;
  'FINSI' ;
*-----
*- Resolution en trace de salinité
*
CHTER2 = RESOUDRE CCC2 FFF ;
CHTER2 = EXCO CHTER2 TH TH ;
THTH1 = EXCO CHTER2 TH SCAL ;
*-----
*- Calcul des flux convectif de salinité
*
  CONCO = 'NOMC' 'SCAL' CHTER2 ;
  CONC1 = 'NOMC' 'SCAL' (TAB.'TRACE') ;
  CONC2 = (TAB.'THETA')*CONCO + ((1.DO-(TAB.'THETA'))*CONC1) ;
  CONC3 = (-1.DO) * CONC2 * QONV1 ;
  TERSC2 = 'DIVU' MODHYBK HYTOT CONC3 MCHYB ;
  TERSC2 = 'NOMC' 'SOUR' TERSC2 ;
*-----
*- Calcul de la salinité au centre des éléments (bilan des flux)
*
TEM0 = HYBP MODHYBK HYTOT CND2A CHTER2 TAB TERSC2 ;
TEM = NOMC SCAL TEM0 ;
*-----
*- Calcul de la densité de l'eau
*
RHO2 = (1. + (kops 0.5 '*' TEM0) ) ;
DRSR = RHO / RHO2 - 1. ;
RHO = RHO2 ;
*-----
*
mess ' ' ;
mess 'ITERATION N° ' ITER;
  erm = maxi (abs drsr) ;
mess 'Evolution RHO :' erm ;
  DIFFTEM = (TEM - TEM0) ;
  MAXTEM = maxi ( RESU (DIFFTEM*VOLELMT));
mess 'max(|TEM - TEM0|)' MAXTEM;
  nrmmmax = maxi (abs VCVC) ;
mess 'Norme de vitesse maxi :' nrmmmax ;
si ( iter > 1);

```



```

    rapport = nrmmax/medium;
    mess 'rapport |v2|/|v1|' rapport;
    AM1 = (rapport - 1.)/DELTAT;
    mess 'd/dt (log(|v|) = 'AM1;
finsi;
    medium = nrmmax;
    CFL = (nrmmax * DELTAT)/1.d2;
mess 'C F L = ' CFL;
    ALPHAM = maxi (ALPHA);
    CFO = (3.d-3 * ALPHAM * DELTAT)/(1.d2 * 1.d2);
mess 'C F O = ' CFO;
*----- Increment de la LISTREELS -----
si (NITER > 1);
    EVOLPERT = EVOLPERT et (prog MAXTEM);
    SINON;
    EVOLPERT = prog MAXTEM;
finsi;
*-----
*      fin de la boucle itérative du  HYDRAULIQUE / TRANSPORT
fin bloc1 ;
***** POISTRAITEMENT *****
*
*- Calcul des pressions élémentaires (bilan des flux)
*
PCEN1 = HYBP MODHYB HYTOT CND1A_P CHTER1_P M_P RCH ;
*
SI (NEG GRAPH2 'N') ;
ampla=MAXI(ABS VCVC);
ampli=0.1/ampla;
VV = VECT VCENT ampli 'VX' 'VY' VERT ;
titre 'VITESSES AUX CENTRES DES ELEMENTS' ampla ;
TRAC VV HYTOT.CENTRE ;
*
PSOM1 = ELNO HYTOT PCEN1 ;
TITRE 'PRESSION (Pa)' ;
TRAC PSOM1 HYTOT.MAILLAGE (CONTOUR HYTOT.MAILLAGE) ;
*
RHO = KCHT HYTOT SCAL CENTRE RHO;
RSOM1 = ELNO HYTOT RHO ;
TITRE 'MASSE VOLUMIQUE' ;
TRAC RSOM1 HYTOT.MAILLAGE (CONTOUR HYTOT.MAILLAGE) ;
*
TEM1 = 'NOMC' TEMO 'SCAL' 'NATURE' 'DISCRET';
TEM1 = 'EXCO' TEM1 'SCAL';
TEM1 = KCHT HYTOT SCAL CENTRE TEM1;
RSOM1 = ELNO HYTOT TEM1 ;
MAS = MAXI (ABS TEMO);
MIS = MINI (ABS TEMO);
TITRE 'TEMPERATURE max = ' MAS ' min = ' MIS;
TRAC (prog 0.1 pas 0.1 0.9)
RSOM1 HYTOT.MAILLAGE (CONTOUR HYTOT.MAILLAGE) ;
*opti donn 5;
liter = prog 0. pas DELTAT (niter*DELTAT);
titre ' evolution de la salinité totale en fonction du temps';
evPER = evol manu 'iteration' liter 'perturbation max' EVOLPERT;
dess evPER;

```

```
*  
finsi ;  
*  
fin;
```

6.4 Cas d'une inclusion circulaire

6.4.1 Formulation Eléments Finis Mixtes avec Hybridation

<i>Opérateurs utilisés</i>	<i>MHYB, MATP, BLOQ, DEPI, HDEB, HVIT et HYBP</i>
<i>Méthodes numériques</i>	<i>EFMH, éléments HYT3 et HYQ4</i>
<i>Nature du problème</i>	<i>Calcul de la charge hydraulique et de la vitesse de Darcy au sein d'un milieu présentant une inclusion circulaire (discontinuité de la conductivité hydraulique). Comparaison avec la solution analytique.</i>
<i>Classification</i>	<i>2D plan, hydrogéologie, cas permanent</i>

Description du problème :

Un cylindre de rayon a , d'axe z et de conductivité hydraulique K_{in} est inclus dans un domaine infini de conductivité K_{out} . La charge h à l'infini tend vers $F_{inf}x$. Par raison de symétrie, l'écoulement est indépendant de la coordonnée z (i.e. 2D plan). On considère donc le plan (x,y) .

La charge hydraulique h et la vitesse de Darcy \vec{q} sont solutions du système d'équations

$$\begin{cases} \vec{q} = -K\nabla h \\ \text{div}\vec{q} = 0 \end{cases}$$

sur le domaine $\Omega = \{(x,y), (x,y) \in [-5a, 5a] \times [0, 3a]\}$. En effet, le flux à travers la droite $y = 0$ étant nul par raison de symétrie, on ne considère que le demi-plan supérieur.

La solution analytique de ce problème est connue [1]. La charge est donnée par

$$\begin{cases} h(x,y) = F_{inf} \frac{K_{out}}{K_{out} + K_{in}} x \text{ si } r \leq a \\ h(x,y) = F_{inf} \left\{ 1 + \frac{a^2}{r^2} \frac{K_{out} - K_{in}}{K_{out} + K_{in}} \right\} x \text{ si } r \geq a \end{cases}$$

où K_{in} (resp. K_{out}) désigne la conductivité hydraulique isotrope du cylindre (resp. du milieu), F_{inf} la densité de flux à l'infini suivant x (nul dans les autres directions). La vitesse de Darcy s'obtient en considérant pour chacune des zones le produit de la conductivité hydraulique par le gradient de charge.

On impose sur les bords du domaine (à l'exception de la droite $y = 0$ où le flux est nul) la charge hydraulique analytique.

Paramètre	Symbole	Valeur	Unité
Rayon du cylindre	a	10.	m
Conductivité hydraulique du cylindre	K_{in}	10^{-10}	$m.s^{-1}$
Conductivité hydraulique du milieu	K_{out}	10^{-12}	$m.s^{-1}$
Densité de flux à l'infini suivant x	F_{inf}	10^3	m^{-1}

Modélisation :

Compte tenu de la discontinuité de conductivité hydraulique, on utilise une méthode de discrétisation spatiale par éléments finis mixtes avec hybridation [2].

Le maillage réalisé est constitué de triangles et de quadrangles (cf figure 6.4.1). Il est non structuré et contient 760 éléments. Le maillage est donc assez grossier et certains éléments quadrangulaires sont assez déformés. Ajoutons que ce maillage n'est pas symétrique par rapport à l'axe $x = 0$.

Analyse des résultats :

Nous présentons la charge et la vitesse calculées (figure 6.4.2), la norme de la vitesse (figure 6.4.3), l'erreur relative sur la charge et sur la norme de la vitesse (figures 6.4.4 et 6.4.5). Pour réaliser les tracés, les valeurs moyennes par élément sont projetées aux sommets.

L'erreur relative maximale est de l'ordre de 0.2% en charge, de 6% en vitesse. On peut remarquer que l'erreur relative sur la vitesse est maximale là où la norme de la vitesse est minimale.

Références :

- [1] *The International Hydrocoin Project, Level 3, Case 7*, pp.249-291, OCDE 1988.
- [2] DABBENE F., *Résolution des équations de Darcy par une méthode d'éléments finis mixtes hybrides*, Rapport interne CEA, DMT93/637, Saclay, 1993.




FIG. 6.4.1 – Inclusion circulaire, maillage




FIG. 6.4.2 – Inclusion circulaire, Charge et Vitesse de Darcy ($H_{\min}=1.2d_4$ et $H_{\max}=1.08d_5$; Isovaleurs : de A=0.2d5 à I=1.d5 par pas de 0.1d5).




FIG. 6.4.3 – Inclusion circulaire, Norme de la vitesse ($V_{\min}=2.2d_{-10}$ et $V_{\max}=2.d_{-9}$; Isovaleurs : de A=0.4d-9 à H=1.8d-9 par pas de 0.2d-9).




FIG. 6.4.4 – Inclusion circulaire, erreur relative en charge (min=1.6d-7 et max=1.9d-3; Isovaleurs : A=1.d-6, B=5.d-6, C=1.d-5, D=5.d-5, E=1.d-4, F=5.d-4, G=1.d-3).




FIG. 6.4.5 – Inclusion circulaire, Erreur relative sur la norme de la vitesse (min=1.d-6 et max=6d-2; Isovaleurs : A=5d-5, B=1d-4, C=5.d-4, D=1.d-3, E=5.d-3, F=1.d-2).

Jeu de données :

```

*
*-- Utilisation du modèle DARCY : problème de l'inclusion
*
*-----
* Un cylindre de rayon R0, d'axe z et de conductivité Kin est inclus
* dans un domaine infini de conductivité Kout. La charge hydraulique
* h tend vers Finf*x à "l'infini".
*-----
* Référence : The international hydrocoin project, level 3, case 7b,
*             pp 249-291, OCDE, 1992.
*-----
*
*-----
* Options générales
*-----
*
GRAPH = '0' ;
'OPTI' 'DIME' 2 'ELEM' 'QUA4' ;
'OPTI' 'ISOV' 'LIGN' ;
'OPTI' 'ECHO' 1 ;
'OPTI' 'TRAC' 'PS' ;
*
*=====
* MAILLAGE
*=====
*
*- Dimensions caractéristiques du domaine de calcul
*
R0 = 10.D0 ;
L0 = 3.D0 ;
L1 = 5.D0 ;

```

```

LX   = L1 * R0 ;
LY   = L0 * R0 ;
MRO  = -1.D0 * R0 ;
MLO  = -1.D0 * L0 ;
ML1  = -1.D0 * L1 ;
MLX  = ML1 * R0 ;
MLY  = 0.D0 * R0 ;
ELIO = R0 / 200.D0 ;
*
*- Définition des points du maillage
*
P0   = 0.D0 0.D0 ;
P1   = MLX MLY ;
P2   = MRO MLY ;
P3   =  R0 MLY ;
P4   =  LX MLY ;
P5   =  LX  LY ;
P6   = 0.D0  LY ;
P7   = MLX  LY ;
P8   = 0.D0  R0 ;
*
*- Définition des droites
*
DM   = R0 / 5.D0 ;
DP   = R0 / 5.D0 ;
*
P1P2 = P1 'DROI' P2 'DINI' DP 'DFIN' DM ;
P2P3 = P2 'DROI' P3 'DINI' DM 'DFIN' DM ;
P2P8 = 'CERC' P2 P0 P8 'DINI' DM 'DFIN' DM ;
P3P4 = P3 'DROI' P4 'DINI' DM 'DFIN' DP ;
P4P5 = P4 'DROI' P5 'DINI' DM 'DFIN' DP ;
P5P6 = P5 'DROI' P6 'DINI' DP 'DFIN' DM ;
P6P7 = P6 'DROI' P7 'DINI' DM 'DFIN' DP ;
P7P1 = P7 'DROI' P1 'DINI' DP 'DFIN' DM ;
P8P3 = 'CERC' P8 P0 P3 'DINI' DM 'DFIN' DM ;
CER1 = P2P8 'ET' P8P3 ;
CER2 = 'INVE' CER1 ;
*
DRBAS = P1P2 'ET' P2P8 'ET' P8P3 'ET' P3P4 ;
DRDRO = P4P5 ;
DRHAU = P5P6 'ET' P6P7 ;
DRGAU = P7P1 ;
CONT3 = DRBAS 'ET' DRDRO 'ET' DRHAU 'ET' DRGAU ;
MA3   = 'SURF' CONT3 ;
CONT2 = P2P3 'ET' CER2 ;
MA2   = 'SURF' CONT2 ;
CONT1 = CONT2 'ET' CONT3 ;
MA1   = MA2 'ET' MA3 ;
*
*- Tables DOMAINE associées aux différentes zones
*
DOMA1 = 'DOMA' MA1 ;
DOMA2 = 'DOMA' MA2 'INCL' DOMA1 ELIO ;
DOMA3 = 'DOMA' MA3 'INCL' DOMA1 ELIO ;
*
*- Coordonnées des différentes classes de points

```

```

*
XC YC = 'COOR' DOMA1.'CENTRE' ;
XF YF = 'COOR' DOMA1.'FACE' ;
XS YS = 'COOR' DOMA1.'SOMMET' ;
RC = XC*XC + (YC*YC) ** 0.5D0 ;
RF = XF*XF + (YF*YF) ** 0.5D0 ;
RS = XS*XS + (YS*YS) ** 0.5D0 ;
*
*- Table DOMAINE associée aux plans frontières du domaine
*- (pour imposition des conditions aux limites de type Dirichlet)
*
FDIRI = DRDRO 'ET' DRHAU 'ET' DRGAU ;
DOMA4 = 'DOMA' FDIRI 'INCL' DOMA1 ELIO ;
*
*=====
* INITIALISATIONS
*=====
*
*- Données physiques
*
Kin = 1.D-10 ;
Kout = 1.D-12 ;
Finf = -1.D3 ;
*
*- Solutions analytiques
*
*   Hin = Finf A x
*   Hout = Finf ( 1. + B/r**2 ) x
*
*   Vin = -Kin Finf ( A , 0)
*   Vout = -Kout Finf ( 1. + B(y2-x2)/r4 ; - B 2xy/r4 )
*
*   avec A = 2 Kout / ( Kout + Kin)
*         B = a**2 (Kout-Kin) / (Kout + Kin)
*
A = 2. * Kout / ( Kout + Kin) ;
B = R0**2. * ( Kout - Kin) / (Kout + Kin) ;
VALO = 60000. ;
*
*- Charge au CENTRE des éléments
*
HCin = A * Finf * XC + VALO ;
HCout = RC ** (-2.d0) * B + 1.d0 * Finf * XC + VALO ;
*
*- Charge au FACE des éléments
*
HFin = A * Finf * XF + VALO ;
HFout = RF ** (-2.DO) * B + 1.DO * Finf * XF + VALO ;
*
*- Vitesse au CENTRE des éléments
*
Vxin = (-1.DO) * Kin * Finf * A ;
Vyin = 0.DO ;
Vin = 'MANU' 'CHPO' DOMA1.'CENTRE' 2 'VX' Vxin 'VY' Vyin ;
Vxout = XC*XC - (YC*YC) * B * (RC ** -4.DO) - 1.DO * Kout * Finf ;

```



```

Vxout = 'NOMC' 'VX' Vxout ;
Vyout = XC * YC * 2.DO * (RC ** -4.DO) * B * Kout * Finf ;
Vyout = 'NOMC' 'VY' Vyout ;
Vout = Vxout + Vyout ;
*
ANA2C = 'REDU' DOMA2.'CENTRE' HCIN ;
ANA3C = 'REDU' DOMA3.'CENTRE' HCOU ;
ANA1C = ANA2C + ANA3C ;
*
VANA2 = 'REDU' DOMA2.'CENTRE' Vin ;
VANA3 = 'REDU' DOMA3.'CENTRE' Vout ;
VANA1 = VANA2 + VANA3 ;
*
*=====
* MODELISATION
*=====
*
*
*- Modèle darcy et conductivité
*
MODL1 = 'MODL' DOMA1 'DARCY' 'ISOTROPE' ;
MODL2 = 'MODL' DOMA2 'DARCY' 'ISOTROPE' ;
MODL3 = 'MODL' DOMA3 'DARCY' 'ISOTROPE' ;
CHKin = 'MANU' 'CHPO' DOMA2.'CENTRE' 1 'K' Kin ;
CHKout = 'MANU' 'CHPO' DOMA3.'CENTRE' 1 'K' Kout ;
CHKtot = CHKin + CHKout ;
MKtot = 'KCHA' DOMA1 'CHAM' CHKtot ;
MATE1 = 'MATE' MODL1 'K' MKtot ;
*
*- Matrices élémentaires
*
MHYB1 = 'MHYB' MODL1 DOMA1 MATE1 ;
MATP1 = 'MATP' MODL1 DOMA1 MHYB1 ;
*
*- Conditions aux limites
*
* Flux nul sur les faces FMX et FMY (CL naturelle)
* Charge imposée sur les autres faces
*
BLOQ1 = 'BLOQ' DOMA4.'CENTRE' 'TH' ;
HIMP = 'REDU' HFout DOMA4.'CENTRE' ;
HIMP = 'NOMC' 'TH' HIMP ;
DEPI1 = 'DEPI' BLOQ1 HIMP ;
*
*- Assemblage matrices élémentaire et conditions aux limites
*
LHS1 = MATP1 'ET' BLOQ1 ;
RHS1 = DEPI1 ;
*
*- Résolution
*
HCAL1 = 'RESOU' LHS1 RHS1 ;
*
*- Post-traitement des inconnues secondaires
*
MCHYB = 'DOMA' DOMA1 'ORIENTAT' ;

```

Chapitre 7

Trajectoires et lignes de courant

7.1 Calcul de trajectoires

7.1.1 Formulations EF/EFMH

<i>Opérateurs utilisés</i>	<i>TRAJ</i>
<i>Méthodes numériques</i>	<i>EFMH, EF</i>
<i>Nature du problème</i>	<i>calcul des trajectoires de particules lâchées dans un champ de vitesses circulaires.</i>
<i>classification</i>	<i>2D plan</i>

Présentation du problème :

On considère un domaine carré maillé, dans lequel on se donne un champ de vitesses circulaires. Pour chaque point le module de la vitesse est égal à la distance au centre du domaine. On se propose de calculer les trajectoires de particules lâchées en différents points de ce domaine. Théoriquement pour un intervalle de temps de 2π , chaque particule doit décrire un cercle complet et pour un temps supérieur à 2π chaque trajectoire devra se superposer à elle même.

Choix des données :

Le domaine étudié est un carré de coté 2. L'origine du maillage est au point (0., 0.). Nous allons calculer les trajectoires de 3 particules lâchées aux points de coordonnées (1.4 , 1.) (1.6 ,1.) et (1.8, 1.). Le temps est choisi de façon à ce que les particules fassent 5 tours. Nous prenons un nombre de Courant de 0.1 ce qui correspond à faire environ 10 pas dans chaque élément. Pour utiliser la formulation EF nous nous donnons un champ de vitesses aux noeuds du maillage. Pour utiliser la formulation EFMH nous nous donnons un champ de vitesses aux noeuds faces du domaine et nous en déduisons les flux aux faces.




FIG. 7.1.1 – Champ de vitesses aux faces, utilisé avec la formulation EFMH





FIG. 7.1.2 – Champ de vitesses aux noeuds, utilisé avec la formulation EF

Résultats :


Nous avons calculé pour chaque point des trajectoires sa distance au centre. Nous avons porté dans le tableau suivant les valeurs maximale et minimale de ces distances.

	Théorique	EFMH	EF	err EFMH	err EF
Max	0.4	0.40787	0.40002	2.10^{-2}	5.10^{-5}
Min	0.4	0.40000	0.39986	1.10^{-5}	3.10^{-4}
Max	0.6	0.60269	0.60000	4.10^{-3}	1.10^{-5}
Min	0.6	0.59720	0.59932	5.10^{-3}	1.10^{-3}
Max	0.8	0.80352	0.80000	4.10^{-3}	1.10^{-5}
Min	0.8	0.79694	0.79959	4.10^{-3}	5.10^{-4}

Nous pouvons constater que les trajectoires se superposent à elles mêmes ce qui est confirmé par le tracé.



test trajectoires formulation hybride



test trajectoires formulation elements finis

FIG. 7.1.3 – Trajectoires obtenues avec la formulation EFMH

FIG. 7.1.4 – Trajectoires obtenues avec la formulation EF

Nous présentons ici les coordonnées de la position de chaque particule au temps 10π , ainsi que l'erreur relative sur la distance parcourue.

	Initial	Final EFMH	final EF	err EFMH	err EF
Coordonnées	1.4 1.	1.377 1.139	1.400 1.000	1.10^{-2}	5.10^{-5}
dist- centre	0.4	0.4019	0.3999	5.10^{-3}	2.10^{-4}
Coordonnées	1.6 1.	1.598 0.979	1.599 1.001	1.10^{-3}	5.10^{-5}
dist- centre	0.6	0.5988	0.5993	2.10^{-3}	1.10^{-3}
Coordonnées	1.8 1.	1.801 0.980	1.800 1.000	8.10^{-4}	2.10^{-5}
dist- centre	0.8	0.8011	0.7996	1.10^{-3}	5.10^{-4}

La différence constatée entre les résultats des deux méthodes tient au fait que, pour cet exemple, la vitesse est connue précisément en chaque point pour la formulation EF, pour la formulation EFMH elle est calculée précisément au centre de chaque élément. (cf rapport DMT/94/707)

Jeu de données :

```

OPTI DIME 2 ELEM QUA4 ;
*
*           DEFINITION DU MAILLAGE
*
TITRE "test trajectoires  " ;
P1 = 0. 0. ; P2 = 2. 0. ;
P3 = 2. 2. ; P4 = 0. 2. ;
LIG = P1 D 10 P2 D 15 P3 D 20 P4 D 25 P1 ;
CARRE = SURF PLAN LIG ;
MAUX= CONT CARRE ;
TRAC CARRE ;
*
*           CREATION DE LA TABLE DOMAINE
*
DOMCAR = DOMA CARRE ;
LIST DOMCAR ;
CHYB1 = DOMA DOMCAR SURFACE      ;
CHYB2 = DOMA DOMCAR NORMALE      ;
MCHYB = DOMA DOMCAR ORIENTAT     ;
*
*           DEFINITION DE LA TABLE DE LACHER
*
LACHER = TABLE ;
LACHER.TEMPS = PROG 0. ;
LACHER.TFINAL = 31.4159265;
* LACHER.CFL = 0.05 ;
LACHER.CFL = 0.1 ;
LACHER.DELTAT = 0. ;
LACHER.1 = (1.4 1.) et (1.6 1.) et (1.8 1.) ;
*
*           ON VA GENERER UNE VITESSE CIRCULAIRE V AU CENTRE DES FACES
*           ET EN DEDUIRE LE FLUX CORRESPONDANT QN
*
TITRE ' Vitesses aux faces des elements ' ;
XX YY = COOR DOMCAR.FACE ;
V1X = YY * (-1.) + 1. ;
V1Y = XX - 1. ;
VVX = NOMC UX V1X ;
VVY = NOMC UY V1Y ;
V = VVX ET VVY ;
VNCH = VECTEUR V 0.1 UX UY ROUGE ;
TRAC VNCH ( CARRE et LACHER.1 ) ;
MOT1 = MOTS UX UY ;
VAVN= PSCA V CHYB2 MOT1 MOT1 ;
VAVN = NOMC SCAL VAVN ;
QN = VAVN* CHYB1 ;
QN = NOMC FLUX QN ;
*

```

```

*           CALCUL DES TRAJECTOIRES AVEC LA FORMULATION MIXTE HYBRIDE
*
UUUU = TRAJ DOMCAR QN MCHYB LACHER ;
*
*           CONTROLE DES RESULTATS
*
XP = PROG 0.4 0.6 0.8 ;
LIST XP ;
II= 0 ;
REPETER BLOC1 3 ;
II=II+1 ;
XP1= EXTR XP II ;
NBPT1=DIME UUUU.MTABX.II ;
JJJ=0 ;
MMX= 0. ;
MNX= XP1;
REPETER BLOC7 NBPT1 ;
JJJ= JJJ+1 ;
XX1=EXTR UUUU.MTABX.II JJJ;
YY1=EXTR UUUU.MTABY.II JJJ ;
TT1=EXTR UUUU.MTABP.II JJJ;
  XX1= XX1 - 1. ;
  YY1= YY1 - 1. ;
DISTC= ( ( XX1*XX1) + (YY1*YY1)) ** 0.5 ;
SI ( DISTC < MNX) ;
  MNX= DISTC ;
FINSI ;
SI ( DISTC > MMX) ;
  MMX= DISTC ;
FINSI ;
FIN BLOC7 ;
LIST MMX ;
LIST MNX ;
DELX= XP1 - XX1;
DISTP= (( DELX*DELX) +(YY1*YY1)) ** 0.5 ;
LIST DISTP ;
LLON= TT1 * XP1 ;
ERRR= DISTP / LLON ;
LIST ERRR ;
FIN BLOC1 ;
*
*           GENERATION D'UN MAILLAGE DE SEGMENTS CROB1 QUI POURRA
*           ETRE MANIPULE PAR LES OPERATEURS DE TRACE
*
TITRE 'test trajectoires formulation hybride ' ;
CROB1= TRTRAJEC UUUU ;
TRAC ( CROB1 ET MAUX) ;
*
*           GENERATION D'UNE VITESSE CIRCULAIRE AUX NOEUDS DU MAILLAGE
*
TITRE 'Vitesses aux noeuds du maillage ' ;
  XX1 YY1 = COOR DOMCAR.MAILLAGE ;
  VX1 = YY1 * (-1.) + 1. ;
  VY1 = XX1 - 1. ;
  VVX1 = NOMC VX VX1 ;
  VVY1 = NOMC VY VY1 ;

```



```

V1 = VVX1 ET VVY1 ;
VNCH1 = VECTEUR V1 0.1 VX VY ROUGE ;
TRAC VNCH1 ( CARRE et LACHER.1 ) ;
*
*      CALCUL DES TRAJECTOIRES AVEC LA FORMULATION ELEMENTS FINIS
*
UUU1 = TRAJ DOMCAR V1 LACHER ;
*
*      CONTROLE DES RESULTATS
*
II= 0 ;
REPETER BLOC2 3 ;
II=II+1 ;
XP1= EXTR XP II ;
NBPT1=DIME UUU1.MTABX.II ;
JJJ=0 ;
MMX= 0. ;
MNX= XP1;
REPETER BLOC8 NBPT1 ;
JJJ= JJJ+1 ;
XX1=EXTR UUU1.MTABX.II JJJ;
YY1=EXTR UUU1.MTABY.II JJJ;
TT1=EXTR UUU1.MTABP.II JJJ;
XX1= XX1 - 1. ;
YY1= YY1 - 1. ;
DISTC= ( ( XX1*XX1) + (YY1*YY1)) ** 0.5 ;
SI ( DISTC < MNX) ;
MNX= DISTC ;
FINSI ;
SI ( DISTC > MMX) ;
MNX= DISTC ;
FINSI ;
FIN BLOC8 ;
LIST MMX ;
LIST MNX ;
DELX= XP1 - XX1;
DISTP= (( DELX*DELX) +(YY1*YY1)) ** 0.5 ;
LIST DISTP ;
LLON= TT1 * XP1 ;
ERRR= DISTP / LLON ;
LIST ERRR ;
FIN BLOC2 ;
*
*      GENERATION D'UN MAILLAGE DE SEGMENTS CROB2 QUI POURRA
*      ETRE MANIPULE PAR LES OPERATEURS DE TRACE
*
TITRE 'test trajectoires formulation elements finis ' ;
CROB2= TRTRAJEC UUU1 ;
TRAC ( CROB2 ET MAUX) ;
FIN ;
$

```

7.2 Lignes de courant

7.2.1 Formulation Eléments Finis Implicite

<i>Opérateurs utilisés</i>	<i>LAPN, FIMP</i>
<i>Méthodes numériques</i>	<i>EF, approximation Q1</i>
<i>Nature du problème</i>	<i>Détermination de lignes de courant dans un domaine fermé à partir de la donnée d'un champ de vitesse. Vérification de la convergence d'ordre 2 de la formulation EF.</i>
<i>Classification</i>	<i>2D plan, diffusion</i>
<i>Jeu de données</i>	<i>/u2/castem/dgibi/lignecourant.dgibi</i>

Rappels concernant la fonction de courant :

Pour un écoulement incompressible plan (stationnaire ou non), l'équation de continuité $\nabla \cdot \mathbf{u} = 0$ montre qu'il existe une fonction $\Psi(x, y, t)$, appelée fonction de courant, définie à une constante près et telle que l'on ait :

$$u = \frac{\partial \Psi}{\partial y}, \quad v = -\frac{\partial \Psi}{\partial x}$$

Les lignes $\Psi = \text{Cte}$ sont à chaque instant les lignes de courant de l'écoulement. De plus, le débit passant entre 2 points A et B d'un écoulement est égal à $\Psi_B - \Psi_A$.

L'exemple qui suit montre comment déterminer la fonction de courant Ψ à partir de la donnée d'un champ de vitesse, issu par exemple de la résolution des équations de Navier-Stokes. Pour un champ de vitesse (u, v) donné sur un domaine Ω , la fonction de courant associée obéit à l'équation de Poisson,

$$\Delta \Psi = \frac{\partial u}{\partial y} - \frac{\partial v}{\partial x}, \quad \forall \mathbf{x} \in \Omega$$

ainsi qu'à des conditions aux limites appropriées sur tout le contour de Ω . Dans le cas d'un domaine fermé simplement connexe, ces conditions se réduisent à :

$$\Psi = 0, \quad \forall \mathbf{x} \in \partial\Omega$$

puisque la frontière est une ligne de courant de l'écoulement.

Description du problème :

On considère le domaine carré $\Omega = \{(x, y), 0 \leq x \leq 1, 0 \leq y \leq 1\}$, sur lequel est définie la fonction de courant analytique Ψ_{exact} ,

$$\Psi_{exact}(x, y) = x(x-1)y^2(y-1)$$

qui vérifie $\Psi = 0$ sur les bords du domaine.

Le champ de vitesse associé est donné par :

$$u_{ex} = x(x-1)y(3y-2), \quad v_{ex} = -y^2(y-1)(2x-1)$$

dont la vorticit   $\omega_{ex} = \frac{\partial u_{ex}}{\partial y} - \frac{\partial v_{ex}}{\partial x}$ est donn  e par :

$$\omega_{ex} = x(x-1)(6y-2) + 2y^2(y-1)$$

On consid  re maintenant une discr  tisation du probl  me :

trouver Ψ telle que $\Delta\Psi = \omega_{ex}$ en tout point de Ω , $\Psi = 0$ sur les bords.

On consid  re une suite $\mathcal{M}_{(k)}$ de maillages de quadrangles de pas d'espace uniformes et d  croissants, $\Delta h_{(k)}$, avec $\Delta h_{(1)} > \Delta h_{(2)} > \dots$. Pour chaque maillage, on d  termine la fonction de courant $\Psi_{num(k)}$ en r  solvant l'  quation de Poisson sur Ψ grace aux op  rateurs LAPN et FIMP, dans une formulation   l  ments Finis Implicite.

Une fois que la solution num  rique $\Psi_{num(k)}$ est calcul  e, on peut   valuer l'erreur entre la solution num  rique et la solution exacte Ψ_{exact} . On choisit pour cela la norme L_2 discr  te,


$$\epsilon_{(k)} = \|\Psi_{num(k)} - \Psi_{exact}\|_{L_2} = \sqrt{\frac{1}{N_{(k)}} \sum_{i=1}^{N_{(k)}} (\Psi_{num}(x_i, y_i) - \Psi_{exact}(x_i, y_i))^2}$$

o   $N_{(k)}$ est le nombre de sommets du maillage $\mathcal{M}_{(k)}$. On sait par ailleurs que pour un sch  ma d'ordre p , l'erreur en norme L_2 est proportionnel    Δh^p . Dans le cas d'  l  ments bilin  aires, on s'attend    une convergence quadratique ($p \approx 2$).

R  sultats :


Les figures 7.2.1 et 7.2.2 montrent sur un maillage 5×5 le champ de vitesse impos   et la fonction de courant calcul  e par r  solution de l'  quation de Poisson. Les figures 7.2.3 et 7.2.4 montrent les r  sultats obtenus sur un maillage 25×25 .

Dans ces 2 cas, ainsi que pour des maillages interm  diaires, on calcule l'erreur entre la fonction de courant calcul  e et la fonction de courant analytique. Les r  sultats de cette   tude de convergence sont report  s sur la figure 7.2.5 : une r  gression lin  aire sur les r  sultats permet de d  terminer la pente de la courbe repr  sent  e en   chelles logarithmiques. La valeur trouv  e, $p \approx 1.93$, est bien en accord avec l'ordre attendu pour une discr  tisation par   l  ments finis Q1 de l'  quation de Poisson.




CHAMP DE VITESSE

FIG. 7.2.1 – Champ de vitesse imposé sur maillage 5×5




FONCTION DE COURANT CALCULEE

FIG. 7.2.2 – Fonction de courant calculée sur maillage 5×5



CHAMP DE VITESSE

FIG. 7.2.3 – Champ de vitesse imposé sur maillage 25×25



FONCTION DE COURANT CALCULEE

FIG. 7.2.4 – Fonction de courant calculée sur maillage 25×25




FIG. 7.2.5 – Etude de convergence : la pente de la courbe d’erreur en fonction du pas du maillage (en échelles logarithmiques) est approximativement égale à 1.9

Jeu de données :

```

GRAPH = FAUX ;

*****
* CALCUL DES LIGNES DE COURANT : ON TESTE LES OPERATEURS LAPN ET FIMP *
* ON VERIFIE QUE L'ERREUR ENTRE PSI_NUM ET PSI_EXACT EVOLUE DE FACON *
* QUADRATIQUE *
* H. PAILLERE/TTMF/JUILLET 1997 *
* ON CONSIDERE LE DOMAINE [0,1]x[0,1] ET LA FONCTION DE COURANT *
* PSI_EXACT = x(x-1)y^2(y-1) *
* ASSOCIEE AU CHAMP DE VITESSE (u,v) DONNE PAR : *
*   u_ex = dPSI/dy et v_ex = -dPSI/dx *
* ON RESOUD L'EQUATION DE POISSON : *
*       d2PSI/dx2 + d2PSI/dy2 = du_ex/dy - dv_ex/dx *
* AVEC COMME C.L PSI=0 SUR LES BORDS DU DOMAINE *
*****

OPTI DIME 2 ;
OPTI ELEM QUA4 ;
OPTI ISOV SULI ;

REPETER BLOC2 6 ;

n = (&BLOC2) ;
NX = 5*n ;
NY = 5*n ;

A1 = 0.0 0.0 ;
A2 = 1.0 0.0 ;
A3 = 1.0 1.0 ;
A4 = 0.0 1.0 ;

DOMTOT = 'DALLER' (A1 'DROI' NX A2) (A2 'DROI' NY A3)
          (A3 'DROI' NX A4) (A4 'DROI' NX A1) 'PLAN' ;

MDOMTOT = CHAN DOMTOT QUAF ;
$DOMTOT = 'MODE' MDOMTOT 'NAVIER_STOKES' LINE;
DOMA $DOMTOT 'IMPR' ;
CNT = CONT DOMTOT ;

XX YY = 'COOR' DOMTOT ;

*****
* EVALUATION DE LA FONCTION DE COURANT EXACTE + CHAMP DE VITESSE *
*****

PSIEX = KOPS XX '*' ((XX-1.0)) ;
PSIEX = KOPS PSIEX '*' (KOPS YY '*' YY) ;
PSIEX = KOPS PSIEX '*' ((YY-1.0)) ;
PSIEX = KCHT $DOMTOT SCAL SOMMET PSIEX ;

SI ( GRAPH ) ;
TRACE PSIEX DOMTOT 14 'TITR'
      'FONCTION DE COURANT EXACTE' ;

```

```

FINSI ;

U = KOPS XX '*' ((XX-1.0)) ;
U = KOPS U '*' YY ;
G = KOPS (YY*3.0) '-' 2.0 ;
U = KOPS U '*' G ;
G = KOPS (XX*2.0) '-' 1.0 ;
V = KOPS G '*' ((YY-1.0)) ;
V = KOPS V '*' YY ;
V = KOPS V '*' YY ;
V = KOPS V '*' (-1.0) ;

U = NOMC 'UX' U 'NATU' 'DISCRET' ;
V = NOMC 'UY' V 'NATU' 'DISCRET' ;
UU = KCHT $DOMTOT VECT SOMMET (U ET V) ;
UNCH = VECT UU 1.0 UX UY VERT ;

SI ( GRAPH ) ;
TRACE UNCH DOMTOT 'TITR' 'CHAMP DE VITESSE' ;
FINSI ;

*****
* ON CALCULE LA FONCTION DE COURANT NUMERIQUEMENT A PARTIR DE LA *
* DONNEE DU CHAMP DE VITESSE *
*****

SW = (-1.) * (KOPS UU 'ROT' $DOMTOT) ;

RK = EQEX $DOMTOT 'OPTI' 'EF' 'IMPL'
ZONE $DOMTOT OPER LAPN 1.0 INCO 'PSI'
ZONE $DOMTOT OPER FIMP sw INCO 'PSI'
'CLIM' 'PSI' 'TIMP' CNT 0. ;

RK.'INCO' = TABLE INCO ;
RK.'INCO'.'PSI' = KCHT $DOMTOT SCAL SOMMET 0. ;
EXEC RK ;

PSI = RK.'INCO'.'PSI' ;

SI ( GRAPH ) ;
TRACE PSI DOMTOT CNT 14 'TITR'
'FONCTION DE COURANT CALCULEE' ;
FINSI ;

*****
* CALCUL DE L'ERREUR ENTRE FONCTION DE COURANT CALCULEE ET *
* FONCTION DE COURANT ANALYTIQUE *
*****

ERO = KOPS PSI '-' PSIEX ;
ER2 = KOPS ERO '*' ERO ;
ERR = 0. ;
NPTD=NBNO DOMTOT ;
REPETER BLOC1 NPTD ;
P1 = (DOMA $DOMTOT SOMMET) POIN &BLOC1 ;
ERR = ERR + ('EXTR' ER2 'SCAL' P1) ;

```

```

FIN BLOC1 ;
ERR = ERR/NPTD ;
ERR = ERR ** 0.5 ;

MESSAGE '===== ' ;
MESSAGE 'ERREUR EN NORME L2 = ' ERR ;
MESSAGE '===== ' ;

SI ( (n 'NEG' 1) ) ;
    LISTE n ;
    PERR = PROG (LOG ERR) ;
    ERRPSI = ERRPSI ET PERR ;
    PXXX = PROG (LOG (1./NX)) ;
    ERRXXX = ERRXXX ET PXXX ;
SINON ;
    ERRPSI = PROG (LOG ERR) ;
    ERRXXX = PROG (LOG (1./NX)) ;
FINSI ;

FIN BLOC2 ;

SI ( GRAPH ) ;
EVOL1 = EVOL 'MANU' 'LOG(DX)' ERRXXX 'LOG(ERR(PESI))' ERRPSI ;
TAB1 = TABLE ;
TAB1.1 = 'MARQ CROI ' ;
TAB1.'TITRE' = TABLE ;
TAB1.'TITRE'. 1 = MOT 'MAILLAGE QUA4' ;
DESS EVOL1 LEGE TAB1 'XBOR' -4. 0. 'YBOR' -11.0 -7.0 'GRIL' ;
FINSI ;

SI ( (MINI ERRPSI) > -10.9 ) ;
    ERREUR 5 ;
FINSI ;

FIN ;

```

Index

BIF, 184, 189

DFDT, 34, 91

FIMP, 321

FPU, 159, 189

FROT, 184, 189

KONV, 34, 200, 232

LAPN, 91, 321

NS, 98, 105, 112, 121, 132, 140, 184, 189

NSKE, 151, 159, 174, 189, 247

PENT, 200, 232

PPRE, 189

PRESSION, 98, 105, 121, 132, 140, 189,
247

PRET, 200

TOIMP, 98, 105, 140

TSCAL, 8, 27, 59, 68, 77, 83, 112, 140,
174, 189, 247